

# Java Source and Bytecode Formalizations in Isabelle: $\mu$ Java

– VerifiCard Project Deliverables –

Gerwin Klein    Tobias Nipkow    David von Oheimb    Leonor Prensa Nieto  
Norbert Schirmer    Martin Strecker

August 16, 2002



# Contents

<b>1</b>	<b>Preface</b>	<b>5</b>
1.1	Introduction . . . . .	5
1.2	Theory Dependencies . . . . .	6
<b>2</b>	<b>Java Source Language</b>	<b>9</b>
2.1	Some Auxiliary Definitions . . . . .	10
2.2	Java types . . . . .	12
2.3	Class Declarations and Programs . . . . .	14
2.4	Relations between Java Types . . . . .	16
2.5	Java Values . . . . .	21
2.6	Program State . . . . .	22
2.7	Expressions and Statements . . . . .	25
2.8	System Classes . . . . .	26
2.9	Well-formedness of Java programs . . . . .	27
2.10	Well-typedness Constraints . . . . .	37
2.11	Conformity Relations for Type Soundness Proof . . . . .	42
<b>3</b>	<b>Java Virtual Machine</b>	<b>49</b>
3.1	State of the JVM . . . . .	50
3.2	Instructions of the JVM . . . . .	53
3.3	JVM Instruction Semantics . . . . .	54
3.4	Exception handling in the JVM . . . . .	59
3.5	Program Execution in the JVM . . . . .	62
3.6	A Defensive JVM . . . . .	63
3.7	System Class Implementations (JVM) . . . . .	68
3.8	Example for generating executable code from JVM semantics . . . . .	73
<b>4</b>	<b>Bytecode Verifier</b>	<b>77</b>
4.1	Semilattices . . . . .	78
4.2	The Error Type . . . . .	85
4.3	Fixed Length Lists . . . . .	92
4.4	Typing and Dataflow Analysis Framework . . . . .	104
4.5	Products as Semilattices . . . . .	105
4.6	More on Semilattices . . . . .	108
4.7	Lifting the Typing Framework to <code>err</code> , <code>app</code> , and <code>eff</code> . . . . .	112
4.8	More about Options . . . . .	118
4.9	The Java Type System as Semilattice . . . . .	125
4.10	The JVM Type System as Semilattice . . . . .	130

4.11	Effect of Instructions on the State Type . . . . .	133
4.12	Monotonicity of <code>eff</code> and <code>app</code> . . . . .	151
4.13	The Bytecode Verifier . . . . .	153
4.14	The Typing Framework for the JVM . . . . .	155
4.15	Kildall's Algorithm . . . . .	165
4.16	Kildall for the JVM . . . . .	176
4.17	The Lightweight Bytecode Verifier . . . . .	181
4.18	Correctness of the LBV . . . . .	188
4.19	Completeness of the LBV . . . . .	192
4.20	LBV for the JVM . . . . .	200
4.21	BV Type Safety Invariant . . . . .	205
4.22	BV Type Safety Proof . . . . .	232
4.23	Welltyped Programs produce no Type Errors . . . . .	300
4.24	Example Welltypings . . . . .	311
4.25	Example for a program with JSR . . . . .	321

# Chapter 1

## Preface

### 1.1 Introduction

This document contains the automatically generated listings of the Isabelle sources for  $\mu$ Java.  $\mu$ Java is a reduced model of JavaCard, dedicated to the study of the interaction of the source language, byte code, the byte code verifier and the compiler. In order to make the Isabelle sources more accessible, this introduction provides a brief survey of the main concepts of  $\mu$ Java.

The  $\mu$ Java **source language** (see Chapter 2) only comprises a part of the original JavaCard language. It models features such as:

- The basic “primitive types” of Java
- Object orientation, in particular classes, and relevant relations on classes (subclass, widening)
- Methods and method signatures
- Inheritance and overriding of methods, dynamic binding
- Representatives of “relevant” expressions and statements
- Generation and propagation of system exceptions

However, the following features are missing in  $\mu$ Java wrt. JavaCard:

- Some primitive types (`byte`, `short`)
- Interfaces and related concepts, arrays
- Most numeric operations, syntactic variants of statements (`do-loop`, `for-loop`)
- Complex block structure, method bodies with multiple returns
- Abrupt termination (`break`, `continue`)
- Class and method modifiers (such as `static` and `public/private` access modifiers)
- User-defined exception classes and an explicit `throw`-statement. Exceptions cannot be caught.

- A “definite assignment” check

In addition, features are missing that are not part of the JavaCard language, such as multithreading and garbage collection. No attempt has been made to model peculiarities of JavaCard such as the applet firewall or the transaction mechanism.

For a more complete Isabelle model of JavaCard, the reader should consult the Bali formalization (<http://isabelle.in.tum.de/verificard/Bali/document.pdf>), which models most of the source language features of JavaCard, however without describing the bytecode level.

The central topics of the source language formalization are:

- Description of the structure of the “runtime environment”, in particular structure of classes and the program state
- Definition of syntax, typing rules and operational semantics of statements and expressions
- Definition of “conformity” (characterizing type safety) and a type safety proof

The  $\mu$ Java **virtual machine** (see Chapter 3) corresponds to the source level in the sense that the same data types are supported and bytecode instructions required for emulating the source level operations are provided. Again, only one representative of different variants of instructions has been selected; for example, there is only one comparison operator. Questions related to the type system for and type correctness of bytecode are described in the chapter on bytecode verification.

The problem of **bytecode verification** (see Chapter 4) is dealt with in several stages:

- First, the notion of “method type” is introduced, which corresponds to the notion of “type” on the source level.
- Well-typedness of instructions wrt. a method type is defined (see Section 4.13). Roughly speaking, determining well-typedness is *type checking*.
- It is shown that bytecode that is well-typed in this sense can be safely executed – a type soundness proof on the bytecode level (Section 4.22).
- Given raw bytecode, one of the purposes of bytecode verification is to determine a method type that is well-typed according to the above definition. Roughly speaking, this is *type inference*. The Isabelle formalization presents bytecode verification as an instance of an abstract dataflow algorithm (Kildall’s algorithm, see Sections 4.15 to 4.20).

Bytecode verification in  $\mu$ Java so far takes into account:

- Operations and branching instructions
- Exceptions
- Object initialization (including constructors)

The `jsr` instruction is not yet accounted for.

## 1.2 Theory Dependencies

Figure 1.1 shows the dependencies between the Isabelle theories in the following sections.




Figure 1.1: Theory Dependency Graph





## Chapter 2

# Java Source Language

## 2.1 Some Auxiliary Definitions

```
theory JBasis = Main:
```

```
lemmas [simp] = Let_def
```

### 2.1.1 unique

```
constdefs
```

```
  unique  :: "('a × 'b) list => bool"
  "unique == distinct ∘ map fst"
```

```
lemma fst_in_set_lemma [rule_format (no_asm)]:
```

```
  "(x, y) : set xys --> x : fst ` set xys"
apply (induct_tac "xys")
apply auto
done
```

```
lemma unique_Nil [simp]: "unique []"
```

```
apply (unfold unique_def)
apply (simp (no_asm))
done
```

```
lemma unique_Cons [simp]: "unique ((x,y)#l) = (unique l & (!y. (x,y) ~: set l))"
```

```
apply (unfold unique_def)
apply (auto dest: fst_in_set_lemma)
done
```

```
lemma unique_append [rule_format (no_asm)]: "unique l' ==> unique l -->
```

```
  (! (x,y):set l. ! (x',y'):set l'. x' ~ = x) --> unique (l @ l')"
apply (induct_tac "l")
apply (auto dest: fst_in_set_lemma)
done
```

```
lemma unique_map_inj [rule_format (no_asm)]:
```

```
  "unique l --> inj f --> unique (map (%(k,x). (f k, g k x)) l)"
apply (induct_tac "l")
apply (auto dest: fst_in_set_lemma simp add: inj_eq)
done
```

### 2.1.2 More about Maps

```
lemma map_of_SomeI [rule_format (no_asm)]:
```

```
  "unique l --> (k, x) : set l --> map_of l k = Some x"
apply (induct_tac "l")
apply auto
done
```

```
lemma Ball_set_table_:
```

```
  "(∀ (x,y) ∈ set l. P x y) --> (∀ x. ∀ y. map_of l x = Some y --> P x y)"
apply (induct_tac "l")
apply (simp_all (no_asm))
apply safe
apply auto
```

done

lemmas Ball\_set\_table = Ball\_set\_table\_ [THEN mp]

lemma table\_of\_remap\_SomeD [rule\_format (no\_asm)]:

"map\_of (map ( $\lambda((k,k'),x). (k,(k',x))$ ) t) k = Some (k',x) -->  
map\_of t (k, k') = Some x"

apply (induct\_tac "t")

apply auto

done

end

## 2.2 Java types

theory Type = JBasis:

typedecl cnam

— exceptions

datatype

```
xcpt
= NullPointer
| ClassCast
| OutOfMemory
| ArrayIndexOutOfBounds
| NegativeArraySize
| ArrayStore
```

— class names

datatype cname

```
= Object
| Xcpt xcpt
| Cname cnam
```

typedecl vnam — variable or field name

typedecl mname — method name

— names for *This* pointer and local/field variables

datatype vname

```
= This
| VName vnam
```

— primitive type, cf. 4.2

datatype prim\_ty

```
= Void — 'result type' of void methods
| Boolean
| Integer
| RetA nat — bytecode return addresses
```

datatype

```
ref_ty — reference type, cf. 4.3
= NullT — null type, cf. 4.1
| ClassT cname — class type
| ArrayT ty — array type
```

and

```
ty — any type, cf. 4.1
= PrimT prim_ty — primitive type
| RefT ref_ty — reference type
```

syntax

```
NT :: "ty"
Class :: "cname => ty"
RA :: "nat => ty"
Array :: "ty => ty" ("_." [90] 90)
```

**translations**

```
"NT"      == "RefT NullT"  
"Class C" == "RefT (ClassT C)"  
"T. []"   == "RefT (ArrayT T)"  
"RA pc"   == "PrimT (RetA pc)"
```

**consts**

```
the_Class :: "ty  $\Rightarrow$  cname"  
isArray  :: "ty  $\Rightarrow$  bool"
```

**recdef the\_Class "{}"**

```
"the_Class (Class C) = C"  
"the_Class (T. []) = Object"
```

**recdef isArray "{}"**

```
"isArray (T. []) = True"  
"isArray T      = False"
```

**end**

## 2.3 Class Declarations and Programs

theory Decl = Type:

types

```
fdecl    = "vname × ty"           — field declaration, cf. 8.3 (, 9.3)
sig      = "mname × ty list"     — signature of a method, cf. 8.4.2
'c mdecl = "sig × ty × 'c"       — method declaration in a class
'c class = "cname × fdecl list × 'c mdecl list"
— class = superclass, fields, methods

'c cdecl = "cname × 'c class"    — class declaration, cf. 8.1
'c prog  = "'c cdecl list"      — program
```

translations

```
"fdecl"  <= (type) "vname × ty"
"sig"    <= (type) "mname × ty list"
"mdecl c" <= (type) "sig × ty × c"
"class c" <= (type) "cname × fdecl list × (c mdecl) list"
"cdecl c" <= (type) "cname × (c class)"
"prog c" <= (type) "(c cdecl) list"
```

constdefs

```
class :: "'c prog => (cname ~> 'c class)"
"class ≡ map_of"
```

```
is_class :: "'c prog => cname => bool"
"is_class G C ≡ class G C ≠ None"
```

lemma finite\_is\_class: "finite {C. is\_class G C}"

```
apply (unfold is_class_def class_def)
apply (fold dom_def)
apply (rule finite_dom_map_of)
done
```

lemma class\_SomeD:

```
"class G C = Some D ⇒ (C,D) ∈ set G"
apply (unfold class_def)
apply (erule map_of_SomeD)
done
```

consts

```
is_type :: "'c prog ⇒ ty ⇒ bool"
isrtype :: "'c prog ⇒ ref_ty ⇒ bool"
```

primrec

```
"is_type G (PrimT T) = True"
  "is_type G (RefT T)  = isrtype G T"
  "isrtype G (NullT)  = True"
  "isrtype G (ClassT C) = is_class G C"
  "isrtype G (ArrayT T) = is_type G T"
```

consts

```
is_RA :: "ty => bool"
recdef is_RA "{}"
  "is_RA (RA pc) = True"
  "is_RA t      = False"
end
```

## 2.4 Relations between Java Types

theory TypeRel = Decl:

consts

```
subcls1 :: "'c prog => (cname × cname) set" — subclass
widen  :: "'c prog => (ty   × ty   ) set" — widening
cast   :: "'c prog => (cname × cname) set" — casting
```

syntax (xsymbols)

```
subcls1 :: "'c prog => [cname, cname] => bool" ("_ ⊢ _ <C1 _" [71,71,71] 70)
subcls  :: "'c prog => [cname, cname] => bool" ("_ ⊢ _ ⊆C _" [71,71,71] 70)
widen   :: "'c prog => [ty   , ty   ] => bool" ("_ ⊢ _ ⊆ _" [71,71,71] 70)
cast    :: "'c prog => [cname, cname] => bool" ("_ ⊢ _ ⊆? _" [71,71,71] 70)
```

syntax

```
subcls1 :: "'c prog => [cname, cname] => bool" ("_ |- _ <=C1 _" [71,71,71] 70)
subcls  :: "'c prog => [cname, cname] => bool" ("_ |- _ <=C _" [71,71,71] 70)
widen   :: "'c prog => [ty   , ty   ] => bool" ("_ |- _ <= _" [71,71,71] 70)
cast    :: "'c prog => [cname, cname] => bool" ("_ |- _ <=? _" [71,71,71] 70)
```

translations

```
"G ⊢ C <C1 D" == "(C,D) ∈ subcls1 G"
"G ⊢ C ⊆C D"  == "(C,D) ∈ (subcls1 G)^*"
"G ⊢ S ⊆ T"   == "(S,T) ∈ widen G"
"G ⊢ C ⊆? D"  == "(C,D) ∈ cast G"
```

— direct subclass, cf. 8.1.3

**inductive** "subcls1 G" intros

```
subcls1I: "[class G C = Some (D,rest); C ≠ Object] ⇒ G ⊢ C <C1 D"
```

— widening, viz. method invocation conversion, cf. 5.3 i.e. sort of syntactic subtyping

**inductive** "widen G" intros

```
refl [intro!, simp]: "G ⊢ T ⊆ T" — identity conv., cf. 5.1.1
subcls: "G ⊢ C ⊆C D ⇒ G ⊢ Class C ⊆ Class D"
null [intro!]: "G ⊢ NT ⊆ RefT R"
arr_obj [intro!, simp]: "G ⊢ T.[] ⊆ Class Object"
array: "G ⊢ RefT S ⊆ RefT T ⇒ G ⊢ RefT S.[] ⊆ RefT T.[]"
```

— casting conversion, cf. 5.5 / 5.1.5

— left out casts on primitive types and arrays

**inductive** "cast G" intros

```
widen: "G ⊢ C ⊆C D ⇒ G ⊢ C ⊆? D"
subcls: "G ⊢ D ⊆C C ⇒ G ⊢ C ⊆? D"
```

lemma subcls1D:

```
"G ⊢ C <C1 D ⇒ C ≠ Object ∧ (∃ fs ms. class G C = Some (D,fs,ms))"
```

apply (erule subcls1.elims)

apply auto

done



lemma subcls1\_def2:

```
"subcls1 G = ( $\Sigma C \in \{C. \text{is\_class } G \ C\} . \{D. C \neq \text{Object} \wedge \text{fst (the (class } G \ C)) = D\}$ )"
  by (auto simp add: is_class_def dest: subcls1D intro: subcls1I)
```

lemma finite\_subcls1: "finite (subcls1 G)"

```
apply(subst subcls1_def2)
apply(rule finite_SigmaI [OF finite_is_class])
apply(rule_tac B = "{fst (the (class } G \ C))}" in finite_subset)
apply auto
done
```

lemma subcls\_is\_class: "(C,D)  $\in$  (subcls1 G)<sup>+</sup> ==> is\_class G C"

```
apply (unfold is_class_def)
apply(erule trancl_trans_induct)
apply (auto dest!: subcls1D)
done
```

lemma subcls\_is\_class2 [rule\_format (no\_asm)]:

```
"G  $\vdash$  C  $\preceq$  C D  $\implies$  is_class G D  $\longrightarrow$  is_class G C"
apply (unfold is_class_def)
apply (erule rtrancl_induct)
apply (drule_tac [2] subcls1D)
apply auto
done
```

consts class\_rec :: "'c prog  $\times$  cname  $\implies$

```
'a  $\implies$  (cname  $\implies$  fdecl list  $\implies$  'c mdecl list  $\implies$  'a  $\implies$  'a)  $\implies$  'a"
```

recdef class\_rec "same\_fst ( $\lambda G. \text{wf ((subcls1 } G)^{-1})$ ) ( $\lambda G. (\text{subcls1 } G)^{-1}$ )"

```
"class_rec (G,C) = ( $\lambda t f. \text{case class } G \ C \text{ of None } \implies \text{arbitrary}$ 
  | Some (D,fs,ms)  $\implies$  if wf ((subcls1 } G)^{-1}) then
  f C fs ms (if C = Object then t else class_rec (G,D) t f) else arbitrary)"
(hints intro: subcls1I)
```

declare class\_rec.simps [simp del]

lemma class\_rec\_lemma: "[[ wf ((subcls1 G)<sup>-1</sup>); class G C = Some (D,fs,ms) ]] ==>

```
class_rec (G,C) t f = f C fs ms (if C=Object then t else class_rec (G,D) t f)"
  apply (rule class_rec.simps [THEN trans [THEN fun_cong [THEN fun_cong]]])
  apply simp
done
```

consts

```
method :: "'c prog  $\times$  cname  $\implies$  ( sig  $\rightsquigarrow$  cname  $\times$  ty  $\times$  'c)"
field   :: "'c prog  $\times$  cname  $\implies$  ( vname  $\rightsquigarrow$  cname  $\times$  ty      )"
fields  :: "'c prog  $\times$  cname  $\implies$  ((vname  $\times$  cname)  $\times$  ty) list"
```

— methods of a class, with inheritance, overriding and hiding, cf. 8.4.6

```
defs method_def: "method  $\equiv$   $\lambda(G,C). \text{class\_rec } (G,C) \text{ empty } (\lambda C \text{ fs ms ts.}$ 
  ts ++ map_of (map ( $\lambda(s,m). (s, (C,m))$ )) ms)"
```

lemma method\_rec\_lemma: "[[class G C = Some (D,fs,ms); wf ((subcls1 G)<sup>-1</sup>)]] ==>

```

method (G,C) = (if C = Object then empty else method (G,D)) ++
  map_of (map (λ(s,m). (s,(C,m))) ms)"
apply (unfold method_def)
apply (simp split del: split_if)
apply (erule (1) class_rec_lemma [THEN trans])
apply auto
done

```

— list of fields of a class, including inherited and hidden ones

```

defs fields_def: "fields ≡ λ(G,C). class_rec (G,C) [] (λC fs ms ts.
  map (λ(fn,ft). ((fn,C),ft)) fs @ ts)"

```

```

lemma fields_rec_lemma: "[|class G C = Some (D,fs,ms); wf ((subcls1 G)^-1)|] ==>
  fields (G,C) =
  map (λ(fn,ft). ((fn,C),ft)) fs @ (if C = Object then [] else fields (G,D))"
apply (unfold fields_def)
apply (simp split del: split_if)
apply (erule (1) class_rec_lemma [THEN trans])
apply auto
done

```

```

defs field_def: "field == map_of o (map (λ((fn,fd),ft). (fn,(fd,ft)))) o fields"

```

```

lemma field_fields:
"field (G,C) fn = Some (fd, fT) ==> map_of (fields (G,C)) (fn, fd) = Some fT"
apply (unfold field_def)
apply (rule table_of_remap_SomeD)
apply simp
done

```

```

lemma widen_PrimT [simp]:
"G ⊢ T ≤ PrimT T' = (T = PrimT T)"
by (rule, erule widen.elims) auto

```

```

lemma widen_PrimT_RefT [iff]: "(G⊢PrimT pT≤RefT rT) = False"
apply (rule iffI)
apply (erule widen.elims)
apply auto
done

```

```

lemma widen_RefT: "G⊢RefT R≤T ==> ∃t. T=RefT t"
apply (ind_cases "G⊢S≤T")
apply auto
done

```

```

lemma widen_RefT2: "G⊢S≤RefT R ==> ∃t. S=RefT t"
apply (ind_cases "G⊢S≤T")
apply auto
done

```

```
lemma widen_Class: "G⊢Class C⊑T ==> ∃D. T=Class D"
```

```
apply (ind_cases "G⊢S⊑T")
apply auto
done
```

```
lemma widen_Class_NullT [iff]: "(G⊢Class C⊑NT) = False"
```

```
apply (rule iffI)
apply (ind_cases "G⊢S⊑T")
apply auto
done
```

```
lemma widen_Class_Class [iff]: "(G⊢Class C⊑ Class D) = (G⊢C⊑C D)"
```

```
apply (rule iffI)
apply (ind_cases "G⊢S⊑T")
apply (auto elim: widen.subcls)
done
```

```
lemma widen_Array: "G⊢S.[]⊑T ==> T=Class Object ∨ (∃T'. T=T'.[] ∧ G⊢S⊑T)"
```

```
apply (ind_cases "G⊢S⊑T")
by auto
```

```
lemma widen_Array2: "G⊢S⊑T.[] ==> S = NT ∨ (∃S'. S=S'.[] ∧ G⊢S'⊑T)"
```

```
apply (ind_cases "G⊢S⊑T")
by auto
```

```
lemma widen_ArrayPrimT: "G⊢PrimT t.[]⊑T ==> T=Class Object ∨ T=PrimT t.[]"
```

```
apply (ind_cases "G⊢S⊑T")
by auto
```

```
lemma widen_ArrayRefT:
```

```
"G⊢RefT t.[]⊑T ==> T=Class Object ∨ (∃s. T=RefT s.[] ∧ G⊢RefT t⊑RefT s)"
```

```
apply (ind_cases "G⊢S⊑T")
by auto
```

```
lemma widen_ArrayRefT_ArrayRefT_eq [simp]:
```

```
"G⊢RefT T.[]⊑RefT T'.[] = G⊢RefT T⊑RefT T'"
```

```
apply (rule iffI)
apply (drule widen_ArrayRefT)
apply simp
apply (erule widen.array)
done
```

```
lemma widen_Array_Array: "G⊢T.[]⊑T'.[] ==> G⊢T⊑T'"
```

```
apply (drule widen_Array)
apply auto
done
```

```
lemma widen_Array_Class: "G⊢S.[]⊑ Class C ==> C=Object"
```

```
by (auto dest: widen_Array)
```

```
lemma widen_Array_NT [iff]:
```

```
"G ⊢ T.[]⊑ NT = False"
```

```
by (auto elim: widen.elims)
```

```

theorem widen_trans [trans]: "[[G⊢S⊆U; G⊢U⊆T]] ⇒ G⊢S⊆T"
proof -
  assume "G⊢S⊆U" thus "∧T. G⊢U⊆T ⇒ G⊢S⊆T"
  proof induct
    case (refl T T') thus "G⊢T⊆T'" .
  next
    case (subcls C D T)
    then obtain E where "T = Class E" by (blast dest: widen_Class)
    with subcls show "G⊢Class C⊆T" by (auto elim: rtrancl_trans)
  next
    case (null R RT)
    then obtain rt where "RT = RefT rt" by (blast dest: widen_RefT)
    thus "G⊢NT⊆RT" by auto
  next
    case (arr_obj T T')
    hence "T' = Class Object"
      by (cases, auto) (erule converse_rtranclE, auto elim: subcls1.elims)
    thus "G ⊢ T.[] ⊆ T'" by simp
  next
    case (array S U T)
    have "G ⊢ RefT U.[] ⊆ T" .
    hence "T = Class Object ∨ (∃T'. T = RefT T'.[])" by cases auto
    with array show "G ⊢ RefT S.[] ⊆ T" by auto
  qed
qed
end

```

## 2.5 Java Values

theory Value = Type:

typeddecl loc\_ — locations, i.e. abstract references on objects

datatype loc  
 = XcptRef xcpt — special locations for pre-allocated system exceptions  
 / Loc loc\_ — usual locations (references on objects)

datatype val  
 = Unit — dummy result value of void methods  
 / Null — null reference  
 / Bool bool — Boolean value  
 / Intg int — integer value, name Intg instead of Int because of clash with HOL/Set.thy  
 / Addr loc — addresses, i.e. locations of objects  
 / RetAddr nat — return address of JSR instruction, for bytecode only

consts  
 the\_Bool :: "val => bool"  
 the\_Intg :: "val => int"  
 the\_Addr :: "val => loc"  
 the\_RetAddr :: "val => nat"

primrec  
 "the\_Bool (Bool b) = b"

primrec  
 "the\_Intg (Intg i) = i"

primrec  
 "the\_Addr (Addr a) = a"

primrec  
 "the\_RetAddr (RetAddr r) = r"

consts  
 defpval :: "prim\_ty => val" — default value for primitive types  
 default\_val :: "ty => val" — default value for all types

primrec  
 "defpval Void = Unit"  
 "defpval Boolean = Bool False"  
 "defpval Integer = Intg 0"  
 "defpval (RetA pc) = RetAddr pc"

primrec  
 "default\_val (PrimT pt) = defpval pt"  
 "default\_val (RefT r) = Null"

end

## 2.6 Program State

**theory** State = TypeRel + Value:

**types**

fields\_ = "(vname × cname ~> val)" — field name, defining class to value  
 entries\_ = "nat ~> val" — array index to value

**datatype**

heap\_entry = Obj cname fields\_ — class instance with class name and fields  
 | Arr ty nat entries\_ — array with component type, length, and entries

**constdefs**

obj\_ty :: "heap\_entry => ty"  
 "obj\_ty entry ≡ case entry of Obj C fs => Class C | Arr T len entries => T.[]"  
  
 init\_vars :: "('a × ty) list => ('a ~> val)"  
 "init\_vars ≡ map\_of ∘ map (λ(n,T). (n,default\_val T))"

**consts**

the\_obj :: "heap\_entry => cname × fields\_"  
 the\_arr :: "heap\_entry => ty × nat × entries\_"

**recdef** the\_obj "{}"

"the\_obj (Obj C fs) = (C,fs)"

**recdef** the\_arr "{}"

"the\_arr (Arr T len entries) = (T,len,entries)"

**types** aheap = "loc ~> heap\_entry" — "heap" used in a translation below

locals = "vname ~> val" — simple state, i.e. variable contents  
 state = "aheap × locals" — heap, local parameter including This  
 xstate = "xcpt option × state" — state including exception information

**syntax**

heap :: "state => aheap"  
 locals :: "state => locals"  
 Norm :: "state => xstate"

**translations**

"heap" => "fst"  
 "locals" => "snd"  
 "Norm s" == "(None,s)"

**constdefs**

new\_Addr :: "aheap => loc × xcpt option"  
 "new\_Addr h == SOME (a,x). (h a = None ∧ x = None) | x = Some OutOfMemory"  
  
 raise\_if :: "bool => xcpt => xcpt option => xcpt option"  
 "raise\_if c x xo == if c ∧ (xo = None) then Some x else xo"  
  
 np :: "val => xcpt option => xcpt option"  
 "np v == raise\_if (v = Null) NullPointer"

```

c_hupd  :: "aheap => xstate => xstate"
"c_hupd h' ==  $\lambda(xo, (h, l)). \text{if } xo = \text{None then } (\text{None}, (h', l)) \text{ else } (xo, (h, l))"$ 

```

```

cast_ok :: "'c prog => cname => aheap => val => bool"
"cast_ok G C h v == v = Null  $\vee G \vdash \text{obj\_ty } (\text{the } (h \text{ (the\_Addr } v))) \preceq \text{Class } C"$ 

```

```

lemma obj_ty_def2 [simp]: "obj_ty (Obj C fs) = Class C"
apply (unfold obj_ty_def)
apply (simp (no_asm))
done

```

```

lemma obj_ty_def3 [simp]: "obj_ty (Arr T len entries) = T.[]"
  by (unfold obj_ty_def) simp

```

```

lemma new_AddrD:
"(a,x) = new_Addr h  $\implies h \ a = \text{None} \wedge x = \text{None} \mid x = \text{Some } \text{OutOfMemory}"$ 
apply (unfold new_Addr_def)
apply (simp add: Pair_fst_snd_eq Eps_split)
apply (rule someI)
apply (rule disjI2)
apply (rule_tac "r" = "snd (?a, Some OutOfMemory)" in trans)
apply auto
done

```

```

lemma raise_if_True [simp]: "raise_if True x y  $\neq \text{None}"$ 
apply (unfold raise_if_def)
apply auto
done

```

```

lemma raise_if_False [simp]: "raise_if False x y = y"
apply (unfold raise_if_def)
apply auto
done

```

```

lemma raise_if_Some [simp]: "raise_if c x (Some y)  $\neq \text{None}"$ 
apply (unfold raise_if_def)
apply auto
done

```

```

lemma raise_if_Some2 [simp]:
  "raise_if c z (if x = None then Some y else x)  $\neq \text{None}"$ 
apply (unfold raise_if_def)
apply (induct_tac "x")
apply auto
done

```

```

lemma raise_if_SomeD [rule_format (no_asm)]:
  "raise_if c x y = Some z  $\implies c \wedge \text{Some } z = \text{Some } x \mid y = \text{Some } z"$ 
apply (unfold raise_if_def)
apply auto
done

```

```

lemma raise_if_NoneD [rule_format (no_asm)]:
  "raise_if c x y = None --> ¬ c ∧ y = None"
apply (unfold raise_if_def)
apply auto
done

lemma np_NoneD [rule_format (no_asm)]:
  "np a' x' = None --> x' = None ∧ a' ≠ Null"
apply (unfold np_def raise_if_def)
apply auto
done

lemma np_None [rule_format (no_asm), simp]: "a' ≠ Null --> np a' x' = x'"
apply (unfold np_def raise_if_def)
apply auto
done

lemma np_Some [simp]: "np a' (Some xc) = Some xc"
apply (unfold np_def raise_if_def)
apply auto
done

lemma np_Null [simp]: "np Null None = Some NullPointer"
apply (unfold np_def raise_if_def)
apply auto
done

lemma np_Addr [simp]: "np (Addr a) None = None"
apply (unfold np_def raise_if_def)
apply auto
done

lemma np_raise_if [simp]: "(np Null (raise_if c xc None)) =
  Some (if c then xc else NullPointer)"
apply (unfold raise_if_def)
apply (simp (no_asm))
done

end

```



## 2.7 Expressions and Statements

theory Term = Value:

**datatype** binop = Eq | Add — function codes for binary operation

**datatype** expr

- = NewC cname — class instance creation
- | Cast cname expr — type cast
- | Lit val — literal value, also references
- | BinOp binop expr expr — binary operation
- | LAcc vname — local (incl. parameter) access
- | LAss vname expr ("\_ :=\_" [90,90]90) — local assign
- | FAcc cname expr vname ("{ }\_ . \_" [10,90,99]90) — field access
- | FAss cname expr vname  
expr ("{ }\_ . \_ :=\_" [10,90,99,90]90) — field ass.
- | Call cname expr mname  
"ty list" "expr list" ("{ }\_ . \_ '( { }\_ )'" [10,90,99,10,10] 90) — method call

**datatype** stmt

- = Skip — empty statement
- | Expr expr — expression statement
- | Comp stmt stmt ("\_ ; \_" [61,60]60)
- | Cond expr stmt stmt ("If '(\_)' \_ Else \_" [80,79,79]70)
- | Loop expr stmt ("While '(\_)' \_" [80,79]70)

end

## 2.8 System Classes

**theory** *SystemClasses* = Decl:

This theory provides definitions for the *Object* class, and the system exceptions. It leaves methods empty (to be instantiated later).

**constdefs**

```

ObjectC_decl :: "'c mdecl list ⇒ 'c cdecl"
"ObjectC_decl ms ≡ (Object, (arbitrary, [],ms))"

NullPointerC_decl :: "'c mdecl list ⇒ 'c cdecl"
"NullPointerC_decl ms ≡ (Xcpt NullPointer, (Object, [],ms))"

ClassCastC_decl :: "'c mdecl list ⇒ 'c cdecl"
"ClassCastC_decl ms ≡ (Xcpt ClassCast, (Object, [],ms))"

OutOfMemoryC_decl :: "'c mdecl list ⇒ 'c cdecl"
"OutOfMemoryC_decl ms ≡ (Xcpt OutOfMemory, (Object, [],ms))"

ArrayIndexOutOfBoundsC_decl :: "'c mdecl list ⇒ 'c cdecl"
"ArrayIndexOutOfBoundsC_decl ms ≡ (Xcpt ArrayIndexOutOfBounds, (Object, [],ms))"

NegativeArraySizeC_decl :: "'c mdecl list ⇒ 'c cdecl"
"NegativeArraySizeC_decl ms ≡ (Xcpt NegativeArraySize, (Object, [],ms))"

ArrayStoreC_decl :: "'c mdecl list ⇒ 'c cdecl"
"ArrayStoreC_decl ms ≡ (Xcpt ArrayStore, (Object, [],ms))"

SystemClasses :: "cname list"
"SystemClasses ≡ [Object, Xcpt NullPointer, Xcpt ClassCast, Xcpt OutOfMemory,
                  Xcpt ArrayIndexOutOfBounds, Xcpt NegativeArraySize,
                  Xcpt ArrayStore]"

```

```

lemmas SystemClass_decl_defs = ObjectC_decl_def NullPointerC_decl_def
                                ClassCastC_decl_def OutOfMemoryC_decl_def
                                ArrayIndexOutOfBoundsC_decl_def
                                NegativeArraySizeC_decl_def
                                ArrayStoreC_decl_def

```

**end**

## 2.9 Well-formedness of Java programs

theory WellForm = TypeRel + SystemClasses:

for static checks on expressions and statements, see WellType.

**improvements over Java Specification 1.0 (cf. 8.4.6.3, 8.4.6.4, 9.4.1):**

- a method implementing or overwriting another method may have a result type that widens to the result type of the other method (instead of identical type)

**simplifications:**

- for uniformity, Object is assumed to be declared like any other class

```
types 'c wf_mb = "'c prog => cname => 'c mdecl => bool"
```

constdefs

```
max_dim :: nat
"max_dim ≡ 255" — c.f. JVMSpec 4.8.1
```

consts

```
dim :: "ty ⇒ nat"
dim_r :: "ref_ty ⇒ nat"
```

primrec

```
"dim (PrimT T) = 0"
"dim (RefT T) = dim_r T"
"dim_r (NullT) = 0"
"dim_r (ClassT C) = 0"
"dim_r (ArrayT T) = Suc (dim T)"
```

consts

```
noRA :: "ty ⇒ bool"
noRA_r :: "ref_ty ⇒ bool"
```

primrec

```
"noRA (PrimT T) = (¬is_RA (PrimT T))"
"noRA (RefT T) = noRA_r T"
"noRA_r (NullT) = True"
"noRA_r (ClassT C) = True"
"noRA_r (ArrayT T) = noRA T"
```

constdefs

```
wf_fdecl :: "'c prog => fdecl => bool"
"wf_fdecl G == λ(fn,ft). is_type G ft ∧ noRA ft ∧ dim ft < max_dim"
```

```
wf_mhead :: "'c prog => sig => ty => bool"
```

```
"wf_mhead G == λ(mn,pTs) rT. (∀T∈set pTs. is_type G T ∧ noRA T ∧ dim T < max_dim) ∧
is_type G rT ∧ noRA rT ∧ dim rT < max_dim"
```

```
wf_mdecl :: "'c wf_mb => 'c wf_mb"
```

```

"wf_mdecl wf_mb G C ==  $\lambda(\text{sig}, rT, \text{mb}). \text{wf\_mhead } G \text{ sig } rT \wedge \text{wf\_mb } G \text{ C } (\text{sig}, rT, \text{mb})"$ 

wf_cdecl :: "'c wf_mb => 'c prog => 'c cdecl => bool"
"wf_cdecl wf_mb G ==
   $\lambda(C, (D, fs, ms)).$ 
  ( $\forall f \in \text{set } fs. \text{wf\_fdecl } G \quad f$ )  $\wedge$  unique fs  $\wedge$ 
  ( $\forall m \in \text{set } ms. \text{wf\_mdecl } wf\_mb \text{ G C } m$ )  $\wedge$  unique ms  $\wedge$ 
  ( $C \neq \text{Object} \longrightarrow \text{is\_class } G \text{ D} \wedge \neg G \vdash D \preceq C \text{ C} \wedge$ 
    ( $\forall (\text{sig}, rT, b) \in \text{set } ms. \forall D' \text{ rT}' \text{ b}'.$ 
       $\text{method}(G, D) \text{ sig} = \text{Some}(D', rT', b') \longrightarrow G \vdash rT' \preceq rT'$ ))"

wf_syscls :: "'c prog => bool"
"wf_syscls G == set SystemClasses  $\subseteq$  fst ' (set G)  $\wedge$  fields (G, Object) = []"

wf_prog :: "'c wf_mb => 'c prog => bool"
"wf_prog wf_mb G ==
  let cs = set G in wf_syscls G  $\wedge$  ( $\forall c \in cs. \text{wf\_cdecl } wf\_mb \text{ G } c$ )  $\wedge$  unique G"

lemma class_wf:
  "[|class G C = Some c; wf_prog wf_mb G|] ==> wf_cdecl wf_mb G (C, c)"
apply (unfold wf_prog_def class_def)
apply (simp)
apply (fast dest: map_of_SomeD)
done

lemma class_Object [simp]:
  "wf_prog wf_mb G ==>  $\exists X \text{ fs } ms. \text{class } G \text{ Object} = \text{Some } (X, fs, ms)"$ 
apply (unfold wf_prog_def wf_syscls_def class_def SystemClasses_def)
apply (auto simp: map_of_SomeI)
done

lemma no_fields_Object [simp]:
  "wf_prog wf_mb G  $\implies$  fields (G, Object) = []"
apply (unfold wf_prog_def wf_syscls_def class_def SystemClasses_def)
apply simp
done

lemma no_field_Object [simp]:
  "wf_prog wf_mb G  $\implies$  field (G, Object) F = None"
by (simp add: field_def)

lemma is_class_Object [simp]: "wf_prog wf_mb G ==> is_class G Object"
apply (unfold is_class_def)
apply (simp (no_asm_simp))
done

lemma is_class_xcpt [simp]: "wf_prog wf_mb G  $\implies$  is_class G (Xcpt x)"
apply (unfold wf_prog_def wf_syscls_def)
apply (simp add: is_class_def class_def SystemClasses_def)
apply clarify
apply (cases x)
apply (auto intro!: map_of_SomeI)
done

```

```

lemma subcls1_wfD: "[|G ⊢ C < C1D; wf_prog wf_mb G|] ==> D ≠ C ∧ ¬(D,C) ∈ (subcls1 G)^+"
apply( frule r_into_trancl)
apply( drule subcls1D)
apply( clarify)
apply( drule (1) class_wf)
apply( unfold wf_cdecl_def)
apply( force simp add: reflcl_trancl [THEN sym] simp del: reflcl_trancl)
done

```

```

lemma wf_cdecl_supD:
"!!r. [[wf_cdecl wf_mb G (C,D,r); C ≠ Object]] ==> is_class G D"
apply (unfold wf_cdecl_def)
apply (auto split add: option.split_asm)
done

```

```

lemma subcls_asym: "[|wf_prog wf_mb G; (C,D) ∈ (subcls1 G)^+|] ==> ¬(D,C) ∈ (subcls1 G)^+"
apply(erule tranclE)
apply(fast dest!: subcls1_wfD )
apply(fast dest!: subcls1_wfD intro: trancl_trans)
done

```

```

lemma subcls_irrefl: "[|wf_prog wf_mb G; (C,D) ∈ (subcls1 G)^+|] ==> C ≠ D"
apply (erule trancl_trans_induct)
apply (auto dest: subcls1_wfD subcls_asym)
done

```

```

lemma acyclic_subcls1: "wf_prog wf_mb G ==> acyclic (subcls1 G)"
apply (unfold acyclic_def)
apply (fast dest: subcls_irrefl)
done

```

```

lemma wf_subcls1: "wf_prog wf_mb G ==> wf ((subcls1 G)^-1)"
apply (rule finite_acyclic_wf)
apply ( subst finite_converse)
apply ( rule finite_subcls1)
apply (subst acyclic_converse)
apply (erule acyclic_subcls1)
done

```

```

lemma subcls_induct:
"[|wf_prog wf_mb G; !!C. ∀D. (C,D) ∈ (subcls1 G)^+ --> P D ==> P C|] ==> P C"
(is "?A ==> PROP ?P ==> _")
proof -
  assume p: "PROP ?P"
  assume ?A thus ?thesis apply -
apply(drule wf_subcls1)
apply(drule wf_trancl)
apply(simp only: trancl_converse)
apply(erule_tac a = C in wf_induct)
apply(rule p)
apply(auto)
done
qed

```

```

lemma subcls1_induct:
  "[[is_class G C; wf_prog wf_mb G; P Object;
    !!C D fs ms. [|C ≠ Object; is_class G C; class G C = Some (D,fs,ms) ∧
      wf_cdecl wf_mb G (C,D,fs,ms) ∧ G⊢C<C1D ∧ is_class G D ∧ P D|] ==> P C
  |] ==> P C"
  (is "?A ==> ?B ==> ?C ==> PROP ?P ==> _")
proof -
  assume p: "PROP ?P"
  assume ?A ?B ?C thus ?thesis apply -
apply(unfold is_class_def)
apply( rule impE)
prefer 2
apply( assumption)
prefer 2
apply( assumption)
apply(erule thin_rl)
apply( rule subcls1_induct)
apply( assumption)
apply( rule impI)
apply( case_tac "C = Object")
apply( fast)
apply safe
apply( frule (1) class_wf)
apply( frule (1) wf_cdecl_supD)

apply( subgoal_tac "G⊢C<C1a")
apply(erule_tac [2] subcls1I)
apply( rule p)
apply(unfold is_class_def)
apply auto
done
qed

lemmas method_rec = wf_subcls1 [THEN [2] method_rec_lemma]

lemmas fields_rec = wf_subcls1 [THEN [2] fields_rec_lemma]

lemma method_Object [simp]:
  "method (G, Object) sig = Some (D, mh, code) ==> wf_prog wf_mb G ==> D = Object"
  apply( frule class_Object, clarify)
  apply( drule method_rec, assumption)
  apply( auto dest: map_of_SomeD)
  done

lemma subcls_C_Object: "[[is_class G C; wf_prog wf_mb G|] ==> G⊢C⊆C Object"
  apply(erule subcls1_induct)
  apply( assumption)
  apply( fast)
  apply(auto dest!: wf_cdecl_supD)
  apply(erule (1) converse_rtrancl_into_rtrancl)
  done

lemma is_type_rTI: "wf_mhead G sig rT ==> is_type G rT"
  apply( unfold wf_mhead_def)

```

```
apply auto
done
```

```
lemma no_RA_rT: "wf_mhead G sig rT ==> noRA rT"
apply (unfold wf_mhead_def)
apply auto
done
```

```
lemma widen_fields_defpl': "[|is_class G C; wf_prog wf_mb G|] ==>
  ∀((fn,fd),fT)∈set (fields (G,C)). G⊢C⊆C fd"
apply(erule subcls1_induct)
apply(assumption)
apply(frule class_Object)
apply(clarify)
apply(frule fields_rec, assumption)
apply(fastsimp)
apply(tactic "safe_tac HOL_cs")
apply(subst fields_rec)
apply(assumption)
apply(assumption)
apply(simp (no_asm) split del: split_if)
apply(rule ballI)
apply(simp (no_asm_simp) only: split_tupled_all)
apply(simp (no_asm))
apply(erule UnE)
apply(force)
apply(erule r_into_rtrancl [THEN rtrancl_trans])
apply auto
done
```

```
lemma widen_fields_defpl:
  "[|((fn,fd),fT) ∈ set (fields (G,C)); wf_prog wf_mb G; is_class G C|] ==>
  G⊢C⊆C fd"
apply(drule (1) widen_fields_defpl')
apply(fast)
done
```

```
lemma unique_fields:
  "[|is_class G C; wf_prog wf_mb G|] ==> unique (fields (G,C))"
apply(erule subcls1_induct)
apply(assumption)
apply(frule class_Object)
apply(clarify)
apply(frule fields_rec, assumption)
apply(drule class_wf, assumption)
apply(simp add: wf_cdecl_def)
apply(safe dest!: wf_cdecl_supD)
apply(drule subcls1_wfD)
apply(assumption)
apply(subst fields_rec)
apply auto
apply(rotate_tac -1)
apply(frule class_wf)
apply auto
```

```

apply( simp add: wf_cdecl_def)
apply( erule unique_append)
apply( rule unique_map_inj)
apply( clarsimp)
apply (rule injI)
apply( simp)
apply(auto dest!: widen_fields_defpl)
done

```

```

lemma fields_mono_lemma [rule_format (no_asm)]:
  "[|wf_prog wf_mb G; (C',C)∈(subcls1 G)^*|] ==>
   x ∈ set (fields (G,C)) --> x ∈ set (fields (G,C'))"
apply(erule converse_rtrancl_induct)
apply( safe dest!: subcls1D)
apply(subst fields_rec)
apply( auto)
done

```

```

lemma fields_mono:
  "[|map_of (fields (G,C)) fn = Some f; G⊢D⊆C C; is_class G D; wf_prog wf_mb G|]
   ==> map_of (fields (G,D)) fn = Some f"
apply (rule map_of_SomeI)
apply (erule (1) unique_fields)
apply (erule (1) fields_mono_lemma)
apply (erule map_of_SomeD)
done

```

```

lemma widen_cfs_fields:
  "[|field (G,C) fn = Some (fd, fT); G⊢D⊆C C; wf_prog wf_mb G|]==>
   map_of (fields (G,D)) (fn, fd) = Some fT"
apply (drule field_fields)
apply (drule rtranclD)
apply safe
apply (frule subcls_is_class)
apply (drule trancl_into_rtrancl)
apply (fast dest: fields_mono)
done

```

```

lemma method_wf_mdecl [rule_format (no_asm)]:
  "wf_prog wf_mb G ==> is_class G C ==>
   method (G,C) sig = Some (md,mh,m)
   --> G⊢C⊆C md ∧ wf_mdecl wf_mb G md (sig,(mh,m))"
apply( erule subcls1_induct)
apply( assumption)
apply( clarify)
apply( frule class_Object)
apply( clarify)
apply( frule method_rec, assumption)
apply( drule class_wf, assumption)
apply( simp add: wf_cdecl_def)
apply( drule map_of_SomeD)
apply( subgoal_tac "md = Object")
apply( fastsimp)
apply( fastsimp)

```



```

apply( clarify)
apply( frule_tac C = C in method_rec)
apply( assumption)
apply( rotate_tac -1)
apply( simp)
apply( drule override_SomeD)
apply( erule disjE)
apply( erule_tac V = "?P --> ?Q" in thin_rl)
apply( frule map_of_SomeD)
apply( clarsimp simp add: wf_cdecl_def)
apply( clarify)
apply( rule rtrancl_trans)
prefer 2
apply( assumption)
apply( rule r_into_rtrancl)
apply( fast intro: subcls1I)
done

lemma subcls_widen_methd [rule_format (no_asm)]:
  "[|G⊢T≤C T'; wf_prog wf_mb G|] ==>
    ∀D rT b. method (G,T') sig = Some (D,rT ,b) -->
    (∃D' rT' b'. method (G,T) sig = Some (D',rT',b') ∧ G⊢rT'≤rT)"
apply( drule rtranclD)
apply( erule disjE)
apply( fast)
apply( erule conjE)
apply( erule trancl_trans_induct)
prefer 2
apply( clarify)
apply( drule spec, drule spec, drule spec, erule (1) impE)
apply( fast elim: widen_trans)
apply( clarify)
apply( drule subcls1D)
apply( clarify)
apply( subst method_rec)
apply( assumption)
apply( unfold override_def)
apply( simp (no_asm_simp) del: split_paired_Ex)
apply( case_tac "∃z. map_of(map (λ(s,m). (s, ?C, m)) ms) sig = Some z")
apply( erule exE)
apply( rotate_tac -1, frule ssubst, erule_tac [2] asm_rl)
prefer 2
apply( rotate_tac -1, frule ssubst, erule_tac [2] asm_rl)
apply( tactic "asm_full_simp_tac (HOL_ss addsimps [not_None_eq RS sym]) 1")
apply( simp_all (no_asm_simp) del: split_paired_Ex)
apply( drule (1) class_wf)
apply( simp (no_asm_simp) only: split_tupled_all)
apply( unfold wf_cdecl_def)
apply( drule map_of_SomeD)
apply auto
done

lemma subtype_widen_methd:
  "[| G⊢ C≤C D; wf_prog wf_mb G;

```

```

    method (G,D) sig = Some (md, rT, b) |]
  ==> ∃ mD' rT' b'. method (G,C) sig= Some(mD',rT',b') ∧ G⊢rT' ≲rT"
apply(auto dest: subcls_widen_methd method_wf_mdecl
      simp add: wf_mdecl_def wf_mhead_def split_def)
done

lemma method_in_md [rule_format (no_asm)]:
  "wf_prog wf_mb G ==> is_class G C ==> ∀D. method (G,C) sig = Some(D,mh,code)
  --> is_class G D ∧ method (G,D) sig = Some(D,mh,code)"
apply (erule (1) subcls1_induct)
apply clarify
apply (frule method_Object, assumption)
apply hypsubst
apply simp
apply (erule conjE)
apply (subst method_rec)
  apply (assumption)
  apply (assumption)
apply (clarify)
apply (erule_tac "x" = "Da" in allE)
apply (clarsimp)
  apply (simp add: map_of_map)
  apply (clarify)
  apply (subst method_rec)
    apply (assumption)
    apply (assumption)
  apply (simp add: override_def map_of_map split add: option.split)
done

lemma widen_methd:
  "[| method (G,C) sig = Some (md,rT,b); wf_prog wf_mb G; G⊢T'' ≲C C|]
  ==> ∃ md' rT' b'. method (G,T'') sig = Some (md',rT',b') ∧ G⊢rT' ≲rT"
apply( drule subcls_widen_methd)
apply auto
done

lemma Call_lemma:
  "[|method (G,C) sig = Some (md,rT,b); G⊢T'' ≲C C; wf_prog wf_mb G;
  class G C = Some y|] ==> ∃T' rT' b. method (G,T'') sig = Some (T',rT',b) ∧
  G⊢rT' ≲rT ∧ G⊢T'' ≲C T' ∧ wf_mhead G sig rT' ∧ wf_mb G T' (sig,rT',b)"
apply( drule (2) widen_methd)
apply( clarify)
apply( frule subcls_is_class2)
apply (unfold is_class_def)
apply (simp (no_asm_simp))
apply( drule method_wf_mdecl)
apply( unfold wf_mdecl_def)
apply( unfold is_class_def)
apply auto
done

lemma fields_is_type_lemma [rule_format (no_asm)]:
  "[|is_class G C; wf_prog wf_mb G|] ==>

```

```

   $\forall f \in \text{set } (\text{fields } (G,C)). \text{is\_type } G \text{ (snd } f) \wedge \text{noRA } (\text{snd } f) \wedge \text{dim } (\text{snd } f) < \text{max\_dim}$ 
  apply(erule (1) subcls1_induct)
  apply(erule class_Object)
  apply(erule clarify)
  apply(erule fields_rec, assumption)
  apply(erule class_wf, assumption)
  apply(simp add: wf_cdecl_def wf_fdecl_def)
  apply(subst fields_rec)
  apply(erule fast)
  apply(erule assumption)
  apply(erule clarsimp)
  apply(erule safe)
  prefer 4
  apply(erule force)
  apply(erule (1) class_wf)
  apply(erule unfold wf_cdecl_def)
  apply(erule clarsimp)
  apply(erule (1) bspec)
  apply(erule unfold wf_fdecl_def)
  apply auto
done

```

```

lemma fields_is_type:
  "[|map_of (fields (G,C)) fn = Some f; wf_prog wf_mb G; is_class G C|] ==>
  is_type G f"
  apply(drule map_of_SomeD)
  apply(drule (2) fields_is_type_lemma)
  apply(auto)
done

```

```

lemma fields_no_RA:
  "[|map_of (fields (G,C)) fn = Some f; wf_prog wf_mb G; is_class G C|] ==>
  noRA f"
  apply(drule map_of_SomeD)
  apply(drule (2) fields_is_type_lemma)
  apply(auto)
done

```

```

lemma fields_dim:
  "[|map_of (fields (G,C)) fn = Some f; wf_prog wf_mb G; is_class G C|] ==>
  dim f < max_dim"
  apply(drule map_of_SomeD)
  apply(drule (2) fields_is_type_lemma)
  apply(auto)
done

```

```

lemma methd:
  "[| wf_prog wf_mb G; (C,S,fs,mdecls)  $\in$  set G; (sig,rT,code)  $\in$  set mdecls |]
  ==> method (G,C) sig = Some(C,rT,code)  $\wedge$  is_class G C"
proof -
  assume wf: "wf_prog wf_mb G" and C: "(C,S,fs,mdecls)  $\in$  set G" and
    m: "(sig,rT,code)  $\in$  set mdecls"
  moreover
  from wf C have "class G C = Some (S,fs,mdecls)"

```

```

    by (auto simp add: wf_prog_def class_def is_class_def intro: map_of_SomeI)
  moreover
  from wf C
  have "unique mdecls" by (unfold wf_prog_def wf_cdecl_def) auto
  hence "unique (map (\(s,m). (s,C,m)) mdecls)" by (induct mdecls, auto)
  with m
  have "map_of (map (\(s,m). (s,C,m)) mdecls) sig = Some (C,rT,code)"
    by (force intro: map_of_SomeI)
  ultimately
  show ?thesis by (auto simp add: is_class_def dest: method_rec)
qed

```

lemma wf\_mb'E:

```

  "[ wf_prog wf_mb G;  $\bigwedge C S fs ms m. [(C,S,fs,ms) \in \text{set } G; m \in \text{set } ms] \implies \text{wf\_mb}' G C m$  ]
]
 $\implies \text{wf\_prog wf\_mb}' G$ "
  apply (simp add: wf_prog_def)
  apply auto
  apply (simp add: wf_cdecl_def wf_mdecl_def)
  apply safe
  apply (drule bspec, assumption) apply simp
  apply (drule bspec, assumption) apply simp
  apply (drule bspec, assumption) apply simp
  apply clarify apply (drule bspec, assumption) apply simp
  apply (drule bspec, assumption) apply simp
  apply (drule bspec, assumption) apply simp
  apply (drule bspec, assumption) apply simp
  apply (drule bspec, assumption) apply simp
  apply (drule bspec, assumption) apply simp
  apply (drule bspec, assumption) apply simp
  apply clarify apply (drule bspec, assumption)+ apply simp
done

```

end

## 2.10 Well-typedness Constraints

**theory WellType = Term + WellForm:**

the formulation of well-typedness of method calls given below (as well as the Java Specification 1.0) is a little too restrictive: It does not allow methods of class Object to be called upon references of interface type.

**simplifications:**

- the type rules include all static checks on expressions and statements, e.g. definedness of names (of parameters, locals, fields, methods)

local variables, including method parameters and This:

**types**

```
lenv    = "vname ~> ty"
'c env  = "'c prog × lenv"
```

**syntax**

```
prg     :: "'c env => 'c prog"
localT  :: "'c env => (vname ~> ty)"
```

**translations**

```
"prg"   => "fst"
"localT" => "snd"
```

**consts**

```
more_spec :: "'c prog => (ty × 'x) × ty list =>
              (ty × 'x) × ty list => bool"
appl_methds :: "'c prog => cname => sig => ((ty × ty) × ty list) set"
max_spec   :: "'c prog => cname => sig => ((ty × ty) × ty list) set"
```

**defs**

```
more_spec_def: "more_spec G == λ((d,h),pTs). λ((d',h'),pTs'). G ⊢ d ≤ d' ∧
                  list_all2 (λT T'. G ⊢ T ≤ T') pTs pTs'"
```

— applicable methods, cf. 15.11.2.1

```
appl_methds_def: "appl_methds G C == λ(mn, pTs).
                  {((Class md,rT),pTs') | md rT mb pTs'.
                    method (G,C) (mn, pTs') = Some (md,rT,mb) ∧
                    list_all2 (λT T'. G ⊢ T ≤ T') pTs pTs'}"
```

— maximally specific methods, cf. 15.11.2.2

```
max_spec_def: "max_spec G C sig == {m. m ∈ appl_methds G C sig ∧
                                       (∀ m' ∈ appl_methds G C sig.
                                        more_spec G m' m --> m' = m)}"
```

**lemma max\_spec2appl\_methds:**

```
"x ∈ max_spec G C sig ==> x ∈ appl_methds G C sig"
```

**apply** (unfold max\_spec\_def)

**apply** (fast)

**done**

```

lemma appl_methsD:
  "((md,rT),pTs')∈appl_methds G C (mn, pTs) ==>
   ∃D b. md = Class D ∧ method (G,C) (mn, pTs') = Some (D,rT,b)
   ∧ list_all2 (λT T'. G⊢T≲T') pTs pTs'"
apply (unfold appl_methds_def)
apply (fast)
done

lemmas max_spec2mheads = insertI1 [THEN [2] equalityD2 [THEN subsetD],
  THEN max_spec2appl_meths, THEN appl_methsD]

consts
  typeof :: "(loc => ty option) => val => ty option"

primrec
  "typeof dt Unit      = Some (PrimT Void)"
  "typeof dt Null      = Some NT"
  "typeof dt (Bool b) = Some (PrimT Boolean)"
  "typeof dt (Intg i) = Some (PrimT Integer)"
  "typeof dt (Addr a) = dt a"
  "typeof dt (RetAddr pc) = Some (RA pc)"

lemma is_type_typeof [rule_format (no_asm), simp]:
  "(∀a. v ≠ Addr a --> (∃T. typeof t v = Some T ∧ is_type G T))"
apply (rule val.induct)
apply auto
done

lemma typeof_empty_is_type [rule_format (no_asm)]:
  "typeof (λa. None) v = Some T → is_type G T"
apply (rule val.induct)
apply auto
done

types
  java_mb = "vname list × (vname × ty) list × stmt × expr"
— method body with parameter names, local variables, block, result expression.
— local variables might include This, which is hidden anyway

consts
  ty_expr  :: "java_mb env => (expr      × ty      ) set"
  ty_exprs:: "java_mb env => (expr list × ty list) set"
  wt_stmt  :: "java_mb env => stmt      set"

syntax (xsymbols)
  ty_expr  :: "java_mb env => [expr      , ty      ] => bool" ("_ ⊢ _ :: _" [51,51,51]50)
  ty_exprs:: "java_mb env => [expr list, ty list] => bool" ("_ ⊢ _ [::] _" [51,51,51]50)
  wt_stmt  :: "java_mb env => stmt      => bool" ("_ ⊢ _ √" [51,51 ]50)

syntax
  ty_expr  :: "java_mb env => [expr      , ty      ] => bool" ("_ ⊢- _ :: _" [51,51,51]50)
  ty_exprs:: "java_mb env => [expr list, ty list] => bool" ("_ ⊢- _ [::] _" [51,51,51]50)

```

```
wt_stmt :: "java_mb env => stmt => bool" ("_ |- _ [ok]" [51,51 ]50)
```

### translations

```
"E⊢e :: T" == "(e,T) ∈ ty_expr E"
"E⊢e[::]T" == "(e,T) ∈ ty_exprs E"
"E⊢c √" == "c ∈ wt_stmt E"
```

### inductive "ty\_expr E" "ty\_exprs E" "wt\_stmt E" intros

```
NewC: "[| is_class (prg E) C |] ==>
  E⊢NewC C::Class C" — cf. 15.8
```

— cf. 15.15

```
Cast: "[| E⊢e::Class C; is_class (prg E) D;
  prg E⊢C⊆? D |] ==>
  E⊢Cast D e::Class D"
```

— cf. 15.7.1

```
Lit: "[| typeof (λv. None) x = Some T |] ==>
  E⊢Lit x::T"
```

— cf. 15.13.1

```
LAcc: "[| localT E v = Some T; is_type (prg E) T |] ==>
  E⊢LAcc v::T"
```

```
BinOp: "[| E⊢e1::T;
  E⊢e2::T;
  if bop = Eq then T' = PrimT Boolean
  else T' = T ∧ T = PrimT Integer |] ==>
  E⊢BinOp bop e1 e2::T'"
```

— cf. 15.25, 15.25.1

```
LAss: "[| v ~ = This;
  E⊢LAcc v::T;
  E⊢e::T';
  prg E⊢T'⊆T |] ==>
  E⊢v:=e::T'"
```

— cf. 15.10.1

```
FAcc: "[| E⊢a::Class C;
  field (prg E,C) fn = Some (fd,fT) |] ==>
  E⊢{fd}a..fn::fT"
```

— cf. 15.25, 15.25.1

```
FAss: "[| E⊢{fd}a..fn::T;
  E⊢v ::T';
  prg E⊢T'⊆T |] ==>
  E⊢{fd}a..fn:=v::T'"
```

— cf. 15.11.1, 15.11.2, 15.11.3

```
Call: "[| E⊢a::Class C;
```

```

    E⊢ps[::]pTs;
    max_spec (prg E) C (mn, pTs) = {((md,rT),pTs')} |] ==>
    E⊢{C}a..mn({pTs'}ps)::rT"

```

— well-typed expression lists

```

— cf. 15.11.???
Nil: "E⊢[][::][]"

```

```

— cf. 15.11.???
Cons:"[| E⊢e::T;
      E⊢es[::]Ts |] ==>
      E⊢e#es[::]T#Ts"

```

— well-typed statements

```

Skip:"E⊢Skip√"

```

```

Expr:"[| E⊢e::T |] ==>
      E⊢Expr e√"

```

```

Comp:"[| E⊢s1√;
        E⊢s2√ |] ==>
      E⊢s1;; s2√"

```

```

— cf. 14.8
Cond:"[| E⊢e::PrimT Boolean;
        E⊢s1√;
        E⊢s2√ |] ==>
      E⊢If(e) s1 Else s2√"

```

```

— cf. 14.10
Loop:"[| E⊢e::PrimT Boolean;
        E⊢s√ |] ==>
      E⊢While(e) s√"

```

**constdefs**

```

wf_java_mdecl :: "java_mb prog => cname => java_mb mdecl => bool"
"wf_java_mdecl G C == λ((mn,pTs),rT,(pns,lvars,blk,res)).
  length pTs = length pns ∧
  distinct pns ∧
  unique lvars ∧
  This ∉ set pns ∧ This ∉ set (map fst lvars) ∧
  (∀pn∈set pns. map_of lvars pn = None) ∧
  (∀(vn,T)∈set lvars. is_type G T) &
  (let E = (G,map_of lvars(pns[↦]pTs)(This↦Class C)) in
   E⊢blk√ ∧ (∃T. E⊢res::T ∧ G⊢T≤rT))"

```

**syntax**

```

wf_java_prog :: "java_mb prog => bool"

```

**translations**

```

"wf_java_prog" == "wf_prog wf_java_mdecl"

```



```

lemma wt_is_type: "wf_prog wf_mb G  $\implies$  ((G,L) $\vdash$ e::T  $\longrightarrow$  is_type G T)  $\wedge$ 
  ((G,L) $\vdash$ es[::]Ts  $\longrightarrow$  Ball (set Ts) (is_type G))  $\wedge$  ((G,L) $\vdash$ c  $\surd$   $\longrightarrow$  True)"
apply (rule ty_expr_ty_exprs_wt_stmt.induct)
apply auto
apply (erule typeof_empty_is_type)
apply (simp split add: split_if_asm)
apply (drule field_fields)
apply (drule (1) fields_is_type)
apply (simp (no_asm_simp))
apply (assumption)
apply (auto dest!: max_spec2mheads method_wf_mdecl is_type_rTI
  simp add: wf_mdecl_def)
done

lemmas ty_expr_is_type = wt_is_type [THEN conjunct1, THEN mp, COMP swap_prem1]

end

```

## 2.11 Conformity Relations for Type Soundness Proof

theory Conform = State + WellType:

types 'c env\_ = "'c prog × (vname  $\rightsquigarrow$  ty)" — same as env of WellType.thy

constdefs

```

hext :: "aheap => aheap => bool" ("_ <=| _" [51,51] 50)
"h<=|h' ≡ ∀ a e. h a = Some e →
  (case e of Obj C fs ⇒ ∃ fs'. h' a = Some (Obj C fs')
   | Arr T l en ⇒ ∃ en'. h' a = Some (Arr T l en'))"

conf :: "'c prog => aheap => val => ty => bool"      ("_,_ |- _ ::<= _" [51,51,51,51]
50)
"G,h|-v::<=T == ∃ T'. typeof (option_map obj_ty o h) v = Some T' ∧ G⊢T'⊆T"

lconf :: "'c prog => aheap => ('a  $\rightsquigarrow$  val) => ('a  $\rightsquigarrow$  ty) => bool"
      ("_,_ |- _ [::<=] _" [51,51,51,51] 50)
"G,h|-vs[::<=]Ts == ∀ n T. Ts n = Some T --> (∃ v. vs n = Some v ∧ G,h|-v::<=T)"

oconf :: "'c prog => aheap => heap_entry => bool" ("_,_ |- _ [ok]" [51,51,51] 50)
"G,h|-entry [ok] ≡ case entry of Obj C fs => G,h|-fs[::<=]map_of (fields (G,C))
  | Arr T len entries => ∀ i < len. ∃ v. entries i = Some
v ∧ G,h|-v::<=T"

hconf :: "'c prog => aheap => bool" ("_ |-h _ [ok]" [51,51] 50)
"G|-h h [ok] == ∀ a obj. h a = Some obj --> G,h|-obj [ok]"

conforms :: "state => java_mb env_ => bool" ("_ ::<= _" [51,51] 50)
"s::<=E == prg E|-h heap s [ok] ∧ prg E,heap s|-locals s[::<=]localT E"

syntax (xsymbols)
hext      :: "aheap => aheap => bool"
          ("_ ≤| _" [51,51] 50)

conf      :: "'c prog => aheap => val => ty => bool"
          ("_,_ ⊢ _ ::≤ _" [51,51,51,51] 50)

lconf     :: "'c prog => aheap => ('a  $\rightsquigarrow$  val) => ('a  $\rightsquigarrow$  ty) => bool"
          ("_,_ ⊢ _ [::≤] _" [51,51,51,51] 50)

oconf     :: "'c prog => aheap => heap_entry => bool"
          ("_,_ ⊢ _ √" [51,51,51] 50)

hconf     :: "'c prog => aheap => bool"
          ("_ ⊢h _ √" [51,51] 50)

conforms  :: "state => java_mb env_ => bool"
          ("_ ::≤ _" [51,51] 50)

```

## 2.11.1 hext

```

lemma hextI:
  "(( $\bigwedge a C fs. h a = \text{Some } (\text{Obj } C fs) \implies \exists fs'. h' a = \text{Some } (\text{Obj } C fs')$ )  $\implies$ 
    ( $\bigwedge a T len en. h a = \text{Some } (\text{Arr } T len en) \implies \exists en'. h' a = \text{Some } (\text{Arr } T len en')$ )  $\implies$ 
     $h \leq |h'$ )"
apply (unfold hext_def)
apply clarify
apply (case_tac e)
apply auto
done

lemma hext_objD: "[| $h \leq |h'$ ;  $h a = \text{Some } (\text{Obj } C fs)$ |]  $\implies \exists fs'. h' a = \text{Some } (\text{Obj } C fs')$ "
apply (unfold hext_def)
apply (force)
done

lemma hext_arrD: "[| $h \leq |h'$ ;  $h a = \text{Some } (\text{Arr } T l en)$ |]  $\implies \exists en'. h' a = \text{Some } (\text{Arr } T l en')$ "
apply (unfold hext_def)
apply force
done

lemma hextE:
  " $h \leq |h' \implies h a = \text{Some } z \implies$ 
    ( $\bigwedge C fs fs'. z = \text{Obj } C fs \implies h' a = \text{Some } (\text{Obj } C fs') \implies P$ )  $\implies$ 
    ( $\bigwedge T l en en'. z = \text{Arr } T l en \implies h' a = \text{Some } (\text{Arr } T l en') \implies P$ )  $\implies$ 
    P"
apply (unfold hext_def)
apply (erule allE, erule allE, erule impE, assumption)
apply (cases z)
  apply clarsimp
  apply blast
apply clarsimp
apply blast
done

lemma hext_refl [simp]: " $h \leq |h$ "
apply (rule hextI)
apply fast+
done

lemma hext_new [simp]: " $h a = \text{None} \implies h \leq |h(a \mapsto x)$ "
apply (rule hextI)
apply auto
done

lemma hext_trans [trans]: " $h \leq |h' \implies h' \leq |h'' \implies h \leq |h''$ "
apply (rule hextI)
apply (fast dest: hext_objD)

```

```

apply (fast_dest: hext_arrD)
done

```

```

lemma hext_upd_obj: "h a = Some (Obj C fs)  $\implies$  h  $\leq$  |h(a  $\mapsto$  Obj C fs)"
apply (rule hextI)
apply auto
done

```

```

lemma hext_upd_arr: "h a = Some (Arr T l en)  $\implies$  h  $\leq$  |h(a  $\mapsto$  Arr T l en)"
apply (rule hextI)
apply auto
done

```

### 2.11.2 conf

```

lemma conf_Null [simp]: "G, h  $\vdash$  Null ::  $\preceq$ T = G  $\vdash$  RefT NullT  $\preceq$ T"
apply (unfold conf_def)
apply (simp (no_asm))
done

```

```

lemma conf_litval [rule_format (no_asm), simp]:
  "typeof ( $\lambda$ v. None) v = Some T  $\longrightarrow$  G, h  $\vdash$  v ::  $\preceq$ T"
apply (unfold conf_def)
apply (rule val.induct)
apply auto
done

```

```

lemma conf_AddrI: "[|h a = Some obj; G  $\vdash$  obj_ty obj  $\preceq$ T|]  $\implies$  G, h  $\vdash$  Addr a ::  $\preceq$ T"
apply (unfold conf_def)
apply (simp)
done

```

```

lemma conf_obj_AddrI: "[|h a = Some (Obj C fs); G  $\vdash$  C  $\preceq$ C D|]  $\implies$  G, h  $\vdash$  Addr a ::  $\preceq$  Class D"
apply (unfold conf_def)
apply (simp)
done

```

```

lemma defval_conf:
  "is_type G T  $\implies$  G, h  $\vdash$  default_val T ::  $\preceq$ T"
apply (unfold conf_def)
apply (cases T)
apply (erule ssubst)
apply (case_tac prim_ty)
apply (auto simp add: widen.null)
done

```

```

lemma conf_upd_obj:
  "h a = Some (Obj C fs)  $\implies$  (G, h(a  $\mapsto$  Obj C fs)  $\vdash$  x ::  $\preceq$ T) = (G, h  $\vdash$  x ::  $\preceq$ T)"
apply (unfold conf_def)
apply (rule val.induct)
apply auto
done

```

```

lemma conf_widen [rule_format (no_asm)]:
  "wf_prog wf_mb G ==> G,h⊢x::⊆T --> G⊢T⊆T' --> G,h⊢x::⊆T'"
apply (unfold conf_def)
apply (rule val.induct)
apply (auto intro: widen_trans)
done

lemma conf_hext [rule_format (no_asm)]: "h⊆|h' ==> G,h⊢v::⊆T --> G,h'⊢v::⊆T"
apply (unfold conf_def)
apply (rule val.induct)
apply (auto elim: hextE)
done

lemma new_locD: "[|h a = None; G,h⊢Addr t::⊆T|] ==> t≠a"
apply (unfold conf_def)
apply auto
done

lemma conf_RefTD [rule_format (no_asm)]:
  "G,h⊢a'::⊆RefT T --> a' = Null |
  (∃a obj T'. a' = Addr a ∧ h a = Some obj ∧ obj_ty obj = T' ∧ G⊢T'⊆RefT T)"
apply (unfold conf_def)
apply (induct_tac "a'")
apply (auto)
done

lemma conf_NullTD: "G,h⊢a'::⊆RefT NullT ==> a' = Null"
apply (drule conf_RefTD)
apply auto
apply (simp add: obj_ty_def)
apply (auto split: heap_entry.splits)
done

lemma non_npD: "[|a' ≠ Null; G,h⊢a'::⊆RefT t|] ==>
  ∃a z. a' = Addr a ∧ h a = Some z ∧ G⊢obj_ty z⊆RefT t"
apply (drule conf_RefTD)
apply auto
done

lemma conf_list_gext_widen [rule_format (no_asm)]:
  "wf_prog wf_mb G ==> ∀Ts Ts'. list_all2 (conf G h) vs Ts -->
  list_all2 (λT T'. G⊢T⊆T') Ts Ts' --> list_all2 (conf G h) vs Ts'"
apply (induct_tac "vs")
  apply (clarsimp)
  apply (clarsimp)
  apply (frule list_all2_lengthD [THEN sym])
  apply (simp (no_asm_use) add: length_Suc_conv)
  apply (safe)
  apply (frule list_all2_lengthD [THEN sym])
  apply (simp (no_asm_use) add: length_Suc_conv)

```

```

apply(clarify)
apply(fast elim: conf_widen)
done

```

### 2.11.3 lconf

```

lemma lconfD: "[| G,h⊢vs[::≲]Ts; Ts n = Some T |] ==> G,h⊢(the (vs n))[::≲]T"
apply (unfold lconf_def)
apply (force)
done

```

```

lemma lconf_hext [elim]: "[| G,h⊢l[::≲]L; h≤|h' |] ==> G,h'⊢l[::≲]L"
apply (unfold lconf_def)
apply (fast elim: conf_hext)
done

```

```

lemma lconf_upd: "!!X. [| G,h⊢l[::≲]lT;
  G,h⊢v[::≲]T; lT va = Some T |] ==> G,h⊢l(va↦v)[::≲]lT"
apply (unfold lconf_def)
apply auto
done

```

```

lemma lconf_init_vars_lemma [rule_format (no_asm)]:
  "∀x. P x --> R (dv x) x ==> (∀x. map_of fs f = Some x --> P x) -->
  (∀T. map_of fs f = Some T -->
  (∃v. map_of (map (λ(f,ft). (f, dv ft)) fs) f = Some v ∧ R v T))"
apply (induct_tac "fs")
apply auto
done

```

```

lemma lconf_init_vars [intro!]:
  "∀n. ∀T. map_of fs n = Some T --> is_type G T ==> G,h⊢init_vars fs[::≲]map_of fs"
apply (unfold lconf_def init_vars_def)
apply auto
apply (rule lconf_init_vars_lemma)
apply (erule_tac [3] asm_rl)
apply (intro strip)
apply (erule defval_conf)
apply auto
done

```

```

lemma lconf_ext: "[|G,s⊢l[::≲]L; G,s⊢v[::≲]T|] ==> G,s⊢l(vn↦v)[::≲]L(vn↦T)"
apply (unfold lconf_def)
apply auto
done

```

```

lemma lconf_ext_list [rule_format (no_asm)]:
  "G,h⊢l[::≲]L ==> ∀vs Ts. distinct vns --> length Ts = length vns -->
  list_all2 (λv T. G,h⊢v[::≲]T) vs Ts --> G,h⊢l(vns[↦]vs)[::≲]L(vns[↦]Ts)"
apply (unfold lconf_def)
apply (induct_tac "vns")
apply (clarsimp)
apply (clarsimp)
apply (frule list_all2_lengthD)

```

```

apply( auto simp add: length_Suc_conv)
done

```

#### 2.11.4 oconf

```

lemma oconf_hext: "G,h⊢obj√ ==> h≤|h' ==> G,h'⊢obj√"
apply (unfold oconf_def)
apply (cases obj)
apply auto
apply (erule allE, erule impE, assumption)
apply clarify
apply (rule exI, rule conjI, assumption)
apply (erule conf_hext)
apply assumption
done

```

```

lemma oconf_obj: "G,h⊢Obj C fs√ =
  (∀T f. map_of(fields (G,C)) f = Some T --> (∃v. fs f = Some v ∧ G,h⊢v::⊆T))"
apply (unfold oconf_def lconf_def)
apply auto
done

```

```

lemmas oconf_objD = oconf_obj [THEN iffD1, THEN spec, THEN spec, THEN mp]

```

#### 2.11.5 hconf

```

lemma hconfD: "[|G⊢h h√; h a = Some obj|] ==> G,h⊢obj√"
apply (unfold hconf_def)
apply (fast)
done

```

```

lemma hconfI: "∀a obj. h a=Some obj --> G,h⊢obj√ ==> G⊢h h√"
apply (unfold hconf_def)
apply (fast)
done

```

#### 2.11.6 conforms

```

lemma conforms_heapD: "(h, l)::⊆(G, lT) ==> G⊢h h√"
apply (unfold conforms_def)
apply (simp)
done

```

```

lemma conforms_localD: "(h, l)::⊆(G, lT) ==> G,h⊢l[::⊆]lT"
apply (unfold conforms_def)
apply (simp)
done

```

```

lemma conformsI: "[|G⊢h h√; G,h⊢l[::⊆]lT|] ==> (h, l)::⊆(G, lT)"
apply (unfold conforms_def)
apply auto
done

```

```

lemma conforms_hext: "[|(h,l)::⊆(G,lT); h≤|h'; G⊢h h'√ |] ==> (h',l)::⊆(G,lT)"
apply( fast dest: conforms_localD elim!: conformsI lconf_hext)

```

48

done

lemma conforms\_upd\_obj:

" $[|(h,l)::\preceq(G, lT); G, h(a \mapsto obj) \vdash obj \surd; h \leq |h(a \mapsto obj)|] \implies (h(a \mapsto obj), l)::\preceq(G, lT)$ "

apply(rule conforms\_hext)

apply auto

apply(rule hconfI)

apply(drule conforms\_heapD)

apply (auto elim: oconf\_hext dest: hconfD simp del: split\_paired\_All)

done

lemma conforms\_upd\_local:

" $[|(h, l)::\preceq(G, lT); G, h \vdash v::\preceq T; lT \text{ va} = \text{Some } T|] \implies (h, l(va \mapsto v))::\preceq(G, lT)$ "

apply (unfold conforms\_def)

apply (auto elim: lconf\_upd)

done

end



## Chapter 3

# Java Virtual Machine

### 3.1 State of the JVM

**theory** *JVMState = Conform:*

For object initialization, we tag each location with the current init status. The tags use an extended type system for object initialization (that gets reused in the BV).

We have either

- usual initialized types, or
- a class that is not yet initialized and has been created by a *New* instruction at a certain line number, or
- a partly initialized class (on which the next super class constructor has to be called). We store the name of the class the super class constructor has to be called of.

**datatype** *init\_ty = Init ty | UnInit cname nat | PartInit cname*

#### 3.1.1 Frame Stack

**types**

```
opstack  = "val list"
locvars  = "val list"
p_count  = nat
ref_upd  = "(val × val)"
```

```
frame = "opstack ×
        locvars ×
        cname ×
        sig ×
        p_count ×
        ref_upd"
```

- operand stack
- local variables (including this pointer and method parameters)
- name of class where current method is defined
- method name + parameter types
- program counter within frame
- ref update for obj init proof

#### 3.1.2 Exceptions

**constdefs**

```
raise_system_xcpt :: "bool ⇒ xcpt ⇒ val option"
"raise_system_xcpt b x ≡ if b then Some (Addr (XcptRef x)) else None"
```

— redefines *State.new\_Addr*:

```
new_Addr :: "aheap ⇒ loc × val option"
"new_Addr h ≡ let (a, x) = State.new_Addr h
               in (a, raise_system_xcpt (x ~= None) OutOfMemory)"
```

**types**

```
init_heap = "loc ⇒ init_ty"
```

— type tags to track init status of objects

```
jvm_state = "val option × aheap × init_heap × frame list"
— exception flag, heap, tag heap, frames
```

a new, blank object with default values in all fields:

**constdefs**

```
blank :: "'c prog ⇒ cname ⇒ heap_entry"
"blank G C ≡ Obj C (init_vars (fields(G,C)))"

blank_arr :: "ty ⇒ nat ⇒ heap_entry"
"blank_arr T len ≡ Arr T len (λx. if x < len then Some (default_val T) else None)"

start_heap :: "'c prog ⇒ aheap"
"start_heap G ≡ empty (XcptRef NullPointer ↦ blank G (Xcpt NullPointer))
(XcptRef ClassCast ↦ blank G (Xcpt ClassCast))
(XcptRef OutOfMemory ↦ blank G (Xcpt OutOfMemory))
(XcptRef ArrayIndexOutOfBounds ↦ blank G (Xcpt ArrayIndexOutOfBounds))
(XcptRef NegativeArraySize ↦ blank G (Xcpt NegativeArraySize))
(XcptRef ArrayStore ↦ blank G (Xcpt ArrayStore))"

start_iheap :: init_heap
"start_iheap ≡ ((((((λx. arbitrary)
(XcptRef NullPointer := Init (Class (Xcpt NullPointer))))
(XcptRef ClassCast := Init (Class (Xcpt ClassCast))))
(XcptRef OutOfMemory := Init (Class ((Xcpt OutOfMemory))))
(XcptRef ArrayIndexOutOfBounds := Init (Class (Xcpt ArrayIndexOutOfBounds))))
(XcptRef NegativeArraySize := Init (Class (Xcpt NegativeArraySize))))
(XcptRef ArrayStore := Init (Class (Xcpt ArrayStore))))"
```

### 3.1.3 Lemmas

**lemma new\_AddrD:**

```
assumes new: "new_Addr hp = (ref, xcp)"
shows "hp ref = None ∧ xcp = None ∨ xcp = Some (Addr (XcptRef OutOfMemory))"
proof -
from new obtain xcpT where old: "State.new_Addr hp = (ref, xcpT)"
by (cases "State.new_Addr hp") (simp add: new_Addr_def)
from this [symmetric]
have "hp ref = None ∧ xcpT = None ∨ xcpT = Some OutOfMemory"
by (rule State.new_AddrD)
with new old show ?thesis
by (auto simp add: new_Addr_def raise_system_xcpt_def)
qed
```

**lemma new\_Addr\_OutOfMemory:**

```
"snd (new_Addr hp) = Some xcp ⇒ xcp = Addr (XcptRef OutOfMemory)"
proof -
obtain ref xp where "new_Addr hp = (ref, xp)" by (cases "new_Addr hp")
moreover assume "snd (new_Addr hp) = Some xcp"
ultimately show ?thesis by (auto dest: new_AddrD)
qed
```

52

end

## 3.2 Instructions of the JVM

theory JVMInstructions = JVMState:

datatype

```

instr = Load nat           — load from local variable
      | Store nat          — store into local variable
      | LitPush val        — push a literal (constant)
      | New cname          — create object
      | Getfield vname cname — Fetch field from object
      | Putfield vname cname — Set field in object
      | Checkcast cname    — Check whether object is of given type
      | Invoke cname mname "(ty list)" — inv. instance meth of an object
      | Invoke_special cname mname "ty list"
                               — no dynamic type lookup, for constructors

      | Return             — return from method
      | Pop                 — pop top element from opstack
      | Dup                 — duplicate top element of opstack
      | Dup_x1              — duplicate next to top element
      | Dup_x2              — duplicate 3rd element
      | Swap                — swap top and next to top element
      | IAdd                — integer addition
      | Goto int            — goto relative address
      | Ifcmpeq int         — branch if int/ref comparison succeeds
      | Throw               — throw top of stack as exception
      | Jsr int             — jump to subroutine
      | Ret nat             — return from subroutine
      | ArrLoad             — load indexed entry from array
      | ArrStore            — store value into indexed array entry
      | ArrLength           — get array length
      | ArrNew ty           — create new 1-dimensional array

```

types

```

bytecode = "instr list"
exception_entry = "p_count × p_count × p_count × cname"
                 — start-pc, end-pc, handler-pc, exception type
exception_table = "exception_entry list"
jvm_method = "nat × nat × bytecode × exception_table"
jvm_prog = "jvm_method prog"

```

end

### 3.3 JVM Instruction Semantics

theory *JVMExecInstr* = *JVMInstructions* + *JVMState*:

the method name of constructors:

consts

*init* :: *mname*

replace a by b in l:

constdefs

*replace* :: "'a ⇒ 'a ⇒ 'a list ⇒ 'a list"

"*replace* a b l == map (λx. if x = a then b else x) l"

some lemmas about replace

lemma *replace\_removes\_elem*:

" $a \neq b \implies a \notin \text{set } (\text{replace } a \ b \ l)$ "

by (unfold *replace\_def*) auto

lemma *replace\_length [simp]*:

"length (*replace* a b l) = length l" by (simp add: *replace\_def*)

lemma *replace\_Nil [iff]*:

"*replace* x y [] = []" by (simp add: *replace\_def*)

lemma *replace\_Cons*:

"*replace* x y (l#ls) = (if l = x then y else l)#(*replace* x y ls)"

by (simp add: *replace\_def*)

lemma *replace\_map*:

"inj f ==> *replace* (f x) (f y) (map f l) = map f (*replace* x y l)"

apply (induct l)

  apply (simp add: *replace\_def*)

apply (simp add: *replace\_def*)

apply clarify

apply (drule injD, assumption)

apply simp

done

lemma *replace\_id*:

" $x \notin \text{set } l \vee x = y \implies \text{replace } x \ y \ l = l$ "

apply (induct l)

  apply (auto simp add: *replace\_def*)

done

single execution step for each instruction:

consts

*exec\_instr* :: "[instr, jvm\_prog, aheap, init\_heap, opstack, locvars,  
                  cname, sig, p\_count, ref\_upd, frame list] => jvm\_state"

primrec

```

"exec_instr (Load idx) G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, ((vars ! idx) # stk, vars, Cl, sig, pc+1, z)#frs)"

"exec_instr (Store idx) G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, (tl stk, vars[idx:=hd stk], Cl, sig, pc+1, z)#frs)"

"exec_instr (LitPush v) G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, (v # stk, vars, Cl, sig, pc+1, z)#frs)"

"exec_instr (New C) G hp ihp stk vars Cl sig pc z frs =
  (let (oref,xp') = new_Addr hp;
       hp'       = if xp'=None then hp(oref ↦ blank G C) else hp;
       ihp'      = if xp'=None then ihp(oref := UnInit C pc) else ihp;
       stk'      = if xp'=None then (Addr oref)#stk else stk;
       pc'      = if xp'=None then pc+1 else pc
   in
  (xp', hp', ihp', (stk', vars, Cl, sig, pc', z)#frs))"

"exec_instr (Getfield F C) G hp ihp stk vars Cl sig pc z frs =
  (let oref      = hd stk;
       xp'      = raise_system_xcpt (oref=None) NullPointer;
       (oc,fs)  = the_obj (the(hp(the_Addr oref)));
       stk'     = if xp'=None then the(fs(F,C))#(tl stk) else tl stk;
       pc'     = if xp'=None then pc+1 else pc
   in
  (xp', hp, ihp, (stk', vars, Cl, sig, pc', z)#frs))"

"exec_instr (Putfield F C) G hp ihp stk vars Cl sig pc z frs =
  (let (fval,oref)= (hd stk, hd(tl stk));
       xp'      = raise_system_xcpt (oref=None) NullPointer;
       a       = the_Addr oref;
       (oc,fs)  = the_obj (the(hp a));
       hp'     = if xp'=None then hp(a ↦ Obj oc (fs((F,C) ↦ fval))) else hp;
       pc'     = if xp'=None then pc+1 else pc
   in
  (xp', hp', ihp, (tl (tl stk), vars, Cl, sig, pc', z)#frs))"

"exec_instr (Checkcast C) G hp ihp stk vars Cl sig pc z frs =
  (let oref      = hd stk;
       xp'      = raise_system_xcpt (¬cast_ok G C hp oref) ClassCast;
       stk'     = if xp'=None then stk else tl stk;
       pc'     = if xp'=None then pc+1 else pc
   in
  (xp', hp, ihp, (stk', vars, Cl, sig, pc', z)#frs))"

"exec_instr (Invoke C mn ps) G hp ihp stk vars Cl sig pc z frs =
  (let n        = length ps;
       args    = take n stk;
       oref    = stk!n;
       xp'     = raise_system_xcpt (oref=None) NullPointer;
       dynT    = the_Class (obj_ty (the(hp (the_Addr oref))));
       (dc,mh,mxs,mxl,c) = the (method (G,dynT) (mn,ps));
       frs'    = (if xp'=None then

```

```

      [[[],oref#(rev args)@(replicate mxl arbitrary),dc,(mn,ps),0,arbitrary]]
      else []
    in
      (xp', hp, ihp, frs'@(stk, vars, Cl, sig, pc, z)#frs))"
— Because exception handling needs the pc of the Invoke instruction,
— Invoke doesn't change stk and pc yet (Return does that).

"exec_instr (Invoke_special C mn ps) G hp ihp stk vars Cl sig pc z frs =
  (let n          = length ps;
      args = take n stk;
          oref      = stk!n;
      addr = the_Addr oref;
          x'        = raise_system_xcpt (oref=None) NullPointer;
          dynT      = the_Class (obj_ty (the(hp addr)));
          (dc,mh,mxs,mxl,c)= the (method (G,C) (mn,ps));
      (addr',x'') = new_Addr hp;
      xp' = if x' = None then x'' else x';
      hp' = if xp' = None then hp(addr' ↦ blank G dynT) else hp;
      ihp' = if C = Object then ihp(addr':= Init (Class dynT))
              else ihp(addr' := PartInit C);
      ihp'' = if xp' = None then ihp' else ihp;
      z' = if C = Object then (Addr addr', Addr addr') else (Addr addr', Null);
          frs'      = (if xp'=None then
              [[[],(Addr addr')#(rev args)@(replicate mxl arbitrary),dc,(mn,ps),0,z']]
              else [])
    in
      (xp', hp', ihp'', frs'@(stk, vars, Cl, sig, pc, z)#frs))"

"exec_instr Return G hp ihp stk0 vars Cl sig0 pc z0 frs =
  (if frs=[] then
    (None, hp, ihp, []))
  else let
    val      = hd stk0;
    (mn,pt) = sig0;
    (stk,loc,C,sig,pc,z) = hd frs;
    (b,c)    = z0;
    (a,c')   = z;
    n        = length pt;
    args     = take n stk;
    addr     = stk!n;
    drpstk   = drop (n+1) stk;
    stk'     = if mn=init then val#replace addr c drpstk else val#drpstk;
    loc'     = if mn=init then replace addr c loc else loc;
    z'       = if mn=init ∧ z = (addr,Null) then (a,c) else z
  in
    (None, hp, ihp, (stk',loc',C,sig,pc+1,z')#tl frs))"
— Return drops arguments from the caller's stack and increases
— the program counter in the caller

— z is only updated if we are in a constructor and have initialized the
— same reference as the constructor in the frame above (otherwise we are
— in the last constructor of the init chain)

```



```

"exec_instr Pop G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, (tl stk, vars, Cl, sig, pc+1, z)#frs)"

"exec_instr Dup G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, (hd stk # stk, vars, Cl, sig, pc+1, z)#frs)"

"exec_instr Dup_x1 G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, (hd stk # hd (tl stk) # hd stk # (tl (tl stk)),
    vars, Cl, sig, pc+1, z)#frs)"

"exec_instr Dup_x2 G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp,
    (hd stk # hd (tl stk) # (hd (tl (tl stk))) # hd stk # (tl (tl (tl stk)))),
    vars, Cl, sig, pc+1, z)#frs)"

"exec_instr Swap G hp ihp stk vars Cl sig pc z frs =
  (let (val1,val2) = (hd stk,hd (tl stk))
  in
    (None, hp, ihp, (val2#val1#(tl (tl stk)), vars, Cl, sig, pc+1, z)#frs))"

"exec_instr IAdd G hp ihp stk vars Cl sig pc z frs =
  (let (val1,val2) = (hd stk,hd (tl stk))
  in
    (None, hp, ihp, (Intg ((the_Intg val1)+(the_Intg val2))#(tl (tl stk)),
    vars, Cl, sig, pc+1, z)#frs))"

"exec_instr (Ifcmpeq i) G hp ihp stk vars Cl sig pc z frs =
  (let (val1,val2) = (hd stk, hd (tl stk));
    pc' = if val1 = val2 then nat(int pc+i) else pc+1
  in
    (None, hp, ihp, (tl (tl stk), vars, Cl, sig, pc', z)#frs))"

"exec_instr (Goto i) G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, (stk, vars, Cl, sig, nat(int pc+i), z)#frs)"

"exec_instr Throw G hp ihp stk vars Cl sig pc z frs =
  (let xcpt = raise_system_xcpt (hd stk = Null) NullPointer;
    xcpt' = if xcpt = None then Some (hd stk) else xcpt
  in
    (xcpt', hp, ihp, (stk, vars, Cl, sig, pc, z)#frs))"

"exec_instr (Jsr i) G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, (RetAddr (pc+1)#stk, vars, Cl, sig, nat(int pc+i), z)#frs)"

"exec_instr (Ret idx) G hp ihp stk vars Cl sig pc z frs =
  (None, hp, ihp, (stk, vars, Cl, sig, the_RetAddr (vars!idx), z)#frs)"

"exec_instr (ArrLoad) G hp ihp stk vars Cl sig pc z frs =
  (let idx = the_Intg (hd stk);
    aref = hd (tl stk);
    xp'' = raise_system_xcpt (aref=Null) NullPointer;
    (T,l,en) = the_arr (the(hp(the_Addr aref)));
  in
    (xp'', hp, ihp, (stk, vars, Cl, sig, pc, z)#frs))"

```

```

      xp' = if xp''=None then raise_system_xcpt (int 1 ≤ idx ∨ idx < 0) ArrayIndexOutOfBounds
else xp'';
      stk' = the (en (nat idx)) # (tl (tl stk));
      pc' = if xp''=None then pc+1 else pc
      in
      (xp', hp, ihp, (stk', vars, Cl, sig, pc', z)#frs))"

```

```

"exec_instr (ArrStore) G hp ihp stk vars Cl sig pc z frs =
  (let val = hd stk;
   idx = the_Intg (hd (tl stk));
   aref = hd (tl (tl stk));
   xp'' = raise_system_xcpt (aref=None) NullPointer;
   a = the_Addr aref;
   (T,l,en) = the_arr (the(hp a));
   xp''' = if xp''=None then raise_system_xcpt (int 1 ≤ idx ∨ idx < 0) ArrayIndexOutOfBounds
else xp'';
   xp' = if xp'''=None then raise_system_xcpt (¬G, hp ⊢ val :: ≤ T) ArrayStore else
xp''';
   hp' = if xp''=None then hp(a ↦ Arr T l (en(nat idx ↦ val))) else
hp;
   pc' = if xp''=None then pc+1 else pc
   in
   (xp', hp', ihp, (tl (tl (tl stk)), vars, Cl, sig, pc', z)#frs))"

```

```

"exec_instr (ArrLength) G hp ihp stk vars Cl sig pc z frs =
  (let aref = hd stk;
   xp' = raise_system_xcpt (aref=None) NullPointer;
   a = the_Addr aref;
   (T,l,en) = the_arr (the(hp a));
   pc' = if xp''=None then pc+1 else pc
   in
   (xp', hp, ihp, (Intg (int l)#(tl stk), vars, Cl, sig, pc', z)#frs))"

```

```

"exec_instr (ArrNew T) G hp ihp stk vars Cl sig pc z frs =
  (let len = the_Intg (hd stk);
   xp''' = raise_system_xcpt (len < 0) NegativeArraySize;
   (aref,xp'') = new_Addr hp;
   xp' = if xp'''=None then xp'' else xp''';
   hp' = if xp''=None then hp(aref ↦ blank_arr T (nat len)) else hp;
   ihp' = if xp''=None then ihp(aref := Init (T.[])) else ihp;
   stk' = (Addr aref)#tl stk;
   pc' = if xp''=None then pc+1 else pc
   in
   (xp', hp', ihp', (stk', vars, Cl, sig, pc', z)#frs))"

```

end

### 3.4 Exception handling in the JVM

theory JVMExceptions = JVMInstructions:

constdefs

```
match_exception_entry :: "jvm_prog ⇒ cname ⇒ p_count ⇒ exception_entry ⇒ bool"
"match_exception_entry G cn pc ee ==
  let (start_pc, end_pc, handler_pc, catch_type) = ee in
  start_pc ≤ pc ∧ pc < end_pc ∧ G ⊢ cn ≤C catch_type"
```

consts

```
match_exception_table :: "jvm_prog ⇒ cname ⇒ p_count ⇒ exception_table
  ⇒ p_count option"
```

primrec

```
"match_exception_table G cn pc [] = None"
"match_exception_table G cn pc (e#es) = (if match_exception_entry G cn pc e
  then Some (fst (snd (snd e)))
  else match_exception_table G cn pc es)"
```

consts

```
cname_of :: "aheap ⇒ val ⇒ cname"
ex_table_of :: "jvm_method ⇒ exception_table"
```

translations

```
"cname_of hp v" == "the_Class (obj_ty (the (hp (the_Addr v))))"
"ex_table_of m" == "snd (snd (snd m))"
```

consts

```
find_handler :: "jvm_prog ⇒ val option ⇒ aheap ⇒ init_heap ⇒ frame list ⇒ jvm_state"
```

primrec

```
"find_handler G xcpt hp ihp [] = (xcpt, hp, ihp, [])"
"find_handler G xcpt hp ihp (fr#frs) =
  (case xcpt of
    None ⇒ (None, hp, ihp, fr#frs)
  | Some xc ⇒
    let (stk, loc, C, sig, pc, r) = fr in
    (case match_exception_table G (cname_of hp xc) pc
      (ex_table_of (snd(snd(the(method (G,C) sig)))))) of
      None ⇒ find_handler G (Some xc) hp ihp frs
    | Some handler_pc ⇒ (None, hp, ihp, ([xc], loc, C, sig, handler_pc, r)#frs)))"
```

Expresses that a value is tagged with an initialized type (only applies to addresses and then only if the heap contains a value for the address)

constdefs

```
is_init :: "aheap ⇒ init_heap ⇒ val ⇒ bool"
```

```
"is_init hp ih v  $\equiv$ 
 $\forall loc. v = Addr\ loc \longrightarrow hp\ loc \neq None \longrightarrow (\exists t. ih\ loc = Init\ t)"$ 
```

System exceptions are allocated in all heaps.

**constdefs**

```
preallocated :: "aheap  $\Rightarrow$  init_heap  $\Rightarrow$  bool"
"preallocated hp ihp  $\equiv \forall x. \exists fs. hp\ (XcptRef\ x) = Some\ (Obj\ (Xcpt\ x)\ fs) \wedge is\_init\ hp\ ihp\ (Addr\ (XcptRef\ x))"$ 
```

**lemma preallocatedD [simp,dest]:**

```
"preallocated hp ihp  $\Longrightarrow \exists fs. hp\ (XcptRef\ x) = Some\ (Obj\ (Xcpt\ x)\ fs) \wedge is\_init\ hp\ ihp\ (Addr\ (XcptRef\ x))"$ 
by (unfold preallocated_def) fast
```

**lemma preallocatedE [elim?]:**

```
"preallocated hp ihp  $\Longrightarrow$ 
( $\wedge fs. hp\ (XcptRef\ x) = Some\ (Obj\ (Xcpt\ x)\ fs) \Longrightarrow is\_init\ hp\ ihp\ (Addr\ (XcptRef\ x))$ )
 $\Longrightarrow P\ hp\ ihp$ 
 $\Longrightarrow P\ hp\ ihp$ "
by fast
```

**lemma cname\_of\_xcp:**

```
"raise_system_xcpt b x = Some xcp  $\Longrightarrow$  preallocated hp ihp
 $\Longrightarrow cname\_of\ hp\ xcp = Xcpt\ x"$ 
```

**proof -**

```
assume "raise_system_xcpt b x = Some xcp"
hence "xcp = Addr (XcptRef x)"
by (simp add: raise_system_xcpt_def split: split_if_asm)
moreover
assume "preallocated hp ihp"
then obtain fs where "hp (XcptRef x) = Some (Obj (Xcpt x) fs)" ..
ultimately show ?thesis by simp
```

**qed**

**lemma preallocated\_start:**

```
"preallocated (start_heap G) start_iheap"
apply (unfold preallocated_def)
apply (unfold start_heap_def start_iheap_def)
apply (rule allI)
apply (case_tac x)
apply (auto simp add: blank_def is_init_def)
done
```

Only program counters that are mentioned in the exception table can be returned by `match_exception_table`

**lemma match\_exception\_table\_in\_et:**

```
"match_exception_table G C pc et = Some pc'  $\Longrightarrow \exists e \in set\ et. pc' = fst\ (snd\ (snd\ e))"$ 
by (induct et) (auto split: split_if_asm)
```

**end**

### 3.5 Program Execution in the JVM

theory *JVMExec* = *JVMExecInstr* + *JVMExceptions*:

consts

```
exec :: "jvm_prog × jvm_state => jvm_state option"
```

— recdef only used for pattern matching

recdef *exec* "{}"

```
"exec (G, xp, hp, ihp, []) = None"
```

```
"exec (G, None, hp, ihp, (stk,loc,C,sig,pc,z)#frs) =
(let
  i = fst(snd(snd(snd(snd(the(method (G,C) sig)))))) ! pc;
  (xcpt', hp', ihp', frs') = exec_instr i G hp ihp stk loc C sig pc z frs
in Some (find_handler G xcpt' hp' ihp' frs'))"
```

```
"exec (G, Some xp, hp, ihp, frs) = None"
```

constdefs

```
exec_all :: "[jvm_prog,jvm_state,jvm_state] => bool"
          ("_ |- _ -jvm-> _" [61,61,61]60)
"G |- s -jvm-> t == (s,t) ∈ {(s,t). exec(G,s) = Some t}~*"
```

syntax (*xsymbols*)

```
exec_all :: "[jvm_prog,jvm_state,jvm_state] => bool"
          ("_ ⊢ _ -jvm→ _" [61,61,61]60)
```

The start configuration of the JVM: in the start heap, we call a method *m* of class *C* in program *G*. The *this* pointer of the frame is set to *Null* to simulate a static method invocation.

constdefs

```
start_state :: "jvm_prog ⇒ cname ⇒ mname ⇒ jvm_state"
"start_state G C m ≡
let (C',rT,mxs,mxl,ins,et) = the (method (G,C) (m,[])) in
(None, start_heap G, start_iheap,
 [[[], Null # replicate mxl arbitrary, C, (m,[]), 0, (Null,Null)]])"
```

end

### 3.6 A Defensive JVM

theory JVMDefensive = JVMExec:

Extend the state space by one element indicating a type error (or other abnormal termination)

datatype 'a type\_error = TypeError | Normal 'a

syntax "fifth" :: "'a × 'b × 'c × 'd × 'e × 'f ⇒ 'e"

translations

"fifth x" == "fst(snd(snd(snd(snd x))))"

consts isAddr :: "val ⇒ bool"

recdef isAddr "{}"

"isAddr (Addr loc) = True"

"isAddr v = False"

consts isIntg :: "val ⇒ bool"

recdef isIntg "{}"

"isIntg (Intg i) = True"

"isIntg v = False"

consts isRetAddr :: "val ⇒ bool"

recdef isRetAddr "{}"

"isRetAddr (RetAddr pc) = True"

"isRetAddr v = False"

constdefs

isRef :: "val ⇒ bool"

"isRef v ≡ v = Null ∨ isAddr v"

consts

check\_instr :: "[instr, jvm\_prog, aheap, init\_heap, opstack, locvars, cname, sig, p\_count, ref\_upd, p\_count, frame list] ⇒ bool"

primrec

"check\_instr (Load idx) G hp ihp stk vars C sig pc z maxpc frs = (idx < length vars)"

"check\_instr (Store idx) G hp ihp stk vars Cl sig pc z maxpc frs = (0 < length stk ∧ idx < length vars)"

"check\_instr (LitPush v) G hp ihp stk vars Cl sig pc z maxpc frs = (¬isAddr v)"

"check\_instr (New C) G hp ihp stk vars Cl sig pc z maxpc frs = is\_class G C"

"check\_instr (Getfield F C) G hp ihp stk vars Cl sig pc z maxpc frs =

(0 < length stk ∧ is\_class G C ∧ field (G,C) F ≠ None ∧

(let (C', T) = the (field (G,C) F); ref = hd stk in

C' = C ∧ isRef ref ∧ (ref ≠ Null →

hp (the\_Addr ref) ≠ None ∧ is\_init hp ihp ref ∧

```

    (let (D,vs) = the_obj (the (hp (the_Addr ref))) in
      G ⊢ D ≤C C ∧ vs (F,C) ≠ None ∧ G, hp ⊢ the (vs (F,C)) :: ≤ T)))"

"check_instr (Putfield F C) G hp ihp stk vars Cl sig pc z maxpc frs =
(1 < length stk ∧ is_class G C ∧ field (G,C) F ≠ None ∧
(let (C', T) = the (field (G,C) F); v = hd stk; ref = hd (tl stk) in
  C' = C ∧ is_init hp ihp v ∧ isRef ref ∧ (ref ≠ Null →
    hp (the_Addr ref) ≠ None ∧ is_init hp ihp ref ∧
    (let (D,vs) = the_obj (the (hp (the_Addr ref))) in
      G ⊢ D ≤C C ∧ G, hp ⊢ v :: ≤ T)))))"

"check_instr (Checkcast C) G hp ihp stk vars Cl sig pc z maxpc frs =
(0 < length stk ∧ is_class G C ∧ isRef (hd stk) ∧ is_init hp ihp (hd stk))"

"check_instr (Invoke C mn ps) G hp ihp stk vars Cl sig pc z maxpc frs =
(length ps < length stk ∧ mn ≠ init ∧
(let n = length ps; v = stk!n in
  isRef v ∧ (v ≠ Null →
    hp (the_Addr v) ≠ None ∧ is_init hp ihp v ∧
    method (G, cname_of hp v) (mn, ps) ≠ None ∧
    list_all2 (λv T. G, hp ⊢ v :: ≤ T ∧ is_init hp ihp v) (rev (take n stk)) ps))))"

"check_instr (Invoke_special C mn ps) G hp ihp stk vars Cl sig pc z maxpc frs =
(length ps < length stk ∧ mn = init ∧
(let n = length ps; ref = stk!n in
  isRef ref ∧ (ref ≠ Null →
    hp (the_Addr ref) ≠ None ∧
    method (G,C) (mn, ps) ≠ None ∧
    fst (the (method (G,C) (mn, ps))) = C ∧
    list_all2 (λv T. G, hp ⊢ v :: ≤ T ∧ is_init hp ihp v) (rev (take n stk)) ps) ∧
    (case ihp (the_Addr ref) of
      Init T ⇒ False
    | UnInit C' pc' ⇒ C' = C
    | PartInit C' ⇒ C' = Cl ∧ G ⊢ C' <C1 C)
  ))"

"check_instr Return G hp ihp stk0 vars Cl sig0 pc z0 maxpc frs =
(0 < length stk0 ∧ (0 < length frs →
  method (G, Cl) sig0 ≠ None ∧
  (let v = hd stk0; (C, rT, body) = the (method (G, Cl) sig0) in
    Cl = C ∧ G, hp ⊢ v :: ≤ rT ∧ is_init hp ihp v) ∧
  (fst sig0 = init →
    snd z0 ≠ Null ∧ isRef (snd z0) ∧ is_init hp ihp (snd z0)))))"

"check_instr Pop G hp ihp stk vars Cl sig pc z maxpc frs =
(0 < length stk)"

"check_instr Dup G hp ihp stk vars Cl sig pc z maxpc frs =
(0 < length stk)"

"check_instr Dup_x1 G hp ihp stk vars Cl sig pc z maxpc frs =
(1 < length stk)"

```



```

"check_instr Dup_x2 G hp ihp stk vars Cl sig pc z maxpc frs =
(2 < length stk)"

"check_instr Swap G hp ihp stk vars Cl sig pc z maxpc frs =
(1 < length stk)"

"check_instr IAdd G hp ihp stk vars Cl sig pc z maxpc frs =
(1 < length stk ∧ isIntg (hd stk) ∧ isIntg (hd (tl stk)))"

"check_instr (Ifcmpeq b) G hp ihp stk vars Cl sig pc z maxpc frs =
(1 < length stk ∧ 0 ≤ int pc+b ∧ nat(int pc+b) < maxpc)"

"check_instr (Goto b) G hp ihp stk vars Cl sig pc z maxpc frs =
(0 ≤ int pc+b ∧ nat(int pc+b) < maxpc)"

"check_instr Throw G hp ihp stk vars Cl sig pc z maxpc frs =
(0 < length stk ∧ isRef (hd stk))"

"check_instr (Jsr b) G hp ihp stk vars Cl sig pc z maxpc frs =
(0 ≤ int pc+b ∧ nat(int pc+b) < maxpc)"

"check_instr (Ret idx) G hp ihp stk vars Cl sig pc z maxpc frs =
(idx < length vars ∧ isRetAddr (vars!idx) ∧ the_RetAddr (vars!idx) < maxpc)"

"check_instr (ArrLoad) G hp ihp stk vars Cl sig pc z maxpc frs =
(1 < length stk ∧
(let idx = hd stk; ref = hd (tl stk) in isIntg idx ∧ isRef ref ∧
ref ≠ Null → isArray (obj_ty (the (hp (the_Addr ref))))))"

"check_instr (ArrStore) G hp ihp stk vars Cl sig pc z maxpc frs =
(2 < length stk ∧
(let idx = hd (tl stk); ref = hd (tl (tl stk)) in isIntg idx ∧ isRef ref ∧
ref ≠ Null → isArray (obj_ty (the (hp (the_Addr ref))))))"

"check_instr (ArrLength) G hp ihp stk vars Cl sig pc z maxpc frs =
(0 < length stk ∧
(let ref = hd stk in isRef ref ∧
ref ≠ Null → isArray (obj_ty (the (hp (the_Addr ref))))))"

"check_instr (ArrNew T) G hp ihp stk vars Cl sig pc z maxpc frs =
(0 < length stk ∧ isIntg (hd stk) ∧ is_type G T)"

```

**constdefs**

```

check :: "jvm_prog ⇒ jvm_state ⇒ bool"
"check G s ≡ let (xcpt, hp, ihp, frs) = s in
  (case frs of [] ⇒ True | (stk,loc,C,sig,pc,z)#frs' ⇒
    (let ins = fifth (the (method (G,C) sig)); i = ins!pc in
      check_instr i G hp ihp stk loc C sig pc z (length ins) frs'))"

exec_d :: "jvm_prog ⇒ jvm_state type_error ⇒ jvm_state option type_error"
"exec_d G s ≡ case s of

```

```

      TypeError ⇒ TypeError
    | Normal s' ⇒ if check G s' then Normal (exec (G, s')) else TypeError"

consts
  "exec_all_d" :: "jvm_prog ⇒ jvm_state type_error ⇒ jvm_state type_error ⇒ bool"
                ("_ |- _ -jvmd->_" [61,61,61]60)

syntax (xsymbols)
  "exec_all_d" :: "jvm_prog ⇒ jvm_state type_error ⇒ jvm_state type_error ⇒ bool"
                ("_ ⊢ _ -jvmd→_" [61,61,61]60)

defs
  exec_all_d_def:
  "G ⊢ s -jvmd→ t ≡
    (s,t) ∈ ({(s,t). exec_d G s = TypeError ∧ t = TypeError} ∪
      {(s,t). ∃t'. exec_d G s = Normal (Some t') ∧ t = Normal t'})*"

declare split_paired_All [simp del]
declare split_paired_Ex [simp del]

lemma [dest!]:
  "(if P then A else B) ≠ B ⇒ P"
  by (cases P, auto)

lemma exec_d_no_errorI [intro]:
  "check G s ⇒ exec_d G (Normal s) ≠ TypeError"
  by (unfold exec_d_def) simp

theorem no_type_error_commutates:
  "exec_d G (Normal s) ≠ TypeError ⇒
  exec_d G (Normal s) = Normal (exec (G, s))"
  by (unfold exec_d_def, auto)

lemma defensive_imp_aggressive:
  "G ⊢ (Normal s) -jvmd→ (Normal t) ⇒ G ⊢ s -jvm→ t"
proof -
  have "∧x y. G ⊢ x -jvmd→ y ⇒ ∀s t. x = Normal s → y = Normal t → G ⊢ s -jvm→
  t"
  apply (unfold exec_all_d_def)
  apply (erule rtrancl_induct)
  apply (simp add: exec_all_def)
  apply (fold exec_all_d_def)
  apply simp
  apply (intro allI impI)
  apply (erule disjE, simp)
  apply (elim exE conjE)
  apply (erule allE, erule impE, assumption)
  apply (simp add: exec_all_def exec_d_def split: type_error.splits split_if_asm)
  apply (rule rtrancl_trans, assumption)
  apply blast
done

```

```
moreover
  assume "G ⊢ (Normal s) -jvmd→ (Normal t)"
  ultimately
    show "G ⊢ s -jvm→ t" by blast
qed
end
```

### 3.7 System Class Implementations (JVM)

**theory** *JVMSystemClasses* = *WellForm* + *JVMExec*:

This theory provides bytecode implementations for the system class library. Each class has a default constructor.

**constdefs**

```

Object_ctor :: "jvm_method mdecl"
"Object_ctor ≡ ((init, []), PrimT Void, (1, 0, [LitPush Unit, Return], []))"

ObjectC :: "jvm_method cdecl"
"ObjectC ≡ ObjectC_decl [Object_ctor]"

Default_ctor :: "jvm_method mdecl"
"Default_ctor ≡
((init, []), PrimT Void, (1,0,[Load 0, Invoke_special Object init [], Return], []))"

NullPointerC :: "jvm_method cdecl"
"NullPointerC ≡ NullPointerC_decl [Default_ctor]"

ClassCastC :: "jvm_method cdecl"
"ClassCastC ≡ ClassCastC_decl [Default_ctor]"

OutOfMemoryC :: "jvm_method cdecl"
"OutOfMemoryC ≡ OutOfMemoryC_decl [Default_ctor]"

NegativeArraySizeC :: "jvm_method cdecl"
"NegativeArraySizeC ≡ NegativeArraySizeC_decl [Default_ctor]"

ArrayIndexOutOfBoundsC :: "jvm_method cdecl"
"ArrayIndexOutOfBoundsC ≡ ArrayIndexOutOfBoundsC_decl [Default_ctor]"

ArrayStoreC :: "jvm_method cdecl"
"ArrayStoreC ≡ ArrayStoreC_decl [Default_ctor]"

JVMSystemClasses :: "jvm_method cdecl list"
"JVMSystemClasses ≡ [ObjectC, NullPointerC, ClassCastC, OutOfMemoryC, NegativeArraySizeC,
ArrayIndexOutOfBoundsC, ArrayStoreC]"

lemmas SystemClassC_defs =
  SystemClass_decl_defs ObjectC_def NullPointerC_def
  OutOfMemoryC_def ClassCastC_def NegativeArraySizeC_def
  ArrayStoreC_def ArrayIndexOutOfBoundsC_def Default_ctor_def Object_ctor_def

lemma fst_mono: "A ⊆ B ⇒ fst ` A ⊆ fst ` B" by fast

locale sys_classes =
  fixes G :: jvm_prog

  assumes set: "set JVMSystemClasses ⊆ set G"
  assumes unique: "unique G"
  assumes wf: "wf ((subcls1 G)-1)"

  notes fields_rec = fields_rec_lemma [OF _ wf]

```

```

notes method_rec = method_rec_lemma [OF _ wf]

lemma (in sys_classes) class_Object [simp]:
  "class G Object = Some (arbitrary, [], [Object_ctor])"
proof -
  from set have "(Object, arbitrary, [], [Object_ctor]) ∈ set G"
    by (auto simp add: JVMSystemClasses_def ObjectC_def ObjectC_decl_def)
  with unique show ?thesis by (simp add: class_def map_of_SomeI)
qed

lemma (in sys_classes) fields_Object [simp]:
  shows "fields (G, Object) = []"
proof -
  from wf
  show ?thesis by (unfold fields_def) (simp add: class_rec.simps)
qed

lemma (in sys_classes) wf_syscls [simp]:
  "wf_syscls G"
  apply (insert set)
  apply (unfold wf_syscls_def SystemClasses_def JVMSystemClasses_def SystemClassC_defs)
  apply (drule fst_mono)
  apply simp
  done

lemma (in sys_classes) class_NullPointer [simp]:
  "class G (Xcpt NullPointer) = Some (Object, [], [Default_ctor])"
  (is "class G ?C = Some ?D")
proof -
  from set have "(?C, ?D) ∈ set G"
    by (auto simp add: JVMSystemClasses_def SystemClassC_defs)
  with unique show ?thesis by (simp add: class_def map_of_SomeI)
qed

lemma (in sys_classes) class_OutOfMemory [simp]:
  "class G (Xcpt OutOfMemory) = Some (Object, [], [Default_ctor])"
  (is "class G ?C = Some ?D")
proof -
  from set have "(?C, ?D) ∈ set G"
    by (auto simp add: JVMSystemClasses_def SystemClassC_defs)
  with unique show ?thesis by (simp add: class_def map_of_SomeI)
qed

lemma (in sys_classes) class_ClassCast [simp]:
  "class G (Xcpt ClassCast) = Some (Object, [], [Default_ctor])"
  (is "class G ?C = Some ?D")
proof -
  from set have "(?C, ?D) ∈ set G"
    by (auto simp add: JVMSystemClasses_def SystemClassC_defs)
  with unique show ?thesis by (simp add: class_def map_of_SomeI)
qed

lemma (in sys_classes) class_ArrayStore [simp]:

```

```

"class G (Xcpt ArrayStore) = Some (Object, [], [Default_ctor])"
(is "class G ?C = Some ?D")
proof -
  from set have "(?C, ?D) ∈ set G"
    by (auto simp add: JVMSystemClasses_def SystemClassC_defs)
  with unique show ?thesis by (simp add: class_def map_of_SomeI)
qed

lemma (in sys_classes) class_NegativeArraySize [simp]:
  "class G (Xcpt NegativeArraySize) = Some (Object, [], [Default_ctor])"
  (is "class G ?C = Some ?D")
proof -
  from set have "(?C, ?D) ∈ set G"
    by (auto simp add: JVMSystemClasses_def SystemClassC_defs)
  with unique show ?thesis by (simp add: class_def map_of_SomeI)
qed

lemma (in sys_classes) class_ArrayIndexOutOfBounds [simp]:
  "class G (Xcpt ArrayIndexOutOfBounds) = Some (Object, [], [Default_ctor])"
  (is "class G ?C = Some ?D")
proof -
  from set have "(?C, ?D) ∈ set G"
    by (auto simp add: JVMSystemClasses_def SystemClassC_defs)
  with unique show ?thesis by (simp add: class_def map_of_SomeI)
qed

lemma (in sys_classes) fields_NullPointer [simp]:
  "fields (G, Xcpt NullPointer) = []"
  by (simp add: fields_rec [OF class_NullPointer])

lemma (in sys_classes) fields_ClassCast [simp]:
  "fields (G, Xcpt ClassCast) = []"
  by (simp add: fields_rec [OF class_ClassCast])

lemma (in sys_classes) fields_OutOfMemory [simp]:
  "fields (G, Xcpt OutOfMemory) = []"
  by (simp add: fields_rec [OF class_OutOfMemory])

lemma (in sys_classes) fields_ArrayIndexOutOfBounds [simp]:
  "fields (G, Xcpt ArrayIndexOutOfBounds) = []"
  by (simp add: fields_rec [OF class_ArrayIndexOutOfBounds])

lemma (in sys_classes) fields_NegativeArraySize [simp]:
  "fields (G, Xcpt NegativeArraySize) = []"
  by (simp add: fields_rec [OF class_NegativeArraySize])

lemma (in sys_classes) fields_ArrayStore [simp]:
  "fields (G, Xcpt ArrayStore) = []"
  by (simp add: fields_rec [OF class_ArrayStore])

lemma (in sys_classes) method_Object [simp]:
  "method (G, Object) sig =
  (if sig = (init, []) then Some (Object, snd Object_ctor) else None)"

```

```

by (auto simp add: method_rec_lemma [OF class_Object wf] Object_ctor_def)

lemma (in sys_classes) method_NullPointer [simp]:
  "method (G, Xcpt NullPointer) sig =
  (if sig = (init,[]) then Some (Xcpt NullPointer, snd Default_ctor) else None)"
  by (auto simp add: method_rec [OF class_NullPointer] Default_ctor_def)

lemma (in sys_classes) method_ClassCast [simp]:
  "method (G, Xcpt ClassCast) sig =
  (if sig = (init,[]) then Some (Xcpt ClassCast, snd Default_ctor) else None)"
  by (auto simp add: method_rec [OF class_ClassCast] Default_ctor_def)

lemma (in sys_classes) method_OutOfMemory [simp]:
  "method (G, Xcpt OutOfMemory) sig =
  (if sig = (init,[]) then Some (Xcpt OutOfMemory, snd Default_ctor) else None)"
  by (auto simp add: method_rec [OF class_OutOfMemory] Default_ctor_def)

lemma (in sys_classes) method_ArrayIndexOutOfBounds [simp]:
  "method (G, Xcpt ArrayIndexOutOfBounds) sig =
  (if sig = (init,[]) then Some (Xcpt ArrayIndexOutOfBounds, snd Default_ctor) else None)"

  by (auto simp add: method_rec [OF class_ArrayIndexOutOfBounds] Default_ctor_def)

lemma (in sys_classes) method_NegativeArraySize [simp]:
  "method (G, Xcpt NegativeArraySize) sig =
  (if sig = (init,[]) then Some (Xcpt NegativeArraySize, snd Default_ctor) else None)"

  by (auto simp add: method_rec [OF class_NegativeArraySize] Default_ctor_def)

lemma (in sys_classes) method_ArrayStore [simp]:
  "method (G, Xcpt ArrayStore) sig =
  (if sig = (init,[]) then Some (Xcpt ArrayStore, snd Default_ctor) else None)"
  by (auto simp add: method_rec [OF class_ArrayStore] Default_ctor_def)

lemmas sys_lemmas =
  sys_classes.fields_Object
  sys_classes.class_Object
  sys_classes.wf_syscls
  sys_classes.class_NullPointer
  sys_classes.class_OutOfMemory
  sys_classes.class_ClassCast
  sys_classes.class_ArrayStore
  sys_classes.class_NegativeArraySize
  sys_classes.class_ArrayIndexOutOfBounds
  sys_classes.fields_NullPointer
  sys_classes.fields_ClassCast
  sys_classes.fields_OutOfMemory
  sys_classes.fields_ArrayIndexOutOfBounds
  sys_classes.fields_NegativeArraySize
  sys_classes.fields_ArrayStore
  sys_classes.method_Object
  sys_classes.method_NullPointer
  sys_classes.method_ClassCast
  sys_classes.method_OutOfMemory

```

```
sys_classes.method_ArrayIndexOutOfBounds  
sys_classes.method_NegativeArraySize  
sys_classes.method_ArrayStore
```

```
end
```



### 3.8 Example for generating executable code from JVM semantics

```

theory JVMListExample = JVMSystemClasses + JVMExec:

consts
  list_nam :: cname
  test_nam :: cname
  append_name :: mname
  makelist_name :: mname
  val_nam :: vname
  next_nam :: vname

constdefs
  list_name :: cname
  "list_name == Cname list_nam"

  test_name :: cname
  "test_name == Cname test_nam"

  val_name :: vname
  "val_name == VName val_nam"

  next_name :: vname
  "next_name == VName next_nam"

append_ins :: bytecode
"append_ins ==
  [Load 0,
   Getfield next_name list_name,
   Dup,
   LitPush Null,
   Ifcmpeq 4,
   Load 1,
   Invoke list_name append_name [Class list_name],
   Return,
   Pop,
   Load 0,
   Load 1,
   Putfield next_name list_name,
   LitPush Unit,
   Return]"

list_class :: "jvm_method class"
"list_class ==
  (Object,
   [(val_name, PrimT Integer), (next_name, Class list_name)],
   [Default_ctor,
    ((append_name, [Class list_name]), PrimT Void,
     (3, 0, append_ins, [(1,2,8,Xcpt NullPointer)]))])"

make_list_ins :: bytecode
"make_list_ins ==
  [New list_name,

```

```

Dup,
Dup,
Invoke_special list_name init [],
Pop,
Store 0,
LitPush (Intg 1),
Putfield val_name list_name,
New list_name,
Dup,
Invoke_special list_name init [],
Pop,
Dup,
Store 1,
LitPush (Intg 2),
Putfield val_name list_name,
New list_name,
Dup,
Invoke_special list_name init [],
Pop,
Dup,
Store 2,
LitPush (Intg 3),
Putfield val_name list_name,
Load 0,
Load 1,
Invoke list_name append_name [Class list_name],
Load 0,
Load 2,
Invoke list_name append_name [Class list_name],
Return]"

test_class :: "jvm_method class"
"test_class ==
  (Object, [], [Default_ctor,
    ((makelist_name, []), PrimT Void, (3, 2, make_list_ins,[]))])"

E :: jvm_prog
"E == JVMSystemClasses @ [(list_name, list_class), (test_name, test_class)]"

types_code
  cnam ("string")
  vnam ("string")
  mname ("string")
  loc_ ("int")

consts_code
  "new_Addr" ("new'_addr {* %x. case x of None => True | Some y => False *}/ {* None *}/
  {* Loc *}")

  "wf" ("true?")

  "arbitrary" ("(raise ERROR)")
  "arbitrary" :: "val × val" ("{* (Unit,Unit) *}")

```





## Chapter 4

# Bytecode Verifier

## 4.1 Semilattices

**theory** *Semilat* = *While\_Combinator*:

**types** 'a ord = "'a  $\Rightarrow$  'a  $\Rightarrow$  bool"  
'a binop = "'a  $\Rightarrow$  'a  $\Rightarrow$  'a"  
'a sl = "'a set  $\times$  'a ord  $\times$  'a binop"

**consts**

@lesub :: "'a  $\Rightarrow$  'a ord  $\Rightarrow$  'a  $\Rightarrow$  bool" ("\_ /<=\_ \_\_ \_)" [50, 1000, 51] 50)  
@lesssub :: "'a  $\Rightarrow$  'a ord  $\Rightarrow$  'a  $\Rightarrow$  bool" ("\_ /<'\_\_ \_)" [50, 1000, 51] 50)

**defs**

lesub\_def: "x <=\_r y == r x y"  
lesssub\_def: "x <\_r y == x <=\_r y  $\wedge$  x  $\neq$  y"

**consts**

@plussub :: "'a  $\Rightarrow$  ('a  $\Rightarrow$  'b  $\Rightarrow$  'c)  $\Rightarrow$  'b  $\Rightarrow$  'c" ("\_ /+\_\_ \_)" [65, 1000, 66] 65)

**defs**

plussub\_def: "x +\_f y  $\equiv$  f x y"

**constdefs**

ord :: "('a  $\times$  'a) set  $\Rightarrow$  'a ord"  
"ord r  $\equiv$   $\lambda$ x y. (x,y):r"

order :: "'a ord  $\Rightarrow$  bool"  
"order r  $\equiv$  ( $\forall$ x. x <=\_r x)  $\wedge$   
( $\forall$ x y. x <=\_r y  $\wedge$  y <=\_r x  $\longrightarrow$  x=y)  $\wedge$   
( $\forall$ x y z. x <=\_r y  $\wedge$  y <=\_r z  $\longrightarrow$  x <=\_r z)"

acc :: "'a ord  $\Rightarrow$  'a set  $\Rightarrow$  bool"  
"acc r A  $\equiv$  wf{(y,x). x  $\in$  A  $\wedge$  y  $\in$  A  $\wedge$  x <\_r y}"

top :: "'a ord  $\Rightarrow$  'a  $\Rightarrow$  bool"  
"top r T  $\equiv$   $\forall$ x. x <=\_r T"

closed :: "'a set  $\Rightarrow$  'a binop  $\Rightarrow$  bool"  
"closed A f  $\equiv$   $\forall$ x  $\in$  A.  $\forall$ y  $\in$  A. x +\_f y  $\in$  A"

semilat :: "'a sl  $\Rightarrow$  bool"  
"semilat  $\equiv$   $\lambda$ (A,r,f). order r  $\wedge$  closed A f  $\wedge$  supremum A r f"

supremum :: "'a set  $\Rightarrow$  'a ord  $\Rightarrow$  'a binop  $\Rightarrow$  bool"  
"supremum A r f  $\equiv$  ( $\forall$ x  $\in$  A.  $\forall$ y  $\in$  A. x <=\_r x +\_f y)  $\wedge$   
( $\forall$ x  $\in$  A.  $\forall$ y  $\in$  A. y <=\_r x +\_f y)  $\wedge$   
( $\forall$ x  $\in$  A.  $\forall$ y  $\in$  A.  $\forall$ z  $\in$  A. x <=\_r z  $\wedge$  y <=\_r z  $\longrightarrow$  x +\_f y <=\_r z)"

is\_ub :: "('a  $\times$  'a) set  $\Rightarrow$  'a  $\Rightarrow$  'a  $\Rightarrow$  'a  $\Rightarrow$  bool"  
"is\_ub r x y u  $\equiv$  (x,u)  $\in$  r  $\wedge$  (y,u)  $\in$  r"

is\_lub :: "('a  $\times$  'a) set  $\Rightarrow$  'a  $\Rightarrow$  'a  $\Rightarrow$  'a  $\Rightarrow$  bool"  
"is\_lub r x y u  $\equiv$  is\_ub r x y u  $\wedge$  ( $\forall$ z. is\_ub r x y z  $\longrightarrow$  (u,z)  $\in$  r)"

some\_lub :: "('a  $\times$  'a) set  $\Rightarrow$  'a  $\Rightarrow$  'a  $\Rightarrow$  'a"

```
"some_lub r x y ≡ SOME z. is_lub r x y z"
```

```
locale (open) semilat =
  fixes A :: "'a set"
    and r :: "'a ord"
    and f :: "'a binop"
  assumes semilat: "semilat(A,r,f)"
```

```
lemma order_refl [simp, intro]:
  "order r ⇒ x ≤r x"
  by (simp add: order_def)
```

```
lemma order_antisym:
  "[ order r; x ≤r y; y ≤r x ] ⇒ x = y"
apply (unfold order_def)
apply (simp (no_asm_simp))
done
```

```
lemma order_trans:
  "[ order r; x ≤r y; y ≤r z ] ⇒ x ≤r z"
apply (unfold order_def)
apply blast
done
```

```
lemma order_less_irrefl [intro, simp]:
  "order r ⇒ ~ x <r x"
apply (unfold order_def lesssub_def)
apply blast
done
```

```
lemma order_less_trans:
  "[ order r; x <r y; y <r z ] ⇒ x <r z"
apply (unfold order_def lesssub_def)
apply blast
done
```

```
lemma topD [simp, intro]:
  "top r T ⇒ x ≤r T"
  by (simp add: top_def)
```

```
lemma top_le_conv [simp]:
  "[ order r; top r T ] ⇒ (T ≤r x) = (x = T)"
  by (blast intro: order_antisym)
```

```
lemma semilat_Def:
  "semilat(A,r,f) == order r ∧ closed A f ∧
    (∀x∈A. ∀y∈A. x ≤r x +f y) ∧
    (∀x∈A. ∀y∈A. y ≤r x +f y) ∧
    (∀x∈A. ∀y∈A. ∀z∈A. x ≤r z ∧ y ≤r z → x +f y ≤r z)"
apply (unfold semilat_def supremum_def split_conv [THEN eq_reflection])
apply (rule refl [THEN eq_reflection])
done
```

```
lemma (in semilat) orderI [simp, intro]:
```

```

"order r"
by (insert semilat) (simp add: semilat_Def)

lemma (in semilat) closedI [simp, intro]:
  "closed A f"
  by (insert semilat) (simp add: semilat_Def)

lemma closedD:
  "[[ closed A f; x:A; y:A ] ]  $\implies$  x +_f y : A"
  by (unfold closed_def) blast

lemma closed_UNIV [simp]: "closed UNIV f"
  by (simp add: closed_def)

lemma (in semilat) closed_f [simp, intro]:
  "[[x:A; y:A] ]  $\implies$  x +_f y : A"
  by (simp add: closedD [OF closedI])

lemma (in semilat) refl_r [intro, simp]:
  "x <=_r x"
  by simp

lemma (in semilat) antisym_r [intro?]:
  "[[ x <=_r y; y <=_r x ] ]  $\implies$  x = y"
  by (rule order_antisym) auto

lemma (in semilat) trans_r [trans, intro?]:
  "[[x <=_r y; y <=_r z] ]  $\implies$  x <=_r z"
  by (auto intro: order_trans)

lemma (in semilat) ub1 [simp, intro?]:
  "[[ x:A; y:A ] ]  $\implies$  x <=_r x +_f y"
  by (insert semilat) (unfold semilat_Def, simp)

lemma (in semilat) ub2 [simp, intro?]:
  "[[ x:A; y:A ] ]  $\implies$  y <=_r x +_f y"
  by (insert semilat) (unfold semilat_Def, simp)

lemma (in semilat) lub [simp, intro?]:
  "[[ x <=_r z; y <=_r z; x:A; y:A; z:A ] ]  $\implies$  x +_f y <=_r z"
  by (insert semilat) (unfold semilat_Def, simp)

lemma (in semilat) plus_le_conv [simp]:
  "[[ x:A; y:A; z:A ] ]  $\implies$  (x +_f y <=_r z) = (x <=_r z  $\wedge$  y <=_r z)"
  by (blast intro: ub1 ub2 lub order_trans)

lemma (in semilat) le_iff_plus_unchanged:
  "[[ x:A; y:A ] ]  $\implies$  (x <=_r y) = (x +_f y = y)"
  apply (rule iffI)
  apply (blast intro: antisym_r refl_r lub ub2)
  apply (erule subst)

```



apply simp  
done

lemma (in semilat) le\_iff\_plus\_unchanged2:  
 "[ x:A; y:A ]  $\implies$  (x  $\leq_r$  y) = (y +\_f x = y)"  
 apply (rule iffI)  
 apply (blast intro: order\_antisym lub order\_refl ub1)  
 apply (erule subst)  
 apply simp  
 done

lemma (in semilat) plus\_assoc [simp]:  
 assumes a: "a  $\in$  A" and b: "b  $\in$  A" and c: "c  $\in$  A"  
 shows "a +\_f (b +\_f c) = a +\_f b +\_f c"  
 proof -  
 from a b have ab: "a +\_f b  $\in$  A" ..  
 from this c have abc: "(a +\_f b) +\_f c  $\in$  A" ..  
 from b c have bc: "b +\_f c  $\in$  A" ..  
 from a this have abc': "a +\_f (b +\_f c)  $\in$  A" ..  
  
 show ?thesis  
 proof  
 show "a +\_f (b +\_f c)  $\leq_r$  (a +\_f b) +\_f c"  
 proof -  
 from a b have "a  $\leq_r$  a +\_f b" ..  
 also from ab c have "...  $\leq_r$  ... +\_f c" ..  
 finally have "a<": "a  $\leq_r$  (a +\_f b) +\_f c" .  
 from a b have "b  $\leq_r$  a +\_f b" ..  
 also from ab c have "...  $\leq_r$  ... +\_f c" ..  
 finally have "b<": "b  $\leq_r$  (a +\_f b) +\_f c" .  
 from ab c have "c<": "c  $\leq_r$  (a +\_f b) +\_f c" ..  
 from "b<" "c<" b c abc have "b +\_f c  $\leq_r$  (a +\_f b) +\_f c" ..  
 from "a<" this a bc abc show ?thesis ..  
 qed  
 show "(a +\_f b) +\_f c  $\leq_r$  a +\_f (b +\_f c)"  
 proof -  
 from b c have "b  $\leq_r$  b +\_f c" ..  
 also from a bc have "...  $\leq_r$  a +\_f ..." ..  
 finally have "b<": "b  $\leq_r$  a +\_f (b +\_f c)" .  
 from b c have "c  $\leq_r$  b +\_f c" ..  
 also from a bc have "...  $\leq_r$  a +\_f ..." ..  
 finally have "c<": "c  $\leq_r$  a +\_f (b +\_f c)" .  
 from a bc have "a<": "a  $\leq_r$  a +\_f (b +\_f c)" ..  
 from "a<" "b<" a b abc' have "a +\_f b  $\leq_r$  a +\_f (b +\_f c)" ..  
 from this "c<" ab c abc' show ?thesis ..  
 qed  
 qed  
 qed

lemma (in semilat) plus\_com\_lemma:  
 "[a  $\in$  A; b  $\in$  A]  $\implies$  a +\_f b  $\leq_r$  b +\_f a"  
 proof -  
 assume a: "a  $\in$  A" and b: "b  $\in$  A"

```

from b a have "a <=_r b +_f a" ..
moreover from b a have "b <=_r b +_f a" ..
moreover note a b
moreover from b a have "b +_f a ∈ A" ..
ultimately show ?thesis ..
qed

```

```

lemma (in semilat) plus_commutative:
  "[[a ∈ A; b ∈ A]] ⇒ a +_f b = b +_f a"
by (blast intro: order_antisym plus_com_lemma)

```

```

lemma is_lubD:
  "is_lub r x y u ⇒ is_ub r x y u ∧ (∀z. is_ub r x y z → (u,z):r)"
by (simp add: is_lub_def)

```

```

lemma is_ubI:
  "[[(x,u) : r; (y,u) : r]] ⇒ is_ub r x y u"
by (simp add: is_ub_def)

```

```

lemma is_ubD:
  "is_ub r x y u ⇒ (x,u) : r ∧ (y,u) : r"
by (simp add: is_ub_def)

```

```

lemma is_lub_bigger1 [iff]:
  "is_lub (r^*) x y y = ((x,y):r^*)"
apply (unfold is_lub_def is_ub_def)
apply blast
done

```

```

lemma is_lub_bigger2 [iff]:
  "is_lub (r^*) x y x = ((y,x):r^*)"
apply (unfold is_lub_def is_ub_def)
apply blast
done

```

```

lemma extend_lub:
  "[[ single_valued r; is_lub (r^*) x y u; (x',x) : r ]]
  ⇒ EX v. is_lub (r^*) x' y v"
apply (unfold is_lub_def is_ub_def)
apply (case_tac "(y,x) : r^*")
  apply (case_tac "(y,x') : r^*")
    apply blast
  apply (blast elim: converse_rtranclE dest: single_valuedD)
apply (rule exI)
apply (rule conjI)
  apply (blast intro: converse_rtrancl_into_rtrancl dest: single_valuedD)
  apply (blast intro: rtrancl_into_rtrancl converse_rtrancl_into_rtrancl
    elim: converse_rtranclE dest: single_valuedD)
done

```

```

lemma single_valued_has_lubs [rule_format]:
  "[[ single_valued r; (x,u) : r^* ]] ⇒ (∀y. (y,u) : r^* →
  (EX z. is_lub (r^*) x y z))"

```

```

apply (erule converse_rtrancl_induct)
  apply clarify
  apply (erule converse_rtrancl_induct)
    apply blast
  apply (blast intro: converse_rtrancl_into_rtrancl)
apply (blast intro: extend_lub)
done

lemma some_lub_conv:
  "[ acyclic r; is_lub (r^*) x y u ]  $\implies$  some_lub (r^*) x y = u"
apply (unfold some_lub_def is_lub_def)
apply (rule someI2)
  apply assumption
apply (blast intro: antisymD dest!: acyclic_impl_antisym_rtrancl)
done

```

```

lemma is_lub_some_lub:
  "[ single_valued r; acyclic r; (x,u):r^*; (y,u):r^* ]
 $\implies$  is_lub (r^*) x y (some_lub (r^*) x y)"
  by (fastsimp dest: single_valued_has_lubs simp add: some_lub_conv)

```

#### 4.1.1 An executable lub-finder

```

constdefs
  exec_lub :: "('a * 'a) set  $\implies$  ('a  $\implies$  'a)  $\implies$  'a binop"
"exec_lub r f x y == while ( $\lambda z. (x,z) \notin r^*$ ) f y"

lemma acyclic_single_valued_finite:
  "[acyclic r; single_valued r; (x,y)  $\in$  r^*]
 $\implies$  finite (r  $\cap$  {a. (x, a)  $\in$  r^*}  $\times$  {b. (b, y)  $\in$  r^*})"
apply (erule converse_rtrancl_induct)
  apply (rule_tac B = "{}" in finite_subset)
    apply (simp only: acyclic_def)
    apply (blast intro: rtrancl_into_trancl2 rtrancl_trancl_trancl)
  apply simp
  apply (rename_tac x x')
  apply (subgoal_tac "r  $\cap$  {a. (x,a)  $\in$  r^*}  $\times$  {b. (b,y)  $\in$  r^*} =
    insert (x,x') (r  $\cap$  {a. (x', a)  $\in$  r^*}  $\times$  {b. (b, y)  $\in$  r^*})")
  apply simp
  apply (blast intro: converse_rtrancl_into_rtrancl
    elim: converse_rtranclE dest: single_valuedD)
done

lemma exec_lub_conv:
  "[ acyclic r;  $\forall x y. (x,y) \in r \implies f x = y$ ; is_lub (r^*) x y u ]  $\implies$ 
  exec_lub r f x y = u"
apply (unfold exec_lub_def)
  apply (rule_tac P = " $\lambda z. (y,z) \in r^* \wedge (z,u) \in r^*$ " and
    r = "(r  $\cap$  {(a,b). (y,a)  $\in$  r^*  $\wedge$  (b,u)  $\in$  r^*})-1" in while_rule)
    apply (blast dest: is_lubD is_ubD)
  apply (erule conjE)
  apply (erule_tac z = u in converse_rtranclE)

```

```

    apply (blast dest: is_lubD is_ubD)
    apply (blast dest: rtrancl_into_rtrancl)
    apply (rename_tac s)
    apply (subgoal_tac "is_ub (r*) x y s")
    prefer 2 apply (simp add: is_ub_def)
    apply (subgoal_tac "(u, s) ∈ r*")
    prefer 2 apply (blast dest: is_lubD)
    apply (erule converse_rtranclE)
    apply blast
    apply (simp only: acyclic_def)
    apply (blast intro: rtrancl_into_trancl2 rtrancl_trancl_trancl)
    apply (rule finite_acyclic_wf)
    apply simp
    apply (erule acyclic_single_valued_finite)
    apply (blast intro: single_valuedI)
    apply (simp add: is_lub_def is_ub_def)
    apply simp
    apply (erule acyclic_subset)
    apply blast
    apply simp
    apply (erule conjE)
    apply (erule_tac z = u in converse_rtranclE)
    apply (blast dest: is_lubD is_ubD)
    apply (blast dest: rtrancl_into_rtrancl)
done

lemma is_lub_exec_lub:
  "[[ single_valued r; acyclic r; (x,u):r^*; (y,u):r^*; ∀ x y. (x,y) ∈ r ⟶ f x = y ]]
  ⟹ is_lub (r^* ) x y (exec_lub r f x y)"
  by (fastsimp dest: single_valued_has_lubs simp add: exec_lub_conv)

end

```

## 4.2 The Error Type

```

theory Err = Semilat:

datatype 'a err = Err | OK 'a

types 'a ebinop = "'a ⇒ 'a ⇒ 'a err"
      'a esl =    "'a set * 'a ord * 'a ebinop"

consts
  ok_val :: "'a err ⇒ 'a"
primrec
  "ok_val (OK x) = x"

constdefs
  lift :: "('a ⇒ 'b err) ⇒ ('a err ⇒ 'b err)"
  "lift f e == case e of Err ⇒ Err | OK x ⇒ f x"

  lift2 :: "('a ⇒ 'b ⇒ 'c err) ⇒ 'a err ⇒ 'b err ⇒ 'c err"
  "lift2 f e1 e2 ==
  case e1 of Err ⇒ Err
           | OK x ⇒ (case e2 of Err ⇒ Err | OK y ⇒ f x y)"

  le :: "'a ord ⇒ 'a err ord"
  "le r e1 e2 ==
  case e2 of Err ⇒ True |
           OK y ⇒ (case e1 of Err ⇒ False | OK x ⇒ x <=_r y)"

  sup :: "('a ⇒ 'b ⇒ 'c) ⇒ ('a err ⇒ 'b err ⇒ 'c err)"
  "sup f == lift2(%x y. OK(x +_f y))"

  err :: "'a set ⇒ 'a err set"
  "err A == insert Err {x . ? y:A. x = OK y}"

  esl :: "'a sl ⇒ 'a esl"
  "esl == %(A,r,f). (A,r, %x y. OK(f x y))"

  sl :: "'a esl ⇒ 'a err sl"
  "sl == %(A,r,f). (err A, le r, lift2 f)"

syntax
  err_semilat :: "'a esl ⇒ bool"
translations
  "err_semilat L" == "semilat(Err.sl L)"

consts
  strict :: "('a ⇒ 'b err) ⇒ ('a err ⇒ 'b err)"
primrec
  "strict f Err = Err"
  "strict f (OK x) = f x"

lemma strict_Some [simp]:
  "(strict f x = OK y) = (∃ z. x = OK z ∧ f z = OK y)"

```

```

    by (cases x, auto)

lemma not_Err_eq:
  "(x ≠ Err) = (∃ a. x = OK a)"
  by (cases x) auto

lemma not_OK_eq:
  "(∀ y. x ≠ OK y) = (x = Err)"
  by (cases x) auto

lemma unfold_lesub_err:
  "e1 <=_(le r) e2 == le r e1 e2"
  by (simp add: lesub_def)

lemma le_err_refl:
  "!x. x <=_r x ==> e <=_(Err.le r) e"
  apply (unfold lesub_def Err.le_def)
  apply (simp split: err.split)
  done

lemma le_err_trans [rule_format]:
  "order r ==> e1 <=_(le r) e2 ==> e2 <=_(le r) e3 ==> e1 <=_(le r) e3"
  apply (unfold unfold_lesub_err le_def)
  apply (simp split: err.split)
  apply (blast intro: order_trans)
  done

lemma le_err_antisym [rule_format]:
  "order r ==> e1 <=_(le r) e2 ==> e2 <=_(le r) e1 ==> e1=e2"
  apply (unfold unfold_lesub_err le_def)
  apply (simp split: err.split)
  apply (blast intro: order_antisym)
  done

lemma OK_le_err_OK:
  "(OK x <=_(le r) OK y) = (x <=_r y)"
  by (simp add: unfold_lesub_err le_def)

lemma order_le_err [iff]:
  "order(le r) = order r"
  apply (rule iffI)
  apply (subst order_def)
  apply (blast dest: order_antisym OK_le_err_OK [THEN iffD2]
    intro: order_trans OK_le_err_OK [THEN iffD1])
  apply (subst order_def)
  apply (blast intro: le_err_refl le_err_trans le_err_antisym
    dest: order_refl)
  done

lemma le_Err [iff]: "e <=_(le r) Err"
  by (simp add: unfold_lesub_err le_def)

lemma Err_le_conv [iff]:
  "Err <=_(le r) e = (e = Err)"

```

```

by (simp add: unfold_lesub_err le_def split: err.split)

lemma le_OK_conv [iff]:
  "e <=_(le r) OK x = (? y. e = OK y & y <=_r x)"
  by (simp add: unfold_lesub_err le_def split: err.split)

lemma OK_le_conv:
  "OK x <=_(le r) e = (e = Err | (? y. e = OK y & x <=_r y))"
  by (simp add: unfold_lesub_err le_def split: err.split)

lemma top_Err [iff]: "top (le r) Err"
  by (simp add: top_def)

lemma OK_less_conv [rule_format, iff]:
  "OK x <_(le r) e = (e=Err | (? y. e = OK y & x <_r y))"
  by (simp add: lesssub_def lesub_def le_def split: err.split)

lemma not_Err_less [rule_format, iff]:
  "~(Err <_(le r) x)"
  by (simp add: lesssub_def lesub_def le_def split: err.split)

lemma semilat_errI [intro]: includes semilat
shows "semilat(err A, Err.le r, lift2(%x y. OK(f x y)))"
apply(insert semilat)
apply (unfold semilat_Def closed_def plussub_def lesub_def
        lift2_def Err.le_def err_def)
apply (simp split: err.split)
done

lemma err_semilat_eslI_aux:
includes semilat shows "err_semilat(esl(A,r,f))"
apply (unfold sl_def esl_def)
apply (simp add: semilat_errI[OF semilat])
done

lemma err_semilat_eslI [intro, simp]:
  " $\bigwedge L. \text{semilat } L \implies \text{err\_semilat}(esl\ L)$ "
by(simp add: err_semilat_eslI_aux split_tupled_all)

lemma Err_in_err [iff]: "Err : err A"
  by (simp add: err_def)

lemma Ok_in_err [iff]: "(OK x : err A) = (x:A)"
  by (auto simp add: err_def)

lemma OK_subset_err [iff]: "(OK ' A  $\subseteq$  err B) = (A  $\subseteq$  B)"
  by (unfold err_def) blast

lemma acc_err [simp, intro!]: "acc r A  $\implies$  acc (le r) (err A)"
apply (unfold acc_def lesub_def le_def lesssub_def)
apply (simp add: wf_eq_minimal split: err.split)
apply clarify
apply (case_tac "Err : Q")
apply blast

```

```

apply (erule_tac x = "{a . OK a : Q}" in allE)
apply (case_tac "x")
  apply fast
apply blast
done

```

#### 4.2.1 lift

```

lemma lift_in_errI:
  "[[ e : err S; !x:S. e = OK x  $\longrightarrow$  f x : err S ] ]  $\implies$  lift f e : err S"
apply (unfold lift_def)
apply (simp split: err.split)
apply blast
done

```

```

lemma Err_lift2 [simp]:
  "Err +_(lift2 f) x = Err"
  by (simp add: lift2_def plussub_def)

```

```

lemma lift2_Err [simp]:
  "x +_(lift2 f) Err = Err"
  by (simp add: lift2_def plussub_def split: err.split)

```

```

lemma OK_lift2_OK [simp]:
  "OK x +_(lift2 f) OK y = x +_f y"
  by (simp add: lift2_def plussub_def split: err.split)

```

#### 4.2.2 sup

```

lemma Err_sup_Err [simp]:
  "Err +_(Err.sup f) x = Err"
  by (simp add: plussub_def Err.sup_def Err.lift2_def)

```

```

lemma Err_sup_Err2 [simp]:
  "x +_(Err.sup f) Err = Err"
  by (simp add: plussub_def Err.sup_def Err.lift2_def split: err.split)

```

```

lemma Err_sup_OK [simp]:
  "OK x +_(Err.sup f) OK y = OK(x +_f y)"
  by (simp add: plussub_def Err.sup_def Err.lift2_def)

```

```

lemma Err_sup_eq_OK_conv [iff]:
  "(Err.sup f ex ey = OK z) = (? x y. ex = OK x & ey = OK y & f x y = z)"
apply (unfold Err.sup_def lift2_def plussub_def)
apply (rule iffI)
  apply (simp split: err.split_asm)
apply clarify
apply simp
done

```

```

lemma Err_sup_eq_Err [iff]:
  "(Err.sup f ex ey = Err) = (ex=Err | ey=Err)"
apply (unfold Err.sup_def lift2_def plussub_def)
apply (simp split: err.split)

```



done

### 4.2.3 semilat (err A) (le r) f

```
lemma semilat_le_err_Err_plus [simp]:
  "[[ x: err A; semilat(err A, le r, f) ] ] ==> Err +_f x = Err"
  by (blast intro: semilat.le_iff_plus_unchanged [THEN iffD1]
      semilat.le_iff_plus_unchanged2 [THEN iffD1])
```

```
lemma semilat_le_err_plus_Err [simp]:
  "[[ x: err A; semilat(err A, le r, f) ] ] ==> x +_f Err = Err"
  by (blast intro: semilat.le_iff_plus_unchanged [THEN iffD1]
      semilat.le_iff_plus_unchanged2 [THEN iffD1])
```

```
lemma semilat_le_err_OK1:
  "[[ x:A; y:A; semilat(err A, le r, f); OK x +_f OK y = OK z ] ]
  ==> x <=_r z"
  apply (rule OK_le_err_OK [THEN iffD1])
  apply (erule subst)
  apply (simp add:semilat.ub1)
  done
```

```
lemma semilat_le_err_OK2:
  "[[ x:A; y:A; semilat(err A, le r, f); OK x +_f OK y = OK z ] ]
  ==> y <=_r z"
  apply (rule OK_le_err_OK [THEN iffD1])
  apply (erule subst)
  apply (simp add:semilat.ub2)
  done
```

```
lemma eq_order_le:
  "[[ x=y; order r ] ] ==> x <=_r y"
  apply (unfold order_def)
  apply blast
  done
```

```
lemma OK_plus_OK_eq_Err_conv [simp]:
  "[[ x:A; y:A; semilat(err A, le r, fe) ] ] ==>
  ((OK x) +_fe (OK y) = Err) = (~(? z:A. x <=_r z & y <=_r z))"
  proof -
    have plus_le_conv3: "\A x y z f r.
      [[ semilat (A,r,f); x +_f y <=_r z; x:A; y:A; z:A ] ]
      ==> x <=_r z \wedge y <=_r z"
      by (rule semilat.plus_le_conv [THEN iffD1])
    case rule_context
    thus ?thesis
    apply (rule_tac iffI)
    apply clarify
    apply (drule OK_le_err_OK [THEN iffD2])
    apply (drule OK_le_err_OK [THEN iffD2])
    apply (drule semilat.lub[of _ _ _ "OK x" _ "OK y"])
    apply assumption
    apply assumption
    apply simp
```

```

    apply simp
    apply simp
    apply simp
    apply (case_tac "(OK x) +_fe (OK y)")
    apply assumption
    apply (rename_tac z)
    apply (subgoal_tac "OK z: err A")
    apply (drule eq_order_le)
    apply (erule semilat.orderI)
    apply (blast dest: plus_le_conv3)
    apply (erule subst)
    apply (blast intro: semilat.closedI closedD)
  done
qed

```

#### 4.2.4 semilat (err(Union AS))

```

lemma all_bex_swap_lemma [iff]:
  "(!x. (? y:A. x = f y) → P x) = (!y:A. P(f y))"
  by blast

```

```

lemma closed_err_Union_lift2I:
  "[[ !A:AS. closed (err A) (lift2 f); AS ~ = {};
    !A:AS.!B:AS. A~B → (!a:A.!b:B. a +_f b = Err) ]]"
  ⇒ closed (err(Union AS)) (lift2 f)"
apply (unfold closed_def err_def)
apply simp
apply clarify
apply simp
apply fast
done

```

If  $AS = \{\}$  the thm collapses to  $order\ r \wedge closed\ \{Err\}\ f \wedge Err\ +_f\ Err = Err$  which may not hold

```

lemma err_semilat_UnionI:
  "[[ !A:AS. err_semilat(A, r, f); AS ~ = {};
    !A:AS.!B:AS. A~B → (!a:A.!b:B. ~ a <=_r b & a +_f b = Err) ]]"
  ⇒ err_semilat(Union AS, r, f)"
apply (unfold semilat_def supremum_def sl_def)
apply (simp add: closed_err_Union_lift2I)
apply (rule conjI)
  apply blast
  apply (simp add: err_def)
apply (rule conjI)
  apply clarify
  apply (rename_tac A a u B b)
  apply (case_tac "A = B")
  apply simp
  apply simp
apply (rule conjI)
  apply clarify
  apply (rename_tac A a u B b)

```

```
  apply (case_tac "A = B")
    apply simp
  apply simp
apply clarify
apply (rename_tac A ya yb B yd z C c a b)
apply (case_tac "A = B")
  apply (case_tac "A = C")
    apply simp
  apply (rotate_tac -1)
  apply simp
apply (rotate_tac -1)
apply (case_tac "B = C")
  apply simp
apply (rotate_tac -1)
apply simp
done

end
```

### 4.3 Fixed Length Lists

theory Listn = Err:

constdefs

```
list :: "nat ⇒ 'a set ⇒ 'a list set"
"list n A == {xs. length xs = n & set xs ≤ A}"
```

```
le :: "'a ord ⇒ ('a list)ord"
"le r == list_all2 (%x y. x ≤r y)"
```

```
syntax "@lesublist" :: "'a list ⇒ 'a ord ⇒ 'a list ⇒ bool"
      ("(_ /≤[_] _)" [50, 0, 51] 50)
syntax "@lessublist" :: "'a list ⇒ 'a ord ⇒ 'a list ⇒ bool"
      ("(_ /<[_] _)" [50, 0, 51] 50)
```

translations

```
"x ≤[r] y" == "x ≤(Listn.le r) y"
"x <[r] y" == "x <(Listn.le r) y"
```

constdefs

```
map2 :: "('a ⇒ 'b ⇒ 'c) ⇒ 'a list ⇒ 'b list ⇒ 'c list"
"map2 f == (%xs ys. map (split f) (zip xs ys))"
```

```
syntax "@plussublist" :: "'a list ⇒ ('a ⇒ 'b ⇒ 'c) ⇒ 'b list ⇒ 'c list"
      ("(_ /+[_] _)" [65, 0, 66] 65)
```

translations "x +<sub>[f]</sub> y" == "x +<sub>(map2 f)</sub> y"

consts coalesce :: "'a err list ⇒ 'a list err"

primrec

```
"coalesce [] = OK[]"
"coalesce (ex#exs) = Err.sup (op #) ex (coalesce exs)"
```

constdefs

```
sl :: "nat ⇒ 'a sl ⇒ 'a list sl"
"sl n == %(A,r,f). (list n A, le r, map2 f)"
```

```
sup :: "('a ⇒ 'b ⇒ 'c err) ⇒ 'a list ⇒ 'b list ⇒ 'c list err"
"sup f == %xs ys. if size xs = size ys then coalesce(xs +[f] ys) else Err"
```

```
upto_esl :: "nat ⇒ 'a esl ⇒ 'a list esl"
"upto_esl m == %(A,r,f). (Union{list n A |n. n ≤ m}, le r, sup f)"
```

lemmas [simp] = set\_update\_subsetI

lemma unfold\_lesub\_list:

```
"xs ≤[r] ys == Listn.le r xs ys"
by (simp add: lesub_def)
```

lemma Nil\_le\_conv [iff]:

```
"([ ] ≤[r] ys) = (ys = [ ])"
```

apply (unfold lesub\_def Listn.le\_def)

apply simp

done

```

lemma Cons_notle_Nil [iff]:
  "~ x#xs <=[r] []"
apply (unfold lesub_def Listn.le_def)
apply simp
done

lemma Cons_le_Cons [iff]:
  "x#xs <=[r] y#ys = (x <=_r y & xs <=[r] ys)"
apply (unfold lesub_def Listn.le_def)
apply simp
done

lemma Cons_less_Conss [simp]:
  "order r  $\implies$ 
  x#xs <_(Listn.le r) y#ys =
  (x <_r y & xs <=[r] ys | x = y & xs <_(Listn.le r) ys)"
apply (unfold lesssub_def)
apply blast
done

lemma list_update_le_cong:
  "[ i < size xs; xs <=[r] ys; x <=_r y ]  $\implies$  xs[i:=x] <=[r] ys[i:=y]"
apply (unfold unfold_lesub_list)
apply (unfold Listn.le_def)
apply (simp add: list_all2_conv_all_nth nth_list_update)
done

lemma le_listD:
  "[ xs <=[r] ys; p < size xs ]  $\implies$  xs!p <=_r ys!p"
apply (unfold Listn.le_def lesub_def)
apply (simp add: list_all2_conv_all_nth)
done

lemma le_list_refl:
  "!x. x <=_r x  $\implies$  xs <=[r] xs"
apply (unfold unfold_lesub_list)
apply (simp add: Listn.le_def list_all2_conv_all_nth)
done

lemma le_list_trans:
  "[ order r; xs <=[r] ys; ys <=[r] zs ]  $\implies$  xs <=[r] zs"
apply (unfold unfold_lesub_list)
apply (simp add: Listn.le_def list_all2_conv_all_nth)
apply clarify
apply simp
apply (blast intro: order_trans)
done

lemma le_list_antisym:
  "[ order r; xs <=[r] ys; ys <=[r] xs ]  $\implies$  xs = ys"
apply (unfold unfold_lesub_list)

```

```

apply (simp add: Listn.le_def list_all2_conv_all_nth)
apply (rule nth_equalityI)
  apply blast
apply clarify
apply simp
apply (blast intro: order_antisym)
done

```

```

lemma order_listI [simp, intro!]:
  "order r  $\implies$  order(Listn.le r)"
apply (subst order_def)
apply (blast intro: le_list_refl le_list_trans le_list_antisym
  dest: order_refl)
done

```

```

lemma lesub_list_impl_same_size [simp]:
  "xs <=[r] ys  $\implies$  size ys = size xs"
apply (unfold Listn.le_def lesub_def)
apply (simp add: list_all2_conv_all_nth)
done

```

```

lemma lesssub_list_impl_same_size:
  "xs <_(Listn.le r) ys  $\implies$  size ys = size xs"
apply (unfold lesssub_def)
apply auto
done

```

```

lemma le_list_appendI:
  " $\bigwedge$  b c d. a <=[r] b  $\implies$  c <=[r] d  $\implies$  a@c <=[r] b@d"
apply (induct a)
  apply simp
apply (case_tac b)
apply auto
done

```

```

lemma le_listI:
  "length a = length b  $\implies$  ( $\bigwedge$ n. n < length a  $\implies$  a!n <=_r b!n)  $\implies$  a <=[r] b"
apply (unfold lesub_def Listn.le_def)
apply (simp add: list_all2_conv_all_nth)
done

```

```

lemma listI:
  "[[ length xs = n; set xs <= A ]  $\implies$  xs : list n A"
apply (unfold list_def)
apply blast
done

```

```

lemma listE_length [simp]:
  "xs : list n A  $\implies$  length xs = n"
apply (unfold list_def)
apply blast
done

```

```

lemma less_lengthI:
  "[ xs : list n A; p < n ] ==> p < length xs"
  by simp

lemma listE_set [simp]:
  "xs : list n A ==> set xs <= A"
apply (unfold list_def)
apply blast
done

lemma list_0 [simp]:
  "list 0 A = {[]}"
apply (unfold list_def)
apply auto
done

lemma in_list_Suc_iff:
  "(xs : list (Suc n) A) = (? y:A. ? ys:list n A. xs = y#ys)"
apply (unfold list_def)
apply (case_tac "xs")
apply auto
done

lemma Cons_in_list_Suc [iff]:
  "(x#xs : list (Suc n) A) = (x:A & xs : list n A)"
apply (simp add: in_list_Suc_iff)
done

lemma list_not_empty:
  "? a. a:A ==> ? xs. xs : list n A"
apply (induct "n")
  apply simp
  apply (simp add: in_list_Suc_iff)
  apply blast
done

lemma nth_in [rule_format, simp]:
  "!i n. length xs = n -> set xs <= A -> i < n -> (xs!i) : A"
apply (induct "xs")
  apply simp
  apply (simp add: nth_Cons split: nat.split)
done

lemma listE_nth_in:
  "[ xs : list n A; i < n ] ==> (xs!i) : A"
  by auto

lemma listn_Cons_Suc [elim!]:
  "l#xs ∈ list n A ==> (∧n'. n = Suc n' ==> l ∈ A ==> xs ∈ list n' A ==> P) ==> P"
  by (cases n) auto

lemma listn_appendE [elim!]:

```

```

  "a@b ∈ list n A ⇒ (∧n1 n2. n=n1+n2 ⇒ a ∈ list n1 A ⇒ b ∈ list n2 A ⇒ P) ⇒
  P"
proof -
  have "∧n. a@b ∈ list n A ⇒ ∃n1 n2. n=n1+n2 ∧ a ∈ list n1 A ∧ b ∈ list n2 A"
    (is "∧n. ?list a n ⇒ ∃n1 n2. ?P a n n1 n2")
  proof (induct a)
    fix n assume "?list [] n"
    hence "?P [] n 0 n" by simp
    thus "∃n1 n2. ?P [] n n1 n2" by fast
  next
    fix n l ls
    assume "?list (l#ls) n"
    then obtain n' where n: "n = Suc n'" "l ∈ A" and "ls@b ∈ list n' A" by fastsimp
    assume "∧n. ls @ b ∈ list n A ⇒ ∃n1 n2. n = n1 + n2 ∧ ls ∈ list n1 A ∧ b ∈ list
    n2 A"
    hence "∃n1 n2. n' = n1 + n2 ∧ ls ∈ list n1 A ∧ b ∈ list n2 A" .
    then obtain n1 n2 where "n' = n1 + n2" "ls ∈ list n1 A" "b ∈ list n2 A" by fast
    with n have "?P (l#ls) n (n1+1) n2" by simp
    thus "∃n1 n2. ?P (l#ls) n n1 n2" by fastsimp
  qed
  moreover
  assume "a@b ∈ list n A" "∧n1 n2. n=n1+n2 ⇒ a ∈ list n1 A ⇒ b ∈ list n2 A ⇒
  P"
  ultimately
  show ?thesis by blast
qed

lemma list_appendI:
  "[a ∈ list x A; b ∈ list y A] ⇒ a @ b ∈ list (x+y) A"
  apply (unfold list_def)
  apply (simp (no_asm))
  apply blast
  done

lemma listt_update_in_list [simp, intro!]:
  "[ xs : list n A; x:A ] ⇒ xs[i := x] : list n A"
  apply (unfold list_def)
  apply simp
  done

lemma list_map [simp]:
  "(map f xs ∈ list (length xs) A) = (f ' set xs ⊆ A)"
  apply (unfold list_def)
  apply simp
  done

lemma [intro]:
  "x ∈ A ⇒ replicate n x ∈ list n A"
  by (induct n, auto)

lemma plus_list_Nil [simp]:
  "[]+[f] xs = []"
  apply (unfold plussub_def map2_def)

```



```
apply simp
done
```

```
lemma plus_list_Cons [simp]:
  "(x#xs) +[f] ys = (case ys of [] => [] | y#ys => (x +_f y)#(xs +[f] ys))"
  by (simp add: plussub_def map2_def split: list.split)
```

```
lemma length_plus_list [rule_format, simp]:
  "!ys. length(xs +[f] ys) = min(length xs) (length ys)"
apply (induct xs)
  apply simp
apply clarify
apply (simp (no_asm_simp) split: list.split)
done
```

```
lemma nth_plus_list [rule_format, simp]:
  "!xs ys i. length xs = n -> length ys = n -> i < n ->
  (xs +[f] ys)!i = (xs!i) +_f (ys!i)"
apply (induct n)
  apply simp
apply clarify
apply (case_tac xs)
  apply simp
apply (force simp add: nth_Cons split: list.split nat.split)
done
```

```
lemma (in semilat) plus_list_ub1 [rule_format]:
  "[[ set xs <= A; set ys <= A; size xs = size ys ]
  => xs <=[r] xs +[f] ys"
apply (unfold unfold_le_sub_list)
apply (simp add: Listn.le_def list_all2_conv_all_nth)
done
```

```
lemma (in semilat) plus_list_ub2:
  "[[set xs <= A; set ys <= A; size xs = size ys ]
  => ys <=[r] xs +[f] ys"
apply (unfold unfold_le_sub_list)
apply (simp add: Listn.le_def list_all2_conv_all_nth)
done
```

```
lemma (in semilat) plus_list_lub [rule_format]:
shows "!xs ys zs. set xs <= A -> set ys <= A -> set zs <= A
  -> size xs = n & size ys = n ->
  xs <=[r] zs & ys <=[r] zs -> xs +[f] ys <=[r] zs"
apply (unfold unfold_le_sub_list)
apply (simp add: Listn.le_def list_all2_conv_all_nth)
done
```

```
lemma (in semilat) list_update_incr [rule_format]:
  "x:A => set xs <= A ->
  (!i. i < size xs -> xs <=[r] xs[i := x +_f xs!i])"
apply (unfold unfold_le_sub_list)
apply (simp add: Listn.le_def list_all2_conv_all_nth)
```

```

apply (induct xs)
  apply simp
apply (simp add: in_list_Suc_iff)
apply clarify
apply (simp add: nth_Cons split: nat.split)
done

lemma acc_le_listI [intro!]:
  "[[ order r; acc r A ] => acc (Listn.le r) (list n A)]"
apply (unfold acc_def)
apply (subgoal_tac
  "wf (UN n. {(ys,xs). xs ∈ list n A & ys ∈ list n A & xs <_(Listn.le r) ys})")
  apply (erule wf_subset)
  apply (blast intro: lesssub_list_impl_same_size)
apply (rule wf_UN)
  prefer 2
  apply clarify
  apply (rename_tac m n)
  apply (case_tac "m=n")
    apply simp
  apply (rule conjI)
    apply (fast intro!: equalsOI dest: not_sym listE_length)
    apply (fast intro!: equalsOI dest: not_sym listE_length)
  apply clarify
  apply (rename_tac n)
  apply (induct_tac n)
    apply (simp add: lesssub_def cong: conj_cong)
  apply (rename_tac k)
  apply (simp add: wf_eq_minimal)
  apply (simp (no_asm) add: in_list_Suc_iff length_Suc_conv cong: conj_cong)
  apply clarify
  apply (rename_tac M m)
  apply (case_tac "∃x∈A. ∃xs∈list k A. x#xs : M")
    prefer 2
    apply (erule thin_rl)
    apply (erule thin_rl)
    apply blast
  apply (erule_tac x = "{a. ∃xs ∈ list k A. a#xs:M}" in alle)
  apply (erule impE)
    apply blast
  apply (thin_tac "∃x∈A. ∃xs∈list k A. ?P x xs")
  apply clarify
  apply (rename_tac maxA xs)
  apply (erule_tac x = "{ys. ys ∈ list (size xs) A & maxA#ys : M}" in alle)
  apply (erule impE)
    apply simp
    apply blast
  apply clarify
  apply (thin_tac "m : M")
  apply (thin_tac "maxA#xs : M")
  apply (rule bexI)
  prefer 2
  apply assumption
  apply clarify

```

```

apply simp
apply blast
done

```

```

lemma acc_upto_listI [intro!]:
  "[ order r; acc r A ] ==> acc (Listn.le r) (Union{list m A | m. m <= n})"
apply (unfold acc_def)
apply (subgoal_tac
  "wf(UN n. {(ys,xs). xs ∈ list n A & ys ∈ list n A & xs <_(Listn.le r) ys})")
apply (erule wf_subset)
apply rule
apply clarsimp
apply (drule lesssub_list_impl_same_size)
apply simp
apply blast
apply (rule wf_UN)
prefer 2
apply clarify
apply (rename_tac m n)
apply (case_tac "m=n")
  apply simp
  apply (rule conjI)
    apply (fast intro!: equalsOI dest: not_sym listE_length)
    apply (fast intro!: equalsOI dest: not_sym listE_length)
  apply clarify
  apply (rename_tac n)
  apply (induct_tac n)
    apply (simp add: lesssub_def cong: conj_cong)
  apply (rename_tac k)
  apply (simp add: wf_eq_minimal)
  apply (simp (no_asm) add: in_list_Suc_iff length_Suc_conv cong: conj_cong)
  apply clarify
  apply (rename_tac M m)
  apply (case_tac "∃x∈A. ∃xs∈list k A. x#xs : M")
    prefer 2
    apply (erule thin_rl)
    apply (erule thin_rl)
    apply blast
  apply (erule_tac x = "{a. ∃xs ∈ list k A. a#xs:M}" in allE)
  apply (erule impE)
    apply blast
  apply (thin_tac "∃x∈A. ∃xs∈list k A. ?P x xs")
  apply clarify
  apply (rename_tac maxA xs)
  apply (erule_tac x = "{ys. ys ∈ list (size xs) A & maxA#ys : M}" in allE)
  apply (erule impE)
    apply simp
    apply blast
  apply clarify
  apply (thin_tac "m : M")
  apply (thin_tac "maxA#xs : M")
  apply (rule bexI)
  prefer 2

```

```

  apply assumption
apply clarify
apply simp
apply blast
done

```

```

lemma closed_listI:
  "closed S f  $\implies$  closed (list n S) (map2 f)"
apply (unfold closed_def)
apply (induct n)
  apply simp
apply clarify
apply (simp add: in_list_Suc_iff)
apply clarify
apply simp
done

```

```

lemma Listn_sl_aux:
includes semilat shows "semilat (Listn.sl n (A,r,f))"
apply (unfold Listn.sl_def)
apply (simp (no_asm) only: semilat_Def split_conv)
apply (rule conjI)
  apply simp
apply (rule conjI)
  apply (simp only: closedI closed_listI)
apply (simp (no_asm) only: list_def)
apply (simp (no_asm_simp) add: plus_list_ub1 plus_list_ub2 plus_list_lub)
done

```

```

lemma Listn_sl: " $\bigwedge L$ . semilat L  $\implies$  semilat (Listn.sl n L)"
  by (simp add: Listn_sl_aux split_tupled_all)

```

```

lemma coalesce_in_err_list [rule_format]:
  "!xes. xes : list n (err A)  $\longrightarrow$  coalesce xes : err(list n A)"
apply (induct n)
  apply simp
apply clarify
apply (simp add: in_list_Suc_iff)
apply clarify
apply (simp (no_asm) add: plussub_def Err.sup_def lift2_def split: err.split)
apply force
done

```

```

lemma lem: " $\bigwedge x$  xs. x +_(op #) xs = x#xs"
  by (simp add: plussub_def)

```

```

lemma coalesce_eq_OK1_D [rule_format]:
  "semilat(err A, Err.le r, lift2 f)  $\implies$ 
  !xs. xs : list n A  $\longrightarrow$  (!ys. ys : list n A  $\longrightarrow$ 
  (!zs. coalesce (xs +[f] ys) = OK zs  $\longrightarrow$  xs <=[r] zs))"
apply (induct n)
  apply simp

```

```

apply clarify
apply (simp add: in_list_Suc_iff)
apply clarify
apply (simp split: err.split_asm add: lem Err.sup_def lift2_def)
apply (force simp add: semilat_le_err_OK1)
done

```

```

lemma coalesce_eq_OK2_D [rule_format]:
  "semilat(err A, Err.le r, lift2 f)  $\implies$ 
  !xs. xs : list n A  $\longrightarrow$  (!ys. ys : list n A  $\longrightarrow$ 
  (!zs. coalesce (xs +[f] ys) = OK zs  $\longrightarrow$  ys <=[r] zs))"

```

```

apply (induct n)
  apply simp
apply clarify
apply (simp add: in_list_Suc_iff)
apply clarify
apply (simp split: err.split_asm add: lem Err.sup_def lift2_def)
apply (force simp add: semilat_le_err_OK2)
done

```

```

lemma lift2_le_ub:
  "[ semilat(err A, Err.le r, lift2 f); x:A; y:A; x +_f y = OK z;
  u:A; x <=_r u; y <=_r u ]  $\implies$  z <=_r u"

```

```

apply (unfold semilat_Def plussub_def err_def)
apply (simp add: lift2_def)
apply clarify
apply (rotate_tac -3)
apply (erule thin_rl)
apply (erule thin_rl)
apply force
done

```

```

lemma coalesce_eq_OK_ub_D [rule_format]:
  "semilat(err A, Err.le r, lift2 f)  $\implies$ 
  !xs. xs : list n A  $\longrightarrow$  (!ys. ys : list n A  $\longrightarrow$ 
  (!zs us. coalesce (xs +[f] ys) = OK zs & xs <=[r] us & ys <=[r] us
  & us : list n A  $\longrightarrow$  zs <=[r] us))"

```

```

apply (induct n)
  apply simp
apply clarify
apply (simp add: in_list_Suc_iff)
apply clarify
apply (simp (no_asm_use) split: err.split_asm add: lem Err.sup_def lift2_def)
apply clarify
apply (rule conjI)
  apply (blast intro: lift2_le_ub)
apply blast
done

```

```

lemma lift2_eq_ErrD:
  "[ x +_f y = Err; semilat(err A, Err.le r, lift2 f); x:A; y:A ]
 $\implies$  ~(? u:A. x <=_r u & y <=_r u)"
  by (simp add: OK_plus_OK_eq_Err_conv [THEN iffD1])

```

```

lemma coalesce_eq_Err_D [rule_format]:
  "[ semilat(err A, Err.le r, lift2 f) ]
  ==> !xs. xs:list n A -> (!ys. ys:list n A ->
    coalesce (xs +[f] ys) = Err ->
    ~(? zs:list n A. xs <=[r] zs & ys <=[r] zs))"
apply (induct n)
  apply simp
apply clarify
apply (simp add: in_list_Suc_iff)
apply clarify
apply (simp split: err.split_asm add: lem Err.sup_def lift2_def)
  apply (blast dest: lift2_eq_ErrD)
apply blast
done

lemma closed_err_lift2_conv:
  "closed (err A) (lift2 f) = (!x:A. !y:A. x +_f y : err A)"
apply (unfold closed_def)
apply (simp add: err_def)
done

lemma closed_map2_list [rule_format]:
  "closed (err A) (lift2 f) ==>
  !xs. xs : list n A -> (!ys. ys : list n A ->
    map2 f xs ys : list n (err A))"
apply (unfold map2_def)
apply (induct n)
  apply simp
apply clarify
apply (simp add: in_list_Suc_iff)
apply clarify
apply (simp add: plussub_def closed_err_lift2_conv)
done

lemma closed_lift2_sup:
  "closed (err A) (lift2 f) ==>
  closed (err (list n A)) (lift2 (sup f))"
  by (fastsimp simp add: closed_def plussub_def sup_def lift2_def
    coalesce_in_err_list closed_map2_list
    split: err.split)

lemma err_semilat_sup:
  "err_semilat (A,r,f) ==>
  err_semilat (list n A, Listn.le r, sup f)"
apply (unfold Err.sl_def)
apply (simp only: split_conv)
apply (simp (no_asm) only: semilat_Def plussub_def)
apply (simp (no_asm_simp) only: semilat.closedI closed_lift2_sup)
apply (rule conjI)
  apply (drule semilat.orderI)
  apply simp
apply (simp (no_asm) only: unfold_lesub_err Err.le_def err_def sup_def lift2_def)
apply (simp (no_asm_simp) add: coalesce_eq_OK1_D coalesce_eq_OK2_D split: err.split)

```

```
apply (blast intro: coalesce_eq_OK_ub_D dest: coalesce_eq_Err_D)
done
```

```
lemma err_semilat_upto_esl:
  " $\wedge L. \text{err\_semitat } L \implies \text{err\_semitat}(upto\_esl\ m\ L)$ "
apply (unfold Listn.upto_esl_def)
apply (simp (no_asm_simp) only: split_tupled_all)
apply simp
apply (fastsimp intro!: err_semitat_UnionI err_semitat_sup
  dest: lesub_list_impl_same_size
  simp add: plussub_def Listn.sup_def)
done
end
```

## 4.4 Typing and Dataflow Analysis Framework

theory *Typing\_Framework* = *Listn*:

The relationship between dataflow analysis and a welltyped-instruction predicate.

types

```
's step_type = "nat ⇒ 's ⇒ (nat × 's) list"
```

constdefs

```
stable :: "'s ord ⇒ 's step_type ⇒ 's list ⇒ nat ⇒ bool"
"stable r step ss p == !(q,s'):set(step p (ss!p)). s' <=_r ss!q"
```

```
stables :: "'s ord ⇒ 's step_type ⇒ 's list ⇒ bool"
"stables r step ss == !p<size ss. stable r step ss p"
```

```
is_bcv :: "'s ord ⇒ 's ⇒ 's step_type
           ⇒ nat ⇒ 's set ⇒ ('s list ⇒ 's list) ⇒ bool"
"is_bcv r T step n A bcv == !ss : list n A.
  (!p<n. (bcv ss)!p ~ = T) =
  (? ts: list n A. ss <=[r] ts & wt_step r T step ts)"
```

```
wt_step ::
"'s ord ⇒ 's ⇒ 's step_type ⇒ 's list ⇒ bool"
"wt_step r T step ts ==
 !p<size(ts). ts!p ~ = T & stable r step ts p"
```

end



## 4.5 Products as Semilattices

```
theory Product = Err:
```

```
constdefs
```

```
le :: "'a ord ⇒ 'b ord ⇒ ('a * 'b) ord"
"le rA rB == %(a,b) (a',b'). a <=_rA a' & b <=_rB b'"

sup :: "'a ebinop ⇒ 'b ebinop ⇒ ('a * 'b) ebinop"
"sup f g == %(a1,b1)(a2,b2). Err.sup Pair (a1+_f a2) (b1+_g b2)"

esl :: "'a esl ⇒ 'b esl ⇒ ('a * 'b) esl"
"esl == %(A,rA,fA) (B,rB,fB). (A <*> B, le rA rB, sup fA fB)"

syntax "@lesubprod" :: "'a*'b ⇒ 'a ord ⇒ 'b ord ⇒ 'b ⇒ bool"
  ("_ /<='(_,_) _)" [50, 0, 0, 51] 50)
translations "p <=(rA,rB) q" == "p <=_ (Product.le rA rB) q"
```

```
lemma unfold_lesub_prod:
```

```
"p <=(rA,rB) q == le rA rB p q"
by (simp add: lesub_def)
```

```
lemma le_prod_Pair_conv [iff]:
```

```
"((a1,b1) <=(rA,rB) (a2,b2)) = (a1 <=_rA a2 & b1 <=_rB b2)"
by (simp add: lesub_def le_def)
```

```
lemma less_prod_Pair_conv:
```

```
"((a1,b1) <_ (Product.le rA rB) (a2,b2)) =
(a1 <_rA a2 & b1 <_rB b2 | a1 <=_rA a2 & b1 <_rB b2)"
apply (unfold lesssub_def)
apply simp
apply blast
done
```

```
lemma order_le_prod [iff]:
```

```
"order(Product.le rA rB) = (order rA & order rB)"
apply (unfold order_def)
apply simp
apply blast
done
```

```
lemma acc_le_prodI [intro!]:
```

```
"[ acc rA A; acc rB B ] ⇒ acc (Product.le rA rB) (A <*> B)"
apply (unfold acc_def)
apply (rule wf_subset)
  apply (erule wf_lex_prod)
  apply assumption
apply (auto simp add: lesssub_def less_prod_Pair_conv lex_prod_def)
done
```

```
lemma closed_lift2_sup:
```

```
"[ closed (err A) (lift2 f); closed (err B) (lift2 g) ] ⇒"
```

```

    closed (err(A<*>B)) (lift2(sup f g))"
  apply (unfold closed_def plussub_def lift2_def err_def sup_def)
  apply (simp split: err.split)
  apply blast
done

```

```

lemma unfold_plussub_lift2:
  "e1 +_(lift2 f) e2 == lift2 f e1 e2"
  by (simp add: plussub_def)

```

```

lemma plus_eq_Err_conv [simp]:
  "[[ x:A; y:A; semilat(err A, Err.le r, lift2 f) ] ]
  ==> (x +_f y = Err) = (~(? z:A. x <=_r z & y <=_r z))"
proof -
  have plus_le_conv2:
    "\^r f z. [[ z : err A; semilat (err A, r, f); OK x : err A; OK y : err A;
      OK x +_f OK y <=_r z ] ] ==> OK x <=_r z ^ OK y <=_r z"
    by (rule semilat.plus_le_conv [THEN iffD1])
  case rule_context
  thus ?thesis
  apply (rule_tac iffI)
  apply clarify
  apply (drule OK_le_err_OK [THEN iffD2])
  apply (drule OK_le_err_OK [THEN iffD2])
  apply (drule semilat.lub[of _ _ "OK x" _ "OK y"])
    apply assumption
    apply assumption
    apply simp
    apply simp
    apply simp
    apply simp
  apply (case_tac "x +_f y")
  apply assumption
  apply (rename_tac "z")
  apply (subgoal_tac "OK z: err A")
  apply (frule plus_le_conv2)
    apply assumption
    apply simp
    apply blast
    apply simp
  apply (blast dest: semilat.orderI order_refl)
  apply blast
  apply (erule subst)
  apply (unfold semilat_def err_def closed_def)
  apply simp
done
qed

```

```

lemma err_semilat_Product_esl:
  "\^L1 L2. [[ err_semilat L1; err_semilat L2 ] ] ==> err_semilat(Product.esl L1 L2)"
  apply (unfold esl_def Err.sl_def)
  apply (simp (no_asm_simp) only: split_tupled_all)
  apply simp

```

```
apply (simp (no_asm) only: semilat_Def)
apply (simp (no_asm_simp) only: semilat.closedI closed_lift2_sup)
apply (simp (no_asm) only: unfold_le_sub_err Err.le_def unfold_plussub_lift2 sup_def)
apply (auto elim: semilat_le_err_OK1 semilat_le_err_OK2
          simp add: lift2_def split: err.split)
apply (blast dest: semilat.orderI)
apply (blast dest: semilat.orderI)

apply (rule OK_le_err_OK [THEN iffD1])
apply (erule subst, subst OK_lift2_OK [symmetric], rule semilat.lub)
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp

apply (rule OK_le_err_OK [THEN iffD1])
apply (erule subst, subst OK_lift2_OK [symmetric], rule semilat.lub)
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
done

end
```

## 4.6 More on Semilattices

theory SemilatAlg = Typing\_Framework + Product:

constdefs

```
lesubstep_type :: "(nat × 's) list ⇒ 's ord ⇒ (nat × 's) list ⇒ bool"
                 ("(_ /<=|_| _)" [50, 0, 51] 50)
"x <=|r| y ≡ ∀(p,s) ∈ set x. ∃s'. (p,s') ∈ set y ∧ s <=_r s'"
```

consts

```
"@plusplussub" :: "'a list ⇒ ('a ⇒ 'a ⇒ 'a) ⇒ 'a ⇒ 'a" ("(_ /++'__ _)" [65, 1000, 66] 65)
```

primrec

```
"[] ++_f y = y"
"(x#xs) ++_f y = xs ++_f (x ++_f y)"
```

constdefs

```
bounded :: "'s step_type ⇒ nat ⇒ 's set ⇒ bool"
"bounded step n A ≡ ∀p<n. ∀s∈A. ∀(q,t)∈set(step p s). q<n"
```

```
pres_type :: "'s step_type ⇒ nat ⇒ 's set ⇒ bool"
"pres_type step n A == ∀s∈A. ∀p<n. ∀(q,s')∈set(step p s). s' ∈ A"
```

```
mono :: "'s ord ⇒ 's step_type ⇒ nat ⇒ 's set ⇒ bool"
"mono r step n A ==
∀s p t. s ∈ A ∧ t ∈ A ∧ p < n ∧ s <=_r t → step p s <=|r| step p t"
```

lemma pres\_typeD:

```
"[ pres_type step n A; s∈A; p<n; (q,s')∈set(step p s) ] ⇒ s' ∈ A"
by (unfold pres_type_def, blast)
```

lemma monoD:

```
"[ mono r step n A; p < n; s∈A; t ∈ A; s <=_r t ] ⇒ step p s <=|r| step p t"
by (unfold mono_def, blast)
```

lemma boundedD:

```
"[ bounded step n A; p < n; s ∈ A; (q,t) ∈ set(step p s) ] ⇒ q < n"
by (unfold bounded_def, blast)
```

lemma lesubstep\_type\_refl [simp, intro]:

```
"(∧x. x <=_r x) ⇒ x <=|r| x"
by (unfold lesubstep_type_def) auto
```

lemma lesub\_step\_typeD:

```
"a <=|r| b ⇒ (x,y) ∈ set a ⇒ ∃y'. (x, y') ∈ set b ∧ y <=_r y'"
by (unfold lesubstep_type_def) blast
```

lemma list\_update\_le\_listI [rule\_format]:

```
"set xs <= A → set ys <= A → xs <=[r] ys → p < size xs →
x <=_r ys!p → semilat(A,r,f) → x∈A →
xs[p := x ++_f xs!p] <=[r] ys"
```

```

apply (unfold Listn.le_def lesub_def semilat_def supremum_def)
apply (simp add: list_all2_conv_all_nth nth_list_update)
done

```

```

lemma plusplus_closed: includes semilat shows
  " $\bigwedge y. \llbracket \text{set } x \subseteq A; y \in A \rrbracket \implies x ++_f y \in A$ "
proof (induct x)
  show " $\bigwedge y. y \in A \implies [] ++_f y \in A$ " by simp
  fix y x xs
  assume y: "y  $\in$  A" and xs: "set (x#xs)  $\subseteq$  A"
  assume IH: " $\bigwedge y. \llbracket \text{set } xs \subseteq A; y \in A \rrbracket \implies xs ++_f y \in A$ "
  from xs obtain x: "x  $\in$  A" and "set xs  $\subseteq$  A" by simp
  from x y have "(x +_f y)  $\in$  A" ..
  with xs have "xs ++_f (x +_f y)  $\in$  A" by - (rule IH)
  thus "(x#xs) ++_f y  $\in$  A" by simp
qed

```

```

lemma (in semilat) pp_ub2:
  " $\bigwedge y. \llbracket \text{set } x \subseteq A; y \in A \rrbracket \implies y \leq_r x ++_f y$ "
proof (induct x)
  from semilat show " $\bigwedge y. y \leq_r [] ++_f y$ " by simp

  fix y a l
  assume y: "y  $\in$  A"
  assume "set (a#l)  $\subseteq$  A"
  then obtain a: "a  $\in$  A" and x: "set l  $\subseteq$  A" by simp
  assume " $\bigwedge y. \llbracket \text{set } l \subseteq A; y \in A \rrbracket \implies y \leq_r l ++_f y$ "
  hence IH: " $\bigwedge y. y \in A \implies y \leq_r l ++_f y$ " .

  from a y have "y  $\leq_r$  a +_f y" ..
  also from a y have "a +_f y  $\in$  A" ..
  hence "(a +_f y)  $\leq_r$  l ++_f (a +_f y)" by (rule IH)
  finally have "y  $\leq_r$  l ++_f (a +_f y)" .
  thus "y  $\leq_r$  (a#l) ++_f y" by simp
qed

```

```

lemma (in semilat) pp_ub1:
  shows " $\bigwedge y. \llbracket \text{set } ls \subseteq A; y \in A; x \in \text{set } ls \rrbracket \implies x \leq_r ls ++_f y$ "
proof (induct ls)
  show " $\bigwedge y. x \in \text{set } [] \implies x \leq_r [] ++_f y$ " by simp

  fix y s ls
  assume "set (s#ls)  $\subseteq$  A"
  then obtain s: "s  $\in$  A" and ls: "set ls  $\subseteq$  A" by simp
  assume y: "y  $\in$  A"

  assume
    " $\bigwedge y. \llbracket \text{set } ls \subseteq A; y \in A; x \in \text{set } ls \rrbracket \implies x \leq_r ls ++_f y$ "
  hence IH: " $\bigwedge y. x \in \text{set } ls \implies y \in A \implies x \leq_r ls ++_f y$ " .

  assume "x  $\in$  set (s#ls)"
  then obtain xls: "x = s  $\vee$  x  $\in$  set ls" by simp

```

```

moreover {
  assume xs: "x = s"
  from s y have "s <=_r s +_f y" ..
  also from s y have "s +_f y ∈ A" ..
  with ls have "(s +_f y) <=_r ls ++_f (s +_f y)" by (rule pp_ub2)
  finally have "s <=_r ls ++_f (s +_f y)" .
  with xs have "x <=_r ls ++_f (s +_f y)" by simp
}
moreover {
  assume "x ∈ set ls"
  hence "∧y. y ∈ A ⇒ x <=_r ls ++_f y" by (rule IH)
  moreover from s y have "s +_f y ∈ A" ..
  ultimately have "x <=_r ls ++_f (s +_f y)" .
}
ultimately
have "x <=_r ls ++_f (s +_f y)" by blast
thus "x <=_r (s#ls) ++_f y" by simp
qed

```

lemma (in semilat) pp\_lub:

```

  assumes "z ∈ A"
  shows
    "∧y. y ∈ A ⇒ set xs ⊆ A ⇒ ∀x ∈ set xs. x <=_r z ⇒ y <=_r z ⇒ xs ++_f y <=_r z"
proof (induct xs)
  fix y assume "y <=_r z" thus "[] ++_f y <=_r z" by simp
next
  fix y l ls assume y: "y ∈ A" and "set (l#ls) ⊆ A"
  then obtain l: "l ∈ A" and ls: "set ls ⊆ A" by auto
  assume "∀x ∈ set (l#ls). x <=_r z"
  then obtain "l <=_r z" and lsz: "∀x ∈ set ls. x <=_r z" by auto
  assume "y <=_r z" have "l +_f y <=_r z" ..
  moreover
  from l y have "l +_f y ∈ A" ..
  moreover
  assume "∧y. y ∈ A ⇒ set ls ⊆ A ⇒ ∀x ∈ set ls. x <=_r z ⇒ y <=_r z
    ⇒ ls ++_f y <=_r z"
  ultimately
  have "ls ++_f (l +_f y) <=_r z" using ls lsz by -
  thus "(l#ls) ++_f y <=_r z" by simp
qed

```

lemma ub1': includes semilat

```

shows "[[∀ (p,s) ∈ set S. s ∈ A; y ∈ A; (a,b) ∈ set S]]
  ⇒ b <=_r map snd [(p', t') ∈ S. p' = a] ++_f y"

```

proof -

```

  let "b <=_r ?map ++_f y" = ?thesis

```

```

  assume "y ∈ A"

```

```

  moreover

```

```

  assume "∀ (p,s) ∈ set S. s ∈ A"

```

```

  hence "set ?map ⊆ A" by auto

```

```
moreover
assume "(a,b) ∈ set S"
hence "b ∈ set ?map" by (induct S, auto)
ultimately
show ?thesis by - (rule pp_ub1)
qed
```

```
lemma plusplus_empty:
  "∀s'. (q, s') ∈ set S ⟶ s' ++_f ss ! q = ss ! q ⟹
  (map snd [(p', t') ∈ S. p' = q] ++_f ss ! q) = ss ! q"
apply (induct S)
apply auto
done
```

```
end
```

## 4.7 Lifting the Typing Framework to err, app, and eff

```

theory Typing_Framework_err = Typing_Framework + SemilatAlg:

constdefs
wt_err_step :: "'s ord  $\Rightarrow$  's err step_type  $\Rightarrow$  's err list  $\Rightarrow$  bool"
"wt_err_step r step ts  $\equiv$  wt_step (Err.le r) Err step ts"

wt_app_eff :: "'s ord  $\Rightarrow$  (nat  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  's step_type  $\Rightarrow$  's list  $\Rightarrow$  bool"
"wt_app_eff r app step ts  $\equiv$ 
   $\forall p < \text{size } ts. \text{app } p (ts!p) \wedge (\forall (q,t) \in \text{set } (\text{step } p (ts!p)). t \leq_r ts!q)$ "

map_snd :: "('b  $\Rightarrow$  'c)  $\Rightarrow$  ('a  $\times$  'b) list  $\Rightarrow$  ('a  $\times$  'c) list"
"map_snd f  $\equiv$  map ( $\lambda(x,y). (x, f y)$ )"

error :: "nat  $\Rightarrow$  (nat  $\times$  'a err) list"
"error n  $\equiv$  map ( $\lambda x. (x, \text{Err})$ ) [0..n]"

err_step :: "nat  $\Rightarrow$  (nat  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  's step_type  $\Rightarrow$  's err step_type"
"err_step n app step p t  $\equiv$ 
  case t of
    Err  $\Rightarrow$  error n
  | OK t'  $\Rightarrow$  if app p t' then map_snd OK (step p t') else error n"

app_mono :: "'s ord  $\Rightarrow$  (nat  $\Rightarrow$  's  $\Rightarrow$  bool)  $\Rightarrow$  nat  $\Rightarrow$  's set  $\Rightarrow$  bool"
"app_mono r app n A  $\equiv$ 
   $\forall s p t. s \in A \wedge t \in A \wedge p < n \wedge s \leq_r t \longrightarrow \text{app } p t \longrightarrow \text{app } p s$ "

lemmas err_step_defs = err_step_def map_snd_def error_def

lemma bounded_err_stepD:
  "bounded (err_step n app step) n (err A)  $\implies$ 
  p < n  $\implies$  a  $\in$  A  $\implies$  app p a  $\implies$  (q,b)  $\in$  set (step p a)  $\implies$ 
  q < n"
  apply (simp add: bounded_def err_step_def)
  apply (erule allE, erule impE, assumption)
  apply (drule_tac x = "OK a" in bspec)
  apply simp
  apply (drule bspec)
  apply (simp add: map_snd_def)
  apply fast
  apply simp
  done

lemma in_map_sndD: "(a,b)  $\in$  set (map_snd f xs)  $\implies$   $\exists b'. (a,b') \in$  set xs"

```



```

apply (induct xs)
apply (auto simp add: map_snd_def)
done

lemma bounded_err_stepI:
  " $\forall p. p < n \longrightarrow (\forall s. \text{app } p \ s \longrightarrow (\forall (q,s') \in \text{set } (\text{step } p \ s). q < n))$ "
   $\implies$  bounded (err_step n app step) n (err A)"
apply (unfold bounded_def)
apply clarify
apply (simp add: err_step_def split: err.splits)
apply (simp add: error_def)
  apply blast
apply (simp split: split_if_asm)
  apply (blast dest: in_map_sndD)
apply (simp add: error_def)
apply blast
done

lemma bounded_lift:
  "bounded step n A  $\implies$  bounded (err_step n app step) n (err A)"
  apply (unfold bounded_def err_step_def error_def)
  apply clarify
  apply (erule allE, erule impE, assumption)
  apply (case_tac s)
  apply (auto simp add: map_snd_def split: split_if_asm)
  done

lemma le_list_map_OK [simp]:
  " $\bigwedge b. \text{map OK } a \leq [\text{Err.le } r] \text{map OK } b = (a \leq [r] b)$ "
  apply (induct a)
  apply simp
  apply simp
  apply (case_tac b)
  apply simp
  apply simp
  done

lemma map_snd_lessI:
  " $x \leq [r] y \implies \text{map_snd OK } x \leq [\text{Err.le } r] \text{map_snd OK } y$ "
  apply (induct x)
  apply (unfold lesubstep_type_def map_snd_def)
  apply auto
  done

```

lemma mono\_lift:

```

"order r  $\implies$  app_mono r app n A  $\implies$  bounded (err_step n app step) n (err A)  $\implies$ 
 $\forall s p t. s \in A \wedge t \in A \wedge p < n \wedge s \leq_r t \implies$  app p t  $\implies$  step p s  $\leq_{|r|}$  step p t
 $\implies$ 
mono (Err.le r) (err_step n app step) n (err A)"
apply (unfold app_mono_def SemilatAlg.mono_def err_step_def)
apply clarify
apply (case_tac s)
  apply simp
  apply simp
  apply (case_tac t)
    apply simp
    apply clarify
    apply (simp add: lesubstep_type_def error_def)
    apply clarify
    apply (drule in_map_sndD)
    apply clarify
    apply (drule bounded_err_stepD, assumption+)
    apply (rule exI [of _ Err])
    apply simp
  apply simp
  apply (erule allE, erule allE, erule allE, erule impE)
    apply (rule conjI, assumption,
      rule conjI, assumption,
      rule conjI, assumption, assumption)
  apply (rule conjI)
  apply clarify
  apply simp
  apply (erule allE, erule allE, erule allE, erule impE)
    apply ((rule conjI, assumption)+, assumption)
  apply (erule impE, assumption)
  apply (rule map_snd_lessI, assumption)
  apply clarify
  apply (simp add: lesubstep_type_def error_def)
  apply clarify
  apply (drule in_map_sndD)
  apply clarify
  apply (drule bounded_err_stepD, assumption+)
  apply (rule exI [of _ Err])
  apply simp
done

```

lemma in\_errorD:

```

"(x,y)  $\in$  set (error n)  $\implies$  y = Err"
by (auto simp add: error_def)

```

lemma pres\_type\_lift:

```

" $\forall s \in A. \forall p. p < n \implies$  app p s  $\implies$  ( $\forall (q, s') \in$  set (step p s).  $s' \in A$ )

```

```

  => pres_type (err_step n app step) n (err A)"
apply (unfold pres_type_def err_step_def)
apply clarify
apply (case_tac b)
  apply simp
apply (case_tac s)
  apply simp
  apply (drule in_errorD)
  apply simp
apply (simp add: map_snd_def split: split_if_asm)
  apply fast
apply (drule in_errorD)
apply simp
done

```

There used to be a condition here that each instruction must have a successor. This is not needed any more, because the definition of `error` now trivially ensures that there is a successor for the critical case where `app` does not hold.

```

lemma wt_err_imp_wt_app_eff:
  assumes wt: "wt_err_step r (err_step (size ts) app step) ts"
  assumes b: "bounded (err_step (size ts) app step) (size ts) (err A)"
  assumes A: "set ts  $\subseteq$  err A"
  shows "wt_app_eff r app step (map ok_val ts)"
proof (unfold wt_app_eff_def, intro strip, rule conjI)
  fix p assume map_ok_val: "p < size (map ok_val ts)"
  hence lp: "p < size ts" by simp
  hence ts: "0 < size ts" by (cases p) auto
  hence err: "(0,Err)  $\in$  set (error (size ts))" by (simp add: error_def)

  from wt lp
  have [intro?]: " $\wedge p. p < size ts \implies ts ! p \neq Err$ "
    by (unfold wt_err_step_def wt_step_def) simp

  show app: "app p (map ok_val ts ! p)"
  proof (rule ccontr)
    from wt lp obtain s where
      OKp: "ts ! p = OK s" and
      less: " $\forall (q,t) \in set (err_step (size ts) app step p (ts!p)). t <=_{(Err.le r)} ts!q$ "
      by (unfold wt_err_step_def wt_step_def stable_def)
        (auto iff: not_Err_eq)
    assume " $\neg$  app p (map ok_val ts ! p)"
    with OKp lp have " $\neg$  app p s" by simp
    with OKp have "err_step (size ts) app step p (ts!p) = error (size ts)"
      by (simp add: err_step_def)
    with err ts obtain q where
      "(q,Err)  $\in$  set (err_step (size ts) app step p (ts!p))" and
      q: "q < size ts" by auto
    with less have "ts!q = Err" by auto
    moreover from q have "ts!q  $\neq$  Err" ..
    ultimately show False by blast
  qed
qed

```

```

show "∀(q,t)∈set(step p (map ok_val ts ! p)). t <=_r map ok_val ts ! q"
proof clarify
  fix q t assume q: "(q,t) ∈ set (step p (map ok_val ts ! p))"

  from wt lp q
  obtain s where
    OKp: "ts ! p = OK s" and
    less: "∀(q,t) ∈ set (err_step (size ts) app step p (ts!p)). t <=(Err.le r) ts!q"
  by (unfold wt_err_step_def wt_step_def stable_def)
    (auto iff: not_Err_eq)

  from map_ok_val have "ts!p ∈ set ts" by auto
  with map_ok_val A OKp have "map ok_val ts ! p ∈ A" by auto

  with b lp app q have lq: "q < size ts" by - (rule bounded_err_stepD)
  hence "ts!q ≠ Err" ..
  then obtain s' where OKq: "ts ! q = OK s'" by (auto iff: not_Err_eq)

  from lp lq OKp OKq app less q
  show "t <=_r map ok_val ts ! q"
    by (auto simp add: err_step_def map_snd_def)
qed
qed

lemma wt_app_eff_imp_wt_err:
  assumes app_eff: "wt_app_eff r app step ts"
  assumes bounded: "bounded (err_step (size ts) app step) (size ts) (err A)"
  assumes A: "set ts ⊆ A"
  shows "wt_err_step r (err_step (size ts) app step) (map OK ts)"
proof (unfold wt_err_step_def wt_step_def, intro strip, rule conjI)
  fix p assume "p < size (map OK ts)"
  hence p: "p < size ts" by simp
  thus "map OK ts ! p ≠ Err" by simp
  { fix q t
    assume q: "(q,t) ∈ set (err_step (size ts) app step p (map OK ts ! p))"
    with p app_eff obtain
      "app p (ts ! p)" "∀(q,t) ∈ set (step p (ts!p)). t <=_r ts!q"
      by (unfold wt_app_eff_def) blast
    moreover
    from A p have "(map OK ts)!p ∈ err A" by simp
    with q p bounded have "q < size ts" by - (rule boundedD)
    hence "map OK ts ! q = OK (ts!q)" by simp
    moreover
    have "p < size ts" by (rule p)
    moreover note q
    ultimately
    have "t <=(Err.le r) map OK ts ! q"
      by (auto simp add: err_step_def map_snd_def)
  }
  thus "stable (Err.le r) (err_step (size ts) app step) (map OK ts) p"
  by (unfold stable_def) blast
qed

```

**end**

## 4.8 More about Options

theory *Opt* = *Err*:

constdefs

```
le :: "'a ord  $\Rightarrow$  'a option ord"
"le r o1 o2 == case o2 of None  $\Rightarrow$  o1=None |
                    Some y  $\Rightarrow$  (case o1 of None  $\Rightarrow$  True
                                | Some x  $\Rightarrow$  x <=_r y)"
```

```
opt :: "'a set  $\Rightarrow$  'a option set"
"opt A == insert None {x . ? y:A. x = Some y}"
```

```
sup :: "'a ebinop  $\Rightarrow$  'a option ebinop"
"sup f o1 o2 ==
  case o1 of None  $\Rightarrow$  OK o2 | Some x  $\Rightarrow$  (case o2 of None  $\Rightarrow$  OK o1
    | Some y  $\Rightarrow$  (case f x y of Err  $\Rightarrow$  Err | OK z  $\Rightarrow$  OK (Some z)))"
```

```
esl :: "'a esl  $\Rightarrow$  'a option esl"
"esl == %(A,r,f). (opt A, le r, sup f)"
```

lemma *unfold\_le\_opt*:

```
"o1 <=_ (le r) o2 =
  (case o2 of None  $\Rightarrow$  o1=None |
    Some y  $\Rightarrow$  (case o1 of None  $\Rightarrow$  True | Some x  $\Rightarrow$  x <=_r y))"
apply (unfold lesub_def le_def)
apply (rule refl)
done
```

lemma *le\_opt\_refl*:

```
"order r  $\Longrightarrow$  o1 <=_ (le r) o1"
by (simp add: unfold_le_opt split: option.split)
```

lemma *le\_opt\_trans* [*rule\_format*]:

```
"order r  $\Longrightarrow$ 
  o1 <=_ (le r) o2  $\longrightarrow$  o2 <=_ (le r) o3  $\longrightarrow$  o1 <=_ (le r) o3"
apply (simp add: unfold_le_opt split: option.split)
apply (blast intro: order_trans)
done
```

lemma *le\_opt\_antisym* [*rule\_format*]:

```
"order r  $\Longrightarrow$  o1 <=_ (le r) o2  $\longrightarrow$  o2 <=_ (le r) o1  $\longrightarrow$  o1=o2"
apply (simp add: unfold_le_opt split: option.split)
apply (blast intro: order_antisym)
done
```

lemma *order\_le\_opt* [*intro!*,*simp*]:

```
"order r  $\Longrightarrow$  order (le r)"
apply (subst order_def)
apply (blast intro: le_opt_refl le_opt_trans le_opt_antisym)
done
```

lemma *None\_bot* [*iff*]:

```
"None <=_ (le r) ox"
```

```

apply (unfold lesub_def le_def)
apply (simp split: option.split)
done

```

```

lemma Some_le [iff]:
  "(Some x <=_(le r) ox) = (? y. ox = Some y & x <=_r y)"
apply (unfold lesub_def le_def)
apply (simp split: option.split)
done

```

```

lemma le_None [iff]:
  "(ox <=_(le r) None) = (ox = None)"
apply (unfold lesub_def le_def)
apply (simp split: option.split)
done

```

```

lemma OK_None_bot [iff]:
  "OK None <=_(Err.le (le r)) x"
  by (simp add: lesub_def Err.le_def le_def split: option.split err.split)

```

```

lemma sup_None1 [iff]:
  "x +_(sup f) None = OK x"
  by (simp add: plussub_def sup_def split: option.split)

```

```

lemma sup_None2 [iff]:
  "None +_(sup f) x = OK x"
  by (simp add: plussub_def sup_def split: option.split)

```

```

lemma None_in_opt [iff]:
  "None : opt A"
  by (simp add: opt_def)

```

```

lemma Some_in_opt [iff]:
  "(Some x : opt A) = (x:A)"
apply (unfold opt_def)
apply auto
done

```

```

lemma semilat_opt [intro, simp]:
  " $\bigwedge L. \text{err\_semilat } L \implies \text{err\_semilat } (\text{Opt.esl } L)$ "
proof (unfold Opt.esl_def Err.sl_def, simp add: split_tupled_all)

```

```

  fix A r f
  assume s: "semilat (err A, Err.le r, lift2 f)"

```

```

  let ?A0 = "err A"
  let ?r0 = "Err.le r"
  let ?f0 = "lift2 f"

```

```

  from s
  obtain

```

```

ord: "order ?r0" and
clo: "closed ?A0 ?f0" and
ub1: " $\forall x \in ?A0. \forall y \in ?A0. x \leq_{?r0} x +_{?f0} y$ " and
ub2: " $\forall x \in ?A0. \forall y \in ?A0. y \leq_{?r0} x +_{?f0} y$ " and
lub: " $\forall x \in ?A0. \forall y \in ?A0. \forall z \in ?A0. x \leq_{?r0} z \wedge y \leq_{?r0} z \longrightarrow x +_{?f0} y \leq_{?r0} z$ "
by (unfold semilat_def supremum_def) simp

```

```

let ?A = "err (opt A)"
let ?r = "Err.le (Opt.le r)"
let ?f = "lift2 (Opt.sup f)"

```

```

from ord
have "order ?r"
  by simp

```

moreover

```

have "closed ?A ?f"
proof (unfold closed_def, intro strip)
  fix x y
  assume x: "x : ?A"
  assume y: "y : ?A"

```

```

{ fix a b
  assume ab: "x = OK a" "y = OK b"

```

```

  with x
  have a: " $\bigwedge c. a = \text{Some } c \implies c : A$ "
    by (clarsimp simp add: opt_def)

```

```

  from ab y
  have b: " $\bigwedge d. b = \text{Some } d \implies d : A$ "
    by (clarsimp simp add: opt_def)

```

```

{ fix c d assume "a = Some c" "b = Some d"
  with ab x y
  have "c:A & d:A"
    by (simp add: err_def opt_def Bex_def)
  with clo
  have "f c d : err A"
    by (simp add: closed_def plussub_def err_def lift2_def)
  moreover
  fix z assume "f c d = OK z"
  ultimately
  have "z : A" by simp
} note f_closed = this

```

```

have "sup f a b : ?A"
proof (cases a)
  case None
  thus ?thesis
    by (simp add: sup_def opt_def) (cases b, simp, simp add: b Bex_def)
next
  case Some

```



```

      thus ?thesis
        by (auto simp add: sup_def opt_def Bex_def a b f_closed split: err.split option.split)
    qed
  }

  thus "x +_?f y : ?A"
    by (simp add: plussub_def lift2_def split: err.split)
qed

moreover

{ fix a b c
  assume "a ∈ opt A" "b ∈ opt A" "a +_(sup f) b = OK c"
  moreover
  from ord have "order r" by simp
  moreover
  { fix x y z
    assume "x ∈ A" "y ∈ A"
    hence "OK x ∈ err A ∧ OK y ∈ err A" by simp
    with ub1 ub2
    have "(OK x) <=_(Err.le r) (OK x) +_(lift2 f) (OK y) ∧
          (OK y) <=_(Err.le r) (OK x) +_(lift2 f) (OK y)"
      by blast
    moreover
    assume "x +_f y = OK z"
    ultimately
    have "x <=_r z ∧ y <=_r z"
      by (auto simp add: plussub_def lift2_def Err.le_def lesub_def)
  }
  ultimately
  have "a <=_(le r) c ∧ b <=_(le r) c"
    by (auto simp add: sup_def le_def lesub_def plussub_def
      dest: order_refl split: option.splits err.splits)
}

hence "(∀x∈?A. ∀y∈?A. x <=_?r x +_?f y) ∧ (∀x∈?A. ∀y∈?A. y <=_?r x +_?f y)"
  by (auto simp add: lesub_def plussub_def Err.le_def lift2_def split: err.split)

moreover

have "∀x∈?A. ∀y∈?A. ∀z∈?A. x <=_?r z ∧ y <=_?r z ⟶ x +_?f y <=_?r z"
proof (intro strip, elim conjE)
  fix x y z
  assume xyz: "x : ?A" "y : ?A" "z : ?A"
  assume xz: "x <=_?r z"
  assume yz: "y <=_?r z"

  { fix a b c
    assume ok: "x = OK a" "y = OK b" "z = OK c"

    { fix d e g
      assume some: "a = Some d" "b = Some e" "c = Some g"

      with ok xyz

```

```

obtain "OK d:err A" "OK e:err A" "OK g:err A"
  by simp
with lub
have "[ (OK d) <=_ (Err.le r) (OK g); (OK e) <=_ (Err.le r) (OK g) ]"
  => "(OK d) +_(lift2 f) (OK e) <=_ (Err.le r) (OK g)"
  by blast
hence "[ d <=_r g; e <=_r g ] => ∃y. d +_f e = OK y ∧ y <=_r g"
  by simp

with ok some xyz xz yz
have "x +_?f y <=_?r z"
  by (auto simp add: sup_def le_def lesub_def lift2_def plussub_def Err.le_def)
} note this [intro!]

from ok xyz xz yz
have "x +_?f y <=_?r z"
  by - (cases a, simp, cases b, simp, cases c, simp, blast)
}

with xyz xz yz
show "x +_?f y <=_?r z"
  by - (cases x, simp, cases y, simp, cases z, simp+)
qed

ultimately

show "semilat (?A,?r,?f)"
  by (unfold semilat_def supremum_def) simp
qed

lemma top_le_opt_Some [iff]:
  "top (le r) (Some T) = top r T"
apply (unfold top_def)
apply (rule iffI)
  apply blast
  apply (rule allI)
  apply (case_tac "x")
  apply simp+
done

lemma Top_le_conv:
  "[ order r; top r T ] => (T <=_r x) = (x = T)"
apply (unfold top_def)
apply (blast intro: order_antisym)
done

lemma acc_le_optI [intro!]:
  "acc r A => acc (le r) (opt A)"
apply (unfold acc_def lesub_def le_def lesssub_def)
apply (simp add: wf_eq_minimal split: option.split)
apply clarify
apply (case_tac "? a. Some a : Q")
  apply (erule_tac x = "{a . Some a : Q}" in allE)

```

```

  apply blast
apply (case_tac "x")
  apply blast
apply blast
done

```

```

lemma option_map_in_optionI:
  "[[ ox : opt S; !x:S. ox = Some x  $\longrightarrow$  f x : S ]]
   $\implies$  option_map f ox : opt S"
apply (unfold option_map_def)
apply (simp split: option.split)
apply blast
done

```

```
end
```

Power Sets are Semilattices `theory SetSemilat = Semilat + Err:`

In fact, arbitrary sets are semilattices, as is trivial to see:

```

lemma "semilat (UNIV, op  $\subseteq$ , op  $\cup$ )"
  by (unfold semilat_def supremum_def order_def plussub_def lesub_def) auto

```

Unfortunately, arbitrary sets do not satisfy the ascending chain condition `acc`, so we work with finite power sets.

```

constdefs
  sl :: "'a set  $\Rightarrow$  'a set err sl"
  "sl A  $\equiv$  Err.sl (esl (Pow A, op  $\subseteq$ , op  $\cup$ ))"

```

The semilattice property is still easy:

```

lemma semilat_set: "semilat (sl A)"
proof -
  have "semilat (Pow A, op  $\subseteq$ , op  $\cup$ )"
    by (unfold semilat_def supremum_def closed_def
          order_def lesub_def plussub_def) auto
  thus ?thesis by (unfold sl_def) fast
qed

```

The ascending chain condition holds when `A` is finite, because then the size of `A` is an upper bound to the length of ascending chains.

```

lemma acc_set: "finite A  $\implies$  acc (op  $\subseteq$ ) (Pow A)"
  apply (unfold acc_def lesssub_def lesub_def)
  apply (simp add: psubset_eq [symmetric])
  apply (rule wf_measure [of " $\lambda x$ . card A - card x", THEN wf_subset])
  apply (simp add: measure_def inv_image_def)
  apply clarify
  apply (frule finite_subset, assumption)
  apply (frule finite_subset) back
  apply assumption
  apply (frule psubset_card_mono, assumption)
  apply (frule card_mono, assumption)
  apply (drule card_mono)

```

124

```
    apply assumption back
    apply arith
done
```

end

## 4.9 The Java Type System as Semilattice

```
theory JType = WellForm + SetSemilat + JVMState:
```

```
consts
```

```
  boundedRA :: "nat × ty ⇒ bool"
  recdef boundedRA "{}"
    "boundedRA (mpc, RA pc) = (pc < mpc)"
    "boundedRA v = True"
```

```
constdefs
```

```
  init_tys :: "'c prog ⇒ nat ⇒ init_ty set"
  "init_tys G maxpc ≡ {Init T          | T. is_type G T ∧ boundedRA (maxpc+1, T) ∧ dim T ≤
max_dim ∧ (isArray T → noRA T)} ∪
    {UnInit C pc | C pc. is_class G C ∧ pc < maxpc} ∪
    {PartInit C  | C. is_class G C}"
```

```
  init_le :: "'c prog ⇒ init_ty ord" ("_ |- _ <=i _" [71,71,71] 70)
  "G |- a <=i b ≡ case a of
    Init t1 ⇒ (case b of Init t2 ⇒ G ⊢ t1 ≤ t2
      | UnInit c n ⇒ False
      | PartInit c ⇒ False)
  | UnInit c1 n1 ⇒ a = b
  | PartInit c1 ⇒ a = b"
```

```
lemma boundedRA_def:
```

```
  "boundedRA (mpc, T) =
  ((∃ pc. T = RA pc ∧ pc < mpc) ∨
  T ∈ RefT ' UNIV ∨
  T ∈ PrimT ' {Void, Integer, Boolean})"
  apply (cases T)
  apply auto
  apply (case_tac prim_ty)
  apply auto
  done
```

```
lemma finite_classes:
```

```
  "finite {C. is_class G C}"
  proof -
    have "{C. is_class G C} = dom (map_of G)"
      by (auto simp add: is_class_def class_def dom_def)
    thus ?thesis by (simp add: finite_dom_map_of)
  qed
```

```
lemma dim_0:
```

```
  "{T. dim T = 0} =
  {Class C | C. True} ∪ {NT} ∪ PrimT ' {Void, Integer, Boolean} ∪ {RA pc | pc. True}"
  apply simp
  apply auto
```

```

    apply (case_tac x)
    apply auto
    apply (case_tac prim_ty)
    apply auto
    apply (case_tac ref_ty)
    apply auto
  done

lemma dim_Suc:
  "(dim T = Suc n) = ( $\exists T'$ . T = T'. []  $\wedge$  dim T' = n)"
  apply (cases T)
  apply simp
  apply (case_tac ref_ty)
  apply auto
  done

lemma boundedRAI1:
  " $\neg$ is_RA T  $\implies$  boundedRA (n,T)"
  apply (cases T)
  apply auto
  apply (case_tac prim_ty)
  apply auto
  done

lemma noRA_isnotRA:
  "noRA T  $\implies$   $\neg$ is_RA T"
  by (cases T, auto)

lemma boundedRAI2:
  "noRA T  $\implies$  boundedRA (n,T)"
  by (simp add: boundedRAI1 noRA_isnotRA)

lemma not_isArray:
  "dim T = 0 = ( $\neg$ isArray T)"
  apply (cases T)
  apply auto
  apply (case_tac ref_ty)
  apply auto
  apply (case_tac ref_ty)
  apply auto
  done

lemma isArray_def:
  "isArray T = ( $\exists T'$ . T = T'. [])"
  apply (cases T)
  apply auto
  apply (case_tac ref_ty)
  apply auto
  done

lemma finite_types:
  "finite {T. is_type G T  $\wedge$  boundedRA (mpc+1,T)  $\wedge$  dim T  $\leq$  n  $\wedge$  (isArray T  $\longrightarrow$  noRA T)}"

```

```

(is "finite (?set n)")
proof (induct n)
  note finite_classes [simp, intro]
  note finite_imageI [simp,intro]

  have "?set 0 =
    {T. dim T = 0} ∩ {T. is_type G T ∧ boundedRA (mpc+1,T)}"
    by (auto simp add: not_isArray)
  also have "{T. dim T = 0} =
    {Class C|C. True} ∪ {NT} ∪ PrimT ' {Void,Integer,Boolean} ∪ {RA pc|pc. True}"
    by (rule dim_0)
  also have "... ∩ {T. is_type G T ∧ boundedRA (mpc+1,T)} =
    {Class C|C. is_class G C} ∪ {NT} ∪ PrimT ' {Void,Integer,Boolean} ∪ {RA pc|pc. pc
  < mpc+1}"
    by auto
  finally have "?set 0 = ..." .
  moreover
  have "{Class C|C. is_class G C} = RefT ' ClassT ' {C. is_class G C}" by auto
  hence "finite {Class C|C. is_class G C}" by simp
  moreover
  have "∀n. {RA pc|pc. pc < Suc n} = {RA n} ∪ {RA pc|pc. pc < n}" by auto
  hence "finite {RA pc|pc. pc < mpc+1}" by (induct mpc) auto
  ultimately
  show "finite (?set 0)" by simp
next
case (Suc n)
have "?set (Suc n) ⊆ ?set n ∪ (RefT o ArrayT) ' ?set n"
proof
  fix T assume "T ∈ ?set (Suc n)"
  then obtain T:
    "is_type G T" "boundedRA (Suc mpc, T)" "dim T ≤ Suc n" "isArray T → noRA T"
    by simp
  { assume "isArray T"
    then obtain T' where "T = T'.[]" by (clarsimp simp add: isArray_def)
    with T have "T ∈ (RefT o ArrayT) ' ?set n" by (auto simp add: boundedRAI2)
  } moreover {
    assume "¬isArray T"
    with T have "T ∈ ?set n" by (auto simp add: not_isArray [symmetric])
  } ultimately
  show "T ∈ ?set n ∪ (RefT o ArrayT) ' ?set n" by (cases) auto
qed
thus "finite (?set (Suc n))" using Suc
  by (blast intro: finite_subset finite_UnI finite_imageI)
qed

```

lemma finite\_init\_tys: "finite (init\_tys G mpc)"

proof -

```

  note finite_classes [simp, intro]
  note finite_imageI [simp,intro]

```

```

let "?P T" =

```

```

  "is_type G T ∧ boundedRA (mpc+1,T) ∧ dim T ≤ max_dim ∧ (isArray T → noRA T)"

```

```

have "{Init T/T. ?P T} = Init ' {T. ?P T}" by auto
moreover
have "finite {T. ?P T}" by (rule finite_types)
ultimately
have "finite {Init T/T. ?P T}" by simp
moreover {
  have " $\forall n. \{pc. pc < Suc\ n\} = \{n\} \cup \{pc. pc < n\}$ " by auto
  hence "finite {pc. pc < mpc}" by (induct mpc) auto
  moreover
  have "{UnInit C pc | C pc. is_class G C  $\wedge$  pc < mpc} =
    ( $\lambda(C,pc). UnInit\ C\ pc$ ) ' ({C. is_class G C}  $\times$  {pc. pc < mpc})"
    by auto
  ultimately
  have "finite {UnInit C pc | C pc. is_class G C  $\wedge$  pc < mpc}" by simp
}
moreover
have "{PartInit C | C. is_class G C} = PartInit ' {C. is_class G C}" by auto
hence "finite {PartInit C | C. is_class G C}" by simp
ultimately
show ?thesis by (unfold init_tys_def) auto
qed

```

```

lemma acc_JType: "acc (op  $\subseteq$ ) (Pow (init_tys G mpc))"
  by (rule acc_set [OF finite_init_tys])

```

```

syntax (xsymbols)
  init_le :: "[code prog,init_ty,init_ty]  $\Rightarrow$  bool" ("_  $\vdash$  _  $\preceq_i$  _" [71,71,71] 70)

```

```

lemma le_Init_Init [iff]:
  "G  $\vdash$  Init t1  $\preceq_i$  Init t2 = G  $\vdash$  t1  $\preceq$  t2"
  by (unfold init_le_def) simp

```

```

lemma le_Init_UnInit [iff]:
  "init_le G (Init t) (UnInit c n) = False"
  by (unfold init_le_def) simp

```

```

lemma le_UnInit_Init [iff]:
  "init_le G (UnInit c n) (Init t) = False"
  by (unfold init_le_def) simp

```

```

lemma le_UnInit_UnInit [iff]:
  "init_le G (UnInit c1 n1) (UnInit c2 n2) = (c1 = c2  $\wedge$  n1 = n2)"
  by (unfold init_le_def) simp

```

```

lemma init_le_Init:
  "(G  $\vdash$  Init t  $\preceq_i$  x) = ( $\exists t'. x = Init\ t' \wedge G \vdash t \preceq t'$ )"
  by (simp add: init_le_def split: init_ty.splits)

```

```

lemma init_le_Init2:
  "(G  $\vdash$  x  $\preceq_i$  (Init t')) = ( $\exists t. x = Init\ t \wedge G \vdash t \preceq t'$ )"
  by (cases x, auto simp add: init_le_def)

```

```

lemma init_le_UnInit [iff]:

```



```

"(G ⊢ UnInit C pc ≤i x) = (x = UnInit C pc)"
by (cases x, auto simp add: init_le_def)

lemma init_le_UnInit2 [iff]:
  "(G ⊢ T ≤i (UnInit C pc)) = (T = UnInit C pc)"
  by (cases T, auto simp add: init_le_def)

lemma init_le_PartInit [iff]:
  "(G ⊢ PartInit C ≤i x) = (x = PartInit C)"
  by (cases x, auto simp add: init_le_def)

lemma init_le_PartInit2 [iff]:
  "G ⊢ x ≤i PartInit C = (x = PartInit C)"
  by (cases x, auto simp add: init_le_def)

lemma init_refl [iff]: "init_le G x x"
  by (auto simp add: init_le_def split: init_ty.split)

lemma init_trans [trans]:
  "[ init_le G a b; init_le G b c ] ⇒ init_le G a c"
  by (auto intro: widen_trans
      simp add: init_le_def split: init_ty.splits)

end

```

## 4.10 The JVM Type System as Semilattice

theory *JVMType* = *Opt* + *Product* + *Listn* + *JType*:

types

```

locvars_type = "init_ty err list"
opstack_type = "init_ty list"
state_type   = "opstack_type × locvars_type"
state_bool   = "state_type × bool"

address_type = "state_bool set"
state        = "address_type err"   — for Kildall
method_type  = "address_type list"  — for BVSpec

class_type   = "sig ⇒ method_type"
prog_type    = "cname ⇒ class_type"

```

constdefs

```

address_types :: "'c prog ⇒ nat ⇒ nat ⇒ nat ⇒ address_type"
"address_types G maxs maxr maxpc ≡
((Union {list n (init_tys G maxpc) |n. n ≤ maxs}) × list maxr (err (init_tys G maxpc)))
× {True,False}"

states :: "'c prog ⇒ nat ⇒ nat ⇒ nat ⇒ state set"
"states G maxs maxr maxpc ≡ err (Pow (address_types G maxs maxr maxpc))"

sl :: "'c prog ⇒ nat ⇒ nat ⇒ nat ⇒ state sl"
"sl G maxs maxr maxpc ≡ SetSemilat.sl (address_types G maxs maxr maxpc)"

```

constdefs

```

le :: "state ord"
"le ≡ Err.le (op ⊆)"

sup :: "state binop"
"sup ≡ lift2 (λx y. OK (x ∪ y))"

```

lemma *sl\_def2*:

```

"sl G mxs mxr mpc ≡ (states G mxs mxr mpc, le, sup)"
by (unfold sl_def SetSemilat.sl_def Err.sl_def
    Err.esl_def sup_def states_def le_def) simp

```

lemma *states\_def2*:

```

"states G mxs mxr mpc ≡ fst (sl G mxs mxr mpc)"
by (unfold sl_def2) simp

```

lemma *le\_def2*:

```

"le ≡ fst (snd (sl G mxs mxr mpc))"
by (unfold sl_def2) simp

```

lemma *sup\_def2*:

```

"sup ≡ snd (snd (sl G mxs mxr mpc))"
by (unfold sl_def2) simp

```

```

lemma finite_list [intro!]:
  assumes A: "finite A"
  shows "finite (list n A)"
proof (induct n)
  have "list 0 A = {[]}" by auto
  thus "finite (list 0 A)" by simp
next
  fix n assume "finite (list n A)"
  moreover
  have "list (Suc n) A = split Cons ' (A × list n A)"
    by (unfold list_def) (auto simp add: length_Suc_conv)
  ultimately
  show "finite (list (Suc n) A)" using A
    by (auto intro: finite_imageI)
qed

lemma finite_Union [intro!]:
  "finite X  $\implies$  ( $\bigwedge x. x \in X \implies$  finite x)  $\implies$  finite (Union X)"
  by (simp add: Union_def)

lemma finite_count [intro!,simp]:
  "finite {f x | x. x  $\leq$  (n:nat)}"
proof (induct n)
  show "finite {f x | x. x  $\leq$  0}" by simp
next
  fix n assume "finite {f x | x. x  $\leq$  n}"
  moreover
  have "{f x | x. x  $\leq$  Suc n} = {f (Suc n)}  $\cup$  {f x | x. x  $\leq$  n}"
  proof
    show "{f (Suc n)}  $\cup$  {f x | x. x  $\leq$  n}  $\subseteq$  {f x | x. x  $\leq$  Suc n}" by auto
    {
      fix y assume "y  $\in$  {f x | x. x  $\leq$  Suc n}"
      then obtain x where y: "y = f x" and "x  $\leq$  Suc n" by auto
      hence "x  $\leq$  n  $\vee$  x = Suc n" by auto
      with y have "y  $\in$  {f (Suc n)}  $\cup$  {f x | x. x  $\leq$  n}" by auto
    }
    thus "{f x | x. x  $\leq$  Suc n}  $\subseteq$  {f (Suc n)}  $\cup$  {f x | x. x  $\leq$  n}" ..
  qed
  ultimately
  show "finite {f x | x. x  $\leq$  Suc n}" by simp
qed

lemma finite_Union_list [intro!]:
  "finite A  $\implies$  finite (Union {list n A | n. n  $\leq$  m})"
  by auto

lemma finite_productI [intro!]:
  "finite A  $\implies$  finite B  $\implies$  finite (A  $\times$  B)"
  by (erule finite_induct) auto

lemma finite_err [iff]:
  "finite (err A) = finite A"
proof -

```

```
have "err A = insert Err (OK'A)" by (unfold err_def) auto
moreover have "inj_on OK A" by (unfold inj_on_def) auto
ultimately show ?thesis by (auto intro: finite_imageI dest: finite_imageD)
qed
```

```
lemma semilat_JVM: "semilat (sl G mxs mxr mpc)"
  by (unfold sl_def) (rule semilat_set)
```

```
lemma finite_address_types: "finite (address_types G mxs mxr mpc)"
  apply (insert finite_init_tys [of G mpc])
  apply (unfold address_types_def)
  apply auto
  done
```

```
lemma acc_JVM: "acc le (states G mxs mxr mpc)"
  apply (unfold states_def le_def)
  apply (rule acc_err)
  apply (rule acc_set)
  apply (rule finite_address_types)
  done
```

```
end
```

## 4.11 Effect of Instructions on the State Type

theory Effect = JVMType + JVExec:

types

succ\_type = "(p\_count × address\_type) list"

consts

the\_RA :: "init\_ty err ⇒ nat"

redef the\_RA "{}"

"the\_RA (OK (Init (RA pc))) = pc"

constdefs

theRA :: "nat ⇒ state\_bool ⇒ nat"

"theRA x s ≡ the\_RA (snd (fst s)!x)"

Program counter of successor instructions:

consts

succs :: "instr ⇒ p\_count ⇒ address\_type ⇒ p\_count list"

primrec

"succs (Load idx) pc s = [pc+1]"  
"succs (Store idx) pc s = [pc+1]"  
"succs (LitPush v) pc s = [pc+1]"  
"succs (Getfield F C) pc s = [pc+1]"  
"succs (Putfield F C) pc s = [pc+1]"  
"succs (New C) pc s = [pc+1]"  
"succs (Checkcast C) pc s = [pc+1]"  
"succs Pop pc s = [pc+1]"  
"succs Dup pc s = [pc+1]"  
"succs Dup\_x1 pc s = [pc+1]"  
"succs Dup\_x2 pc s = [pc+1]"  
"succs Swap pc s = [pc+1]"  
"succs IAdd pc s = [pc+1]"  
"succs (Ifcmpeq b) pc s = [pc+1, nat (int pc + b)]"  
"succs (Goto b) pc s = [nat (int pc + b)]"  
"succs Return pc s = []"  
"succs (Invoke C mn fpTs) pc s = [pc+1]"  
"succs (Invoke\_special C mn fpTs) pc s = [pc+1]"  
"succs Throw pc s = []"  
"succs (Jsr b) pc s = [nat (int pc + b)]"  
"succs (Ret x) pc s = (SOME l. set l = theRA x ' s)"  
"succs ArrLoad pc s = [pc+1]"  
"succs ArrStore pc s = [pc+1]"  
"succs ArrLength pc s = [pc+1]"  
"succs (ArrNew C) pc s = [pc+1]"

```

consts theClass :: "init_ty  $\Rightarrow$  ty"
primrec
  "theClass (PartInit C) = Class C"
  "theClass (UnInit C pc) = Class C"

```

Effect of instruction on the state type:

```

consts
eff' :: "instr  $\times$  jvm_prog  $\times$  p_count  $\times$  state_type  $\Rightarrow$  state_type"

recdef eff' "{}"
"eff' (Load idx, G, pc, (ST, LT))           = (ok_val (LT ! idx) # ST, LT)"
"eff' (Store idx, G, pc, (ts#ST, LT))       = (ST, LT[idx:= OK ts])"
"eff' (LitPush v, G, pc, (ST, LT))          = (Init (the (typeof ( $\lambda$ v. None) v))#ST, LT)"
"eff' (Getfield F C, G, pc, (oT#ST, LT))    = (Init (snd (the (field (G,C) F)))#ST, LT)"
"eff' (Putfield F C, G, pc, (vT#oT#ST, LT)) = (ST,LT)"
"eff' (New C, G, pc, (ST,LT))               = (UnInit C pc # ST, replace (OK (UnInit C
pc)) Err LT)"
"eff' (Checkcast C,G,pc,(Init (RefT t)#ST,LT)) = (Init (Class C) # ST,LT)"
"eff' (Pop, G, pc, (ts#ST,LT))              = (ST,LT)"
"eff' (Dup, G, pc, (ts#ST,LT))              = (ts#ts#ST,LT)"
"eff' (Dup_x1, G, pc, (ts1#ts2#ST,LT))      = (ts1#ts2#ts1#ST,LT)"
"eff' (Dup_x2, G, pc, (ts1#ts2#ts3#ST,LT)) = (ts1#ts2#ts3#ts1#ST,LT)"
"eff' (Swap, G, pc, (ts1#ts2#ST,LT))       = (ts2#ts1#ST,LT)"
"eff' (IAdd, G, pc, (t1#t2#ST,LT))         = (Init (PrimT Integer)#ST,LT)"
"eff' (Ifcmpeq b, G, pc, (ts1#ts2#ST,LT))  = (ST,LT)"
"eff' (Goto b, G, pc, s)                   = s"
  — Return has no successor instruction in the same method:
"eff' (Return, G, pc, s)                   = s"
  — Throw always terminates abruptly:
"eff' (Throw, G, pc, s)                     = s"
"eff' (Jsr t, G, pc, (ST,LT))               = ((Init (RA (pc+1)))#ST,LT)"
"eff' (Ret x, G, pc, s)                     = s"
"eff' (ArrLoad, G, pc, (idx#Init (T.[])#ST,LT)) = (Init T#ST,LT)"
"eff' (ArrStore, G, pc, (vT#idx#aT#ST,LT))  = (ST,LT)"
"eff' (ArrLength, G, pc, (aT#ST,LT))        = (Init (PrimT Integer)#ST,LT)"
"eff' (ArrNew T, G, pc, (len#ST,LT))        = (Init (T.[])#ST,LT)"
"eff' (Invoke C mn fpTs, G, pc, (ST,LT)) =
  (let ST' = drop (length fpTs) ST;
      X = hd ST';
      ST'' = tl ST';
      rT = fst (snd (the (method (G,C) (mn,fpTs))))
  in ((Init rT)#ST'', LT))"
"eff' (Invoke_special C mn fpTs, G, pc, (ST,LT)) =
  (let ST' = drop (length fpTs) ST;
      X = hd ST';
      N = Init (theClass X);
      ST'' = replace X N (tl ST');

```

```

    LT' = replace (OK X) (OK N) LT;
    rT  = fst (snd (the (method (G,C) (mn,fpTs))))
  in ((Init rT)#ST'', LT')"

```

For `Invoke_special` only: mark when invoking a constructor on a partly initialized class. `app` will check that we call the right constructor.

**constdefs**

```

  eff_bool :: "instr ⇒ jvm_prog ⇒ p_count ⇒ state_bool ⇒ state_bool"
  "eff_bool i G pc == λ((ST,LT),z). (eff'(i,G,pc,(ST,LT))),
  if ∃C p D. i = Invoke_special C init p ∧ ST!length p = PartInit D then True else z)"

```

For exception handling:

**consts**

```

  match_any :: "jvm_prog ⇒ p_count ⇒ exception_table ⇒ cname list"

```

**primrec**

```

  "match_any G pc [] = []"
  "match_any G pc (e#es) = (let (start_pc, end_pc, handler_pc, catch_type) = e;
    es' = match_any G pc es
  in
    if start_pc ≤ pc ∧ pc < end_pc then catch_type#es' else es')"

```

**consts**

```

  match :: "jvm_prog ⇒ cname ⇒ p_count ⇒ exception_table ⇒ cname list"

```

**primrec**

```

  "match G X pc [] = []"
  "match G X pc (e#es) =
  (if match_exception_entry G X pc e then [X] else match G X pc es)"

```

**lemma match\_some\_entry:**

```

  "match G X pc et = (if ∃e ∈ set et. match_exception_entry G X pc e then [X] else [])"
  by (induct et) auto

```

**consts**

```

  xcpt_names :: "instr × jvm_prog × p_count × exception_table ⇒ cname list"

```

**reodef xcpt\_names "{}"**

```

  "xcpt_names (Getfield F C, G, pc, et) = match G (Xcpt NullPointer) pc et"
  "xcpt_names (Putfield F C, G, pc, et) = match G (Xcpt NullPointer) pc et"
  "xcpt_names (New C, G, pc, et) = match G (Xcpt OutOfMemory) pc et"
  "xcpt_names (Checkcast C, G, pc, et) = match G (Xcpt ClassCast) pc et"
  "xcpt_names (ArrLoad, G, pc, et) = match G (Xcpt ArrayIndexOutOfBounds) pc et @
  (match G (Xcpt NullPointer) pc et)"
  "xcpt_names (ArrStore, G, pc, et) = match G (Xcpt ArrayIndexOutOfBounds) pc et @
  (match G (Xcpt NullPointer) pc et) @ (match G (Xcpt ArrayStore) pc et)"
  "xcpt_names (ArrLength, G, pc, et) = match G (Xcpt NullPointer) pc et"
  "xcpt_names (ArrNew C, G, pc, et) = match G (Xcpt OutOfMemory) pc et @ (match G
  (Xcpt NegativeArraySize) pc et)"
  "xcpt_names (Throw, G, pc, et) = match_any G pc et"
  "xcpt_names (Invoke C m p, G, pc, et) = match_any G pc et"

```

```

"xcpt_names (Invoke_special C m p, G, pc, et) = match_any G pc et"
"xcpt_names (i, G, pc, et)                = []"

```

**consts**

```
theIdx :: "instr ⇒ nat"
```

**primrec**

```

"theIdx (Load idx) = idx"
"theIdx (Store idx) = idx"
"theIdx (Ret idx)  = idx"

```

**constdefs**

```

xcpt_eff :: "instr ⇒ jvm_prog ⇒ p_count ⇒ address_type ⇒ exception_table ⇒ succ_type"
"xcpt_eff i G pc at et ≡
  map (λC. (the (match_exception_table G C pc et), (λs. ([[Init (Class C)], snd (fst
s)), snd s))'at ))
  (xcpt_names (i,G,pc,et))"

```

```

norm_eff :: "instr ⇒ jvm_prog ⇒ p_count ⇒ p_count ⇒ address_type ⇒ address_type"
"norm_eff i G pc pc' at ≡ (eff_bool i G pc) ' (if ∃idx. i = Ret idx then
  {s. s∈at ∧ pc' = theRA (theIdx i) s} else at)"

```

Putting it all together:

**constdefs**

```

eff :: "instr ⇒ jvm_prog ⇒ p_count ⇒ exception_table ⇒ address_type ⇒ succ_type"
"eff i G pc et at ≡ (map (λpc'. (pc',norm_eff i G pc pc' at)) (succs i pc at)) @ (xcpt_eff
i G pc at et)"

```

Some small helpers for direct executability

**constdefs**

```

isPrimT :: "ty ⇒ bool"
"isPrimT T ≡ case T of PrimT T' ⇒ True | RefT T' ⇒ False"

```

```

isRefT :: "ty ⇒ bool"
"isRefT T ≡ case T of PrimT T' ⇒ False | RefT T' ⇒ True"

```

**lemma isPrimT [simp]:**

```
"isPrimT T = (∃T'. T = PrimT T')" by (simp add: isPrimT_def split: ty.splits)
```

**lemma isRefT [simp]:**

```
"isRefT T = (∃T'. T = RefT T')" by (simp add: isRefT_def split: ty.splits)
```

Conditions under which eff is applicable:

**consts**

```
app' :: "instr × jvm_prog × cname × p_count × nat × ty × state_type ⇒ bool"
```

**recdef app' "{}"**

```
"app' (Load idx, G, C', pc, maxs, rT, s)
```



```

= (idx < length (snd s) ∧ (snd s) ! idx ≠ Err ∧ length (fst s) < maxs)"

"app' (Store idx, G, C', pc, maxs, rT, (ts#ST, LT))
  = (idx < length LT)"

"app' (LitPush v, G, C', pc, maxs, rT, s)
  = (length (fst s) < maxs ∧ typeof (λt. None) v ∈ {Some NT} ∪ (Some◦PrimT) '{Boolean,Void,Integer}')"

"app' (Getfield F C, G, C', pc, maxs, rT, (oT#ST, LT))
  = (is_class G C ∧ field (G,C) F ≠ None ∧ fst (the (field (G,C) F)) = C ∧
    G ⊢ oT ≤i Init (Class C))"

"app' (Putfield F C, G, C', pc, maxs, rT, (vT#oT#ST, LT))
  = (is_class G C ∧ field (G,C) F ≠ None ∧ fst (the (field (G,C) F)) = C ∧
    G ⊢ oT ≤i Init (Class C) ∧ G ⊢ vT ≤i Init (snd (the (field (G,C) F))))"

"app' (New C, G, C', pc, maxs, rT, s)
  = (is_class G C ∧ length (fst s) < maxs ∧ UnInit C pc ∉ set (fst s))"

"app' (Checkcast C, G, C', pc, maxs, rT, (Init (RefT rt)#ST,LT))
  = is_class G C"

"app' (Pop, G, C', pc, maxs, rT, (ts#ST,LT))           = True"
"app' (Dup, G, C', pc, maxs, rT, (ts#ST,LT))           = (1+length ST < maxs)"
"app' (Dup_x1, G, C', pc, maxs, rT, (ts1#ts2#ST,LT))  = (2+length ST < maxs)"
"app' (Dup_x2, G, C', pc, maxs, rT, (ts1#ts2#ts3#ST,LT)) = (3+length ST < maxs)"
"app' (Swap, G, C', pc, maxs, rT, (ts1#ts2#ST,LT))    = True"

"app' (IAdd, G, C', pc, maxs, rT, (t1#t2#ST,LT))
  = (t1 = Init (PrimT Integer) ∧ t1 = t2)"

"app' (Ifcmpeq b, G, C', pc, maxs, rT, (Init ts#Init ts'#ST,LT))
  = (0 ≤ int pc + b ∧ (isPrimT ts → ts' = ts) ∧ (isRefT ts → isRefT ts'))"

"app' (Goto b, G, C', pc, maxs, rT, s)                 = (0 ≤ int pc + b)"
"app' (Return, G, C', pc, maxs, rT, (T#ST,LT))        = (G ⊢ T ≤i Init rT)"
"app' (Throw, G, C', pc, maxs, rT, (Init T#ST,LT))    = isRefT T"
"app' (Jsr b, G, C', pc, maxs, rT, (ST,LT))           = (0 ≤ int pc + b ∧ length ST < maxs)"
"app' (Ret x, G, C', pc, maxs, rT, (ST,LT))           = (x < length LT ∧ (∃ r. LT!x=OK (Init
(RA r))))"

"app' (ArrLoad, G, C', pc, maxs, rT, (idx#Init (T.[])#ST,LT)) =
  (idx = Init (PrimT Integer))"

"app' (ArrStore, G, C', pc, maxs, rT, (T#idx#Init (T'.[])#ST,LT)) =
  (idx = Init (PrimT Integer) ∧ G ⊢ T ≤i Init T'"

"app' (ArrLength, G, C', pc, maxs, rT, (Init (T.[])#ST, LT)) = True"

```

```
"app' (ArrNew T, G, C', pc, maxs, rT, (len#ST,LT)) =
  (len = Init (PrimT Integer)  $\wedge$  is_type G T  $\wedge$  noRA T  $\wedge$  dim T+1 < max_dim)"
```

```
"app' (Invoke C mn fpTs, G, C', pc, maxs, rT, s) =
  (length fpTs < length (fst s)  $\wedge$  mn  $\neq$  init  $\wedge$ 
  (let apTs = rev (take (length fpTs) (fst s));
    X = hd (drop (length fpTs) (fst s))
  in is_class G C  $\wedge$ 
  list_all2 ( $\lambda$ aT fT. G  $\vdash$  aT  $\preceq_i$  (Init fT)) apTs fpTs  $\wedge$ 
  G  $\vdash$  X  $\preceq_i$  Init (Class C))  $\wedge$ 
  method (G,C) (mn,fpTs)  $\neq$  None)"
```

```
"app' (Invoke_special C mn fpTs, G, C', pc, maxs, rT, s) =
  (length fpTs < length (fst s)  $\wedge$  mn = init  $\wedge$ 
  (let apTs = rev (take (length fpTs) (fst s));
    X = (fst s)!length fpTs
  in is_class G C  $\wedge$ 
  list_all2 ( $\lambda$ aT fT. G  $\vdash$  aT  $\preceq_i$  (Init fT)) apTs fpTs  $\wedge$ 
  ( $\exists$ rT' b. method (G,C) (mn,fpTs) = Some (C,rT',b))  $\wedge$ 
  (( $\exists$ pc. X = UnInit C pc)  $\vee$  (X = PartInit C'  $\wedge$  G  $\vdash$  C'  $\prec_{C1}$  C))))"
```

- C' is the current class, the constructor must be called on the
- superclass (if partly initialized) or on the exact class that is
- to be constructed (if not yet initialized at all).
- In JCVm *Invoke\_special* may also call another constructor of the same
- class (C = C'  $\vee$  C = super C')

```
"app' (i,G,pc,maxs,rT,s) = False"
```

#### constdefs

```
xcpt_app :: "instr  $\Rightarrow$  jvm_prog  $\Rightarrow$  nat  $\Rightarrow$  exception_table  $\Rightarrow$  bool"
"xcpt_app i G pc et  $\equiv$   $\forall$ C $\in$ set(xcpt_names (i,G,pc,et)). is_class G C"
```

#### constdefs

```
app :: "instr  $\Rightarrow$  jvm_prog  $\Rightarrow$  cname  $\Rightarrow$  p_count  $\Rightarrow$  nat  $\Rightarrow$  nat  $\Rightarrow$  ty  $\Rightarrow$  bool  $\Rightarrow$ 
  exception_table  $\Rightarrow$  address_type  $\Rightarrow$  bool"
"app i G C' pc mxs mpc rT ini et at  $\equiv$  ( $\forall$ (s,z)  $\in$  at.
  xcpt_app i G pc et  $\wedge$ 
  app' (i,G,C',pc,mxs,rT,s)  $\wedge$ 
  (ini  $\wedge$  i = Return  $\longrightarrow$  z)  $\wedge$ 
  ( $\forall$ C m p. i = Invoke_special C m p  $\wedge$  (fst s)!length p = PartInit C'  $\longrightarrow$   $\neg$ z))  $\wedge$ 
  ( $\forall$ (pc',s')  $\in$  set (eff i G pc et at). pc' < mpc)"
```

#### lemma match\_any\_match\_table:

```
"C  $\in$  set (match_any G pc et)  $\implies$  match_exception_table G C pc et  $\neq$  None"
```

```

apply (induct et)
  apply simp
apply simp
apply clarify
apply (simp split: split_if_asm)
apply (auto simp add: match_exception_entry_def)
done

```

**lemma** match\_X\_match\_table:

```

"C ∈ set (match G X pc et) ⇒ match_exception_table G C pc et ≠ None"
apply (induct et)
  apply simp
apply (simp split: split_if_asm)
done

```

**lemma** xcpt\_names\_in\_et:

```

"C ∈ set (xcpt_names (i,G,pc,et)) ⇒
∃e ∈ set et. the (match_exception_table G C pc et) = fst (snd (snd e))"
apply (cases i)
apply (auto dest!: match_any_match_table match_X_match_table
          dest: match_exception_table_in_et)
done

```

**lemma** length\_casesE1:

```

"((xs,y),z) ∈ at ⇒
(xs = [] ⇒ P []) ⇒
(∧l. xs = [l] ⇒ P [l]) ⇒
(∧l l'. xs = [l,l'] ⇒ P [l,l']) ⇒
(∧l l' ls. xs = l#l'#ls ⇒ P (l#l'#ls))
⇒ P xs"
apply (cases xs)
apply auto
apply (rename_tac ls)
apply (case_tac ls)
apply auto
done

```

**lemmas** [simp] = app\_def xcpt\_app\_def

simp rules for app

```

lemma appNone[simp]: "app i G C' pc maxs mpc rT ini et {} = (∀(pc',s') ∈ set (eff i G
pc et {}). pc' < mpc)"
  by simp

```

**lemmas** eff\_defs [simp] = eff\_def norm\_eff\_def eff\_bool\_def xcpt\_eff\_def

**lemma** appLoad[simp]:

```

"app (Load idx) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧ (∀s ∈ at.

```

```

(∃ ST LT z. s = ((ST,LT),z) ∧ idx < length LT ∧ LT!idx ≠ Err ∧ length ST < maxs)))"

  by auto

lemma appStore[simp]:
"app (Store idx) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧ (∀ s ∈ at. ∃ ts ST LT
z. s = ((ts#ST,LT),z) ∧ idx < length LT))"
  apply auto
  apply (auto dest!: bspec elim!: length_casesE1)
  done

lemma appLitPush[simp]:
"app (LitPush v) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀ s ∈ at. ∃ ST LT z. s = ((ST,LT),z) ∧ length ST < maxs ∧
  typeof (λv. None) v ∈ {Some NT} ∪ (Some◦PrimT){Void,Boolean,Integer}))"
  by auto

lemma appGetField[simp]:
"app (Getfield F C) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀ x ∈ set (match G (Xcpt NullPointer) pc et). the (match_exception_table G x pc et)
< mpc) ∧
(∀ s ∈ at. ∃ oT vT ST LT z. s = ((oT#ST, LT),z) ∧ is_class G C ∧
field (G,C) F = Some (C,vT) ∧ G ⊢ oT ≤i (Init (Class C)) ∧
(∀ x ∈ set (match G (Xcpt NullPointer) pc et). is_class G x)))"
  apply rule
  defer
  apply (clarsimp, (drule bspec, assumption?)+)
  apply (auto elim!: length_casesE1)
  done

lemma appPutField[simp]:
"app (Putfield F C) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀ x ∈ set (match G (Xcpt NullPointer) pc et). the (match_exception_table G x pc et)
< mpc) ∧
(∀ s ∈ at. ∃ vT vT' oT ST LT z. s = ((vT#oT#ST, LT),z) ∧ is_class G C ∧
field (G,C) F = Some (C, vT') ∧
G ⊢ oT ≤i Init (Class C) ∧ G ⊢ vT ≤i Init vT' ∧
(∀ x ∈ set (match G (Xcpt NullPointer) pc et). is_class G x)))"
  apply rule
  defer
  apply (clarsimp, (drule bspec, assumption?)+)
  apply (auto elim!: length_casesE1)
  done

lemma appNew[simp]:
"app (New C) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀ x ∈ set (match G (Xcpt OutOfMemory) pc et). the (match_exception_table G x pc et)
< mpc) ∧
(∀ s ∈ at. ∃ ST LT z. s=((ST,LT),z) ∧
is_class G C ∧ length ST < maxs ∧
UnInit C pc ∉ set ST ∧
(∀ x ∈ set (match G (Xcpt OutOfMemory) pc et). is_class G x)))"

```

```

by auto

lemma appCheckcast[simp]:
"app (Checkcast C) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀x ∈ set (match G (Xcpt ClassCast) pc et). the (match_exception_table G x pc et) <
mpc) ∧
(∀s ∈ at. ∃rT ST LT z. s = ((Init (RefT rT)#ST,LT),z) ∧ is_class G C ∧
(∀x ∈ set (match G (Xcpt ClassCast) pc et). is_class G x)))"
apply rule
apply clarsimp
defer
apply clarsimp
apply (drule bspec, assumption)
apply clarsimp
apply (drule bspec, assumption)
apply clarsimp
apply (case_tac a)
apply auto
apply (case_tac aa)
apply auto
apply (case_tac ty)
apply auto
apply (case_tac aa)
apply auto
apply (case_tac ty)
apply auto
done

lemma appPop[simp]:
"app Pop G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧ (∀s ∈ at. ∃ts ST LT z. s =
((ts#ST,LT),z)))"
by auto (auto dest!: bspec elim!: length_casesE1)

lemma appDup[simp]:
"app Dup G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀s ∈ at. ∃ts ST LT z. s = ((ts#ST,LT),z) ∧ 1+length ST < maxs))"
by auto (auto dest!: bspec elim!: length_casesE1)

lemma appDup_x1[simp]:
"app Dup_x1 G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀s ∈ at. ∃ts1 ts2 ST LT z. s = ((ts1#ts2#ST,LT),z) ∧ 2+length ST < maxs))"
by auto (auto dest!: bspec elim!: length_casesE1)

lemma appDup_x2[simp]:
"app Dup_x2 G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀s ∈ at. ∃ts1 ts2 ts3 ST LT z. s = ((ts1#ts2#ts3#ST,LT),z) ∧ 3+length ST < maxs))"
apply auto
apply (auto dest!: bspec elim!: length_casesE1)
apply (case_tac ls, auto)
done

```

lemma appSwap[simp]:

```
"app Swap G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀s ∈ at. ∃ts1 ts2 ST LT z. s = ((ts1#ts2#ST,LT),z)))"
  by auto (auto dest!: bspec elim!: length_casesE1)
```

lemma appIAdd[simp]:

```
"app IAdd G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
(∀s ∈ at. ∃ST LT z. s = ((Init (PrimT Integer)#Init (PrimT Integer)#ST,LT),z)))"
  by auto (auto dest!: bspec elim!: length_casesE1)
```

lemma appIfcmpeq[simp]:

```
"app (Ifcmpeq b) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧ nat (int pc+b) < mpc
∧
(∀s ∈ at. ∃ts1 ts2 ST LT z. s = ((Init ts1#Init ts2#ST,LT),z) ∧ 0 ≤ b + int pc ∧
((∃p. ts1 = PrimT p ∧ ts2 = PrimT p) ∨
(∃r r'. ts1 = RefT r ∧ ts2 = RefT r')))))"
  apply auto
  apply (auto dest!: bspec)
  apply (case_tac aa, simp)
  apply (case_tac list)
  apply (case_tac a, simp, simp, simp)
  apply (case_tac a)
  apply simp
  apply (case_tac ab)
  apply auto
  apply (case_tac ty, auto)
  apply (case_tac ty, auto)
  apply (case_tac ty, auto)
  apply (case_tac ty, auto)
  apply (case_tac ty, auto)
  done
```

lemma appReturn[simp]:

```
"app Return G C' pc maxs mpc rT ini et at =
(∀s ∈ at. ∃T ST LT z. s = ((T#ST,LT),z) ∧ (G ⊢ T ≤i Init rT) ∧ (ini → z))"
  apply auto
  apply (auto dest!: bspec)
  apply (erule length_casesE1)
  apply auto
  apply (erule length_casesE1)
  apply auto
  done
```

lemma appGoto[simp]:

```
"app (Goto b) G C' pc maxs mpc rT ini et at =
(nat (int pc + b) < mpc ∧ (at ≠ {} → 0 ≤ int pc + b))"
  by auto
```

lemma appThrow[simp]:

```
"app Throw G C' pc maxs mpc rT ini et at =
  (( $\forall C \in \text{set (match\_any G pc et)}$ ). the (match\_exception\_table G C pc et) < mpc)  $\wedge$ 
  ( $\forall s \in \text{at}$ .  $\exists ST LT z r$ .  $s = ((\text{Init (RefT r)}\#ST,LT),z) \wedge (\forall C \in \text{set (match\_any G pc et)}$ ).
  is\_class G C)))"
  apply auto
  apply (drule bspec, assumption)
  apply clarsimp
  apply (case_tac a)
  apply auto
  apply (case_tac aa)
  apply auto
  done
```

lemma appJsrr[simp]:

```
"app (Jsrr b) G C' pc maxs mpc rT ini et at = (nat (int pc + b) < mpc  $\wedge$  ( $\forall s \in \text{at}$ .  $\exists ST
LT z$ .  $s = ((ST,LT),z) \wedge 0 \leq \text{int pc} + b \wedge \text{length ST} < \text{maxs}$ ))"
  by auto
```

lemma set\_SOME\_lists:

```
"finite s  $\implies$  set (SOME l. set l = s) = s"
  apply (erule finite_induct)
  apply simp
  apply (rule_tac a="x#(SOME l. set l = F)" in someI2)
  apply auto
  done
```

lemma appRet[simp]:

```
"finite at  $\implies$ 
  app (Ret x) G C' pc maxs mpc rT ini et at = ( $\forall s \in \text{at}$ .  $\exists ST LT z$ .  $s = ((ST,LT),z) \wedge
x < \text{length LT} \wedge (\exists r$ .  $LT!x=OK (\text{Init (RA r)}) \wedge r < \text{mpc}$ ))"
  apply (auto simp add: set_SOME_lists finite_imageI theRA_def)
  apply (drule bspec, assumption)+
  apply simp
  apply (drule bspec, assumption)+
  apply auto
  done
```

lemma appArrLoad[simp]:

```
"app ArrLoad G C' pc maxs mpc rT ini et at = (Suc pc < mpc  $\wedge$ 
  ( $\forall x \in \text{set (match G (Xcpt NullPointer) pc et)}$ ). the (match\_exception\_table G x pc et)
< mpc)  $\wedge$ 
  ( $\forall x \in \text{set (match G (Xcpt ArrayIndexOutOfBounds) pc et)}$ ). the (match\_exception\_table G
x pc et) < mpc)  $\wedge$ 
  ( $\forall s \in \text{at}$ .  $\exists T ST LT z$ .  $s = ((\text{Init (PrimT Integer)}\#\text{Init (T.[])}\#ST,LT),z) \wedge$ 
  ( $\forall x \in \text{set (match G (Xcpt NullPointer) pc et)}$ ). is\_class G x)  $\wedge$ 
  ( $\forall x \in \text{set (match G (Xcpt ArrayIndexOutOfBounds) pc et)}$ ). is\_class G x)))"
  apply auto
  apply (drule bspec, assumption)
  apply clarsimp
  apply (case_tac a)
  apply simp
```

```

apply (case_tac list)
  apply simp
apply (case_tac ab)
apply (case_tac ty)
  apply simp
apply (case_tac ref_ty)
apply auto
done

```

lemma appArrStore[simp]:

```

"app ArrStore G C' pc maxs mpc rT ini et at = (Suc pc < mpc ^
  (∀x ∈ set (match G (Xcpt NullPointer) pc et). the (match_exception_table G x pc et)
< mpc) ^
  (∀x ∈ set (match G (Xcpt ArrayIndexOutOfBounds) pc et). the (match_exception_table G
x pc et) < mpc) ^
  (∀x ∈ set (match G (Xcpt ArrayStore) pc et). the (match_exception_table G x pc et) <
mpc) ^
  (∀s ∈ at. ∃T T' ST LT z. s=((T#Init (PrimT Integer)#Init (T'.[])#ST,LT),z) ^
  G ⊢ T ≤i Init T' ^
  (∀x ∈ set (match G (Xcpt NullPointer) pc et). is_class G x) ^
  (∀x ∈ set (match G (Xcpt ArrayStore) pc et). is_class G x) ^
  (∀x ∈ set (match G (Xcpt ArrayIndexOutOfBounds) pc et). is_class G x)))"
apply auto
apply (drule bspec, assumption)
apply clarsimp
apply (case_tac a, simp)
apply (case_tac list, simp)
apply (case_tac lista, simp)
apply simp
apply (case_tac ac)
apply (case_tac ty)
  apply simp
apply (case_tac ref_ty)
apply auto
done

```

lemma appArrLength[simp]:

```

"app ArrLength G C' pc maxs mpc rT ini et at = (Suc pc < mpc ^
  (∀x ∈ set (match G (Xcpt NullPointer) pc et). the (match_exception_table G x pc et)
< mpc) ^
  (∀s ∈ at. ∃T ST LT z. s=((Init (T.[])#ST,LT),z) ^
  (∀x ∈ set (match G (Xcpt NullPointer) pc et). is_class G x)))"
apply auto
apply (drule bspec, assumption)
apply clarsimp
apply (case_tac a, simp)
apply (case_tac aa)
apply (case_tac ty, simp)
apply (case_tac ref_ty)
apply auto
done

```

lemma appArrNew[simp]:



```

"app (ArrNew T) G C' pc maxs mpc rT ini et at = (Suc pc < mpc ∧
  (∀x ∈ set (match G (Xcpt NegativeArraySize) pc et). the (match_exception_table G x pc
et) < mpc) ∧
  (∀x ∈ set (match G (Xcpt OutOfMemory) pc et). the (match_exception_table G x pc et)
< mpc) ∧
  (∀s ∈ at. ∃ST LT z. s=((Init (PrimT Integer)#ST,LT),z) ∧
  is_type G T ∧ noRA T ∧ dim T + 1 < max_dim ∧
  (∀x ∈ set (match G (Xcpt NegativeArraySize) pc et). is_class G x) ∧
  (∀x ∈ set (match G (Xcpt OutOfMemory) pc et). is_class G x)))"
apply rule
defer
apply fastsimp
apply clarsimp
apply (rule conjI)
  apply clarsimp
  apply (drule bspec, rule UnI2, rule imageI, assumption)
  apply fastsimp
apply (rule conjI)
  apply clarsimp
  apply (drule bspec, rule UnI1, rule imageI, assumption)
  apply fastsimp
apply clarsimp
apply (drule bspec, assumption)
apply clarsimp
apply (case_tac a)
apply auto
done

```

lemma appInvoke[simp]:

```

"app (Invoke C mn fpTs) G C' pc maxs mpc rT ini et at =
  ((Suc pc < mpc ∧
  (∀C ∈ set (match_any G pc et). the (match_exception_table G C pc et) < mpc)) ∧
  (∀s ∈ at. ∃apTs X ST LT mD' rT' b' z.
  s = (((rev apTs) @ (X # ST), LT), z) ∧ mn ≠ init ∧
  length apTs = length fpTs ∧ is_class G C ∧
  (∀(aT,fT)∈set(zip apTs fpTs). G ⊢ aT ≤i (Init fT)) ∧
  method (G,C) (mn,fpTs) = Some (mD', rT', b') ∧ (G ⊢ X ≤i Init (Class C)) ∧
  (∀C ∈ set (match_any G pc et). is_class G C)))"
(is "?app at = (?Q ∧ (∀s ∈ at. ?P s))")

```

proof -

note list\_all2\_def[simp]

{ fix a b z

assume app: "?app at" and at: "(a,b),z) ∈ at"

have "?P ((a,b),z)"

proof -

from app and at

have "a = (rev (rev (take (length fpTs) a))) @ (drop (length fpTs) a) ∧  
length fpTs < length a" (is "?a ∧ ?1")

by (auto simp add: app\_def)

hence "?a ∧ 0 < length (drop (length fpTs) a)" (is "?a ∧ ?1")

by auto

hence "?a ∧ ?1 ∧ length (rev (take (length fpTs) a)) = length fpTs"

```

    by (auto simp add: min_def)
  then obtain apTs ST where
    "a = rev apTs @ ST ∧ length apTs = length fpTs ∧ 0 < length ST"
  by blast
  hence "a = rev apTs @ ST ∧ length apTs = length fpTs ∧ ST ≠ []"
  by blast
  then obtain X ST' where
    "a = rev apTs @ X # ST'" "length apTs = length fpTs"
  by (simp add: neq_Nil_conv) blast
  with app and at show ?thesis by fastsimp
qed } note x = this
have "?app at ⇒ (∀s ∈ at. ?P s)" by clarify (rule x)
hence "?app at ⇒ ?Q ∧ (∀s ∈ at. ?P s)" by auto
moreover
have "?Q ∧ (∀s ∈ at. ?P s) ⇒ ?app at"
  apply clarsimp
  apply (drule bspec, assumption)
  apply (clarsimp simp add: min_def)
  done
ultimately
show ?thesis by (rule iffI)
qed

lemma appInvoke_special[simp]:
"app (Invoke_special C mn fpTs) G C' pc maxs mpc rT ini et at =
((Suc pc < mpc ∧
(∀C ∈ set (match_any G pc et). the (match_exception_table G C pc et) < mpc)) ∧
(∀s ∈ at. ∃apTs X ST LT rT' b' z.
s = ((rev apTs) @ X # ST, LT), z) ∧ mn = init ∧
length apTs = length fpTs ∧ is_class G C ∧
(∀(aT,fT)∈set(zip apTs fpTs). G ⊢ aT ≼i (Init fT)) ∧
method (G,C) (mn,fpTs) = Some (C, rT', b') ∧
((∃pc. X = UnInit C pc) ∨ (X = PartInit C' ∧ G ⊢ C' <C1 C ∧ ¬z)) ∧
(∀C ∈ set (match_any G pc et). is_class G C)))"
(is "?app at = (?Q ∧ (∀s ∈ at. ?P s))")
proof -
  note list_all2_def [simp]
  { fix a b z
    assume app: "?app at" and at: "(a,b),z) ∈ at"
    have "?P ((a,b),z)"
    proof -
      from app and at
      have "a = (rev (rev (take (length fpTs) a))) @ (drop (length fpTs) a) ∧
        length fpTs < length a" (is "?a ∧ ?1")
      by (auto simp add: app_def)
      hence "?a ∧ 0 < length (drop (length fpTs) a)" (is "?a ∧ ?1")
      by auto
      hence "?a ∧ ?1 ∧ length (rev (take (length fpTs) a)) = length fpTs"
      by (auto simp add: min_def)
    then obtain apTs ST where
      "a = rev apTs @ ST ∧ length apTs = length fpTs ∧ 0 < length ST"
    by blast
    hence "a = rev apTs @ ST ∧ length apTs = length fpTs ∧ ST ≠ []"
    by blast
  }

```

```

then obtain X ST' where
  "a = rev apTs @ X # ST'" "length apTs = length fpTs"
  by (simp add: neq_Nil_conv) blast
  with app and at show ?thesis by (fastsimp simp add: nth_append)
qed } note x = this
have "?app at  $\implies \forall s \in at. ?P s$ " by clarify (rule x)
hence "?app at  $\implies ?Q \wedge (\forall s \in at. ?P s)$ " by auto
moreover
have "?Q  $\wedge (\forall s \in at. ?P s) \implies ?app at$ "
  apply clarsimp
  apply (drule bspec, assumption)
  apply (fastsimp simp add: nth_append min_def)
  done
ultimately
show ?thesis by (rule iffI)
qed

```

```

lemma replace_map_OK:
  "replace (OK x) (OK y) (map OK l) = map OK (replace x y l)"
proof -
  have "inj OK" by (blast intro: datatype_injI)
  thus ?thesis by (rule replace_map)
qed

```

```

lemma effNone:
  "(pc', s')  $\in$  set (eff i G pc et {})  $\implies s' = \{\}$ "
  by (auto simp add: eff_def xcpt_eff_def norm_eff_def split: split_if_asm)

```

```

lemmas app_simps =
  appNone appLoad appStore appLitPush appGetField appPutField appNew
  appCheckcast appPop appDup appDup_x1 appDup_x2 appSwap appIAdd appIfcmpeq
  appReturn appGoto appThrow appJsr appRet appInvoke appInvoke_special

```

#### 4.11.1 Code generator setup

```

declare list_all2_Nil [code]
declare list_all2_Cons [code]

```

```

lemma xcpt_app_lemma [code]:
  "xcpt_app i G pc et = list_all (is_class G) (xcpt_names (i, G, pc, et))"
  by (simp add: list_all_conv)

```

```

constdefs
  set_filter :: "('a  $\implies$  bool)  $\implies$  'a set  $\implies$  'a set"
  "set_filter P A  $\equiv$  {s. s  $\in$  A  $\wedge$  P s}"

  tolist :: "'a set  $\implies$  'a list"
  "tolist s  $\equiv$  (SOME l. set l = s)"

```

```

lemma [code]:
  "succs (Ret x) pc s = tolist (theRA x ' s)"
  apply (simp add: tolist_def)

```

```

done

consts
  isRet :: "instr ⇒ bool"
recdef isRet "{}"
  "isRet (Ret r) = True"
  "isRet i = False"

lemma [code]:
  "norm_eff i G pc pc' at = eff_bool i G pc ' (if isRet i then set_filter (λs. pc' = theRA
(theIdx i) s) at else at)"
  apply (cases i)
  apply (auto simp add: norm_eff_def set_filter_def)
done

consts_code
  "set_filter" ("filter")
  "tolist"      ("(fn x => x)")

lemma [code]:
  "app' (Ifcmpeq b, G, C', pc, maxs, rT, Init ts # Init ts' # ST, LT) =
(0 ≤ int pc + b ∧ (if isPrimT ts then ts' = ts else True) ∧ (if isRefT ts then isRefT
ts' else True))"
  apply simp
done

consts
  isUninitC :: "init_ty ⇒ cname ⇒ bool"
primrec
  "isUninitC (Init T) C = False"
  "isUninitC (UnInit C' pc) C = (C=C')"
  "isUninitC (PartInit D) C = False"

lemma [code]:
  "app' (Invoke_special C mn fpTs, G, C', pc, maxs, rT, s) =
(length fpTs < length (fst s) ∧
mn = init ∧
(let apTs = rev (take (length fpTs) (fst s)); X = fst s ! length fpTs
in is_class G C ∧
list_all2 (λaT fT. G ⊢ aT ≤i Init fT) apTs fpTs ∧
method (G, C) (mn, fpTs) ≠ None ∧
(let (C'', rT', b) = the (method (G, C) (mn, fpTs)) in
C = C'' ∧
(isUninitC X C ∨ X = PartInit C' ∧ G ⊢ C' <C1 C))))"
  apply auto
  apply (cases "fst s ! length fpTs")
  apply auto
  apply (cases "fst s ! length fpTs")
  apply auto
done

consts
  isOKInitRA :: "init_ty err ⇒ bool"

```

```

recdef isOKInitRA "{}"
  "isOKInitRA (OK (Init (RA r))) = True"
  "isOKInitRA z = False"

lemma [code]:
  "app' (Ret x, G, C', pc, maxs, rT, ST, LT) = (x < length LT ∧ isOKInitRA (LT!x))"
  apply simp
  apply auto
  apply (cases "LT!x")
  apply simp
  apply simp
  apply (case_tac a)
  apply auto
  apply (case_tac ty)
  apply auto
  apply (case_tac prim_ty)
  apply auto
  done

consts
  isInv_spcl :: "instr ⇒ bool"
recdef
  isInv_spcl "{}"
  "isInv_spcl (Invoke_special C m p) = True"
  "isInv_spcl i = False"

consts
  mNam :: "instr ⇒ mname"
recdef
  mNam "{}"
  "mNam (Invoke_special C m p) = m"

consts
  pLen :: "instr ⇒ nat"
recdef
  pLen "{}"
  "pLen (Invoke_special C m p) = length p"

consts
  isPartInit :: "init_ty ⇒ bool"
recdef
  isPartInit "{}"
  "isPartInit (PartInit D) = True"
  "isPartInit T = False"

lemma [code]:
  "app i G C' pc mxs mpc rT ini et at =
  ((∀ (s, z) ∈ at.
  xcpt_app i G pc et ∧
  app' (i, G, C', pc, mxs, rT, s) ∧
  (if ini ∧ i = Return then z else True) ∧

```

```

      (if isInv_spcl i ∧ fst s ! (pLen i) = PartInit C' then ¬ z else True)) ∧
      (∀(pc', s') ∈ set (eff i G pc et at). pc' < mpc))"
  apply (simp add: split_beta app_def)
  apply (cases i)
  apply auto
  done

```

```

lemma [code]:
  "eff_bool i G pc = (λ((ST, LT), z). (eff' (i, G, pc, ST, LT),
  if isInv_spcl i ∧ mNam i = init ∧ isPartInit(ST ! (pLen i)) then True else z))"
  apply (auto simp add: eff_bool_def split_def)
  apply (cases i)
  apply auto
  apply (rule ext)
  apply auto
  apply (case_tac "a!length list")
  apply auto
  done

```

```

lemmas [simp del] = app_def xcpt_app_def

```

```

end

```

## 4.12 Monotonicity of eff and app

```
theory EffectMono = Effect:
```

```
lemmas eff_defs [simp del]
```

```
lemma in_address_types_finite:
```

```
"x  $\subseteq$  address_types G mxs mxr mpc  $\implies$  finite x"
  apply (drule finite_subset)
  apply (rule finite_address_types)
  apply assumption
  done
```

```
lemma set_SOME_lists:
```

```
"finite s  $\implies$  set (SOME l. set l = s) = s"
  apply (erule finite_induct)
  apply simp
  apply (rule_tac a="x#(SOME l. set l = F)" in someI2)
  apply auto
  done
```

```
lemma succs_mono:
```

```
  assumes s: "s  $\subseteq$  s'"
  assumes finite: "finite s'"
  shows "set (succs i pc s)  $\subseteq$  set (succs i pc s')"
proof (cases i)
  case (Ret pc)
  from finite have "finite (theRA pc ' s')" by (rule finite_imageI)
  moreover
  from s finite have "finite s" by (rule finite_subset)
  hence "finite (theRA pc ' s)" by (rule finite_imageI)
  ultimately
  show ?thesis using s Ret by (auto simp add: set_SOME_lists)
qed auto
```

```
lemma succs_eff_mono:
```

```
"[s  $\subseteq$  s'; finite s';  $\forall$  (pc', s')  $\in$  set (eff i G pc et s'). pc' < mpc]  $\implies$ 
 $\forall$  (pc', s')  $\in$  set (eff i G pc et s). pc' < mpc"
  apply (unfold eff_def xcpt_eff_def)
  apply auto
  apply (drule bspec)
  apply simp
  apply (rule disjI1)
  apply (rule imageI)
  apply (erule subsetD [OF succs_mono])
  apply assumption
  apply assumption
  apply simp
  done
```

```
lemma app_mono:
```

```
"[s ⊆ s'; finite s'; app i G C pc m mpc rT ini et s'] ⇒ app i G C pc m mpc rT ini et s"
```

```
  apply (unfold app_def)
  apply clarsimp
  apply (rule conjI)
  prefer 2
  apply (rule succs_eff_mono, assumption+)
  apply auto
  apply (drule_tac x="((a,b),True)" in bspec)
  apply auto
  done
```

```
end
```



## 4.13 The Bytecode Verifier

**theory** BVSpec = Effect:

This theory contains a specification of the BV. The specification describes correct typings of method bodies; it corresponds to type *checking*.

**constdefs**

— The method type only contains declared classes:

```
check_types :: "[jvm_prog ⇒ nat ⇒ nat ⇒ nat ⇒ state list ⇒ bool]"
"check_types G mxs mxr mpc phi ≡ set phi ⊆ states G mxs mxr mpc"
```

— An instruction is welltyped if it is applicable and its effect

— is compatible with the type at all successor instructions:

```
wt_instr :: "[instr, jvm_prog, cname, ty, method_type, nat, bool,
             p_count, exception_table, p_count] ⇒ bool"
"wt_instr i G C rT phi mxs ini max_pc et pc ≡
 app i G C pc mxs max_pc rT ini et (phi!pc) ∧
 (∀ (pc', s') ∈ set (eff i G pc et (phi!pc)). pc' < max_pc ∧ s' ⊆ phi!pc)"
```

— The type at pc=0 conforms to the method calling convention:

```
wt_start :: "[jvm_prog, cname, mname, ty list, nat, method_type] ⇒ bool"
"wt_start G C mn pTs mxl phi ≡
let
  this = OK (if mn = init ∧ C ≠ Object then PartInit C else Init (Class C));
  start = (([], this#(map OK (map Init pTs))@(replicate mxl Err)), C=Object)
in
  start ∈ phi!0"
```

— A method is welltyped if the body is not empty, if execution does not

— leave the body, if the method type covers all instructions and mentions

— declared classes only, if the method calling convention is respected, and

— if all instructions are welltyped.

```
wt_method :: "[jvm_prog, cname, mname, ty list, ty, nat, nat,
              instr list, exception_table, method_type] ⇒ bool"
"wt_method G C mn pTs rT mxs mxl ins et phi ≡
  let max_pc = length ins in
  0 < max_pc ∧
  length phi = length ins ∧
  check_types G mxs (1+length pTs+mxl) max_pc (map OK phi) ∧
  wt_start G C mn pTs mxl phi ∧
  (∀ pc. pc < max_pc → wt_instr (ins!pc) G C rT phi mxs (mn=init) max_pc et pc)"
```

```
wt_jvm_prog :: "[jvm_prog, prog_type] ⇒ bool"
```

```
"wt_jvm_prog G phi ≡
```

```
  wf_prog (λG C (sig, rT, (maxs, maxl, b, et)).
```

```
    wt_method G C (fst sig) (snd sig) rT maxs maxl b et (phi C sig)) G"
```

**lemma** wt\_jvm\_progD:

```
"wt_jvm_prog G phi ⇒ (∃ wt. wf_prog wt G)"
```

```
by (unfold wt_jvm_prog_def, blast)
```

```

lemma wt_jvm_prog_impl_wt_instr:
  "[[ wt_jvm_prog G phi; is_class G C;
    method (G,C) sig = Some (C,rT,maxs,maxl,ins,et); pc < length ins ]]
  ==> wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig=init) (length ins) et pc"
by (unfold wt_jvm_prog_def, drule method_wf_mdecl,
    simp, simp, simp add: wf_mdecl_def wt_method_def)

```

```

lemma wt_jvm_prog_impl_wt_start:
  "[[ wt_jvm_prog G phi; is_class G C;
    method (G,C) sig = Some (C,rT,maxs,maxl,ins,et) ]] ==>
  0 < (length ins) ^ wt_start G C (fst sig) (snd sig) maxl (phi C sig)"
by (unfold wt_jvm_prog_def, drule method_wf_mdecl,
    simp, simp, simp add: wf_mdecl_def wt_method_def)

```

end

## 4.14 The Typing Framework for the JVM

```
theory Typing_Framework_JVM = Typing_Framework_err + JVMType + EffectMono + BVSpec:
```

```
constdefs
```

```
  exec :: "jvm_prog ⇒ cname ⇒ nat ⇒ ty ⇒ bool ⇒ exception_table ⇒ instr list ⇒
           state step_type"
  "exec G C mxs rT ini et bs ==
  err_step (size bs) (λpc. app (bs!pc) G C pc mxs (size bs) rT ini et) (λpc. eff (bs!pc)
  G pc et)"
```

```
locale (open) JVM_sl =
```

```
  fixes wf_mb and G and C and mxs and mxl
  fixes pTs :: "ty list" and mn and bs and et and rT
```

```
  fixes mxr and A and r and app and eff and step
  defines [simp]: "mxr ≡ 1+length pTs+mxl"
  defines [simp]: "A ≡ states G mxs mxr (length bs)"
  defines [simp]: "r ≡ JVMType.le"
```

```
  defines [simp]: "app ≡ λpc. Effect.app (bs!pc) G C pc mxs (size bs) rT (mn=init) et"
  defines [simp]: "eff ≡ λpc. Effect.eff (bs!pc) G pc et"
  defines [simp]: "step ≡ exec G C mxs rT (mn=init) et bs"
```

```
locale (open) start_context = JVM_sl +
```

```
  assumes wf: "wf_prog wf_mb G"
  assumes C: "is_class G C"
  assumes pTs: "Init ' set pTs ⊆ init_tys G (size bs)"
```

```
  fixes this and first :: state_bool and start
  defines [simp]:
    "this ≡ OK (if mn=init ∧ C ≠ Object then PartInit C else Init (Class C))"
  defines [simp]:
    "first ≡ (([], this#(map (OK∘Init) pTs)@(replicate mxl Err)), C=Object)"
  defines [simp]:
    "start ≡ OK {first}#(replicate (size bs - 1) (OK {}))"
```

### 4.14.1 Connecting JVM and Framework

```
lemma special_ex_swap_lemma [iff]:
```

```
  "(? X. (? n. X = A n & P n) & Q X) = (? n. Q(A n) & P n)"
  by blast
```

```
lemmas [iff del] = not_None_eq
```

```
lemma replace_in_setI:
```

```
  "∧n. ls ∈ list n A ⇒ b ∈ A ⇒ replace a b ls ∈ list n A"
  by (induct ls) (auto simp add: replace_def)
```

```

lemma in_list_Ex [iff]:
  "( $\exists n. xs \in \text{list } n \ A \ \wedge \ P \ xs \ n$ ) = (set xs  $\subseteq$  A  $\wedge$  P xs (length xs))"
  apply (unfold list_def)
  apply auto
  done

```

```

lemma eff_pres_type:
  "[[wf_prog wf_mb S; s  $\subseteq$  address_types S maxs maxr (length bs);
   p < length bs; app (bs!p) S C p maxs (length bs) rT ini et s;
   (a, b)  $\in$  set (eff (bs ! p) S p et s)]]
  $\implies$   $\forall x \in b. x \in \text{address\_types } S \ \text{maxs } \text{maxr } (\text{length } bs)"
  apply clarify
  apply (unfold eff_def xcpt_eff_def norm_eff_def eff_bool_def)
  apply (case_tac "bs!p")$ 
```

— load

```

apply (clarsimp simp add: not_Err_eq address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (rotate_tac -1)
apply clarsimp
apply (drule listE_nth_in, assumption)
apply fastsimp

```

— store

```

apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply fastsimp

```

— litpush

```

apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply clarsimp
apply (rotate_tac -3) — for n = length ab
apply (fastsimp simp add: init_tys_def max_dim_def)

```

— new

```

apply clarsimp
apply (erule disjE)
  apply (clarsimp simp add: address_types_def)
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply clarsimp
  apply (rotate_tac -4)
  apply (fastsimp simp add: replace_in_setI init_tys_def)
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)

```

```

apply clarsimp
apply (rule_tac x=1 in exI)
apply (fastsimp simp add: init_tys_def max_dim_def)

— getfield
apply clarsimp
apply (erule disjE)
  apply (clarsimp simp add: address_types_def)
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply clarsimp
  apply (rule_tac x="Suc n'" in exI)
  apply (drule field_fields)
  apply simp
  apply (frule fields_is_type, assumption, assumption)
  apply (frule fields_no_RA, assumption, assumption)
  apply (frule fields_dim, assumption, assumption)
  apply (simp (no_asm) add: init_tys_def)
  apply simp
  apply (rule boundedRAI2)
  apply assumption
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (simp add: match_some_entry split: split_if_asm)
apply (rule_tac x=1 in exI)
apply fastsimp

— putfield
apply clarsimp
apply (erule disjE)
  apply (clarsimp simp add: address_types_def)
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply (clarsimp simp add: init_tys_def)
  apply fastsimp
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (simp add: match_some_entry split: split_if_asm)
apply (rule_tac x=1 in exI)
apply fastsimp

— checkcast
apply clarsimp
apply (erule disjE)
  apply (clarsimp simp add: address_types_def)
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply (clarsimp simp add: init_tys_def)
  apply fastsimp
apply (clarsimp simp add: address_types_def)

```

```

apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (rule_tac x=1 in exI)
apply fastsimp

— invoke
apply clarsimp
apply (erule disjE)
  apply (clarsimp simp add: address_types_def)
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply (clarsimp simp add: init_tys_def)
  apply (drule method_wf_mdecl, assumption+)
  apply (clarsimp simp add: wf_mdecl_def wf_mhead_def)
  apply (rule in_list_Ex [THEN iffD2])
  apply (simp add: boundedRAI2)
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (rule_tac x=1 in exI)
apply fastsimp

— invoke_special
apply clarsimp
apply (erule disjE)
  apply (clarsimp simp add: address_types_def)
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply (clarsimp simp add: init_tys_def)
  apply (drule method_wf_mdecl, assumption+)
  apply (clarsimp simp add: wf_mdecl_def wf_mhead_def)
  apply (rule conjI)
    apply (rule_tac x="Suc (length ST)" in exI)
    apply (fastsimp intro: replace_in_setI subcls_is_class boundedRAI2)
  apply (fastsimp intro: replace_in_setI subcls_is_class boundedRAI2)
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (rule_tac x=1 in exI)
apply (fastsimp simp add: max_dim_def)

— return
apply fastsimp

— pop
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply fastsimp

```

```

— dup
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (rule_tac x="n'+2" in exI)
apply simp
apply (drule listE_length)+
apply fastsimp

— dup_x1
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (rule_tac x="Suc (Suc (Suc (length ST)))" in exI)
apply simp
apply (drule listE_length)+
apply fastsimp

— dup_x2
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (rule_tac x="Suc (Suc (Suc (Suc (length ST))))" in exI)
apply simp
apply (drule listE_length)+
apply fastsimp

— swap
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply fastsimp

— iadd
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply fastsimp

— goto
apply fastsimp

— icmpeq
apply (clarsimp simp add: address_types_def)
apply fastsimp

— throw
apply (clarsimp simp add: address_types_def init_tys_def)
apply (drule bspec, assumption)

```

```

apply (drule subsetD, assumption)
apply clarsimp
apply (rule_tac x=1 in exI)
apply (fastsimp simp add: max_dim_def)

— jsr
apply (clarsimp simp add: address_types_def)
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: init_tys_def)
apply (rotate_tac -3)
apply (fastsimp simp add: max_dim_def)

— ret
apply fastsimp

— ArrLoad
apply clarsimp
apply (erule disjE)
  apply clarsimp
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply (clarsimp simp add: address_types_def)
  apply (rule_tac x="Suc (length ST)" in exI)
  apply (fastsimp simp add: init_tys_def boundedRAI2)
apply clarsimp
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: address_types_def)
apply (rule_tac x=1 in exI)
apply (fastsimp simp add: init_tys_def max_dim_def)

— ArrStore
apply clarsimp
apply (erule disjE)
  apply clarsimp
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply (clarsimp simp add: address_types_def)
  apply (rule_tac x="length ST" in exI)
  apply fastsimp
apply clarsimp
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: address_types_def)
apply (rule_tac x=1 in exI)
apply (fastsimp simp add: init_tys_def max_dim_def)

— ArrLength
apply clarsimp
apply (erule disjE)
  apply clarsimp
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)

```



```

  apply (clarsimp simp add: address_types_def)
  apply (rule_tac x="Suc (length ST)" in exI)
  apply (fastsimp simp add: init_tys_def max_dim_def)
apply clarsimp
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: address_types_def)
apply (rule_tac x=1 in exI)
apply (fastsimp simp add: init_tys_def max_dim_def)

— ArrNew
apply clarsimp
apply (erule disjE)
  apply clarsimp
  apply (drule bspec, assumption)
  apply (drule subsetD, assumption)
  apply (clarsimp simp add: address_types_def)
  apply (rule_tac x="Suc (length ST)" in exI)
  apply (fastsimp simp add: init_tys_def)
apply clarsimp
apply (drule bspec, assumption)
apply (drule subsetD, assumption)
apply (clarsimp simp add: address_types_def)
apply (rule_tac x=1 in exI)
apply (fastsimp simp add: init_tys_def max_dim_def)
done

lemmas [iff] = not_None_eq

lemma app_mono:
  "app_mono (op  $\subseteq$ ) ( $\lambda pc. app (bs!pc) G C pc maxs (size bs) rT ini et) (length bs) (Pow (address\_types G mxs mxr mpc))"$ 
  apply (unfold app_mono_def lesub_def)
  apply clarify
  apply (drule in_address_types_finite)+
  apply (blast intro: EffectMono.app_mono)
  done

theorem exec_pres_type:
  "wf_prog wf_mb S  $\implies$ 
  pres_type (exec S C maxs rT ini et bs) (size bs) (states S maxs maxr (length bs))"
  apply (unfold exec_def states_def)
  apply (rule pres_type_lift)
  apply clarify
  apply (blast dest: eff_pres_type)
  done

declare split_paired_All [simp del]

declare eff_defs [simp del]

```

```

lemma eff_mono:
  "[wf_prog wf_mb G; t  $\subseteq$  address_types G maxs maxr (length bs); p < length bs; s  $\subseteq$  t;
  app (bs!p) G C p maxs (size bs) rT ini et t]]
   $\implies$  eff (bs ! p) G p et s  $\leq$  |op  $\subseteq$  | eff (bs ! p) G p et t"
  apply (unfold lesubstep_type_def lesub_def)
  apply clarify
  apply (frule in_address_types_finite)
  apply (frule EffectMono.app_mono, assumption, assumption)
  apply (frule subset_trans, assumption)
  apply (drule eff_pres_type, assumption, assumption, assumption, assumption)
  apply (frule finite_subset, assumption)
  apply (unfold eff_def)
  apply clarsimp
  apply (erule disjE)
  defer
    apply (unfold xcpt_eff_def)
    apply clarsimp
    apply blast
  apply clarsimp
  apply (rule exI)
  apply (rule conjI)
    apply (rule disjI1)
    apply (rule imageI)
    apply (rule subsetD [OF succs_mono], assumption+)
  apply (unfold norm_eff_def)
  apply clarsimp
  apply (rule conjI)
    apply clarsimp
    apply (simp add: set_SOME_lists finite_imageI)
    apply (rule imageI)
    apply clarsimp
    apply (rule subsetD, assumption+)
  apply clarsimp
  apply (rule imageI)
  apply (rule subsetD, assumption+)
done

```

```

lemma bounded_exec:
  "bounded (exec G C maxs rT ini et bs) (size bs) (states G maxs maxr (size bs))"
  apply (unfold bounded_def exec_def err_step_def app_def)
  apply (auto simp add: error_def map_snd_def split: err.splits split_if_asm)
done

```

```

theorem exec_mono:
  "wf_prog wf_mb G  $\implies$ 
  mono JVMType.le (exec G C maxs rT ini et bs) (size bs) (states G maxs maxr (size bs))"

  apply (insert bounded_exec [of G C maxs rT ini et bs maxr])
  apply (unfold exec_def JVMType.le_def JVMType.states_def)
  apply (rule mono_lift)
    apply (unfold order_def lesub_def)
    apply blast
  apply (rule app_mono)

```

```

  apply assumption
  apply clarify
  apply (rule eff_mono)
  apply assumption+
  done

```

```

lemma map_id [rule_format]:
  "( $\forall n < \text{length } xs. f (g (xs!n)) = xs!n$ )  $\longrightarrow$  map f (map g xs) = xs"
  by (induct xs, auto)

```

```

lemma is_type_pTs:
  "[ wf_prog wf_mb G; (C,S,fs,mdecls)  $\in$  set G; ((mn,pTs),rT,code)  $\in$  set mdecls ]
 $\implies$  Init'set pTs  $\subseteq$  init_tys G mpc"

```

proof

```

  assume "wf_prog wf_mb G"
    "(C,S,fs,mdecls)  $\in$  set G"
    "((mn,pTs),rT,code)  $\in$  set mdecls"
  hence "wf_mdecl wf_mb G C ((mn,pTs),rT,code)"
    by (unfold wf_prog_def wf_cdecl_def) auto
  hence " $\forall t \in \text{set pTs}. \text{is\_type } G t \wedge \text{noRA } t \wedge \text{dim } t < \text{max\_dim}$ "
    by (unfold wf_mdecl_def wf_mhead_def) auto
  moreover
  fix t assume "t  $\in$  Init'set pTs"
  then obtain t' where t': "t = Init t'" and "t'  $\in$  set pTs" by auto
  ultimately
  have "is_type G t'  $\wedge$  noRA t'  $\wedge$  dim t' < max_dim" by blast
  with t' show "t  $\in$  init_tys G mpc" by (auto simp add: init_tys_def boundedRAI2)
qed

```

```

lemma (in JVM_sl) wt_method_def2:
  "wt_method G C mn pTs rT mxs mxl bs et phi =
  (bs  $\neq$  []  $\wedge$ 
  length phi = length bs  $\wedge$ 
  check_types G mxs mxr (size bs) (map OK phi)  $\wedge$ 
  wt_start G C mn pTs mxl phi  $\wedge$ 
  wt_app_eff (op  $\subseteq$ ) app_eff phi)"
  apply (unfold wt_method_def wt_app_eff_def wt_instr_def lesub_def)
  apply rule
  apply fastsimp
  apply clarsimp
  apply (rule conjI)
  defer
  apply fastsimp
  apply (insert bounded_exec [of G C mxs rT "mn=init" et bs "Suc (length pTs + mxl)"])
  apply (unfold exec_def bounded_def err_step_def)
  apply (erule allE, erule impE, assumption)+
  apply clarsimp
  apply (drule_tac x = "OK (phi!pc)" in bspec)
  apply (fastsimp simp add: check_types_def)
  apply (fastsimp simp add: map_snd_def)
  done

```

```

lemma jvm_prog_lift:
  assumes wf:
    "wf_prog ( $\lambda G C bd. P G C bd$ ) G"

  assumes rule:
    " $\wedge wf\_mb C mn pTs C rT maxs maxl b et bd.$ 
     wf_prog wf_mb G  $\implies$ 
     method (G,C) (mn,pTs) = Some (C,rT,maxs,maxl,b,et)  $\implies$ 
     is_class G C  $\implies$ 
     Init'set pTs  $\subseteq$  init_tys G (length b)  $\implies$ 
     bd = ((mn,pTs),rT,maxs,maxl,b,et)  $\implies$ 
     P G C bd  $\implies$ 
     Q G C bd"

  shows
    "wf_prog ( $\lambda G C bd. Q G C bd$ ) G"
proof -
  from wf show ?thesis
  apply (unfold wf_prog_def wf_cdecl_def)
  apply clarsimp
  apply (drule bspec, assumption)
  apply (unfold wf_mdecl_def)
  apply clarsimp
  apply (drule bspec, assumption)
  apply clarsimp
  apply (frule methd [OF wf], assumption+)
  apply (frule is_type_pTs [OF wf], assumption+)
  apply clarify
  apply (drule rule [OF wf], assumption+)
  apply (rule refl)
  apply assumption+
  done
qed

end

```

## 4.15 Kildall's Algorithm

```
theory Kildall = SemilatAlg + While_Combinator:
```

```
consts
```

```
iter :: "'s binop ⇒ 's step_type ⇒
        's list ⇒ nat set ⇒ 's list × nat set"
propa :: "'s binop ⇒ (nat × 's) list ⇒ 's list ⇒ nat set ⇒ 's list * nat set"
```

```
primrec
```

```
"propa f []      ss w = (ss,w)"
"propa f (q'#qs) ss w = (let (q,t) = q';
                           u = t +_f ss!q;
                           w' = (if u = ss!q then w else insert q w)
                           in propa f qs (ss[q := u] w'))"
```

```
defs iter_def:
```

```
"iter f step ss w ==
  while (λ(ss,w). w ≠ {})
    (λ(ss,w). let p = SOME p. p ∈ w
              in propa f (step p (ss!p)) ss (w-{p}))
  (ss,w)"
```

```
constdefs
```

```
unstables :: "'s ord ⇒ 's step_type ⇒ 's list ⇒ nat set"
"unstables r step ss == {p. p < size ss ∧ ¬stable r step ss p}"
```

```
kildall :: "'s ord ⇒ 's binop ⇒ 's step_type ⇒ 's list ⇒ 's list"
"kildall r f step ss == fst(iter f step ss (unstables r step ss))"
```

```
consts merges :: "'s binop ⇒ (nat × 's) list ⇒ 's list ⇒ 's list"
```

```
primrec
```

```
"merges f []      ss = ss"
"merges f (p'#ps) ss = (let (p,s) = p' in merges f ps (ss[p := s +_f ss!p]))"
```

```
lemmas [simp] = Let_def semilat.le_iff_plus_unchanged [symmetric]
```

```
lemma (in semilat) nth_merges:
```

```
"∧ss. [p < length ss; ss ∈ list n A; ∀(p,t)∈set ps. p<n ∧ t∈A] ⇒
  (merges f ps ss)!p = map snd [(p',t') ∈ ps. p'=p] ++_f ss!p"
(is "∧ss. [_; _; ?steptype ps] ⇒ ?P ss ps")
```

```
proof (induct ps)
```

```
  show "∧ss. ?P ss []" by simp
```

```
  fix ss p' ps'
```

```
  assume ss: "ss ∈ list n A"
```

```
  assume l: "p < length ss"
```

```
  assume "?steptype (p'#ps)'"
```

```
  then obtain a b where
```

```
    p': "p'=(a,b)" and ab: "a<n" "b∈A" and "?steptype ps'"
```

```
    by (cases p', auto)
```

assume " $\bigwedge ss. p < \text{length } ss \implies ss \in \text{list } n \ A \implies ?\text{steptype } ps' \implies ?P \ ss \ ps'$ "  
 hence IH: " $\bigwedge ss. ss \in \text{list } n \ A \implies p < \text{length } ss \implies ?P \ ss \ ps'$ " .

```

from ss ab
have "ss[a := b +_f ss!a] ∈ list n A" by (simp add: closedD)
moreover
from calculation
have "p < length (ss[a := b +_f ss!a])" by simp
ultimately
have "?P (ss[a := b +_f ss!a]) ps'" by (rule IH)
with p' 1
show "?P ss (p'#ps'" by simp
qed

```

```

lemma length_merges [rule_format, simp]:
  "∀ss. size(merges f ps ss) = size ss"
  by (induct_tac ps, auto)

```

```

lemma (in semilat) merges_preserves_type_lemma:
  shows "∀xs. xs ∈ list n A ⟶ (∀(p,x) ∈ set ps. p < n ∧ x ∈ A)
        ⟶ merges f ps xs ∈ list n A"
  apply (insert closedI)
  apply (unfold closed_def)
  apply (induct_tac ps)
  apply simp
  apply clarsimp
  done

```

```

lemma (in semilat) merges_preserves_type [simp]:
  "[ xs ∈ list n A; ∀(p,x) ∈ set ps. p < n ∧ x ∈ A ]
  ⟹ merges f ps xs ∈ list n A"
  by (simp add: merges_preserves_type_lemma)

```

```

lemma (in semilat) merges_pres_type [simp]:
  "[ xs ∈ list n A; s ∈ A; p < n; pres_type step n A; bounded step n A ]
  ⟹ merges f (step p s) xs ∈ list n A"
  by (blast dest: pres_typeD boundedD merges_preserves_type)

```

```

lemma (in semilat) merges_incr_lemma:
  "∀xs. xs ∈ list n A ⟶ (∀(p,x) ∈ set ps. p < size xs ∧ x ∈ A) ⟶ xs <=[r] merges f ps
  xs"
  apply (induct_tac ps)
  apply simp
  apply simp
  apply clarify
  apply (rule order_trans)
  apply simp
  apply (erule list_update_incr)
  apply simp
  apply simp

```

```
apply (blast intro!: listE_set intro: closedD listE_length [THEN nth_in])
done
```

```
lemma (in semilat) merges_incr:
  "[ xs ∈ list n A; ∀ (p,x) ∈ set ps. p < size xs ∧ x ∈ A ]
  ⇒ xs <=[r] merges f ps xs"
  by (simp add: merges_incr_lemma)
```

```
lemma (in semilat) merges_same_conv [rule_format]:
  "(∀ xs. xs ∈ list n A → (∀ (p,x) ∈ set ps. p < size xs ∧ x ∈ A) →
   (merges f ps xs = xs) = (∀ (p,x) ∈ set ps. x <=_r xs!p))"
  apply (induct_tac ps)
  apply simp
  apply clarsimp
  apply (rename_tac p x ps xs)
  apply (rule iffI)
  apply (rule context_conjI)
  apply (subgoal_tac "xs[p := x +_f xs!p] <=[r] xs")
  apply (force dest!: le_listD simp add: nth_list_update)
  apply (erule subst, rule merges_incr)
  apply (blast intro!: listE_set intro: closedD listE_length [THEN nth_in])
  apply clarify
  apply (rule conjI)
  apply simp
  apply blast
  apply blast
  apply clarify
  apply (rotate_tac -2)
  apply (erule allE)
  apply (erule impE)
  apply assumption
  apply (erule impE)
  apply assumption
  apply (drule bspec)
  apply assumption
  apply (simp add: le_iff_plus_unchanged [THEN iffD1] list_update_same_conv [THEN iffD2])
  apply blast
  apply clarify
  apply (simp add: le_iff_plus_unchanged [THEN iffD1] list_update_same_conv [THEN iffD2])
done
```

```
lemma (in semilat) list_update_le_listI [rule_format]:
  "set xs <= A → set ys <= A → xs <=[r] ys → p < size xs →
  x <=_r ys!p → x ∈ A → xs[p := x +_f xs!p] <=[r] ys"
  apply (insert semilat)
  apply (unfold Listn.le_def lesub_def semilat_def supremum_def)
  apply (simp add: list_all2_conv_all_nth nth_list_update)
done
```

```
lemma (in semilat) merges_pres_le_ub:
  shows "[ set ts <= A; set ss <= A;
  ∀ (p,t) ∈ set ps. t <=_r ts!p ∧ t ∈ A ∧ p < size ts; ss <=[r] ts ]
```

```

    => merges f ps ss <=[r] ts"
proof -
  { fix t ts ps
    have
      " $\bigwedge qs. \llbracket \text{set } ts \leq A; \forall (p,t) \in \text{set } ps. t \leq_r ts!p \wedge t \in A \wedge p < \text{size } ts \rrbracket \implies$ "
      set qs <= set ps  $\longrightarrow$ 
      ( $\forall ss. \text{set } ss \leq A \longrightarrow ss \leq [r] ts \longrightarrow \text{merges } f qs ss \leq [r] ts$ )"
    apply (induct_tac qs)
    apply simp
    apply (simp (no_asm_simp))
    apply clarify
    apply (rotate_tac -2)
    apply simp
    apply (erule allE, erule impE, erule_tac [2] mp)
    apply (drule bspec, assumption)
    apply (simp add: closedD)
    apply (drule bspec, assumption)
    apply (simp add: list_update_le_listI)
    done
  } note this [dest]

  case rule_context
  thus ?thesis by blast
qed

```

```

lemma decomp_propa:
  " $\bigwedge ss w. (\forall (q,t) \in \text{set } qs. q < \text{size } ss) \implies$ "
  propa f qs ss w =
  (merges f qs ss, {q.  $\exists t. (q,t) \in \text{set } qs \wedge t +_f ss!q \neq ss!q$ } Un w)"
  apply (induct qs)
  apply simp
  apply (simp (no_asm))
  apply clarify
  apply simp
  apply (rule conjI)
  apply (simp add: nth_list_update)
  apply blast
  apply (simp add: nth_list_update)
  apply blast
  done

```

```

lemma (in semilat) stable_pres_lemma:
  shows " $\llbracket \text{pres\_type } \text{step } n A; \text{bounded } \text{step } n A;$ "
  ss  $\in \text{list } n A; p \in w; \forall q \in w. q < n;$ 
   $\forall q. q < n \longrightarrow q \notin w \longrightarrow \text{stable } r \text{ step } ss q; q < n;$ 
   $\forall s'. (q,s') \in \text{set } (\text{step } p (ss ! p)) \longrightarrow s' +_f ss ! q = ss ! q;$ 
   $q \notin w \vee q = p \rrbracket$ 
  => stable r step (merges f (step p (ss!p)) ss) q"

```



```

apply (unfold stable_def)
apply (subgoal_tac " $\forall s'. (q, s') \in \text{set } (\text{step } p \text{ } (ss!p)) \longrightarrow s' : A$ ")
prefer 2
apply clarify
apply (erule pres_typeD)
  prefer 3 apply assumption
  apply (rule listE_nth_in)
  apply assumption
  apply simp
  apply simp
apply simp
apply clarify
apply (subst nth_merges)
  apply simp
  apply (frule boundedD, assumption)
  prefer 2 apply assumption
  apply (rule listE_nth_in)
  prefer 2 apply assumption
  apply (rule merges_pres_type, assumption)
  apply (rule listE_nth_in, assumption)
  apply blast
  apply blast
  apply assumption
  apply assumption
  apply assumption
  apply assumption
  apply clarify
  apply (rule conjI)
  apply (erule boundedD)
  prefer 3 apply assumption
  apply blast
  apply (erule listE_nth_in)
  apply blast
  apply (erule pres_typeD)
  prefer 3 apply assumption
  apply simp
  apply simp
apply (subgoal_tac " $q < \text{length } ss$ ")
prefer 2 apply simp
  apply (frule nth_merges [of q _ _ "step p (ss!p)"])
apply assumption
  apply clarify
  apply (rule conjI)
  apply (erule boundedD)
  prefer 3 apply assumption
  apply blast
  apply (erule listE_nth_in)
  apply blast
  apply (erule pres_typeD)
  prefer 3 apply assumption
  apply simp
  apply simp
apply (drule_tac P = " $\lambda x. (a, b) \in \text{set } (\text{step } q \text{ } x)$ " in subst)
  apply assumption

```

```

apply (simp add: plusplus_empty)
apply (cases "q ∈ w")
  apply simp
  apply (rule ub1')
    apply assumption
    apply clarify
    apply (rule pres_typeD)
      apply assumption
      prefer 3 apply assumption
      apply (blast intro: listE_nth_in)
      apply (blast intro: pres_typeD)
      apply (blast intro: listE_nth_in dest: boundedD)
    apply assumption
  apply simp
  apply (erule allE, erule impE, assumption, erule impE, assumption)
  apply (rule order_trans)
    apply simp
  defer
  apply (rule pp_ub2)
    apply simp
    apply clarify
    apply simp
    apply (rule pres_typeD)
      apply assumption
      prefer 3 apply assumption
      apply (blast intro: listE_nth_in)
      apply (blast intro: pres_typeD)
      apply (blast intro: listE_nth_in dest: boundedD)
  apply blast
done

```

lemma (in semilat) merges\_bounded\_lemma:

```

"[[ mono r step n A; bounded step n A;
  ∀ (p',s') ∈ set (step p (ss!p)). s' ∈ A; ss ∈ list n A; ts ∈ list n A; p < n;
  ss <=[r] ts; ∀ p. p < n → stable r step ts p ]]
⇒ merges f (step p (ss!p)) ss <=[r] ts"

```

```

apply (unfold stable_def)
apply (rule merges_pres_le_ub)
  apply simp
  apply simp
  prefer 2 apply assumption

  apply clarsimp
  apply (drule boundedD, assumption)
  prefer 2 apply assumption
  apply (erule listE_nth_in)
  apply blast
  apply (erule allE, erule impE, assumption)
  apply (drule bspec, assumption)
  apply simp

  apply (drule monoD [of _ _ _ p "ss!p" "ts!p"])
  apply assumption

```

```

    apply simp
    apply simp
    apply (simp add: le_listD)

  apply (drule lesub_step_typeD, assumption)
  apply clarify
  apply (drule bspec, assumption)
  apply simp
  apply (blast intro: order_trans)
done

lemma termination_lemma: includes semilat
shows "[ ss ∈ list n A; ∀(q,t)∈set qs. q<n ∧ t∈A; p∈w ] ⇒
  ss <[r] merges f qs ss ∨
  merges f qs ss = ss ∧ {q. ∃t. (q,t)∈set qs ∧ t+_f ss!q ≠ ss!q} Un (w-{p}) < w"
apply(insert semilat)
  apply (unfold lesssub_def)
  apply (simp (no_asm_simp) add: merges_incr)
  apply (rule impI)
  apply (rule merges_same_conv [THEN iffD1, elim_format])
  apply assumption+
  defer
  apply (rule sym, assumption)
  defer apply simp
  apply (subgoal_tac "∀q t. ¬((q, t) ∈ set qs ∧ t+_f ss ! q ≠ ss ! q)")
  apply (blast intro!: psubsetI elim: equalityE)
  apply clarsimp
  apply (drule bspec, assumption)
  apply (drule bspec, assumption)
  apply clarsimp
done

lemma iter_properties[rule_format]: includes semilat
shows "[ acc r A; pres_type step n A; mono r step n A;
  bounded step n A; ∀p∈w0. p < n; ss0 ∈ list n A;
  ∀p<n. p ∉ w0 → stable r step ss0 p ] ⇒
  iter f step ss0 w0 = (ss',w')
  →
  ss' ∈ list n A ∧ stables r step ss' ∧ ss0 <=[r] ss' ∧
  (∀ts∈list n A. ss0 <=[r] ts ∧ stables r step ts → ss' <=[r] ts)"
apply(insert semilat)
  apply (unfold iter_def stables_def)
  apply (rule_tac P = "λ(ss,w).
  ss ∈ list n A ∧ (∀p<n. p ∉ w → stable r step ss p) ∧ ss0 <=[r] ss ∧
  (∀ts∈list n A. ss0 <=[r] ts ∧ stables r step ts → ss <=[r] ts) ∧
  (∀p∈w. p < n)" and
  r = "{(ss',ss) . ss ∈ list n A ∧ ss' ∈ list n A ∧ ss <[r] ss'} <[*lex*] finite_psubset"
  in while_rule)
— Invariant holds initially:
  apply (simp add:stables_def)

— Invariant is preserved:
  apply(simp add: stables_def split_paired_all)

```

```

apply(rename_tac ss w)
apply(subgoal_tac "(SOME p. p ∈ w) ∈ w")
  prefer 2 apply (fast intro: someI)
apply(subgoal_tac "∀(q,t) ∈ set (step (SOME p. p ∈ w) (ss ! (SOME p. p ∈ w))). q < length
ss ∧ t ∈ A")
  prefer 2
  apply clarify
  apply (rule conjI)
  apply clarsimp
  apply (rule boundedD, assumption)
    prefer 3 apply assumption
    apply blast
  apply (erule listE_nth_in)
  apply blast
  apply (erule pres_typeD)
  prefer 3
  apply assumption
  apply (erule listE_nth_in)
  apply blast
  apply blast
  apply (subst decomp_propa)
  apply blast
  apply simp
  apply (rule conjI)
  apply (rule merges_preserves_type)
  apply blast
  apply clarify
  apply (rule conjI)
  apply clarsimp
  apply (rule boundedD, assumption)
    prefer 3 apply assumption
    apply blast
  apply (erule listE_nth_in)
  apply blast
  apply (erule pres_typeD)
  prefer 3
  apply assumption
  apply (erule listE_nth_in)
  apply blast
  apply blast
  apply (rule conjI)
  apply clarify
  apply (blast intro!: stable_pres_lemma)
  apply (rule conjI)
  apply (blast intro!: merges_incr intro: le_list_trans)
  apply (rule conjI)
  apply clarsimp
  apply (blast intro!: merges_bounded_lemma)
  apply clarsimp
  apply (erule disjE)
  apply clarify
  apply (erule boundedD)
    prefer 3 apply assumption
  apply blast

```

```

apply (erule listE_nth_in)
apply blast
apply blast

```

— Postcondition holds upon termination:

```

apply(clarsimp simp add: stables_def split_paired_all)

```

— Well-foundedness of the termination relation:

```

apply (rule wf_lex_prod)
  apply (drule orderI [THEN acc_le_listI])
  apply (simp only: acc_def lesssub_def)
apply (rule wf_finite_psubset)

```

— Loop decreases along termination relation:

```

apply(simp add: stables_def split_paired_all)
apply(rename_tac ss w)
apply(subgoal_tac "(SOME p. p ∈ w) ∈ w")
  prefer 2 apply (fast intro: someI)
apply(subgoal_tac "∀(q,t) ∈ set (step (SOME p. p ∈ w) (ss ! (SOME p. p ∈ w))). q < length
ss ∧ t ∈ A")
  prefer 2
  apply clarify
  apply (rule conjI)
  apply clarsimp
  apply (rule boundedD, assumption)
  prefer 3 apply assumption
  apply blast
  apply (erule listE_nth_in)
  apply blast
  apply (erule pres_typeD)
  prefer 3
  apply assumption
  apply (erule listE_nth_in)
  apply blast
  apply blast
  apply (subst decomp_propa)
  apply blast
  apply clarify
  apply (simp del: listE_length
    add: lex_prod_def finite_psubset_def
    bounded_nat_set_is_finite)
  apply (frule merges_preserves_type) back
  apply (drule listE_length) back
  apply (rotate_tac -1)
  apply simp
  apply simp
  apply (frule termination_lemma)
  apply assumption back
  defer
  apply assumption+
  defer
  apply clarsimp
  apply (rule conjI)
  apply (erule boundedD)

```

```

    prefer 3 apply assumption
  apply blast
  apply (erule listE_nth_in)
  apply blast
  apply blast
done

```

```

lemma kildall_properties: includes semilat
shows "[[ acc r A; pres_type step n A; mono r step n A;
  bounded step n A; ss0 ∈ list n A ]] ⇒
  kildall r f step ss0 ∈ list n A ∧
  stables r step (kildall r f step ss0) ∧
  ss0 <=[r] kildall r f step ss0 ∧
  (∀ ts ∈ list n A. ss0 <=[r] ts ∧ stables r step ts →
    kildall r f step ss0 <=[r] ts)"
apply (unfold kildall_def)
apply (case_tac "iter f step ss0 (unstables r step ss0)")
apply (simp)
apply (rule iter_properties)
by (simp_all add: unstables_def stable_def)

```

```

lemma is_bcv_kildall: includes semilat
shows "[[ acc r A; top r T; pres_type step n A; bounded step n A; mono r step n A ]]
  ⇒ is_bcv r T step n A (kildall r f step)"
apply (unfold is_bcv_def wt_step_def)
apply (insert semilat kildall_properties[of A])
apply (simp add: stables_def)
apply clarify
apply (subgoal_tac "kildall r f step ss ∈ list n A")
  prefer 2 apply (simp(no_asm_simp))
apply (rule iffI)
  apply (rule_tac x = "kildall r f step ss" in bexI)
  apply (rule conjI)
  apply (blast)
  apply (simp (no_asm_simp))
  apply (assumption)
apply clarify
apply (subgoal_tac "kildall r f step ss!p <=_r ts!p")
  apply simp
apply (blast intro!: le_listD less_lengthI)
done

```

#### 4.15.1 Code generator setup

Kildall's algorithm is executable. The following sections gives alternative, directly executable implementations for those parts of the specification that Isabelle's ML code generator does not understand without help.

```

lemma unstables_exec [code]:
  "unstables r step ss = (UN p: {..size ss}. if ¬stable r step ss p then {p} else {})"
  apply (unfold unstables_def)
  apply (rule equalityI)

```

```

apply (rule subsetI)
apply (erule CollectE)
apply (erule conjE)
apply (rule UN_I)
apply simp
apply simp
apply (rule subsetI)
apply (erule UN_E)
apply (case_tac " $\neg$  stable r step ss p")
apply simp+
done

```

```
lemmas [code] = lessThan_0 lessThan_Suc
```

```
constdefs
```

```

some_elem :: "'a set  $\Rightarrow$  'a"
"some_elem == (%S. SOME x. x : S)"

```

```
lemma iter_exec [code]:
```

```

"iter f step ss w =
while (%(ss,w). w  $\neq$  {})
  (%(ss,w). let p = some_elem w
            in propa f (step p (ss!p)) ss (w-{p}))
(ss,w)"
by (unfold iter_def some_elem_def, rule refl)

```

The work list sets of the algorithm are implemented in ML as lists:

```
types_code
```

```
set ("_ list")
```

```
consts_code
```

```

"wf"      ("true?")
"{}"      ("[]")
"insert"  ("(_ ins _)")
"op :"    ("(_ mem _)")
"op Un"   ("(_ union _)")
"image"   ("map")
"UNION"   ("(fn A => fn f => flat (map f A))")
"Bex"     ("(fn A => fn f => exists f A)")
"Ball"    ("(fn A => fn f => forall f A)")
"some_elem" ("hd")
"op -" :: "'a set  $\Rightarrow$  'a set  $\Rightarrow$  'a set" ("(_ \ _)")

```

```
lemmas [code ind] = rtrancl_refl converse_rtrancl_into_rtrancl
```

```
end
```

## 4.16 Kildall for the JVM

theory JVM = Typing\_Framework\_JVM + Kildall:

constdefs

```
kiljvm :: "jvm_prog ⇒ cname ⇒ nat ⇒ nat ⇒ ty ⇒ bool ⇒ exception_table ⇒
instr list ⇒ state list ⇒ state list"
"kiljvm G C maxs maxr rT ini et bs ≡
kildall (JVMTy.le) (JVMTy.sup) (exec G C maxs rT ini et bs)"
```

```
c_kil :: "jvm_prog ⇒ cname ⇒ ty list ⇒ ty ⇒ bool ⇒ nat ⇒ nat
⇒ exception_table ⇒ instr list ⇒ state list"
"c_kil G C pTs rT ini mxs mxl et bs ≡
let this = OK (if ini ∧ C ≠ Object then PartInit C else Init (Class C));
first = ([], this#(map (OK∘Init) pTs)@(replicate mxl Err)), C=Object);
start = OK {first}#(replicate (size bs - 1) (OK {}))
in kiljvm G C mxs (1+size pTs+mxl) rT ini et bs start"
```

```
wt_kil :: "jvm_prog ⇒ cname ⇒ ty list ⇒ ty ⇒ bool ⇒ nat ⇒ nat
⇒ exception_table ⇒ instr list ⇒ bool"
"wt_kil G C pTs rT ini mxs mxl et bs ≡
0 < size bs ∧ (∀ n < size bs. c_kil G C pTs rT ini mxs mxl et bs ! n ≠ Err)"
```

```
wt_jvm_prog_kildall :: "jvm_prog ⇒ bool"
"wt_jvm_prog_kildall G ≡
wf_prog (λG C (sig,rT,(maxs,maxl,b,et)).
wt_kil G C (snd sig) rT (fst sig=init) maxs maxl et b) G"
```

theorem (in start\_context) is\_bcv\_kiljvm:

```
"is_bcv r Err step (size bs) A (kiljvm G C mxs mxr rT (mn=init) et bs)"
apply simp
apply (insert wf)
apply (unfold kiljvm_def)
apply (rule is_bcv_kildall)
apply (fold sl_def2, rule semilat_JVM)
apply (rule acc_JVM)
apply (simp add: JVMTy.le_def)
apply (erule exec_pres_type)
apply (rule bounded_exec)
apply (erule exec_mono)
done
```

lemma subset\_replicate: "set (replicate n x) ⊆ {x}"  
by (induct n) auto

lemma in\_set\_replicate:

"x ∈ set (replicate n y) ⇒ x = y"

proof -



```

assume "x ∈ set (replicate n y)"
also have "set (replicate n y) ⊆ {y}" by (rule subset_replicate)
finally have "x ∈ {y}" .
thus ?thesis by simp
qed

```

```

lemma (in start_context) start_in_A [intro?]:
  "0 < size bs ⇒ start ∈ list (size bs) A"
  apply (insert pTs C)
  apply (simp split del: split_if)
  apply (unfold states_def address_types_def init_tys_def max_dim_def)
  apply (auto simp add: map_compose split
    intro!: listI list_appendI dest!: in_set_replicate)
  apply force+
  done

```

```

theorem (in start_context) wt_kil_correct:
  assumes wtk: "wt_kil G C pTs rT (mn=init) mxs mxl et bs"
  shows "∃phi. wt_method G C mn pTs rT mxs mxl bs et phi"

```

proof -

from wtk obtain res where

```

result: "res = kiljvm G C mxs mxr rT (mn=init) et bs start" and
success: "∀n < size bs. res!n ≠ Err" and
instrs: "0 < size bs"
by (unfold wt_kil_def c_kil_def) (simp add: map_compose)

```

have bcv:

```

"is_bcv r Err step (size bs) A (kiljvm G C mxs mxr rT (mn=init) et bs)"
by (rule is_bcv_kiljvm)

```

from instrs have "start ∈ list (size bs) A" ..

with bcv success result have

```

"∃ts∈list (size bs) A. start <=[r] ts ∧ wt_step r Err step ts"
by (unfold is_bcv_def) blast

```

then obtain phi' where

```

in_A: "phi' ∈ list (size bs) A" and
s: "start <=[r] phi'" and
w: "wt_step r Err step phi'"
by blast

```

hence wt\_err\_step: "wt\_err\_step (op ⊆) step phi'"

```

by (simp add: wt_err_step_def JVMType.le_def)

```

from in\_A have l: "size phi' = size bs" by simp

moreover {

```

from in_A have "check_types G mxs mxr (size bs) phi'" by (simp add: check_types_def)
also from w
have [symmetric]: "map OK (map ok_val phi') = phi'"
by (auto intro!: map_id simp add: wt_step_def not_Err_eq)
finally have "check_types G mxs mxr (size bs) (map OK (map ok_val phi'))" .

```

}

moreover {

```

from s have "start!0 <=_r phi'!0" by (rule le_listD) simp
moreover

```

```

from instrs w l
have "phi'!0 ≠ Err" by (unfold wt_step_def) simp
then obtain phi0 where "phi'!0 = OK phi0" by (auto simp add: not_Err_eq)
ultimately
have "wt_start G C mn pTs mxl (map ok_val phi')" using l instrs
  by (unfold wt_start_def)
  (simp add: map_compose lesub_def JVMType.le_def Err.le_def)
}
moreover
have bounded: "bounded step (length bs) A" by simp (rule bounded_exec)
from in_A have "set phi' ⊆ A" by simp
with wt_err_step bounded
have "wt_app_eff (op ⊆) app eff (map ok_val phi')"
  by (auto intro: wt_err_imp_wt_app_eff simp add: l exec_def states_def)
ultimately
have "wt_method G C mn pTs rT mxs mxl bs et (map ok_val phi')"
  using instrs by (simp add: wt_method_def2)
thus ?thesis by blast
qed

```

theorem (in start\_context) wt\_kil\_complete:

assumes wtm: "wt\_method G C mn pTs rT mxs mxl bs et phi"

shows "wt\_kil G C pTs rT (mn=init) mxs mxl et bs"

proof -

from wtm obtain

```

instrs: "0 < size bs" and
length: "length phi = length bs" and
ck_type: "check_types G mxs mxr (size bs) (map OK phi)" and
wt_start: "wt_start G C mn pTs mxl phi" and
app_eff: "wt_app_eff (op ⊆) app eff phi"
by (simp add: wt_method_def2)

```

from ck\_type

have in\_A: "set (map OK phi) ⊆ A"

by (simp add: check\_types\_def)

have bounded: "bounded step (size bs) A"

by simp (rule bounded\_exec)

with app\_eff in\_A

have "wt\_err\_step (op ⊆) (err\_step (size phi) app eff) (map OK phi)"

by - (erule wt\_app\_eff\_imp\_wt\_err,  
auto simp add: exec\_def length states\_def)

hence wt\_err: "wt\_err\_step (op ⊆) step (map OK phi)"

by (simp add: exec\_def length)

have is\_bcv:

"is\_bcv r Err step (size bs) A (kiljvm G C mxs mxr rT (mn=init) et bs)"

by (rule is\_bcv\_kiljvm)

moreover

from instrs have "start ∈ list (size bs) A" ..

moreover

let ?phi = "map OK phi"

have less\_phi: "start ≤<sub>[r]</sub> ?phi"

proof (rule le\_listI)

from length instrs

```

  show "length start = length (map OK phi)" by simp
next
  fix n
  from wt_start have "ok_val (start!0)  $\subseteq$  phi!0"
    by (simp add: wt_start_def map_compose)
  moreover from instrs length have "0 < length phi" by simp
  ultimately have "start!0  $\leq_r$  ?phi!0"
    by (simp add: JVMType.le_def Err.le_def lesub_def)
  moreover {
    fix n'
    have "OK {}  $\leq_r$  ?phi!n"
      by (auto simp add: JVMType.le_def Err.le_def lesub_def
        split: err.splits)
    hence "[ n = Suc n'; n < size start ]
       $\implies$  start!n  $\leq_r$  ?phi!n" by simp
  }
  ultimately
  show "n < size start  $\implies$  start!n  $\leq_r$  ?phi!n" by (cases n, blast+)
qed
moreover
from ck_type length
have "?phi  $\in$  list (size bs) A"
  by (auto intro!: listI simp add: check_types_def)
moreover
from wt_err have "wt_step r Err step ?phi"
  by (simp add: wt_err_step_def JVMType.le_def)
ultimately
have " $\forall p. p < size bs \implies$  kiljvm G C mxs mxr rT (mn=init) et bs start ! p  $\neq$  Err"
  by (unfold is_bcv_def) blast
with instrs
show "wt_kil G C pTs rT (mn=init) mxs mxl et bs"
  by (unfold c_kil_def wt_kil_def) (simp add: map_compose)
qed

```

**theorem** *jvm\_kildall\_correct*:

"wt\_jvm\_prog\_kildall G = ( $\exists$ Phi. wt\_jvm\_prog G Phi)"

**proof**

let ?Phi = " $\lambda C$  sig. let (C,rT,(maxs,maxl,ins,et)) = the (method (G,C) sig) in  
 SOME phi. wt\_method G C (fst sig) (snd sig) rT maxs maxl ins et phi"

— soundness

assume "wt\_jvm\_prog\_kildall G"

hence "wt\_jvm\_prog G ?Phi"

  apply (unfold wt\_jvm\_prog\_def wt\_jvm\_prog\_kildall\_def)

  apply (erule jvm\_prog\_lift)

  apply (auto dest!: start\_context.wt\_kil\_correct intro: someI)

  done

thus " $\exists$ Phi. wt\_jvm\_prog G Phi" by fast

**next**

— completeness

assume " $\exists$ Phi. wt\_jvm\_prog G Phi"

thus "wt\_jvm\_prog\_kildall G"

  apply (clarify)

180

```
    apply (unfold wt_jvm_prog_def wt_jvm_prog_kildall_def)
    apply (erule jvm_prog_lift)
    apply (auto intro: start_context.wt_kil_complete)
  done
qed

end
```

## 4.17 The Lightweight Bytecode Verifier

```
theory LBVSpec = SemilatAlg + Opt:
```

```
types
```

```
's certificate = "'s list"
```

```
consts
```

```
merge :: "'s certificate ⇒ 's binop ⇒ 's ord ⇒ 's ⇒ nat ⇒ (nat × 's) list ⇒ 's  
⇒ 's"
```

```
primrec
```

```
"merge cert f r T pc [] x = x"  
"merge cert f r T pc (s#ss) x = merge cert f r T pc ss (let (pc',s') = s in  
if pc'=pc+1 then s' +_f x  
else if s' <=_r (cert!pc') then x  
else T)"
```

```
constdefs
```

```
wtl_inst :: "'s certificate ⇒ 's binop ⇒ 's ord ⇒ 's ⇒  
's step_type ⇒ nat ⇒ 's ⇒ 's"  
"wtl_inst cert f r T step pc s ≡ merge cert f r T pc (step pc s) (cert!(pc+1))"
```

```
wtl_cert :: "'s certificate ⇒ 's binop ⇒ 's ord ⇒ 's ⇒ 's ⇒  
's step_type ⇒ nat ⇒ 's ⇒ 's"
```

```
"wtl_cert cert f r T B step pc s ≡  
if cert!pc = B then  
wtl_inst cert f r T step pc s  
else  
if s <=_r (cert!pc) then wtl_inst cert f r T step pc (cert!pc) else T"
```

```
consts
```

```
wtl_inst_list :: "'a list ⇒ 's certificate ⇒ 's binop ⇒ 's ord ⇒ 's ⇒ 's ⇒  
's step_type ⇒ nat ⇒ 's ⇒ 's"
```

```
primrec
```

```
"wtl_inst_list [] cert f r T B step pc s = s"  
"wtl_inst_list (i#is) cert f r T B step pc s =  
(let s' = wtl_cert cert f r T B step pc s in  
if s' = T ∨ s = T then T else wtl_inst_list is cert f r T B step (pc+1) s')"
```

```
constdefs
```

```
cert_ok :: "'s certificate ⇒ nat ⇒ 's ⇒ 's ⇒ 's set ⇒ bool"  
"cert_ok cert n T B A ≡ (∀i < n. cert!i ∈ A ∧ cert!i ≠ T) ∧ (cert!n = B)"
```

```
constdefs
```

```
bottom :: "'a ord ⇒ 'a ⇒ bool"  
"bottom r B ≡ ∀x. B <=_r x"
```

```
locale (open) lbv = semilat +
```

```
fixes T :: "'a" ("⊤")
```

```
fixes B :: "'a" ("⊥")
```

```
fixes step :: "'a step_type"
```

```
assumes top: "top r ⊤"
```

```
assumes T_A: "⊤ ∈ A"
```

```

assumes bot: "bottom r  $\perp$ "
assumes B_A: " $\perp \in A$ "

fixes merge :: "'a certificate  $\Rightarrow$  nat  $\Rightarrow$  (nat  $\times$  'a) list  $\Rightarrow$  'a  $\Rightarrow$  'a"
defines mrg_def: "merge cert  $\equiv$  LBVSpec.merge cert f r  $\top$ "

fixes wti :: "'a certificate  $\Rightarrow$  nat  $\Rightarrow$  'a  $\Rightarrow$  'a"
defines wti_def: "wti cert  $\equiv$  wtl_inst cert f r  $\top$  step"

fixes wtc :: "'a certificate  $\Rightarrow$  nat  $\Rightarrow$  'a  $\Rightarrow$  'a"
defines wtc_def: "wtc cert  $\equiv$  wtl_cert cert f r  $\top$   $\perp$  step"

fixes wtl :: "'b list  $\Rightarrow$  'a certificate  $\Rightarrow$  nat  $\Rightarrow$  'a  $\Rightarrow$  'a"
defines wtl_def: "wtl ins cert  $\equiv$  wtl_inst_list ins cert f r  $\top$   $\perp$  step"

lemma (in lbv) wti:
  "wti c pc s  $\equiv$  merge c pc (step pc s) (c!(pc+1))"
  by (simp add: wti_def mrg_def wtl_inst_def)

lemma (in lbv) wtc:
  "wtc c pc s  $\equiv$  if c!pc =  $\perp$  then wti c pc s else if s  $\leq_r$  c!pc then wti c pc (c!pc)
  else  $\top$ "
  by (unfold wtc_def wti_def wtl_cert_def)

lemma cert_okD1 [intro?]:
  "cert_ok c n T B A  $\implies$  pc < n  $\implies$  c!pc  $\in$  A"
  by (unfold cert_ok_def) fast

lemma cert_okD2 [intro?]:
  "cert_ok c n T B A  $\implies$  c!n = B"
  by (simp add: cert_ok_def)

lemma cert_okD3 [intro?]:
  "cert_ok c n T B A  $\implies$  B  $\in$  A  $\implies$  pc < n  $\implies$  c!Suc pc  $\in$  A"
  by (drule Suc_leI) (auto simp add: le_eq_less_or_eq dest: cert_okD1 cert_okD2)

lemma cert_okD4 [intro?]:
  "cert_ok c n T B A  $\implies$  pc < n  $\implies$  c!pc  $\neq$  T"
  by (simp add: cert_ok_def)

declare Let_def [simp]

```

#### 4.17.1 more semilattice lemmas

```

lemma (in lbv) sup_top [simp, elim]:
  assumes x: "x  $\in$  A"
  shows "x +_f  $\top$  =  $\top$ "
proof -
  from top have "x +_f  $\top$   $\leq_r$   $\top$ " ..
  moreover from x have " $\top$   $\leq_r$  x +_f  $\top$ " ..
  ultimately show ?thesis ..
qed

```

```
lemma (in lbv) plusplussup_top [simp, elim]:
  "set xs  $\subseteq$  A  $\implies$  xs ++_f  $\top$  =  $\top$ "
  by (induct xs) auto
```

```
lemma (in semilat) pp_ub1':
  assumes S: "snd' set S  $\subseteq$  A"
  assumes y: "y  $\in$  A" and ab: "(a, b)  $\in$  set S"
  shows "b  $\leq_r$  map snd [(p', t')  $\in$  S . p' = a] ++_f y"
proof -
  from S have " $\forall$  (x,y)  $\in$  set S. y  $\in$  A" by auto
  with semilat y ab show ?thesis by - (rule ub1')
qed
```

```
lemma (in lbv) bottom_le [simp, intro]:
  " $\perp$   $\leq_r$  x"
  by (insert bot) (simp add: bottom_def)
```

```
lemma (in lbv) le_bottom [simp]:
  "x  $\leq_r$   $\perp$  = (x =  $\perp$ )"
  by (blast intro: antisym_r)
```

#### 4.17.2 merge

```
lemma (in lbv) merge_Nil [simp]:
  "merge c pc [] x = x" by (simp add: mrg_def)
```

```
lemma (in lbv) merge_Cons [simp]:
  "merge c pc (l#ls) x = merge c pc ls (if fst l=pc+1 then snd l +_f x
    else if snd l  $\leq_r$  (c!fst l) then x
    else  $\top$ )"
  by (simp add: mrg_def split_beta)
```

```
lemma (in lbv) merge_Err [simp]:
  "snd' set ss  $\subseteq$  A  $\implies$  merge c pc ss  $\top$  =  $\top$ "
  by (induct ss) auto
```

```
lemma (in lbv) merge_not_top:
  " $\bigwedge$ x. snd' set ss  $\subseteq$  A  $\implies$  merge c pc ss x  $\neq$   $\top$   $\implies$ 
 $\forall$  (pc', s')  $\in$  set ss. (pc'  $\neq$  pc+1  $\longrightarrow$  s'  $\leq_r$  (c!pc'))"
  (is " $\bigwedge$ x. ?set ss  $\implies$  ?merge ss x  $\implies$  ?P ss")
```

```
proof (induct ss)
```

```
  show "?P []" by simp
```

```
next
```

```
  fix x ls l
```

```
  assume "?set (l#ls)" then obtain set: "snd' set ls  $\subseteq$  A" by simp
```

```
  assume merge: "?merge (l#ls) x"
```

```
  moreover
```

```
  obtain pc' s' where [simp]: "l = (pc', s')" by (cases l)
```

```
  ultimately
```

```
  obtain x' where "?merge ls x'" by simp
```

```
  assume " $\bigwedge$ x. ?set ls  $\implies$  ?merge ls x  $\implies$  ?P ls" hence "?P ls" .
```

```

    moreover
    from merge set
    have "pc' ≠ pc+1 → s' ≤r (c!pc'" by (simp split: split_if_asm)
    ultimately
    show "?P (l#ls)" by simp
qed

```

```

lemma (in lbv) merge_def:
  shows
    "∧x. x ∈ A ⇒ snd' set ss ⊆ A ⇒
    merge c pc ss x =
    (if ∃ (pc', s') ∈ set ss. pc' ≠ pc+1 → s' ≤r c!pc' then
      map snd [(p', t') ∈ ss. p' = pc+1] ++_f x
    else ⊤)"
    (is "∧x. _ ⇒ _ ⇒ ?merge ss x = ?if ss x" is "∧x. _ ⇒ _ ⇒ ?P ss x")
  proof (induct ss)
    fix x show "?P [] x" by simp
  next
    fix x assume x: "x ∈ A"
    fix l: "nat × 'a" and ls
    assume "snd' set (l#ls) ⊆ A"
    then obtain l: "snd l ∈ A" and ls: "snd' set ls ⊆ A" by auto
    assume "∧x. x ∈ A ⇒ snd' set ls ⊆ A ⇒ ?P ls x"
    hence IH: "∧x. x ∈ A ⇒ ?P ls x" .
    obtain pc' s' where [simp]: "l = (pc', s'" by (cases l)
    hence "?merge (l#ls) x = ?merge ls
      (if pc' = pc+1 then s' ++_f x else if s' ≤r c!pc' then x else ⊤)"
      (is "?merge (l#ls) x = ?merge ls ?if'")
      by simp
    also have "... = ?if ls ?if'"
  proof -
    from l have "s' ∈ A" by simp
    with x have "s' ++_f x ∈ A" by simp
    with x have "?if' ∈ A" by auto
    hence "?P ls ?if'" by (rule IH) thus ?thesis by simp
  qed
  also have "... = ?if (l#ls) x"
  proof (cases "∃ (pc', s') ∈ set (l#ls). pc' ≠ pc+1 → s' ≤r c!pc'")
    case True
    hence "∃ (pc', s') ∈ set ls. pc' ≠ pc+1 → s' ≤r c!pc'" by auto
    moreover
    from True have
      "map snd [(p', t') ∈ ls . p' = pc+1] ++_f ?if' =
      (map snd [(p', t') ∈ l#ls . p' = pc+1] ++_f x)"
      by simp
    ultimately
    show ?thesis using True by simp
  next
    case False
    moreover
    from ls have "set (map snd [(p', t') ∈ ls . p' = Suc pc]) ⊆ A" by auto
    ultimately show ?thesis by auto
  qed

```



finally show "?P (l#ls) x" .  
qed

```
lemma (in lbv) merge_not_top_s:
  assumes x: "x ∈ A" and ss: "snd' set ss ⊆ A"
  assumes m: "merge c pc ss x ≠ ⊤"
  shows "merge c pc ss x = (map snd [(p',t') ∈ ss. p'=pc+1] ++_f x)"
proof -
  from ss m have "∀(pc',s') ∈ set ss. (pc' ≠ pc+1 → s' <=_r c!pc')"
    by (rule merge_not_top)
  with x ss m show ?thesis by - (drule merge_def, auto split: split_if_asm)
qed
```

### 4.17.3 wtl-inst-list

lemmas [iff] = not\_Err\_eq

```
lemma (in lbv) wtl_Nil [simp]: "wtl [] c pc s = s"
  by (simp add: wtl_def)
```

```
lemma (in lbv) wtl_Cons [simp]:
  "wtl (i#is) c pc s =
  (let s' = wtc c pc s in if s' = ⊤ ∨ s = ⊤ then ⊤ else wtl is c (pc+1) s')"
  by (simp add: wtl_def wtc_def)
```

```
lemma (in lbv) wtl_Cons_not_top:
  "wtl (i#is) c pc s ≠ ⊤ =
  (wtc c pc s ≠ ⊤ ∧ s ≠ ⊤ ∧ wtl is c (pc+1) (wtc c pc s) ≠ ⊤)"
  by (auto simp del: split_paired_Ex)
```

```
lemma (in lbv) wtl_top [simp]: "wtl ls c pc ⊤ = ⊤"
  by (cases ls) auto
```

```
lemma (in lbv) wtl_not_top:
  "wtl ls c pc s ≠ ⊤ ⇒ s ≠ ⊤"
  by (cases "s=⊤") auto
```

```
lemma (in lbv) wtl_append [simp]:
  "∧pc s. wtl (a@b) c pc s = wtl b c (pc+length a) (wtl a c pc s)"
  by (induct a) auto
```

```
lemma (in lbv) wtl_take:
  "wtl is c pc s ≠ ⊤ ⇒ wtl (take pc' is) c pc s ≠ ⊤"
  (is "?wtl is ≠ _ ⇒ _")
```

```
proof -
  assume "?wtl is ≠ ⊤"
  hence "?wtl (take pc' is @ drop pc' is) ≠ ⊤" by simp
  thus ?thesis by (auto dest!: wtl_not_top simp del: append_take_drop_id)
qed
```

```
lemma take_Suc:
  "∀n. n < length l → take (Suc n) l = (take n l)@[l!n]" (is "?P l")
proof (induct l)
  show "?P []" by simp
```

```

next
  fix x xs assume IH: "?P xs"
  show "?P (x#xs)"
  proof (intro strip)
    fix n assume "n < length (x#xs)"
    with IH show "take (Suc n) (x # xs) = take n (x # xs) @ [(x # xs) ! n]"
      by (cases n, auto)
  qed
qed

lemma (in lbv) wtl_Suc:
  assumes suc: "pc+1 < length is"
  assumes wtl: "wtl (take pc is) c 0 s  $\neq$   $\top$ "
  shows "wtl (take (pc+1) is) c 0 s = wtc c pc (wtl (take pc is) c 0 s)"
proof -
  from suc have "take (pc+1) is=(take pc is)@[is!pc]" by (simp add: take_Suc)
  with suc wtl show ?thesis by (simp add: min_def)
qed

lemma (in lbv) wtl_all:
  assumes all: "wtl is c 0 s  $\neq$   $\top$ " (is "?wtl is  $\neq$  _")
  assumes pc: "pc < length is"
  shows "wtc c pc (wtl (take pc is) c 0 s)  $\neq$   $\top$ "
proof -
  from pc have "0 < length (drop pc is)" by simp
  then obtain i r where Cons: "drop pc is = i#r"
    by (auto simp add: neq_Nil_conv simp del: length_drop)
  hence "i#r = drop pc is" ..
  with all have take: "?wtl (take pc is@i#r)  $\neq$   $\top$ " by simp
  from pc have "is!pc = drop pc is ! 0" by simp
  with Cons have "is!pc = i" by simp
  with take pc show ?thesis by (auto simp add: min_def split: split_if_asm)
qed

```

#### 4.17.4 preserves-type

```

lemma (in lbv) merge_pres:
  assumes s0: "snd'set ss  $\subseteq$  A" and x: "x  $\in$  A"
  shows "merge c pc ss x  $\in$  A"
proof -
  from s0 have "set (map snd [(p', t') $\in$ ss . p'=pc+1])  $\subseteq$  A" by auto
  with x have "(map snd [(p', t') $\in$ ss . p'=pc+1] ++_f x)  $\in$  A"
    by (auto intro!: plusplus_closed)
  with s0 x show ?thesis by (simp add: merge_def T_A)
qed

```

```

lemma pres_typeD2:
  "pres_type step n A  $\implies$  s  $\in$  A  $\implies$  p < n  $\implies$  snd'set (step p s)  $\subseteq$  A"
  by auto (drule pres_typeD)

```

```

lemma (in lbv) wti_pres [intro?]:
  assumes pres: "pres_type step n A"

```

```

  assumes cert: "c!(pc+1) ∈ A"
  assumes s_pc: "s ∈ A" "pc < n"
  shows "wti c pc s ∈ A"
proof -
  from pres s_pc have "snd'set (step pc s) ⊆ A" by (rule pres_typeD2)
  with cert show ?thesis by (simp add: wti merge_pres)
qed

```

```

lemma (in lbv) wtc_pres:
  assumes "pres_type step n A"
  assumes "c!pc ∈ A" and "c!(pc+1) ∈ A"
  assumes "s ∈ A" and "pc < n"
  shows "wtc c pc s ∈ A"
proof -
  have "wti c pc s ∈ A" ..
  moreover have "wti c pc (c!pc) ∈ A" ..
  ultimately show ?thesis using T_A by (simp add: wtc)
qed

```

```

lemma (in lbv) wtl_pres:
  assumes pres: "pres_type step (length is) A"
  assumes cert: "cert_ok c (length is) ⊤ ⊥ A"
  assumes s: "s ∈ A"
  assumes all: "wtl is c 0 s ≠ ⊤"
  shows "pc < length is ⇒ wtl (take pc is) c 0 s ∈ A"
  (is "?len pc ⇒ ?wtl pc ∈ A")
proof (induct pc)
  from s show "?wtl 0 ∈ A" by simp
next
  fix n assume "Suc n < length is"
  then obtain n: "n < length is" by simp
  assume "n < length is ⇒ ?wtl n ∈ A"
  hence "?wtl n ∈ A" .
  moreover
  from cert have "c!n ∈ A" by (rule cert_okD1)
  moreover
  have n1: "n+1 < length is" by simp
  with cert have "c!(n+1) ∈ A" by (rule cert_okD1)
  ultimately
  have "wtc c n (?wtl n) ∈ A" by - (rule wtc_pres)
  also
  from all n have "?wtl n ≠ ⊤" by - (rule wtl_take)
  with n1 have "wtc c n (?wtl n) = ?wtl (n+1)" by (rule wtl_Suc [symmetric])
  finally show "?wtl (Suc n) ∈ A" by simp
qed

```

end

## 4.18 Correctness of the LBV

theory LBVCorrect = LBVSpec + Typing\_Framework:

```

locale (open) lbvs = lbv +
  fixes s0  :: 'a
  fixes c   :: "'a list"
  fixes ins :: "'b list"
  fixes phi :: "'a list" ("φ")
  defines phi_def:
    "φ ≡ map (λpc. if c!pc = ⊥ then wtl (take pc ins) c 0 s0 else c!pc)
      [0..length ins]"

  assumes bounded: "bounded step (length ins) A"
  assumes cert: "cert_ok c (length ins) ⊤ ⊥ A"
  assumes pres: "pres_type step (length ins) A"

lemma (in lbvs) phi_None [intro?]:
  "[ pc < length ins; c!pc = ⊥ ] ⇒ φ ! pc = wtl (take pc ins) c 0 s0"
  by (simp add: phi_def)

lemma (in lbvs) phi_Some [intro?]:
  "[ pc < length ins; c!pc ≠ ⊥ ] ⇒ φ ! pc = c ! pc"
  by (simp add: phi_def)

lemma (in lbvs) phi_len [simp]:
  "length φ = length ins"
  by (simp add: phi_def)

lemma (in lbvs) wtl_suc_pc:
  assumes all: "wtl ins c 0 s0 ≠ ⊤"
  assumes pc: "pc+1 < length ins"
  shows "wtl (take (pc+1) ins) c 0 s0 ≤r φ!(pc+1)"
proof -
  from all pc
  have "wtc c (pc+1) (wtl (take (pc+1) ins) c 0 s0) ≠ ⊤" by (rule wtl_all)
  with pc show ?thesis by (simp add: phi_def wtc split: split_if_asm)
qed

lemma (in lbvs) wtl_stable:
  assumes wtl: "wtl ins c 0 s0 ≠ ⊤"
  assumes s0: "s0 ∈ A"
  assumes pc: "pc < length ins"
  shows "stable r step φ pc"
proof (unfold stable_def, clarify)
  fix pc' s' assume step: "(pc',s') ∈ set (step pc (φ ! pc))"
    (is "(pc',s') ∈ set (?step pc)")

  have tkpc: "wtl (take pc ins) c 0 s0 ≠ ⊤" (is "?s1 ≠ _") by (rule wtl_take)
  have s2: "wtl (take (pc+1) ins) c 0 s0 ≠ ⊤" (is "?s2 ≠ _") by (rule wtl_take)

```

```

from wtl pc have wt_s1: "wtc c pc ?s1  $\neq$   $\top$ " by (rule wtl_all)

have c_Some: " $\forall$  pc t. pc < length ins  $\longrightarrow$  c!pc  $\neq$   $\perp$   $\longrightarrow$   $\varphi$ !pc = c!pc"
  by (simp add: phi_def)
have c_None: "c!pc =  $\perp$   $\implies$   $\varphi$ !pc = ?s1" ..

from wt_s1 pc c_None c_Some
have inst: "wtc c pc ?s1 = wti c pc ( $\varphi$ !pc)"
  by (simp add: wtc_split: split_if_asm)

have "?s1  $\in$  A" by (rule wtl_pres)
with pc c_Some cert c_None
have phi_in_A: " $\varphi$ !pc  $\in$  A" by (cases "c!pc =  $\perp$ ") (auto dest: cert_okD1)
with pc pres
have step_in_A: "snd'set (?step pc)  $\subseteq$  A" by (auto dest: pres_typeD2)

from bounded pc phi_in_A step have pc': "pc' < length ins" by (rule boundedD)

show "s'  $\leq_r$   $\varphi$ !pc'"
proof (cases "pc' = pc+1")
  case True
  with pc' cert
  have cert_in_A: "c!(pc+1)  $\in$  A" by (auto dest: cert_okD1)
  from True pc' have pc1: "pc+1 < length ins" by simp
  with tkpc have "?s2 = wtc c pc ?s1" by - (rule wtl_Suc)
  with inst
  have merge: "?s2 = merge c pc (?step pc) (c!(pc+1))" by (simp add: wti)
  also
  from s2 merge have "...  $\neq$   $\top$ " (is "?merge  $\neq$  _") by simp
  with cert_in_A step_in_A
  have "?merge = (map snd [(p',t') $\in$ ?step pc. p'=pc+1] ++_f (c!(pc+1)))"
    by (rule merge_not_top_s)
  finally
  have "s'  $\leq_r$  ?s2" using step_in_A cert_in_A True step
    by (auto intro: pp_ub1')
  also
  from wtl pc1 have "?s2  $\leq_r$   $\varphi$ !(pc+1)" by (rule wtl_suc_pc)
  also note True [symmetric]
  finally show ?thesis by simp
next
  case False
  from wt_s1 inst
  have "merge c pc (?step pc) (c!(pc+1))  $\neq$   $\top$ " by (simp add: wti)
  with step_in_A
  have " $\forall$  (pc', s') $\in$ set (?step pc). pc'  $\neq$  pc+1  $\longrightarrow$  s'  $\leq_r$  c!pc'"
    by - (rule merge_not_top)
  with step False
  have ok: "s'  $\leq_r$  c!pc'" by blast
  moreover
  from ok
  have "c!pc' =  $\perp$   $\implies$  s' =  $\perp$ " by simp
  moreover
  from c_Some pc'
  have "c!pc'  $\neq$   $\perp$   $\implies$   $\varphi$ !pc' = c!pc'" by auto

```

```

ultimately
  show ?thesis by (cases "c!pc' =  $\perp$ ") auto
qed

```

```

lemma (in lbvs) phi_not_top:
  assumes wtl: "wtl ins c 0 s0  $\neq$   $\top$ "
  assumes pc: "pc < length ins"
  shows " $\varphi!$ pc  $\neq$   $\top$ "
proof (cases "c!pc =  $\perp$ ")
  case False with pc
  have " $\varphi!$ pc = c!pc" ..
  also from cert pc have "...  $\neq$   $\top$ " by (rule cert_okD4)
  finally show ?thesis .
next
  case True with pc
  have " $\varphi!$ pc = wtl (take pc ins) c 0 s0" ..
  also from wtl have "...  $\neq$   $\top$ " by (rule wtl_take)
  finally show ?thesis .
qed

```

```

lemma (in lbvs) phi_in_A:
  assumes wtl: "wtl ins c 0 s0  $\neq$   $\top$ "
  assumes s0: "s0  $\in$  A"
  shows " $\varphi \in$  list (length ins) A"
proof -
  { fix x assume "x  $\in$  set  $\varphi$ "
    then obtain xs ys where " $\varphi$  = xs @ x # ys"
      by (auto simp add: in_set_conv_decomp)
    then obtain pc where pc: "pc < length  $\varphi$ " and x: " $\varphi!$ pc = x"
      by (simp add: that [of "length xs"] nth_append)

    from wtl s0 pc
    have "wtl (take pc ins) c 0 s0  $\in$  A" by (auto intro!: wtl_pres)
    moreover
    from pc have "pc < length ins" by simp
    with cert have "c!pc  $\in$  A" ..
    ultimately
    have " $\varphi!$ pc  $\in$  A" using pc by (simp add: phi_def)
    hence "x  $\in$  A" using x by simp
  }
  hence "set  $\varphi \subseteq$  A" ..
  thus ?thesis by (unfold list_def) simp
qed

```

```

lemma (in lbvs) phi0:
  assumes wtl: "wtl ins c 0 s0  $\neq$   $\top$ "
  assumes 0: "0 < length ins"
  shows "s0  $\leq_r$   $\varphi!0$ "
proof (cases "c!0 =  $\perp$ ")
  case True
  with 0 have " $\varphi!0$  = wtl (take 0 ins) c 0 s0" ..

```

```

    moreover have "wtl (take 0 ins) c 0 s0 = s0" by simp
    ultimately have " $\varphi!0 = s0$ " by simp
    thus ?thesis by simp
next
  case False
  with 0 have " $\varphi!0 = c!0$ " ..
  moreover
  have "wtl (take 1 ins) c 0 s0  $\neq \top$ " by (rule wtl_take)
  with 0 False
  have " $s0 \leq_r c!0$ " by (auto simp add: neq_Nil_conv wtc split: split_if_asm)
  ultimately
  show ?thesis by simp
qed

```

```

theorem (in lbvs) wtl_sound:
  assumes "wtl ins c 0 s0  $\neq \top$ "
  assumes " $s0 \in A$ "
  shows " $\exists ts. wt\_step\ r\ \top\ step\ ts$ "
proof -
  have "wt_step r  $\top$  step  $\varphi$ "
  proof (unfold wt_step_def, intro strip conjI)
    fix pc assume " $pc < length\ \varphi$ "
    then obtain " $pc < length\ ins$ " by simp
    show " $\varphi!pc \neq \top$ " by (rule phi_not_top)
    show "stable r step  $\varphi$  pc" by (rule wtl_stable)
  qed
  thus ?thesis ..
qed

```

```

theorem (in lbvs) wtl_sound_strong:
  assumes "wtl ins c 0 s0  $\neq \top$ "
  assumes " $s0 \in A$ "
  assumes " $0 < length\ ins$ "
  shows " $\exists ts \in list\ (length\ ins)\ A. wt\_step\ r\ \top\ step\ ts \wedge s0 \leq_r\ ts!0$ "
proof -
  have " $\varphi \in list\ (length\ ins)\ A$ " by (rule phi_in_A)
  moreover
  have "wt_step r  $\top$  step  $\varphi$ "
  proof (unfold wt_step_def, intro strip conjI)
    fix pc assume " $pc < length\ \varphi$ "
    then obtain " $pc < length\ ins$ " by simp
    show " $\varphi!pc \neq \top$ " by (rule phi_not_top)
    show "stable r step  $\varphi$  pc" by (rule wtl_stable)
  qed
  moreover
  have " $s0 \leq_r \varphi!0$ " by (rule phi0)
  ultimately
  show ?thesis by fast
qed
end

```

## 4.19 Completeness of the LBV

theory LBVComplete = LBVSpec + Typing\_Framework:

constdefs

```
is_target :: "[ 's step_type, 's list, nat ] ⇒ bool"
"is_target step phi pc' ≡
  ∃ pc s'. pc' ≠ pc+1 ∧ pc < length phi ∧ (pc',s') ∈ set (step pc (phi!pc))"

make_cert :: "[ 's step_type, 's list, 's ] ⇒ 's certificate"
"make_cert step phi B ≡
  map (λpc. if is_target step phi pc then phi!pc else B) [0..length phi() @ [B]]"
```

constdefs

```
list_ex :: "('a ⇒ bool) ⇒ 'a list ⇒ bool"
"list_ex P xs ≡ ∃ x ∈ set xs. P x"
```

lemma [code]: "list\_ex P [] = False" by (simp add: list\_ex\_def)

lemma [code]: "list\_ex P (x#xs) = (P x ∨ list\_ex P xs)" by (simp add: list\_ex\_def)

lemma [code]:

```
"is_target step phi pc' =
list_ex (λpc. pc' ≠ pc+1 ∧ pc' mem (map fst (step pc (phi!pc)))) [0..length phi()]"
apply (simp add: list_ex_def is_target_def set_mem_eq)
apply force
done
```

locale (open) lbvc = lbv +

```
fixes phi :: "'a list" ("φ")
fixes c    :: "'a list"
defines cert_def: "c ≡ make_cert step φ ⊥"
```

```
assumes mono: "mono r step (length φ) A"
assumes pres: "pres_type step (length φ) A"
assumes phi:  "∀pc < length φ. φ!pc ∈ A ∧ φ!pc ≠ ⊥"
assumes bounded: "bounded step (length φ) A"
```

```
assumes B_neq_T: "⊥ ≠ ⊤"
```

lemma (in lbvc) cert: "cert\_ok c (length φ) ⊤ ⊥ A"

proof (unfold cert\_ok\_def, intro strip conjI)

note [simp] = make\_cert\_def cert\_def nth\_append

show "c!length φ = ⊥" by simp

fix pc assume pc: "pc < length φ"

from pc phi B\_A show "c!pc ∈ A" by simp

from pc phi B\_neq\_T show "c!pc ≠ ⊤" by simp

qed

lemmas [simp del] = split\_paired\_Ex



```

lemma (in lbvc) cert_target [intro?]:
  "[ (pc',s') ∈ set (step pc (φ!pc));
    pc' ≠ pc+1; pc < length φ; pc' < length φ ]
  ⇒ c!pc' = φ!pc'"
  by (auto simp add: cert_def make_cert_def nth_append is_target_def)

```

```

lemma (in lbvc) cert_approx [intro?]:
  "[ pc < length φ; c!pc ≠ ⊥ ]
  ⇒ c!pc = φ!pc"
  by (auto simp add: cert_def make_cert_def nth_append)

```

```

lemma (in lbv) le_top [simp, intro]:
  "x ≤r ⊤"
  by (insert top) simp

```

```

lemma (in lbv) merge_mono:
  assumes less: "ss2 ≤r ss1"
  assumes x: "x ∈ A"
  assumes ss1: "snd'set ss1 ⊆ A"
  assumes ss2: "snd'set ss2 ⊆ A"
  shows "merge c pc ss2 x ≤r merge c pc ss1 x" (is "?s2 ≤r ?s1")

```

proof-

```

have "?s1 = ⊤ ⇒ ?thesis" by simp

```

```

moreover {

```

```

  assume merge: "?s1 ≠ ⊤"

```

```

  from x ss1 have "?s1 =

```

```

    (if ∃ (pc', s') ∈ set ss1. pc' ≠ pc + 1 → s' ≤r c!pc'
    then (map snd [(p', t') ∈ ss1 . p' = pc+1]) ++_f x
    else ⊤)"

```

```

  by (rule merge_def)

```

```

with merge obtain

```

```

  app: "∃ (pc', s') ∈ set ss1. pc' ≠ pc+1 → s' ≤r c!pc'"
  (is "?app ss1") and

```

```

  sum: "(map snd [(p', t') ∈ ss1 . p' = pc+1] ++_f x) = ?s1"
  (is "?map ss1 ++_f x = _" is "?sum ss1 = _")

```

```

  by (simp split: split_if_asm)

```

```

from app less

```

```

have "?app ss2" by (blast dest: trans_r lesub_step_typeD)

```

```

moreover {

```

```

  from ss1 have map1: "set (?map ss1) ⊆ A" by auto

```

```

  with x have "?sum ss1 ∈ A" by (auto intro!: plusplus_closed)

```

```

  with sum have "?s1 ∈ A" by simp

```

```

  moreover

```

```

  have mapD: "∧ x ss. x ∈ set (?map ss) ⇒ ∃ p. (p, x) ∈ set ss ∧ p = pc+1" by auto

```

```

  from x map1

```

```

  have "∧ x ∈ set (?map ss1). x ≤r ?sum ss1"

```

```

  by clarify (rule pp_ub1)

```

```

  with sum have "∧ x ∈ set (?map ss1). x ≤r ?s1" by simp

```

```

  with less have "∧ x ∈ set (?map ss2). x ≤r ?s1"

```

```

  by (fastsimp dest!: mapD lesub_step_typeD intro: trans_r)

```

```

  moreover

```

```

    from map1 x have "x <=_r (?sum ss1)" by (rule pp_ub2)
    with sum have "x <=_r ?s1" by simp
    moreover
    from ss2 have "set (?map ss2)  $\subseteq$  A" by auto
    ultimately
    have "?sum ss2 <=_r ?s1" using x by - (rule pp_lub)
  }
  moreover
  from x ss2 have
    "?s2 =
    (if  $\forall (pc', s') \in \text{set } ss2. pc' \neq pc + 1 \longrightarrow s' <=_r c!pc'$ 
    then map snd [(p', t')  $\in$  ss2 . p' = pc + 1] ++_f x
    else  $\top$ )"
    by (rule merge_def)
  ultimately have ?thesis by simp
}
ultimately show ?thesis by (cases "?s1 =  $\top$ ") auto
qed

```

```

lemma (in lbvc) wti_mono:
  assumes less: "s2 <=_r s1"
  assumes pc: "pc < length  $\varphi$ "
  assumes s1: "s1  $\in$  A"
  assumes s2: "s2  $\in$  A"
  shows "wti c pc s2 <=_r wti c pc s1" (is "?s2' <=_r ?s1'")
proof -
  from mono s2 have "step pc s2 <=|r| step pc s1" by - (rule monoD)
  moreover
  from pc cert have "c!Suc pc  $\in$  A" by - (rule cert_okD3)
  moreover
  from pres s1 pc
  have "snd'set (step pc s1)  $\subseteq$  A" by (rule pres_typeD2)
  moreover
  from pres s2 pc
  have "snd'set (step pc s2)  $\subseteq$  A" by (rule pres_typeD2)
  ultimately
  show ?thesis by (simp add: wti merge_mono)
qed

```

```

lemma (in lbvc) wtc_mono:
  assumes less: "s2 <=_r s1"
  assumes pc: "pc < length  $\varphi$ "
  assumes s1: "s1  $\in$  A"
  assumes s2: "s2  $\in$  A"
  shows "wtc c pc s2 <=_r wtc c pc s1" (is "?s2' <=_r ?s1'")
proof (cases "c!pc =  $\perp$ ")
  case True
  moreover have "wti c pc s2 <=_r wti c pc s1" by (rule wti_mono)
  ultimately show ?thesis by (simp add: wtc)
next
  case False
  have "?s1' =  $\top \implies$  ?thesis" by simp
  moreover {

```

```

    assume "?s1' ≠ ⊤"
    with False have c: "s1 ≤r c!pc" by (simp add: wtc split: split_if_asm)
    with less have "s2 ≤r c!pc" ..
    with False c have ?thesis by (simp add: wtc)
  }
  ultimately show ?thesis by (cases "?s1' = ⊤") auto
qed

```

```

lemma (in lbv) top_le_conv [simp]:
  "⊤ ≤r x = (x = ⊤)"
  by (insert semilat) (simp add: top top_le_conv)

```

```

lemma (in lbv) neq_top [simp, elim]:
  "[ x ≤r y; y ≠ ⊤ ] ⇒ x ≠ ⊤"
  by (cases "x = ⊤") auto

```

```

lemma (in lbvc) stable_wti:
  assumes stable: "stable r step φ pc"
  assumes pc: "pc < length φ"
  shows "wti c pc (φ!pc) ≠ ⊤"
proof -
  let ?step = "step pc (φ!pc)"
  from stable
  have less: "∀ (q,s') ∈ set ?step. s' ≤r φ!q" by (simp add: stable_def)

  from cert pc
  have cert_suc: "c!Suc pc ∈ A" by - (rule cert_okD3)
  moreover
  from phi pc have phi_in_A: "φ!pc ∈ A" by simp
  with pres pc
  have stepA: "snd' set ?step ⊆ A" by - (rule pres_typeD2)
  ultimately
  have "merge c pc ?step (c!Suc pc) =
    (if ∀ (pc',s') ∈ set ?step. pc' ≠ pc+1 → s' ≤r c!pc'
     then map snd [(p',t') ∈ ?step.p'=pc+1] ++_f c!Suc pc
     else ⊤)" by (rule merge_def)
  moreover {
    fix pc' s' assume s': "(pc', s') ∈ set ?step" and suc_pc: "pc' ≠ pc+1"
    with less have "s' ≤r φ!pc'" by auto
    also
    from bounded pc phi_in_A s' have "pc' < length φ" by (rule boundedD)
    with s' suc_pc pc have "c!pc' = φ!pc'" ..
    hence "φ!pc' = c!pc'" ..
    finally have "s' ≤r c!pc'" .
  } hence "∀ (pc',s') ∈ set ?step. pc' ≠ pc+1 → s' ≤r c!pc'" by auto
  moreover
  from pc have "Suc pc = length φ ∨ Suc pc < length φ" by auto
  hence "map snd [(p',t') ∈ ?step.p'=pc+1] ++_f c!Suc pc ≠ ⊤"
    (is "?map ++_f _ ≠ _")
proof (rule disjE)
  assume pc': "Suc pc = length φ"
  with cert have "c!Suc pc = ⊥" by (simp add: cert_okD2)

```

```

    moreover
    from pc' bounded pc phi_in_A
    have " $\forall (p', t') \in \text{set } ?\text{step}. p' \neq pc+1$ " by clarify (drule boundedD, auto)
    hence " $[(p', t') \in ?\text{step}. p' = pc+1] = []$ " by (blast intro: filter_False)
    hence "?map = []" by simp
    ultimately show ?thesis by (simp add: B_neq_T)
  next
    assume pc': "Suc pc < length  $\varphi$ "
    from pc' phi have " $\varphi! \text{Suc } pc \in A$ " by simp
    moreover note cert_suc
    moreover from stepA
    have "set ?map  $\subseteq A$ " by auto
    moreover
    have " $\bigwedge s. s \in \text{set } ?\text{map} \implies \exists t. (\text{Suc } pc, t) \in \text{set } ?\text{step}$ " by auto
    with less have " $\forall s' \in \text{set } ?\text{map}. s' \leq_r \varphi! \text{Suc } pc$ " by auto
    moreover
    from pc' have " $c! \text{Suc } pc \leq_r \varphi! \text{Suc } pc$ "
      by (cases " $c! \text{Suc } pc = \perp$ ") (auto dest: cert_approx)
    ultimately
    have "?map ++_f c!Suc pc  $\leq_r \varphi! \text{Suc } pc$ " by (rule pp_lub)
    moreover
    from pc' phi have " $\varphi! \text{Suc } pc \neq \top$ " by simp
    ultimately
    show ?thesis by auto
  qed
  ultimately
  have "merge c pc ?step (c!Suc pc)  $\neq \top$ " by simp
  thus ?thesis by (simp add: wti)
qed

lemma (in lbvc) wti_less:
  assumes stable: "stable r step  $\varphi$  pc"
  assumes suc_pc: "Suc pc < length  $\varphi$ "
  shows "wti c pc ( $\varphi!pc$ )  $\leq_r \varphi! \text{Suc } pc$ " (is "?wti  $\leq_r$  _")
proof -
  let ?step = "step pc ( $\varphi!pc$ )"

  from stable
  have less: " $\forall (q, s') \in \text{set } ?\text{step}. s' \leq_r \varphi!q$ " by (simp add: stable_def)

  from suc_pc have pc: "pc < length  $\varphi$ " by simp
  with cert have cert_suc: " $c! \text{Suc } pc \in A$ " by - (rule cert_okD3)
  moreover
  from phi pc have " $\varphi!pc \in A$ " by simp
  with pres pc have stepA: "snd 'set ?step  $\subseteq A$ " by - (rule pres_typeD2)
  moreover
  from stable pc have "?wti  $\neq \top$ " by (rule stable_wti)
  hence "merge c pc ?step (c!Suc pc)  $\neq \top$ " by (simp add: wti)
  ultimately
  have "merge c pc ?step (c!Suc pc) =
    map snd [(p', t')  $\in ?\text{step}. p' = pc+1$ ] ++_f c!Suc pc" by (rule merge_not_top_s)
  hence "?wti = ..." (is "_ = (?map ++_f _)" is "_ = ?sum") by (simp add: wti)
  also {
    from suc_pc phi have " $\varphi! \text{Suc } pc \in A$ " by simp
  }

```

```

    moreover note cert_suc
    moreover from stepA have "set ?map  $\subseteq$  A" by auto
    moreover
    have " $\bigwedge s. s \in \text{set ?map} \implies \exists t. (\text{Suc } pc, t) \in \text{set ?step}$ " by auto
    with less have " $\forall s' \in \text{set ?map}. s' \leq_r \varphi! \text{Suc } pc$ " by auto
    moreover
    from suc_pc have "c!Suc pc  $\leq_r \varphi! \text{Suc } pc$ "
      by (cases "c!Suc pc =  $\perp$ ") (auto dest: cert_approx)
    ultimately
    have "?sum  $\leq_r \varphi! \text{Suc } pc$ " by (rule pp_lub)
  }
  finally show ?thesis .
qed

```

```

lemma (in lbvc) stable_wtc:
  assumes stable: "stable r step phi pc"
  assumes pc: "pc < length  $\varphi$ "
  shows "wtc c pc ( $\varphi!pc$ )  $\neq$   $\top$ "
proof -
  have wti: "wti c pc ( $\varphi!pc$ )  $\neq$   $\top$ " by (rule stable_wti)
  show ?thesis
  proof (cases "c!pc =  $\perp$ ")
    case True with wti show ?thesis by (simp add: wtc)
  next
    case False
    with pc have "c!pc =  $\varphi!pc$ " ..
    with False wti show ?thesis by (simp add: wtc)
  qed
qed

```

```

lemma (in lbvc) wtc_less:
  assumes stable: "stable r step  $\varphi$  pc"
  assumes suc_pc: "Suc pc < length  $\varphi$ "
  shows "wtc c pc ( $\varphi!pc$ )  $\leq_r \varphi! \text{Suc } pc$ " (is "?wtc  $\leq_r$  _")
proof (cases "c!pc =  $\perp$ ")
  case True
  moreover have "wti c pc ( $\varphi!pc$ )  $\leq_r \varphi! \text{Suc } pc$ " by (rule wti_less)
  ultimately show ?thesis by (simp add: wtc)
next
  case False
  from suc_pc have pc: "pc < length  $\varphi$ " by simp
  hence "?wtc  $\neq$   $\top$ " by - (rule stable_wtc)
  with False have "?wtc = wti c pc (c!pc)"
    by (unfold wtc) (simp split: split_if_asm)
  also from pc False have "c!pc =  $\varphi!pc$ " ..
  finally have "?wtc = wti c pc ( $\varphi!pc$ )" .
  also have "wti c pc ( $\varphi!pc$ )  $\leq_r \varphi! \text{Suc } pc$ " by (rule wti_less)
  finally show ?thesis .
qed

```

```

lemma (in lbvc) wt_step_wtl_lemma:
  assumes wt_step: "wt_step r  $\top$  step  $\varphi$ "
  shows " $\bigwedge pc s. pc + \text{length } ls = \text{length } \varphi \implies s \leq_r \varphi!pc \implies s \in A \implies s \neq \top \implies$ "

```

```

      wtl ls c pc s ≠ T"
    (is "∧pc s. _ ⇒ _ ⇒ _ ⇒ _ ⇒ ?wtl ls pc s ≠ _")
  proof (induct ls)
    fix pc s assume "s≠T" thus "?wtl [] pc s ≠ T" by simp
  next
    fix pc s i ls
    assume "∧pc s. pc+length ls=length φ ⇒ s ≤_r φ!pc ⇒ s ∈ A ⇒ s≠T ⇒
      ?wtl ls pc s ≠ T"
    moreover
    assume pc_1: "pc + length (i#ls) = length φ"
    hence suc_pc_1: "Suc pc + length ls = length φ" by simp
    ultimately
    have IH: "∧s. s ≤_r φ!Suc pc ⇒ s ∈ A ⇒ s ≠ T ⇒ ?wtl ls (Suc pc) s ≠ T" .

    from pc_1 obtain pc: "pc < length φ" by simp
    with wt_step have stable: "stable r step φ pc" by (simp add: wt_step_def)
    moreover
    assume s_phi: "s ≤_r φ!pc"
    ultimately
    have wt_phi: "wtc c pc (φ!pc) ≠ T" by - (rule stable_wtc)

    from phi pc have phi_pc: "φ!pc ∈ A" by simp
    moreover
    assume s: "s ∈ A"
    ultimately
    have wt_s_phi: "wtc c pc s ≤_r wtc c pc (φ!pc)" using s_phi by - (rule wtc_mono)
    with wt_phi have wt_s: "wtc c pc s ≠ T" by simp
    moreover
    assume s: "s ≠ T"
    ultimately
    have "ls = [] ⇒ ?wtl (i#ls) pc s ≠ T" by simp
    moreover {
      assume "ls ≠ []"
      with pc_1 have suc_pc: "Suc pc < length φ" by (auto simp add: neq_Nil_conv)
      with stable have "wtc c pc (phi!pc) ≤_r φ!Suc pc" by (rule wtc_less)
      with wt_s_phi have "wtc c pc s ≤_r φ!Suc pc" by (rule trans_r)
      moreover
      from cert suc_pc have "c!pc ∈ A" "c!(pc+1) ∈ A"
        by (auto simp add: cert_ok_def)
      with pres have "wtc c pc s ∈ A" by (rule wtc_pres)
      ultimately
      have "?wtl ls (Suc pc) (wtc c pc s) ≠ T" using IH wt_s by blast
      with s wt_s have "?wtl (i#ls) pc s ≠ T" by simp
    }
    ultimately show "?wtl (i#ls) pc s ≠ T" by (cases ls) blast+
  qed

```

theorem (in lbvc) wtl\_complete:

assumes "wt\_step r T step φ"

assumes "s ≤\_r φ!0" and "s ∈ A" and "s ≠ T" and "length ins = length phi"

shows "wtl ins c 0 s ≠ T"

proof -

have "0+length ins = length phi" by simp

```
  thus ?thesis by - (rule wt_step_wtl_lemma)
qed
```

```
end
```

## 4.20 LBV for the JVM

```
theory LBVJVM = LBVCorrect + LBVComplete + Typing_Framework_JVM:
```

```
types prog_cert = "cname  $\Rightarrow$  sig  $\Rightarrow$  state list"
```

```
constdefs
```

```
check_cert :: "jvm_prog  $\Rightarrow$  nat  $\Rightarrow$  nat  $\Rightarrow$  nat  $\Rightarrow$  state list  $\Rightarrow$  bool"
"check_cert G mxs mxr n cert  $\equiv$  check_types G mxs mxr n cert  $\wedge$  length cert = n+1  $\wedge$ 
  ( $\forall i < n$ . cert!i  $\neq$  Err)  $\wedge$  cert!n = OK {}"
```

```
lbvjvm :: "jvm_prog  $\Rightarrow$  cname  $\Rightarrow$  nat  $\Rightarrow$  nat  $\Rightarrow$  ty  $\Rightarrow$  bool  $\Rightarrow$  exception_table  $\Rightarrow$ 
  state list  $\Rightarrow$  instr list  $\Rightarrow$  state  $\Rightarrow$  state"
```

```
"lbvjvm G C mxs mxr rT ini et cert bs  $\equiv$ 
```

```
wtl_inst_list bs cert JVMType.sup JVMType.le Err (OK {}) (exec G C mxs rT ini et bs)
0"
```

```
wt_lbv :: "jvm_prog  $\Rightarrow$  cname  $\Rightarrow$  mname  $\Rightarrow$  ty list  $\Rightarrow$  ty  $\Rightarrow$  nat  $\Rightarrow$  nat  $\Rightarrow$ 
  exception_table  $\Rightarrow$  state list  $\Rightarrow$  instr list  $\Rightarrow$  bool"
```

```
"wt_lbv G C mn pTs rT mxs mxl et cert ins  $\equiv$ 
```

```
check_cert G mxs (1+size pTs+mxl) (length ins) cert  $\wedge$ 
0 < size ins  $\wedge$ 
```

```
(let this = OK (if mn=init  $\wedge$  C  $\neq$  Object then PartInit C else Init (Class C));
```

```
start = {([],this#(map (OK $\circ$ Init) pTs)@(replicate mxl Err)),C=Object});
```

```
result = lbvjvm G C mxs (1+size pTs+mxl) rT (mn=init) et cert ins (OK start)
```

```
in result  $\neq$  Err)"
```

```
wt_jvm_prog_lbv :: "jvm_prog  $\Rightarrow$  prog_cert  $\Rightarrow$  bool"
```

```
"wt_jvm_prog_lbv G cert  $\equiv$ 
```

```
wf_prog ( $\lambda G C$  (sig,rT,(mxs,maxl,b,et)). wt_lbv G C (fst sig) (snd sig) rT mxs maxl
et (cert C sig) b) G"
```

```
mk_cert :: "jvm_prog  $\Rightarrow$  cname  $\Rightarrow$  nat  $\Rightarrow$  ty  $\Rightarrow$  bool  $\Rightarrow$  exception_table  $\Rightarrow$  instr list
 $\Rightarrow$  method_type  $\Rightarrow$  state list"
```

```
"mk_cert G C mxs rT ini et bs phi  $\equiv$  make_cert (exec G C mxs rT ini et bs) (map OK
phi) (OK {})"
```

```
prg_cert :: "jvm_prog  $\Rightarrow$  prog_type  $\Rightarrow$  prog_cert"
```

```
"prg_cert G phi C sig  $\equiv$  let (C,rT,(mxs,maxl,ins,et)) = the (method (G,C) sig) in
  mk_cert G C mxs rT (fst sig=init) et ins (phi C sig)"
```

```
lemma check_certD:
```

```
"check_cert G mxs mxr n cert  $\implies$  cert_ok cert n Err (OK {}) (states G mxs mxr n)"
by (unfold cert_ok_def check_cert_def check_types_def) auto
```

```
lemma (in start_context) first_in_A: "OK {first}  $\in$  A"
```

```
apply (insert pTs C)
```

```
apply (simp add: states_def address_types_def init_tys_def max_dim_def)
```

```
apply (auto intro!: list_appendI)
```

```
apply force+
```

```
done
```



```

lemma (in start_context) wt_lbv_wt_step:
  assumes lbv: "wt_lbv G C mn pTs rT mxs mxl et cert bs"
  defines [simp]: "f ≡ JVMType.sup"
  shows "∃ ts ∈ list (size bs) A. wt_step r Err step ts ∧ OK {first} <=_r ts!0"
proof -
  have "semilat (JVMType.sl G mxs mxr (size bs))" by (rule semilat_JVM)
  hence "semilat (A, r, f)" by (simp add: sl_def2)
  moreover
  have "top r Err" by (simp add: JVMType.le_def)
  moreover
  have "Err ∈ A" by (simp add: states_def)
  moreover
  have "bottom r (OK {})"
    by (simp add: JVMType.le_def bottom_def lesub_def Err.le_def split: err.split)
  moreover
  have "OK {} ∈ A" by (simp add: states_def)
  moreover
  have "bounded step (length bs) A" by (simp add: bounded_exec)
  moreover
  from lbv
  have "cert_ok cert (length bs) Err (OK {}) A"
    by (unfold wt_lbv_def) (auto dest: check_certD)
  moreover
  from wf
  have "pres_type step (length bs) A" by simp (rule exec_pres_type)
  moreover
  from lbv
  have "wtl_inst_list bs cert f r Err (OK {}) step 0 (OK {first}) ≠ Err"
    by (simp add: wt_lbv_def lbvjvm_def)
  moreover
  note first_in_A
  moreover
  from lbv have "0 < length bs" by (simp add: wt_lbv_def)
  ultimately
  show ?thesis by (rule lbvs.wtl_sound_strong)
qed

```

```

lemma in_list:
  "(xs ∈ list n A) = (length xs = n ∧ set xs ⊆ A)"
  by (unfold list_def) auto

```

```

lemma (in start_context) wt_lbv_wt_method:
  assumes lbv: "wt_lbv G C mn pTs rT mxs mxl et cert bs"

  shows "∃ phi. wt_method G C mn pTs rT mxs mxl bs et phi"
proof -
  from lbv have l: "bs ≠ []" by (simp add: wt_lbv_def)
  moreover
  from wf lbv C pTs
  obtain phi where
    list: "phi ∈ list (length bs) A" and
    step: "wt_step r Err step phi" and
    start: "OK {first} <=_r phi!0"
  by (blast dest: wt_lbv_wt_step)

```

```

from list have [simp]: "length phi = length bs" by simp
have "length (map ok_val phi) = length bs" by simp
moreover
from 1 have 0: "0 < length phi" by simp
with step obtain phi0 where "phi!0 = OK phi0"
  by (unfold wt_step_def) blast
with start 0
have "wt_start G C mn pTs mxl (map ok_val phi)"
  by (simp add: wt_start_def JVMType.le_def lesub_def map_compose Err.le_def)
moreover {
  from list
  have "check_types G mxs mxr (length bs) phi"
    by (simp add: check_types_def)
  also from step
  have [symmetric]: "map OK (map ok_val phi) = phi"
    by (auto intro!: map_id simp add: wt_step_def)
  finally have "check_types G mxs mxr (length bs) (map OK (map ok_val phi))" .
}
moreover {
  have "bounded (err_step (length phi) app eff) (length bs) A"
    by (simp, fold exec_def) (rule bounded_exec)
  moreover
  from list have "set phi  $\subseteq$  A" by simp
  moreover
  from step
  have "wt_err_step (op  $\subseteq$ ) step phi"
    by (simp add: wt_err_step_def JVMType.le_def)
  ultimately
  have "wt_app_eff (op  $\subseteq$ ) app eff (map ok_val phi)"
    by (auto intro: wt_err_imp_wt_app_eff simp add: exec_def states_def)
}
ultimately
have "wt_method G C mn pTs rT mxs mxl bs et (map ok_val phi)"
  by (simp add: wt_method_def2)
thus ?thesis ..
qed

```

```

lemma (in start_context) wt_method_wt_lbv:
  assumes wt: "wt_method G C mn pTs rT mxs mxl bs et phi"

  defines [simp]: "cert  $\equiv$  mk_cert G C mxs rT (mn=init) et bs phi"
  defines [simp]: "f  $\equiv$  JVMType.sup"

  shows "wt_lbv G C mn pTs rT mxs mxl et cert bs"
proof -
  let ?phi = "map OK phi"
  let ?cert = "make_cert step ?phi (OK {})"

  from wt obtain
    0: "0 < length bs" and
    length: "length bs = length ?phi" and
    ck_types: "check_types G mxs mxr (length bs) ?phi" and
    wt_start: "wt_start G C mn pTs mxl phi" and

```

```

app_eff:    "wt_app_eff (op  $\subseteq$ ) app_eff phi"
by (simp add: wt_method_def2)

have "semilat (JVMTy.pe.sl G mxs mxr (size bs))" by (rule semilat_JVM)
hence "semilat (A, r, f)" by (simp add: sl_def2)
moreover
have "top r Err" by (simp add: JVMTy.pe.le_def)
moreover
have "Err  $\in$  A" by (simp add: states_def)
moreover
have "bottom r (OK {})"
  by (simp add: JVMTy.pe.le_def bottom_def Err.le_def lesub_def split: err.splits)
moreover
have "OK {}  $\in$  A" by (simp add: states_def)
moreover
have bounded: "bounded step (length bs) A" by (simp add: bounded_exec)
with wf
have "mono r step (length bs) A" by simp (rule exec_mono)
hence "mono r step (length ?phi) A" by (simp add: length)
moreover
from wf have "pres_type step (length bs) A" by simp (rule exec_pres_type)
hence "pres_type step (length ?phi) A" by (simp add: length)
moreover
from ck_types
have phi_in_A: "set ?phi  $\subseteq$  A" by (simp add: check_types_def)
hence " $\forall pc. pc < \text{length } ?phi \longrightarrow ?phi!pc \in A \wedge ?phi!pc \neq \text{Err}$ " by auto
moreover
from bounded
have "bounded step (length ?phi) A" by (simp add: length)
moreover
have "OK {}  $\neq$  Err" by simp
moreover
from bounded length phi_in_A app_eff
have "wt_err_step (op  $\subseteq$ ) step ?phi"
  by (auto intro: wt_app_eff_imp_wt_err simp add: exec_def states_def)
hence "wt_step r Err step ?phi"
  by (simp add: wt_err_step_def JVMTy.pe.le_def)
moreover
from 0 length have "0 < length phi" by auto
hence "?phi!0 = OK (phi!0)" by simp
with wt_start have "OK {first}  $\leq_r$  ?phi!0"
  by (clarsimp simp add: wt_start_def lesub_def Err.le_def JVMTy.pe.le_def map_compose)

moreover
note first_in_A
moreover
have "OK {first}  $\neq$  Err" by simp
moreover
note length
ultimately
have "wtl_inst_list bs ?cert f r Err (OK {first}) step 0 (OK {first})  $\neq$  Err"
  by (rule lbvc.wtl_complete)
moreover

```

```

from 0 length have "phi ≠ []" by auto
moreover
from ck_types
have "check_types G mxs mxr (length bs) ?cert"
  by (force simp add: make_cert_def check_types_def states_def)
moreover
note 0 length
ultimately
show ?thesis by (simp add: wt_lbv_def lbvjvm_def mk_cert_def
                      check_cert_def make_cert_def nth_append)
qed

theorem jvm_lbv_correct:
  "wt_jvm_prog_lbv G Cert  $\implies$   $\exists$ Phi. wt_jvm_prog G Phi"
proof -
  let ?Phi = " $\lambda$ C sig. let (C,rT,(maxs,maxl,bs,et)) = the (method (G,C) sig) in
    SOME phi. wt_method G C (fst sig) (snd sig) rT maxs maxl bs et phi"

  assume "wt_jvm_prog_lbv G Cert"
  hence "wt_jvm_prog G ?Phi"
    apply (unfold wt_jvm_prog_def wt_jvm_prog_lbv_def)
    apply (erule jvm_prog_lift)
    apply (auto dest!: start_context.wt_lbv_wt_method intro: someI)
    done
  thus ?thesis by blast
qed

theorem jvm_lbv_complete:
  "wt_jvm_prog G Phi  $\implies$  wt_jvm_prog_lbv G (prg_cert G Phi)"
  apply (unfold wt_jvm_prog_def wt_jvm_prog_lbv_def)
  apply (erule jvm_prog_lift)
  apply (auto simp add: prg_cert_def intro: start_context.wt_method_wt_lbv)
  done

end

```

## 4.21 BV Type Safety Invariant

**theory** *Correct* = *BVSpec* + *JVMExec*:

The fields of all objects contain only fully initialized objects:

**constdefs**

```

l_init :: "'c prog ⇒ aheap ⇒ init_heap ⇒ ('a ~> val) ⇒ ('a ~> ty) ⇒ bool"
"l_init G hp ih vs Ts ≡
∀n T. Ts n = Some T ⟶ (∃v. vs n = Some v ∧ is_init hp ih v)"

o_init :: "'c prog ⇒ aheap ⇒ init_heap ⇒ heap_entry ⇒ bool"
"o_init G hp ih entry ≡ case entry of Obj C fs ⇒ l_init G hp ih fs (map_of (fields
(G,C)))
| Arr T l en ⇒ l_init G hp ih en (λx. if x < l
then Some T else None)"

```

```

h_init :: "'c prog ⇒ aheap ⇒ init_heap ⇒ bool"
"h_init G h ih ≡ ∀a obj. h a = Some obj ⟶ o_init G h ih obj"

```

Type and init information conforms. For uninitialized objects dynamic+static type must be the same

**constdefs**

```

iconf :: "'c prog ⇒ aheap ⇒ init_heap ⇒ val ⇒ init_ty ⇒ bool"
("_,_,_ ⊢ _ :: ≤i _" [51,51,51,51,51] 50)
"G,h,ih ⊢ v :: ≤i T ==
case T of Init ty ⇒ G,h⊢v::≤ty ∧ is_init h ih v
| UnInit C pc ⇒ G,h⊢v::≤(Class C) ∧
(∃l fs. v = Addr l ∧ h l = Some (Obj C fs) ∧ ih l = T)
| PartInit C ⇒ G,h⊢v::≤(Class C) ∧
(∃l D fs. v = Addr l ∧ h l = Some (Obj D fs) ∧ ih l = T)"

```

Alias analysis for uninitialized objects is correct. If two values have the same uninitialized type, they must be equal and marked with this type on the type tag heap.

**constdefs**

```

corr_val :: "[val, init_ty, init_heap] ⇒ bool"
"corr_val v T ihp == ∃l. v = Addr l ∧ ihp l = T"

corr_loc :: "[val list, locvars_type, init_heap, val, init_ty] ⇒ bool"
"corr_loc loc LT ihp v T ==
list_all2 (λl t. t = OK T ⟶ l = v ∧ corr_val v T ihp) loc LT"

corr_stk :: "[opstack, opstack_type, init_heap, val, init_ty] ⇒ bool"
"corr_stk stk ST ihp v T == corr_loc stk (map OK ST) ihp v T"

corresponds :: "[opstack, val list, state_type, init_heap, val, init_ty] ⇒ bool"
"corresponds stk loc s ihp v T ==
corr_stk stk (fst s) ihp v T ∧ corr_loc loc (snd s) ihp v T"

```

```

consistent_init :: "[opstack, val list, state_type, init_heap] ⇒ bool"
"consistent_init stk loc s ihp ==
(∀ C pc. ∃ v. corresponds stk loc s ihp v (UnInit C pc)) ∧
(∀ C. ∃ v. corresponds stk loc s ihp v (PartInit C))"

```

The values on stack and local variables conform to their static type:

#### constdefs

```

approx_val :: "[jvm_prog, aheap, init_heap, val, init_ty err] ⇒ bool"
"approx_val G h i v any == case any of Err ⇒ True | OK T ⇒ G, h, i ⊢ v :: ⊆i T"

approx_loc :: "[jvm_prog, aheap, init_heap, val list, locvars_type] ⇒ bool"
"approx_loc G hp i loc LT == list_all2 (approx_val G hp i) loc LT"

approx_stk :: "[jvm_prog, aheap, init_heap, opstack, opstack_type] ⇒ bool"
"approx_stk G hp i stk ST == approx_loc G hp i stk (map OK ST)"

```

A call frame is correct, if stack and local variables conform, if the types UnInit and PartInit are used consistently, if the **this** pointer in constructors is tagged correctly and its type is only used for the **this** pointer, if the *pc* is inside the method, and if the number of local variables is correct.

#### constdefs

```

correct_frame :: "[jvm_prog, aheap, init_heap, state_type, nat, bytecode]
⇒ frame ⇒ bool"
"correct_frame G hp i == λ(ST, LT) maxl ins (stk, loc, C, sig, pc, r).
approx_stk G hp i stk ST ∧
approx_loc G hp i loc LT ∧
consistent_init stk loc (ST, LT) i ∧
(fst sig = init →
corresponds stk loc (ST, LT) i (fst r) (PartInit C) ∧
(∃ l. fst r = Addr l ∧ hp l ≠ None ∧
(i l = PartInit C ∨ (∃ C'. i l = Init (Class C'))))) ∧
pc < length ins ∧
length loc = length(snd sig) + maxl + 1"

```

The reference update between constructors must be correct. In this predicate,

- a** is the **this** pointer of the calling constructor (the reference to be initialized), it must be of type UnInit or PartInit.
- b** is the **this** pointer of the current constructor, it must be partly initialized up to the current class.
- c** is the fully initialized object it can be null (before super is called), or must be a fully initialized object with type of the original (calling) class (z from the BV is true iff c contains an object)
- C** is the class where the current constructor is declared

#### constdefs

```

constructor_ok :: "[jvm_prog, aheap, init_heap, val, cname, bool, ref_upd] ⇒ bool"

```

```

"constructor_ok G hp ih a C z ==  $\lambda(b,c).$ 
z = (c  $\neq$  Null)  $\wedge$  ( $\exists C' D pc l l' fs1 fs2.$  a = Addr l  $\wedge$  b = Addr l'  $\wedge$ 
(ih l = UnInit C' pc  $\vee$  ih l = PartInit D)  $\wedge$  hp l = Some (Obj C' fs1)  $\wedge$ 
(ih l' = PartInit C  $\vee$  ih l' = Init (Class C'))  $\wedge$  hp l' = Some (Obj C' fs2)  $\wedge$ 
(c  $\neq$  Null  $\longrightarrow$ 
( $\exists loc. c = Addr loc \wedge (\exists fs3. hp loc = Some (Obj C' fs3)) \wedge ih loc = Init (Class C'))))$ "

```

The whole call frame stack must be correct (the topmost frame is handled separately)

**consts**

```

correct_frames :: "[jvm_prog, aheap, init_heap, prog_type, ty, sig,
                  bool, ref_upd, frame list]  $\Rightarrow$  bool"

```

**primrec**

```

"correct_frames G hp i phi rT0 sig0 z r [] = True"

```

```

"correct_frames G hp i phi rT0 sig0 z0 r0 (f#frs) =
  (let (stk, loc, C, sig, pc, r) = f in
( $\exists ST LT z rT maxs maxl ins et.$ 
  ((ST, LT), z)  $\in$  phi C sig ! pc  $\wedge$  is_class G C  $\wedge$ 
  method (G, C) sig = Some(C, rT, (maxs, maxl, ins, et))  $\wedge$ 
  ( $\exists C' mn pTs. (ins!pc = Invoke C' mn pTs \vee ins!pc = Invoke\_special C' mn pTs)$ 

```

$\wedge$

```

  (mn, pTs) = sig0  $\wedge$ 
  ( $\exists apTs T ST'.$ 
  ST = (rev apTs) @ T # ST'  $\wedge$ 
  length apTs = length pTs  $\wedge$ 
  ( $\exists D' rT' body'. method (G, C') sig0 = Some(D', rT', body') \wedge G \vdash rT0 \preceq rT') \wedge$ 
  (mn = init  $\longrightarrow$  constructor_ok G hp i (stk!length apTs) C' z0 r0)  $\wedge$ 
  correct_frame G hp i (ST, LT) maxl ins f  $\wedge$ 
  correct_frames G hp i phi rT sig z r frs))))"

```

Invariant for the whole program state:

**constdefs**

```

correct_state :: "[jvm_prog, prog_type, jvm_state]  $\Rightarrow$  bool"
  ("_,_ |-JVM _ [ok]" [51,51] 50)

```

```

"correct_state G phi ==  $\lambda(xp, hp, ihp, frs).$ 

```

```

  case xp of
  None  $\Rightarrow$  (case frs of
    []  $\Rightarrow$  True
  | (f#fs)  $\Rightarrow$  G $\vdash$ h hp $\surd$   $\wedge$  h_init G hp ihp  $\wedge$  preallocated hp ihp  $\wedge$ 
    (let (stk, loc, C, sig, pc, r) = f
      in  $\exists rT maxs maxl ins et s z.$ 
      is_class G C  $\wedge$ 
      method (G, C) sig = Some(C, rT, (maxs, maxl, ins, et))  $\wedge$ 
      (s, z)  $\in$  phi C sig ! pc  $\wedge$ 
      correct_frame G hp ihp s maxl ins f  $\wedge$ 
      correct_frames G hp ihp phi rT sig z r fs))
  | Some x  $\Rightarrow$  frs = [])"

```

**syntax** (xsymbols)

```
correct_state :: "[jvm_prog,prog_type,jvm_state] => bool"
               ("_,_ ⊢JVM _ √" [51,51] 50)
```

```
lemma constructor_ok_field_update:
```

```
"[[ constructor_ok G hp ihp x C' z r; hp a = Some (Obj C od) ]]
=> constructor_ok G (hp(a↦Obj C od')) ihp x C' z r"
apply (cases r)
apply (unfold constructor_ok_def)
apply (cases "∃y. x = Addr y")
  defer
  apply simp
apply clarify
apply simp
apply (rule conjI)
  apply clarsimp
  apply blast
apply (rule impI)
apply (rule conjI)
  apply clarsimp
  apply blast
apply (rule impI)
apply (rule conjI)
  apply blast
apply (rule impI, erule impE, assumption)
apply (elim exE conjE)
apply simp
apply (rule exI)
apply (rule conjI)
  defer
  apply (rule impI)
  apply (rule conjI)
    apply (rule refl)
  apply blast
apply (rule impI)
apply simp
done
```

```
lemma constructor_ok_array_update:
```

```
"[[ constructor_ok G hp ihp x C' z r; hp a = Some (Arr T l en) ]]
=> constructor_ok G (hp(a↦Arr T l en')) ihp x C' z r"
apply (cases r)
apply (unfold constructor_ok_def)
apply (cases "∃y. x = Addr y")
  defer
  apply simp
apply clarify
apply simp
apply (rule conjI)
  apply clarsimp
  apply (rule impI)
  apply (rule conjI)
    apply clarsimp
  apply (rule impI)
```



```

apply (rule conjI)
  apply blast
apply (rule impI, erule impE, assumption)
apply (elim exE conjE)
apply simp
apply (rule exI)
apply (rule conjI)
  defer
  apply (rule impI)
  apply (rule conjI)
    apply (rule refl)
  apply blast
apply auto
done

lemma constructor_ok_newref:
  "[ hp x = None; constructor_ok G hp ihp v C' z r ]
  ⇒ constructor_ok G (hp(x↦obj)) (ihp(x := T)) v C' z r"
apply (cases r)
apply (unfold constructor_ok_def)
apply clarify
apply (rule conjI)
  apply (rule refl)
apply clarsimp
apply (rule conjI)
  apply clarsimp
apply (rule impI)
apply (rule conjI)
  apply clarsimp
apply (rule impI)
apply (rule conjI)
  apply blast
apply (rule impI, erule impE, assumption)
apply (elim exE conjE)
apply (intro exI)
apply (rule conjI)
  defer
  apply (rule impI)
  apply (rule conjI, assumption)
  apply blast
apply clarsimp
done

lemma constructor_ok_pass_val:
  "[ constructor_ok G hp ihp x C' True r;
    constructor_ok G hp ihp v C z0 r0; x = fst r0 ]
  ⇒ constructor_ok G hp ihp v C True (x, snd r)"
apply (cases r, cases r0)
apply (unfold constructor_ok_def)
apply simp
apply (elim conjE exE)

```

```

apply simp
apply (erule disjE)
  apply clarsimp
apply clarsimp
done

```

```

lemma is_init_default_val:
  "is_init hp ihp (default_val T)"
  apply (simp add: is_init_def)
  apply (cases T)
    apply (case_tac prim_ty)
  apply auto
done

```

#### 4.21.1 approx-val

```

lemma iconf_widen:
  "G, hp, ihp ⊢ xcp :: ≤i T ⇒ G ⊢ T ≤i T' ⇒ wf_prog wf_mb G
  ⇒ G, hp, ihp ⊢ xcp :: ≤i T'"
  apply (cases T')
  apply (auto simp add: init_le_Init2 iconf_def elim: conf_widen)
done

```

```

lemma is_init [elim!]:
  "G, hp, ihp ⊢ v :: ≤i Init T ⇒ is_init hp ihp v"
  by (simp add: iconf_def)

```

```

lemma approx_val_Err:
  "approx_val G hp ihp x Err"
  by (simp add: approx_val_def)

```

```

lemma approx_val_Null:
  "approx_val G hp ihp Null (OK (Init (RefT x)))"
  by (auto simp add: approx_val_def iconf_def is_init_def)

```

```

lemma approx_val_widen:
  "wf_prog wt G ⇒ approx_val G hp ihp v (OK T) ⇒ G ⊢ T ≤i T'
  ⇒ approx_val G hp ihp v (OK T'"
  by (unfold approx_val_def) (auto intro: iconf_widen)

```

```

lemma conf_notNone:
  "G, hp ⊢ Addr loc :: ≤ ty ⇒ hp loc ≠ None"
  by (unfold conf_def) auto

```

```

lemma approx_val_imp_approx_val_sup_heap [rule_format]:
  "approx_val G hp ihp v at → hp ≤l hp' → approx_val G hp' ihp v at"
  apply (simp add: approx_val_def iconf_def is_init_def
    split: err.split init_ty.split)
  apply (fast dest: conf_notNone elim: hextE intro: conf_hext)
done

```

```

lemma approx_val_heap_update:
  "[ hp a = Some obj'; G, hp ⊢ v :: ≤T; obj_ty obj = obj_ty obj' ]
  ⇒ G, hp(a ↦ obj) ⊢ v :: ≤T"
  by (cases v, auto simp add: obj_ty_def conf_def)

```

#### 4.21.2 approx-loc

```

lemma approx_loc_Cons [iff]:
  "approx_loc G hp ihp (s#xs) (l#ls) =
  (approx_val G hp ihp s l ∧ approx_loc G hp ihp xs ls)"
  by (simp add: approx_loc_def)

```

```

lemma approx_loc_Nil [simp, intro!]:
  "approx_loc G hp ihp [] []"
  by (simp add: approx_loc_def)

```

```

lemma approx_loc_len [elim]:
  "approx_loc G hp ihp loc LT ⇒ length loc = length LT"
  by (unfold approx_loc_def list_all2_def) simp

```

```

lemma approx_loc_replace_Err:
  "approx_loc G hp ihp loc LT ⇒ approx_loc G hp ihp loc (replace v Err LT)"
  by (clarsimp simp add: approx_loc_def list_all2_conv_all_nth replace_def
    approx_val_Err)

```

```

lemma assConv_approx_stk_imp_approx_loc [rule_format]:
  "wf_prog wt G ⇒ (∀ (t, t') ∈ set (zip tys_n ts). G ⊢ t ≤i t')
  → length tys_n = length ts → approx_stk G hp ihp s tys_n →
  approx_loc G hp ihp s (map OK ts)"
  apply (unfold approx_stk_def approx_loc_def list_all2_def)
  apply (clarsimp simp add: all_set_conv_all_nth)
  apply (rule approx_val_widen)
  apply auto
  done

```

```

lemma approx_loc_imp_approx_loc_sup_heap:
  "[ approx_loc G hp ihp lvars lt; hp ≤l hp' ]
  ⇒ approx_loc G hp' ihp lvars lt"
  apply (unfold approx_loc_def list_all2_def)
  apply (auto intro: approx_val_imp_approx_val_sup_heap)
  done

```

```

lemma approx_val_newref:
  "h loc = None ⇒
  approx_val G (h(loc ↦ (Obj C fs))) (ih(loc := UnInit C pc)) (Addr loc) (OK (UnInit C pc))"
  by (auto simp add: approx_val_def iconf_def is_init_def intro: conf_obj_AddrI)

```

```

lemma approx_val_newref_false:
  "[ h l = None; approx_val G h ih (Addr l) (OK T) ] ⇒ False"
  by (simp add: approx_val_def iconf_def conf_def
    split: init_ty.split_asm ty.split_asm)

```

```

lemma approx_val_imp_approx_val_newref:
  "[[ approx_val G hp ihp v T; hp loc = None ]]"
  => approx_val G hp (ihp(loc:=X)) v T"
  apply (cases "v = Addr loc")
  apply (auto simp add: approx_val_def iconf_def is_init_def conf_def
    split: err.split init_ty.split val.split)
  apply (erule allE, erule disjE)
  apply auto
  done

lemma approx_loc_newref:
  "[[ approx_loc G hp ihp lvars lt; hp loc = None ]]"
  => approx_loc G hp (ihp(loc:=X)) lvars lt"
  apply (unfold approx_loc_def list_all2_def)
  apply (auto intro: approx_val_imp_approx_val_newref)
  done

lemma approx_loc_newref_Err:
  "[[ approx_loc G hp ihp loc LT; hp l = None; i < length loc; loc!i=Addr l ]]"
  => LT!i = Err"
  apply (cases "LT!i", simp)
  apply (clarsimp simp add: approx_loc_def list_all2_conv_all_nth)
  apply (erule allE, erule impE, assumption)
  apply (erule approx_val_newref_false)
  apply simp
  done

lemma approx_loc_newref_all_Err:
  "[[ approx_loc G hp ihp loc LT; hp l = None ]]"
  => list_all2 ( $\lambda v T. v = \text{Addr } l \longrightarrow T = \text{Err}$ ) loc LT"
  apply (simp only: list_all2_conv_all_nth)
  apply (rule conjI)
  apply (simp add: approx_loc_def list_all2_def)
  apply clarify
  apply (rule approx_loc_newref_Err, assumption+)
  done

lemma approx_loc_imp_approx_loc_subst [rule_format]:
  " $\forall \text{loc idx } x X. (\text{approx\_loc } G \text{ hp ihp loc } LT) \longrightarrow (\text{approx\_val } G \text{ hp ihp } x X) \longrightarrow (\text{approx\_loc } G \text{ hp ihp } (\text{loc}[\text{idx}:=x]) (LT[\text{idx}:=X]))$ "
  apply (unfold approx_loc_def list_all2_def)
  apply (auto dest: subsetD [OF set_update_subset_insert] simp add: zip_update)
  done

lemma approx_loc_Err [iff]:
  "approx_loc G hp ihp (replicate n v) (replicate n Err)"
  by (induct n) (auto simp add: approx_val_def)

```

lemmas [cong] = conj\_cong

```
lemma approx_loc_append [rule_format]:
  "∀ L1 L2. length l1=length L1 →
  approx_loc G hp ihp (l1@L2) (L1@L2) =
  (approx_loc G hp ihp l1 L1 ∧ approx_loc G hp ihp l2 L2)"
  apply (unfold approx_loc_def list_all2_def)
  apply simp
  apply blast
  done
```

lemmas [cong del] = conj\_cong

```
lemma approx_val_Err_or_same:
  "[ approx_val G hp ihp v (OK X); X = UnInit C pc ∨ X = PartInit D;
  approx_val G hp ihp v X' ]
  ⇒ X' = Err ∨ X' = OK X"
  apply (simp add: approx_val_def split: err.split_asm)
  apply (erule disjE)
  apply (clarsimp simp add: iconf_def is_init_def split: init_ty.split_asm)
  apply (clarsimp simp add: iconf_def is_init_def split: init_ty.split_asm)
  done
```

```
lemma approx_loc_replace:
  "[ approx_loc G hp ihp loc LT; approx_val G hp ihp x' (OK X');
  approx_val G hp ihp x (OK X);
  X = UnInit C pc ∨ X = PartInit D; corr_loc loc LT ihp x X ]
  ⇒ approx_loc G hp ihp (replace x x' loc) (replace (OK X) (OK X') LT)"
  apply (simp add: approx_loc_def corr_loc_def replace_def list_all2_conv_all_nth)
  apply clarsimp
  apply (erule allE, erule impE, assumption)+
  apply simp
  apply (drule approx_val_Err_or_same, assumption+)
  apply (simp add: approx_val_Err)
  done
```

### 4.21.3 approx-stk

```
lemma list_all2_approx:
  "∧ s. list_all2 (approx_val G hp ihp) s (map OK S) =
  list_all2 (iconf G hp ihp) s S"
  apply (induct S)
  apply (auto simp add: list_all2_Cons2 approx_val_def)
  done
```

```
lemma list_all2_iconf_widen:
  "wf_prog mb G ⇒
  list_all2 (iconf G hp ihp) a b ⇒
  list_all2 (λx y. G ⊢ x ≼i Init y) b c ⇒
  list_all2 (λv T. G, hp ⊢ v :: ≼ T ∧ is_init hp ihp v) a c"
  apply (rule list_all2_trans)
  defer
  apply assumption
  apply assumption
```

```

    apply (drule iconf_widen, assumption+)
    apply (simp add: iconf_def)
  done

lemma approx_stk_rev_lem:
  "approx_stk G hp ihp (rev s) (rev t) = approx_stk G hp ihp s t"
  apply (unfold approx_stk_def approx_loc_def list_all2_def)
  apply (auto simp add: zip_rev sym [OF rev_map])
  done

lemma approx_stk_rev:
  "approx_stk G hp ihp (rev s) t = approx_stk G hp ihp s (rev t)"
  by (auto intro: subst [OF approx_stk_rev_lem])

lemma approx_stk_imp_approx_stk_sup_heap [rule_format]:
  "∀ lvars. approx_stk G hp ihp lvars lt → hp ≤| hp'
  → approx_stk G hp' ihp lvars lt"
  by (auto intro: approx_loc_imp_approx_loc_sup_heap simp add: approx_stk_def)

lemma approx_stk_Nil [iff]:
  "approx_stk G hp ihp [] []"
  by (simp add: approx_stk_def approx_loc_def)

lemma approx_stk_Cons [iff]:
  "approx_stk G hp ihp (x # stk) (S#ST) =
  (approx_val G hp ihp x (OK S) ∧ approx_stk G hp ihp stk ST)"
  by (simp add: approx_stk_def approx_loc_def)

lemma approx_stk_Cons_lemma [iff]:
  "approx_stk G hp ihp stk (S#ST') =
  (∃ s stk'. stk = s#stk' ∧ approx_val G hp ihp s (OK S) ∧
  approx_stk G hp ihp stk' ST'"
  by (simp add: list_all2_Cons2 approx_stk_def approx_loc_def)

lemma approx_stk_len [elim]:
  "approx_stk G hp ihp stk ST ⇒ length stk = length ST"
  by (unfold approx_stk_def) (simp add: approx_loc_len)

lemma approx_stk_append_lemma:
  "approx_stk G hp ihp stk (S@ST') ⇒
  (∃ s stk'. stk = s@stk' ∧ length s = length S ∧ length stk' = length ST' ∧
  approx_stk G hp ihp s S ∧ approx_stk G hp ihp stk' ST'"
  by (simp add: list_all2_append2 approx_stk_def approx_loc_def)

lemma approx_stk_newref:
  "[[ approx_stk G hp ihp lvars lt; hp loc = None ]
  ⇒ approx_stk G hp (ihp(loc:=X)) lvars lt"
  by (auto simp add: approx_stk_def intro: approx_loc_newref)

lemma newref_notin_stk:
  "[[ approx_stk G hp ihp stk ST; hp l = None ]
  ⇒ Addr l ∉ set stk"

```

proof

```

assume "Addr l ∈ set stk"
then obtain a b where
  stk: "stk = a @ (Addr l) # b"
  by (clarsimp simp add: in_set_conv_decomp)
hence "stk!(length a) = Addr l" by (simp add: nth_append)
with stk obtain i where
  "stk!i = Addr l" and i: "i < length stk" by clarsimp
moreover
assume "approx_stk G hp ihp stk ST"
hence l: "approx_loc G hp ihp stk (map OK ST)" by (unfold approx_stk_def)
moreover
assume "hp l = None"
moreover
from l
have "length stk = length ST" by (simp add: approx_loc_def list_all2_def)
with i
have "map OK ST ! i ≠ Err" by simp
ultimately
show False by (auto dest: approx_loc_newref_Err)
qed

```

#### 4.21.4 corresponds

```

lemma corr_loc_empty [simp]:
  "corr_loc [] [] ihp v T"
  by (simp add: corr_loc_def)

```

```

lemma corr_loc_cons:
  "corr_loc (s#loc) (L#LT) ihp v T =
  ((L = OK T → s = v ∧ corr_val v T ihp) ∧ corr_loc loc LT ihp v T)"
  by (simp add: corr_loc_def)

```

```

lemma corr_loc_start:
  "∧loc.
  [∀x ∈ set LT. x = Err ∨ (∃t. x = OK (Init t));
  length loc = length LT; T = UnInit C pc ∨ T = PartInit C']
  ⇒ corr_loc loc LT ihp v T" (is "PROP ?P LT")

```

proof (induct LT)

```
show "PROP ?P []" by (simp add: corr_loc_def)
```

```
fix L LS assume IH: "PROP ?P LS"
```

```
show "PROP ?P (L#LS)"
```

proof -

```
fix loc: "val list"
```

```
assume "length loc = length (L # LS)"
```

```
then obtain l ls where
```

```
loc: "loc = l#ls" and len: "length ls = length LS"
```

```
by (auto simp add: length_Suc_conv)
```

```
assume "∀x∈set (L#LS). x = Err ∨ (∃t. x = OK (Init t))"
```

```
then obtain
```

```
first: "L = Err ∨ (∃t. L = OK (Init t))" and
```

```
rest: "∀x∈set LS. x = Err ∨ (∃t. x = OK (Init t))"
```

```
by auto
```

```
assume T: "T = UnInit C pc ∨ T = PartInit C'"
```

```

with rest len IH
have "corr_loc ls LS ihp v T" by blast
with T first
have "corr_loc (l#ls) (L # LS) ihp v T"
  by (clarsimp simp add: corr_loc_def)
with loc
show "corr_loc loc (L # LS) ihp v T" by simp
qed
qed

lemma consistent_init_start:
  "[[ LTO = OK (Init (Class C)) # map OK (map Init pTs) @ replicate mxl' Err;
    loc = (oX # rev opTs @ replicate mxl' arbitrary);
    length pTs = length opTs ]]"
  => consistent_init [] loc ([], LTO) ihp"
  (is "[[ PROP ?LTO; PROP ?loc; PROP _ ]]" => PROP _)
proof -
  assume LTO: "PROP ?LTO" and "PROP ?loc" "length pTs = length opTs"
  hence len: "length loc = length LTO" by simp
  from LTO have no_UnInit: "∀x ∈ set LTO. x = Err ∨ (∃t. x = OK (Init t))"
    by (auto dest: in_set_replicated)
  with len show ?thesis
    by (simp add: consistent_init_def corresponds_def corr_stk_def)
      (blast intro: corr_loc_start)
qed

lemma corresponds_stk_cons:
  "corresponds (s#stk) loc (S#ST,LT) ihp v T =
  ((S = T → s = v ∧ corr_val v T ihp) ∧ corresponds stk loc (ST,LT) ihp v T)"
  by (simp add: corresponds_def corr_stk_def corr_loc_def)

lemma corresponds_loc_nth:
  "[[ corresponds stk loc (ST,LT) ihp v T; n < length LT; LT!n = OK X ]]"
  X = T → (loc!n) = v ∧ corr_val v T ihp"
  by (simp add: corresponds_def corr_loc_def list_all2_conv_all_nth)

lemma consistent_init_loc_nth:
  "[[ consistent_init stk loc (ST,LT) ihp; n < length LT; LT!n = OK X ]]"
  => consistent_init (loc!n#stk) loc (X#ST,LT) ihp"
  apply (simp add: consistent_init_def corresponds_stk_cons)
  apply (intro strip conjI)
  apply (elim alle exE conjE)
  apply (intro exI conjI corresponds_loc_nth)
  apply assumption+
  apply (elim alle exE conjE)
  apply (intro exI conjI corresponds_loc_nth)
  apply assumption+
  done

lemma consistent_init_corresponds_stk_cons:
  "[[ consistent_init (s#stk) loc (S#ST,LT) ihp;
    S = UnInit C pc ∨ S = PartInit C' ]]"
  => corresponds (s#stk) loc (S#ST, LT) ihp s S"
  by (simp add: consistent_init_def corresponds_stk_cons) blast

```



```

lemma consistent_init_corresponds_loc:
  "[ consistent_init stk loc (ST,LT) ihp; LT!n = OK T;
    T = UnInit C pc ∨ T = PartInit C'; n < length LT ]
  ⇒ corresponds stk loc (ST,LT) ihp (loc!n) T"
proof -
  assume "consistent_init stk loc (ST,LT) ihp"
    "T = UnInit C pc ∨ T = PartInit C'"
  then obtain v where
    corr: "corresponds stk loc (ST,LT) ihp v T"
    by (simp add: consistent_init_def) blast
  moreover
  assume "LT!n = OK T" "n < length LT"
  ultimately
  have "v = (loc!n)" by (simp add: corresponds_loc_nth)
  with corr
  show ?thesis by simp
qed

lemma consistent_init_pop:
  "consistent_init (s#stk) loc (S#ST,LT) ihp
  ⇒ consistent_init stk loc (ST,LT) ihp"
  by (simp add: consistent_init_def corresponds_stk_cons) fast

lemma consistent_init_Init_stk:
  "consistent_init stk loc (ST,LT) ihp ⇒
  consistent_init (s#stk) loc ((Init T')#ST,LT) ihp"
  by (simp add: consistent_init_def corresponds_def corr_stk_def corr_loc_def)

lemma corr_loc_set:
  "[ corr_loc loc LT ihp v T; OK T ∈ set LT ] ⇒ corr_val v T ihp"
  by (auto simp add: corr_loc_def in_set_conv_decomp list_all2_append2
    list_all2_Cons2)

lemma corresponds_var_upd_UnInit:
  "[ corresponds stk loc (ST,LT) ihp v T; n < length LT;
    OK T ∈ set (map OK ST) ∪ set LT ]
  ⇒ corresponds stk (loc[n:=v]) (ST,LT[n:=OK T]) ihp v T"
proof -
  assume "corresponds stk loc (ST,LT) ihp v T"
  then obtain
    stk: "corr_stk stk ST ihp v T" and
    loc: "corr_loc loc LT ihp v T"
    by (simp add: corresponds_def)
  from loc
  obtain
    len: "length loc = length LT" and
    "∀i. i < length loc → LT ! i = OK T → loc ! i = v ∧ corr_val v T ihp"
    (is "?all loc LT")
    by (simp add: corr_loc_def list_all2_conv_all_nth)
  moreover
  assume "OK T ∈ set (map OK ST) ∪ set LT"
  hence ihp: "corr_val v T ihp"
  proof

```

```

    assume "OK T ∈ set LT"
    with loc show ?thesis by (rule corr_loc_set)
  next
    assume "OK T ∈ set (map OK ST)"
    with stk show ?thesis by (unfold corr_stk_def) (rule corr_loc_set)
  qed
  moreover
  assume "n < length LT"
  ultimately
  have "?all (loc[n:= v]) (LT[n:= OK T])"
    by (simp add: nth_list_update)
  with len
  have "corr_loc (loc[n:= v]) (LT[n:= OK T]) ihp v T"
    by (simp add: corr_loc_def list_all2_conv_all_nth)
  with stk
  show ?thesis
    by (simp add: corresponds_def)
  qed

lemma corresponds_var_upd_UnInit2:
  "[[ corresponds stk loc (ST,LT) ihp v T; n < length LT; T' ≠ T ]]
  ⇒ corresponds stk (loc[n:= v']) (ST,LT[n:= OK T']) ihp v T"
  by (auto simp add: corresponds_def corr_loc_def
      list_all2_conv_all_nth nth_list_update)

lemma corresponds_var_upd:
  "[[ corresponds (s#stk) loc (S#ST, LT) ihp v T; idx < length LT ]]
  ⇒ corresponds stk (loc[idx := s]) (ST, LT[idx := OK S]) ihp v T"
  apply (cases "S = T")
  apply (drule corresponds_var_upd_UnInit, assumption)
  apply simp
  apply (simp only: corresponds_stk_cons)
  apply (drule corresponds_var_upd_UnInit2, assumption+)
  apply (simp only: corresponds_stk_cons)
  apply blast
  done

lemma consistent_init_conv:
  "consistent_init stk loc s ihp =
  (∀T. ((∃C pc. T = UnInit C pc) ∨ (∃C. T = PartInit C)) →
  (∃v. corresponds stk loc s ihp v T))"
  by (unfold consistent_init_def) blast

lemma consistent_init_store:
  "[[ consistent_init (s#stk) loc (S#ST,LT) ihp; n < length LT ]]
  ⇒ consistent_init stk (loc[n:= s]) (ST,LT[n:= OK S]) ihp"
  apply (simp only: consistent_init_conv)
  apply (blast intro: corresponds_var_upd)
  done

lemma corr_loc_newT:
  "[[ OK T ∉ set LT; length loc = length LT ]] ⇒ corr_loc loc LT ihp v T"
  by (auto dest: nth_mem simp add: corr_loc_def list_all2_conv_all_nth)

```

```

lemma corr_loc_new_val:
  "[ corr_loc loc LT ihp v T; Addr l ∉ set loc ]
  ⇒ corr_loc loc LT (ihp(l:= T')) v T"
proof -
  assume new: "Addr l ∉ set loc"
  assume "corr_loc loc LT ihp v T"
  then obtain
    len: "length loc = length LT" and
    all: "∀i. i < length loc → LT ! i = OK T → loc ! i = v ∧ corr_val v T ihp"
    by (simp add: corr_loc_def list_all2_conv_all_nth)
  show ?thesis
  proof (unfold corr_loc_def, simp only: list_all2_conv_all_nth,
        intro strip conjI)
    from len show "length loc = length LT" .
    fix i assume i: "i < length loc" and "LT!i = OK T"
    with all
    have l: "loc!i = v ∧ corr_val v T ihp" by blast
    thus "loc!i = v" by blast
    from i l new
    have "v ≠ Addr l" by (auto dest: nth_mem)
    with l
    show "corr_val v T (ihp(l:= T'))"
      by (auto simp add: corr_val_def split: init_ty.split)
  qed
qed

```

```

lemma corr_loc_Err_val:
  "[ corr_loc loc LT ihp v T; list_all2 (λv T. v = Addr l → T = Err) loc LT ]
  ⇒ corr_loc loc LT (ihp(l:= T')) v T"
  apply (simp add: corr_loc_def)
  apply (simp only: list_all2_conv_all_nth)
  apply clarify
  apply (simp (no_asm))
  apply clarify
  apply (erule_tac x = i in allE, erule impE)
  apply simp
  apply (erule impE, assumption)
  apply simp
  apply (unfold corr_val_def)
  apply (elim conjE exE)
  apply (rule_tac x = la in exI)
  apply clarsimp
  apply (erule allE, erule impE, assumption)
  apply (cases T)
  apply auto
  done

```

```

lemma corr_loc_len:
  "corr_loc loc LT ihp v T ⇒ length loc = length LT"
  by (simp add: corr_loc_def list_all2_conv_all_nth)

```

```

lemma corresponds_newT:
  "[ corresponds stk loc (ST,LT) ihp' v' T'; OK T ∉ set (map OK ST) ∪ set LT ]

```

```

⇒ corresponds stk loc (ST,LT) ihp v T"
apply (clarsimp simp add: corresponds_def corr_stk_def)
apply (rule conjI)
  apply (rule corr_loc_newT, simp)
  apply (simp add: corr_loc_len)
apply (rule corr_loc_newT, assumption)
apply (simp add: corr_loc_len)
done

```

```

lemma corresponds_new_val:
  "[[ corresponds stk loc (ST,LT) ihp v T; Addr l ∉ set stk;
    list_all2 (λv T. v = Addr l → T = Err) loc LT ]]"
  ⇒ corresponds stk loc (ST,LT) (ihp(l:= T')) v T"
  apply (simp add: corresponds_def corr_stk_def)
  apply (blast intro: corr_loc_new_val corr_loc_Err_val)
done

```

```

lemma corresponds_newref:
  "[[ corresponds stk loc (ST, LT) ihp v T;
    OK (UnInit C pc) ∉ set (map OK ST) ∪ set LT;
    Addr l ∉ set stk; list_all2 (λv T. v = Addr l → T = Err) loc LT ]]"
  ⇒ ∃v. corresponds (Addr l#stk) loc ((UnInit C pc)#ST,LT)
    (ihp(l:= UnInit C pc)) v T"
  apply (simp add: corresponds_stk_cons)
  apply (cases "T = UnInit C pc")
  apply simp
  apply (rule conjI)
  apply (unfold corr_val_def)
  apply simp
  apply (rule corresponds_newT, assumption)
  apply simp
  apply (blast intro: corresponds_new_val)
done

```

```

lemma consistent_init_new_val_lemma:
  "[[ consistent_init stk loc (ST,LT) ihp;
    Addr l ∉ set stk; list_all2 (λv T. v = Addr l → T = Err) loc LT ]]"
  ⇒ consistent_init stk loc (ST,LT) (ihp(l:= T'))"
  by (unfold consistent_init_def) (blast intro: corresponds_new_val)

```

```

lemma consistent_init_newref_lemma:
  "[[ consistent_init stk loc (ST,LT) ihp;
    OK (UnInit C pc) ∉ set (map OK ST) ∪ set LT;
    Addr l ∉ set stk; list_all2 (λv T. v = Addr l → T = Err) loc LT ]]"
  ⇒ consistent_init (Addr l#stk) loc ((UnInit C pc)#ST,LT) (ihp(l:= UnInit C pc))"
  by (unfold consistent_init_def) (blast intro: corresponds_newref)

```

```

lemma consistent_init_newref:
  "[[ consistent_init stk loc (ST,LT) ihp;
    approx_loc G hp ihp loc LT;
    hp l = None;
    approx_stk G hp ihp stk ST;
    OK (UnInit C pc) ∉ set (map OK ST) ∪ set LT ]]"

```

```

⇒ consistent_init (Addr l#stk) loc ((UnInit C pc)#ST,LT) (ihp(l:= UnInit C pc))"
apply (drule approx_loc_newref_all_Err, assumption)
apply (drule newref_notin_stk, assumption)
apply (rule consistent_init_newref_lemma)
apply assumption+
done

```

lemma corresponds\_new\_val2:

```

"[[corresponds stk loc (ST, LT) ihp v T; approx_loc G hp ihp loc LT; hp l = None;
  approx_stk G hp ihp stk ST]]
⇒ corresponds stk loc (ST, LT) (ihp(l := T')) v T"
apply (drule approx_loc_newref_all_Err, assumption)
apply (drule newref_notin_stk, assumption)
apply (rule corresponds_new_val)
apply assumption+
done

```

lemma consistent\_init\_new\_val:

```

"[[ consistent_init stk loc (ST,LT) ihp;
  approx_loc G hp ihp loc LT; hp l = None;
  approx_stk G hp ihp stk ST ]]
⇒ consistent_init stk loc (ST,LT) (ihp(l:= T'))"
apply (drule approx_loc_newref_all_Err, assumption)
apply (drule newref_notin_stk, assumption)
apply (rule consistent_init_new_val_lemma)
apply assumption+
done

```

lemma corr\_loc\_replace\_type:

```

"[[ corr_loc loc LT ihp v T; T ≠ (Init T'') ]]
⇒ corr_loc loc (replace (OK T') (OK (Init T'')) LT) ihp v T"
apply (unfold corr_loc_def replace_def)
apply (simp add: list_all2_conv_all_nth)
apply blast
done

```

lemma corr\_stk\_replace\_type:

```

"[[ corr_stk loc ST ihp v T; T ≠ (Init T'') ]]
⇒ corr_stk loc (replace T' (Init T'') ST) ihp v T"

```

proof -

```

assume "corr_stk loc ST ihp v T"
hence "corr_loc loc (map OK ST) ihp v T" by (unfold corr_stk_def)
moreover
assume "T ≠ (Init T'')"
ultimately
have "corr_loc loc (replace (OK T') (OK (Init T'')) (map OK ST)) ihp v T"
  by (rule corr_loc_replace_type)
moreover
have "replace (OK T') (OK (Init T'')) (map OK ST) =
  map OK (replace T' (Init T'') ST)"
  by (rule replace_map_OK)
ultimately

```

```

    have "corr_loc loc (map OK (replace T' (Init T'') ST)) ihp v T" by simp
    thus ?thesis by (unfold corr_stk_def)
qed

```

```

lemma consistent_init_replace_type:
  "[[ consistent_init stk loc (ST,LT) ihp ]]"
  ⇒ consistent_init stk loc
    (replace T (Init T') ST, replace (OK T) (OK (Init T'')) LT) ihp"
apply (unfold consistent_init_def corresponds_def)
apply simp
apply (blast intro: corr_stk_replace_type corr_loc_replace_type)
done

```

```

lemma corr_loc_replace:
  "∧loc.
  [[ corr_loc loc LT ihp v T; approx_loc G hp ihp loc LT;
    approx_val G hp ihp oX (OK TO); T ≠ T'; T = UnInit C pc ∨ T = PartInit D ]]"
  ⇒ corr_loc (replace oX x loc) (replace (OK TO) (OK T') LT) ihp v T"
apply (frule corr_loc_len)
apply (induct LT)
  apply (simp add: replace_def)
  apply (clarsimp simp add: length_Suc_conv)
  apply (clarsimp simp add: corr_loc_cons replace_Cons split del: split_if)
  apply (simp split: split_if_asm)
  apply clarify
  apply (drule approx_val_Err_or_same, assumption+)
  apply simp
done

```

```

lemma corr_stk_replace:
  "[[ corr_stk stk ST ihp v T; approx_stk G hp ihp stk ST;
    approx_val G hp ihp oX (OK TO); T ≠ T';
    T = UnInit C pc ∨ T = PartInit D ]]"
  ⇒ corr_stk (replace oX x stk) (replace TO T' ST) ihp v T"
apply (unfold corr_stk_def approx_stk_def)
apply (drule corr_loc_replace, assumption+)
apply (simp add: replace_map_OK)
done

```

```

lemma corresponds_replace:
  "[[ corresponds stk loc (ST,LT) ihp v T;
    approx_loc G hp ihp loc LT;
    approx_stk G hp ihp stk ST;
    approx_val G hp ihp oX (OK TO);
    T ≠ T'; T = UnInit C pc ∨ T = PartInit D ]]"
  ⇒ corresponds (replace oX x stk) (replace oX x loc)
    (replace TO T' ST, replace (OK TO) (OK T') LT) ihp v T"
apply (clarsimp simp add: corresponds_def)
apply (rule conjI)
  apply (rule corr_stk_replace)
  apply assumption+
  apply (rule corr_loc_replace)
  apply assumption+
done

```

```

lemma consistent_init_replace:
  "[[ consistent_init stk loc (ST,LT) ihp;
    approx_loc G hp ihp loc LT;
    approx_stk G hp ihp stk ST;
    approx_val G hp ihp oX (OK TO);
    T' = Init C' ]]"
  => consistent_init (replace oX x stk) (replace oX x loc)
    (replace TO T' ST, replace (OK TO) (OK T') LT) ihp"
  apply (simp only: consistent_init_conv)
  apply clarify apply (blast intro: corresponds_replace)
  done

lemma corr_loc_replace_Err:
  "corr_loc loc LT ihp v T => corr_loc loc (replace T' Err LT) ihp v T"
  by (simp add: corr_loc_def list_all2_conv_all_nth replace_def)

lemma corresponds_replace_Err:
  "corresponds stk loc (ST, LT) ihp v T
  => corresponds stk loc (ST, replace T' Err LT) ihp v T"
  by (simp add: corresponds_def corr_loc_replace_Err)

lemma consistent_init_replace_Err:
  "consistent_init stk loc (ST, LT) ihp
  => consistent_init stk loc (ST, replace T' Err LT) ihp"
  by (unfold consistent_init_def, blast intro: corresponds_replace_Err)

lemma corresponds_append:
  "\^X. [[ corresponds (x@stk) loc (X@ST,LT) ihp v T; length X = length x ]]"
  => corresponds stk loc (ST,LT) ihp v T"
  apply (induct x)
  apply simp
  apply (clarsimp simp add: length_Suc_conv)
  apply (simp add: corresponds_stk_cons)
  apply blast
  done

lemma consistent_init_append:
  "\^X. [[ consistent_init (x@stk) loc (X@ST,LT) ihp; length X = length x ]]"
  => consistent_init stk loc (ST,LT) ihp"
  apply (induct x)
  apply simp
  apply (clarsimp simp add: length_Suc_conv)
  apply (drule consistent_init_pop)
  apply blast
  done

lemma corresponds_xcp:
  "corresponds stk loc (ST, LT) ihp v T => T = UnInit C pc \vee T = PartInit D
  => corresponds [x] loc ([Init X], LT) ihp v T"
  apply (simp add: corresponds_def corr_stk_def corr_loc_def)
  apply blast
  done

```

lemma consistent\_init\_xcp:

```
"consistent_init stk loc (ST,LT) ihp
⇒ consistent_init [x] loc ([Init X], LT) ihp"
apply (unfold consistent_init_def)
apply (blast intro: corresponds_xcp)
done
```

lemma corresponds\_pop:

```
"corresponds (s#stk) loc (S#ST,LT) ihp v T
⇒ corresponds stk loc (ST,LT) ihp v T"
by (simp add: corresponds_stk_cons)
```

lemma consistent\_init\_Dup:

```
"consistent_init (s#stk) loc (S#ST,LT) ihp
⇒ consistent_init (s#s#stk) loc (S#S#ST,LT) ihp"
apply (simp only: consistent_init_conv)
apply clarify
apply (simp add: corresponds_stk_cons)
apply blast
done
```

lemma corresponds\_Dup:

```
"corresponds (s#stk) loc (S#ST,LT) ihp v T
⇒ corresponds (s#s#stk) loc (S#S#ST,LT) ihp v T"
by (simp add: corresponds_stk_cons)
```

lemma corresponds\_Dup\_x1:

```
"corresponds (s1#s2#stk) loc (S1#S2#ST,LT) ihp v T
⇒ corresponds (s1#s2#s1#stk) loc (S1#S2#S1#ST,LT) ihp v T"
by (simp add: corresponds_stk_cons)
```

lemma consistent\_init\_Dup\_x1:

```
"consistent_init (s1#s2#stk) loc (S1#S2#ST,LT) ihp
⇒ consistent_init (s1#s2#s1#stk) loc (S1#S2#S1#ST,LT) ihp"
by (unfold consistent_init_def, blast intro: corresponds_Dup_x1)
```

lemma corresponds\_Dup\_x2:

```
"corresponds (s1#s2#s3#stk) loc (S1#S2#S3#ST,LT) ihp v T
⇒ corresponds (s1#s2#s3#s1#stk) loc (S1#S2#S3#S1#ST,LT) ihp v T"
by (simp add: corresponds_stk_cons)
```

lemma consistent\_init\_Dup\_x2:

```
"consistent_init (s1#s2#s3#stk) loc (S1#S2#S3#ST,LT) ihp
⇒ consistent_init (s1#s2#s3#s1#stk) loc (S1#S2#S3#S1#ST,LT) ihp"
by (unfold consistent_init_def, blast intro: corresponds_Dup_x2)
```

lemma corresponds\_Swap:

```
"corresponds (s1#s2#stk) loc (S1#S2#ST,LT) ihp v T
⇒ corresponds (s2#s1#stk) loc (S2#S1#ST,LT) ihp v T"
by (simp add: corresponds_stk_cons)
```

lemma consistent\_init\_Swap:

```
"consistent_init (s1#s2#stk) loc (S1#S2#ST,LT) ihp
⇒ consistent_init (s2#s1#stk) loc (S2#S1#ST,LT) ihp"
```



by (unfold consistent\_init\_def, blast intro: corresponds\_Swap)

#### 4.21.5 oconf

lemma oconf\_blank:

"[ is\_class G D; wf\_prog wf\_mb G ]  $\implies$  G, hp  $\vdash$  blank G D  $\checkmark$ "  
by (auto simp add: oconf\_def blank\_def dest: fields\_is\_type)

lemma oconf\_field\_update:

"[map\_of (fields (G, oT)) FD = Some T; G, hp  $\vdash$  v ::  $\leq$  T; G, hp  $\vdash$  (Obj oT fs)  $\checkmark$  ]  
 $\implies$  G, hp  $\vdash$  Obj oT (fs(FD  $\mapsto$  v))  $\checkmark$ "  
by (simp add: oconf\_def lconf\_def)

lemma oconf\_imp\_oconf\_heap\_newref [rule\_format]:

"[hp oref = None; G, hp  $\vdash$  obj  $\checkmark$ ; G, hp  $\vdash$  obj'  $\checkmark$ ]  $\implies$  G, hp(oref  $\mapsto$  obj')  $\vdash$  obj  $\checkmark$ "  
apply (unfold oconf\_def lconf\_def)  
apply (simp split: heap\_entry.splits)  
apply (fast intro: conf\_hext hext\_new)+  
done

lemma oconf\_imp\_oconf\_heap\_update [rule\_format]:

"hp a = Some obj'  $\longrightarrow$  obj\_ty obj' = obj\_ty obj''  $\longrightarrow$  G, hp  $\vdash$  obj  $\checkmark$   
 $\longrightarrow$  G, hp(a  $\mapsto$  obj'')  $\vdash$  obj  $\checkmark$ "  
apply (unfold oconf\_def lconf\_def)  
apply (simp split: heap\_entry.splits)  
apply (force intro: approx\_val\_heap\_update)  
done

#### 4.21.6 hconf

lemma hconf\_imp\_hconf\_newref [rule\_format]:

"hp oref = None  $\longrightarrow$  G  $\vdash$  h hp  $\checkmark$   $\longrightarrow$  G, hp  $\vdash$  obj  $\checkmark$   $\longrightarrow$  G  $\vdash$  h hp(oref  $\mapsto$  obj)  $\checkmark$ "  
apply (simp add: hconf\_def)  
apply (fast intro: oconf\_imp\_oconf\_heap\_newref)  
done

lemma hconf\_imp\_hconf\_field\_update [rule\_format]:

"map\_of (fields (G, oT)) (F, D) = Some T  $\wedge$  hp oloc = Some (Obj oT fs)  $\wedge$   
G, hp  $\vdash$  v ::  $\leq$  T  $\wedge$  G  $\vdash$  h hp  $\checkmark$   $\longrightarrow$  G  $\vdash$  h hp(oloc  $\mapsto$  Obj oT (fs((F, D)  $\mapsto$  v)))  $\checkmark$ "  
apply (simp add: hconf\_def)  
apply (force intro: oconf\_imp\_oconf\_heap\_update oconf\_field\_update  
simp add: obj\_ty\_def)  
done

#### 4.21.7 h-init

lemma h\_init\_field\_update:

"[ h\_init G hp ihp; hp x = Some (Obj C fs); is\_init hp ihp v ]  
 $\implies$  h\_init G (hp(x  $\mapsto$  Obj C (fs(X  $\mapsto$  v)))) ihp"  
apply (unfold h\_init\_def o\_init\_def l\_init\_def)  
apply clarsimp  
apply (rule conjI)  
apply clarify  
apply (rule conjI)  
apply (clarsimp simp add: is\_init\_def)

```

apply clarsimp
apply (elim allE, erule impE, assumption)
apply clarsimp
apply (elim allE,
      erule impE, rule exI, assumption)
apply (clarsimp simp add: is_init_def)
apply (clarsimp split: heap_entry.split)
apply (rule conjI)
apply clarify
apply (elim allE, erule impE, assumption, simp,
      elim allE,
      erule impE, rule exI, assumption)
apply (clarsimp simp add: is_init_def)
apply clarify
apply (elim allE, erule impE, assumption) back
apply simp
apply (erule allE, erule impE, simp)
apply (clarsimp simp add: is_init_def)
done

```

lemma oconf\_arrD:

```

"[[G,h ⊢ Arr T len en √; n < len]]
⇒ ∃ v. en n = Some v ∧ G,h ⊢ v :: ≤ T"
by (unfold oconf_def) simp

```

lemma h\_init\_newref:

```

"[[ hp r = None; G ⊢ h hp √; h_init G hp ihp ]]
⇒ h_init G (hp(r↦blank G X)) (ihp(r := T'))"
apply (unfold h_init_def o_init_def l_init_def blank_def)
apply clarsimp
apply (rule conjI)
apply clarsimp
apply (simp add: init_vars_def map_of_map)
apply (rule is_init_default_val)
apply (clarsimp split: heap_entry.split)
apply (rule conjI)
apply clarify
apply (elim allE, erule impE, assumption)
apply simp
apply (elim allE, erule impE)
  apply fastsimp
apply clarsimp
apply (simp add: is_init_def)
apply (drule hconfD, assumption)
apply (drule oconf_objD, assumption)
apply clarsimp
apply (drule new_locD, assumption)
apply simp
apply clarify
apply (elim allE, erule impE, assumption)
apply simp
apply (elim allE, erule impE)
  apply fastsimp
apply clarsimp

```

```

apply (simp add: is_init_def)
apply (drule hconfD, assumption)
apply (drule oconf_arrD, assumption)
apply clarsimp
apply (drule new_locD, assumption)
apply simp
done

```

#### 4.21.8 preallocated

lemma preallocated\_field\_update:

```

"[[ hp a = Some (Obj oT fs); G⊢h hp√; preallocated hp ihp ]]
⇒ preallocated (hp(a ↦ Obj oT fs')) ihp"
apply (unfold preallocated_def)
apply (rule allI)
apply (erule_tac x=x in allE)
apply (simp add: is_init_def)
apply (rule ccontr)
apply (unfold hconf_def)
apply (erule allE, erule allE, erule impE, assumption)
apply (unfold oconf_def lconf_def)
apply (simp del: split_paired_All)
done

```

lemma preallocated\_array\_update:

```

"[[ hp a = Some (Arr T l en);
   G⊢h hp√; preallocated hp ihp ]]
⇒ preallocated (hp(a ↦ Arr T l en')) ihp"
apply (unfold preallocated_def)
apply (rule allI)
apply (erule_tac x=x in allE)
apply (simp add: is_init_def)
apply (rule ccontr)
apply (unfold hconf_def)
apply (erule allE, erule allE, erule impE, assumption)
apply (unfold oconf_def lconf_def)
apply (simp del: split_paired_All)
done

```

lemma preallocated\_newref:

```

assumes none: "hp oref = None" and alloc: "preallocated hp ihp"
shows "preallocated (hp(oref↦obj)) (ihp(oref:=T)) "
proof (cases oref)
  case (XcptRef x)
  with none alloc have "False" by (auto elim: preallocatedE [of _ _ x])
  thus ?thesis ..
next
  case (Loc l)
  with alloc show ?thesis by (simp add: preallocated_def is_init_def)
qed

```

### 4.21.9 constructor-ok

```

lemma correct_frames_ctor_ok:
  "correct_frames G hp ihp phi rT sig z r frs
  ⇒ frs = [] ∨ (fst sig = init → (∃ a C'. constructor_ok G hp ihp a C' z r))"
  apply (cases frs)
  apply simp
  apply simp
  apply clarify
  apply simp
  apply blast
done

```

### 4.21.10 correct-frame

```

lemma correct_frameE [elim?]:
  "correct_frame G hp ihp (ST,LT) maxl ins (stk, loc, C, sig, pc, r) ⇒
  (⟦approx_stk G hp ihp stk ST; approx_loc G hp ihp loc LT;
  consistent_init stk loc (ST, LT) ihp;
  fst sig = init →
  corresponds stk loc (ST, LT) ihp (fst r) (PartInit C) ∧
  (∃ l. fst r = Addr l ∧ hp l ≠ None ∧ (ihp l = PartInit C ∨ (∃ C'. ihp l = Init
  (Class C'))))⟧);
  pc < length ins; length loc = length (snd sig) + maxl + 1⟧
  ⇒ P)
  ⇒ P"
  apply (unfold correct_frame_def)
  apply fast
done

```

### 4.21.11 correct-frames

```

lemmas [simp del] = fun_upd_apply
lemmas [split del] = split_if

```

```

lemma correct_frames_imp_correct_frames_field_update [rule_format]:
  "∀ rT C sig z r. correct_frames G hp ihp phi rT sig z r frs →
  hp a = Some (Obj C od) → map_of (fields (G, C)) fl = Some fd →
  G, hp ⊢ v :: ≤ fd →
  correct_frames G (hp(a ↦ Obj C (od(fl ↦ v)))) ihp phi rT sig z r frs"
  apply (induct frs)
  apply simp
  apply clarify
  apply simp
  apply clarify
  apply (unfold correct_frame_def)
  apply (simp (no_asm_use))
  apply clarify
  apply (intro exI conjI)
  apply assumption
  apply assumption+
  apply (rule refl)
  apply assumption+
  apply (rule impI)
  apply (rule constructor_ok_field_update)

```

```

  apply (erule impE, assumption)
  apply simp
  apply assumption
  apply (rule approx_stk_imp_approx_stk_sup_heap, assumption)
  apply (rule hext_upd_obj, assumption)
  apply (rule approx_loc_imp_approx_loc_sup_heap, assumption)
  apply (rule hext_upd_obj, assumption)
  apply assumption+
  apply (rule impI)
  apply (erule impE, assumption, erule conjE)
  apply (rule conjI)
  apply assumption
  apply (elim exE conjE)
  apply (rule exI)
  apply (rule conjI)
  apply assumption
  apply (rule conjI)
  apply (simp add: fun_upd_apply)
  apply (case_tac "l = a")
  apply simp
  apply simp
  apply assumption+
  apply blast
done

```

```

lemma correct_frames_array_update [rule_format]:
  "∀rT C sig z r. correct_frames G hp ihp phi rT sig z r frs →
  hp a = Some (Arr T l en) →
  G, hp ⊢ v :: ≤fd →
  correct_frames G (hp(a ↦ Arr T l en')) ihp phi rT sig z r frs"
  apply (induct frs)
  apply simp
  apply clarify
  apply simp
  apply clarify
  apply (unfold correct_frame_def)
  apply (simp (no_asm_use))
  apply clarify
  apply (intro exI conjI)
  apply assumption
  apply assumption+
  apply (rule refl)
  apply assumption+
  apply (rule impI)
  apply (rule constructor_ok_array_update)
  apply (erule impE, assumption)
  apply simp
  apply assumption
  apply (rule approx_stk_imp_approx_stk_sup_heap, assumption)
  apply (rule hext_upd_arr, assumption)
  apply (rule approx_loc_imp_approx_loc_sup_heap, assumption)
  apply (rule hext_upd_arr, assumption)
  apply assumption+

```

```

apply (rule impI)
apply (erule impE, assumption, erule conjE)
apply (rule conjI)
  apply assumption
apply (elim exE conjE)
apply (rule exI)
apply (rule conjI)
  apply assumption
apply (rule conjI)
  apply (simp add: fun_upd_apply)
  apply (case_tac "la = a")
    apply simp
    apply simp
apply assumption+
apply blast
done

```

```

lemma correct_frames_imp_correct_frames_newref [rule_format]:
  "∀rT sig z r. hp x = None → correct_frames G hp ihp phi rT sig z r frs
  → correct_frames G (hp(x→obj)) (ihp(x:=T)) phi rT sig z r frs"
  apply (induct frs)
    apply simp
    apply clarify
    apply simp
    apply clarify
    apply (unfold correct_frame_def)
    apply (simp (no_asm_use))
    apply clarify
    apply (intro exI conjI)
      apply assumption+
      apply (rule refl)
      apply assumption+
    apply (rule impI)
    apply (erule impE, assumption)
    apply (rule constructor_ok_newref, assumption)
    apply simp
    apply (drule approx_stk_newref, assumption)
    apply (rule approx_stk_imp_approx_stk_sup_heap, assumption)
    apply (rule hext_new, assumption)
    apply (drule approx_loc_newref, assumption)
    apply (rule approx_loc_imp_approx_loc_sup_heap, assumption)
    apply (rule hext_new, assumption)
    apply (rule consistent_init_new_val, assumption+)
    apply (rule impI, erule impE, assumption, erule conjE)
    apply (rule conjI)
      apply (rule corresponds_new_val2, assumption+)
    apply (elim conjE exE)
    apply (simp add: fun_upd_apply)
    apply (case_tac "l = x")
      apply simp
      apply simp
    apply assumption+
    apply blast

```

done

lemmas [simp add] = fun\_upd\_apply

lemmas [split add] = split\_if

#### 4.21.12 correct-state

lemma correct\_stateE [elim?]:

```
"G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ⇒
  (∧rT maxs maxl ins et ST LT z.
    [[G ⊢h hp √; h_init G hp ihp; preallocated hp ihp; is_class G C; method (G, C)
sig = Some (C, rT, maxs, maxl, ins, et);
      ((ST,LT), z) ∈ phi C sig ! pc; correct_frame G hp ihp (ST,LT) maxl ins (stk,
loc, C, sig, pc, r);
      correct_frames G hp ihp phi rT sig z r frs]] ⇒ P)
⇒ P"
apply (unfold correct_state_def)
apply simp
apply blast
done
```

lemma correct\_stateE2 [elim?]:

```
"G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ⇒
method (G,C) sig = Some (C,rT,maxs,maxl,ins,et) ⇒
(∧ST LT z.
  [[G ⊢h hp √; h_init G hp ihp; preallocated hp ihp; is_class G C;
    ((ST,LT), z) ∈ phi C sig ! pc;
    correct_frame G hp ihp (ST,LT) maxl ins (stk, loc, C, sig, pc, r);
    correct_frames G hp ihp phi rT sig z r frs]] ⇒ P)
⇒ P"
apply (erule correct_stateE)
apply simp
done
```

lemma correct\_stateI [intro?]:

```
"[[G ⊢h hp √; h_init G hp ihp; preallocated hp ihp; is_class G C;
  method (G, C) sig = Some (C, rT, maxs, maxl, ins, et);
  (s',z) ∈ phi C sig ! pc;
  correct_frame G hp ihp s' maxl ins (stk, loc, C, sig, pc, r);
  correct_frames G hp ihp phi rT sig z r frs]]
⇒ G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"
apply (unfold correct_state_def)
apply (cases s')
apply simp
apply fast
done
```

end

## 4.22 BV Type Safety Proof

`theory BVSpecTypeSafe = Correct:`

This theory contains proof that the specification of the bytecode verifier only admits type safe programs.

### 4.22.1 Preliminaries

Simp and intro setup for the type safety proof:

```
lemmas defs1 = correct_state_def correct_frame_def wt_instr_def
           eff_def norm_eff_def eff_bool_def
```

```
lemmas correctE = correct_stateE2 correct_frameE
```

```
declare eff_defs [simp del]
```

```
lemmas [simp del] = split_paired_All
```

If we have a welltyped program and a conforming state, we can directly infer that the current instruction is well typed:

```
lemma wt_jvm_prog_impl_wt_instr_cor:
  "[[ wt_jvm_prog G phi; method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
    G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]]
  ⇒ wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc"
  apply (elim correct_stateE correct_frameE)
  apply simp
  apply (blast intro: wt_jvm_prog_impl_wt_instr)
  done
```

### 4.22.2 Exception Handling

Exceptions don't touch anything except the stack:

```
lemma exec_instr_xcpt:
  "(fst (exec_instr i G hp ihp stk vars Cl sig pc z frs) = Some xcp)
  = (∃stk'. exec_instr i G hp ihp stk vars Cl sig pc z frs =
    (Some xcp, hp, ihp, (stk', vars, Cl, sig, pc, z)#frs))"
  by (cases i, auto simp add: split_beta split: split_if_asm)
```

Relates `match_any` from the Bytecode Verifier with `match_exception_table` from the operational semantics:

```
lemma in_match_any:
  "match_exception_table G xcpt pc et = Some pc' ⇒
  ∃C. C ∈ set (match_any G pc et) ∧ G ⊢ xcpt ≤C C ∧
  match_exception_table G C pc et = Some pc'"
  (is "PROP ?P et" is "?match et ⇒ ?match_any et")
  proof (induct et)
```



```

show "PROP ?P []" by simp

fix e es
assume IH: "PROP ?P es"
assume match: "?match (e#es)"

obtain start_pc end_pc handler_pc catch_type where
  [simp]: "e = (start_pc, end_pc, handler_pc, catch_type)" by (cases e)

from IH match
show "?match_any (e#es)"
proof (cases "match_exception_entry G xcpt pc e")
  case False
  with match
  have "match_exception_table G xcpt pc es = Some pc'" by simp
  with IH
  obtain C where
    set: "C ∈ set (match_any G pc es)" and
    C: "G ⊢ xcpt ≤C C" and
    m: "match_exception_table G C pc es = Some pc'" by blast

  from set
  have "C ∈ set (match_any G pc (e#es))" by simp
  moreover
  from False C
  have "¬ match_exception_entry G C pc e"
    by - (erule contrapos_nn,
      auto simp add: match_exception_entry_def elim: rtrancl_trans)
  with m
  have "match_exception_table G C pc (e#es) = Some pc'" by simp
  moreover note C
  ultimately
  show ?thesis by blast
next
  case True with match
  have "match_exception_entry G catch_type pc e"
    by (simp add: match_exception_entry_def)
  moreover
  from True match
  obtain
    "start_pc ≤ pc"
    "pc < end_pc"
    "G ⊢ xcpt ≤C catch_type"
    "handler_pc = pc'"
    by (simp add: match_exception_entry_def)
  ultimately
  show ?thesis by auto
qed

```

qed

```
lemma match_et_imp_match:
  "match_exception_table G X pc et = Some handler
  ⇒ match G X pc et = [X]"
  apply (simp add: match_some_entry)
  apply (induct et)
  apply (auto split: split_if_asm)
  done
```

```
lemma imageE2:
  "x ∈ A ⇒ y = f x ⇒ y ∈ f ` A"
  by simp
```

```
lemma subcls_obj [iff]:
  "(G ⊢ Object ≤C C) = (C = Object)"
  apply auto
  apply (erule converse_rtranclE, simp)
  apply (erule subcls1.elims)
  apply simp
  done
```

We can prove separately that the recursive search for exception handlers (*find\_handler*) in the frame stack results in a conforming state (if there was no matching exception handler in the current frame). We require that the exception is a valid, initialized heap address, and that the state before the exception occurred conforms.

```
lemma uncaught_xcpt_correct:
  "∧f. [ wt_jvm_prog G phi; xcp = Addr adr; hp adr = Some T; is_init hp ihp xcp;
        G,phi ⊢JVM (None, hp, ihp, f#frs)√ ]
  ⇒ G,phi ⊢JVM (find_handler G (Some xcp) hp ihp frs)√"
  (is "∧f. [ ?wt; ?adr; ?hp; ?init; ?correct (None, hp, ihp, f#frs) ] ⇒ ?correct (?find
  frs)")
  proof (induct frs)
    — the base case is trivial, as it should be
    show "?correct (?find [])" by (simp add: correct_state_def)

    — we will need both forms wt_jvm_prog and wf_prog later
    assume wt: ?wt
    then obtain mb where wf: "wf_prog mb G" by (simp add: wt_jvm_prog_def)

    — these do not change in the induction:
    assume adr: ?adr
    assume hp: ?hp
    assume init: ?init

    — the assumption for the cons case:
    fix f f' frs' assume cr: "?correct (None, hp, ihp, f#f'#frs)'"
```

— the induction hypothesis as produced by Isabelle, immediatly simplified with the fixed assumptions above

```
assume " $\bigwedge f. \llbracket ?wt; ?adr; ?hp; ?init; ?correct (None, hp, ihp, f\#frs') \rrbracket \implies ?correct (?find frs')$ "
```

```
with wt adr hp init
```

```
have IH: " $\bigwedge f. ?correct (None, hp, ihp, f\#frs') \implies ?correct (?find frs')$ " by blast
```

```
obtain stk loc C sig pc r where f' [simp]: "f' = (stk,loc,C,sig,pc,r)"
```

```
by (cases f')
```

```
from cr have cr': "?correct (None, hp, ihp, f'\#frs')"
```

```
by (unfold correct_state_def, clarsimp) blast
```

```
then obtain rT maxs maxl ins et where
```

```
meth: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
```

```
by (fastsimp elim!: correct_stateE)
```

```
hence [simp]: "ex_table_of (snd (snd (the (method (G, C) sig)))) = et" by simp
```

```
show "?correct (?find (f'\#frs'))"
```

```
proof (cases "match_exception_table G (cname_of hp xcp) pc et")
```

```
case None
```

```
with cr' IH
```

```
show ?thesis by simp
```

```
next
```

```
fix handler_pc
```

```
assume match: "match_exception_table G (cname_of hp xcp) pc et = Some handler_pc"
```

```
(is "?match (cname_of hp xcp) = _")
```

```
from wt meth cr' [simplified]
```

```
have wti: "wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc"
```

```
by (rule wt_jvm_prog_impl_wt_instr_cor)
```

```
from cr meth
```

```
obtain C' mn pts ST LT z where
```

```
ins: "ins ! pc = Invoke C' mn pts  $\vee$  ins ! pc = Invoke_special C' mn pts" and
```

```
phi: "((ST, LT),z)  $\in$  phi C sig ! pc" and
```

```
frm: "correct_frame G hp ihp (ST,LT) maxl ins (stk, loc, C, sig, pc, r)" and
```

```
frms: "correct_frames G hp ihp phi rT sig z r frs'"
```

```
by (simp add: correct_state_def) fast
```

```
from match
```

```
obtain D where
```

```
in_any: "D  $\in$  set (match_any G pc et)" and
```

```
D: "G  $\vdash$  cname_of hp xcp  $\preceq_C$  D" and
```

```
match': "?match D = Some handler_pc"
```

```
by (blast dest: in_match_any)
```

```

from ins have "xcpt_names (ins ! pc, G, pc, et) = match_any G pc et" by auto
with wti phi have
  "\forall D \in set (match_any G pc et). the (?match D) < length ins \wedge
  (([Init (Class D)], LT), z) \in phi C sig ! the (?match D)"
  apply -
  apply (unfold wt_instr_def xcpt_eff_def eff_def)
  apply clarsimp
  apply (drule bspec)
  apply (rule UnI2)
  apply (rule imageI)
  apply assumption
  apply clarsimp
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  done
with in_any match' obtain
  pc: "handler_pc < length ins"
  "(([Init (Class D)], LT), z) \in phi C sig ! handler_pc"
  by auto

from frm obtain
  len: "length loc = 1 + length (snd sig) + maxl" and
  loc: "approx_loc G hp ihp loc LT" and
  csi: "consistent_init stk loc (ST, LT) ihp" and
  cor: "(fst sig = init \longrightarrow
  corresponds stk loc (ST, LT) ihp (fst r) (PartInit C) \wedge
  (\exists l. fst r = Addr l \wedge hp l \neq None \wedge (ihp l = PartInit C \vee (\exists C'. ihp l =
Init (Class C'))))))"
  by (fastsimp elim!: correct_frameE)

let ?f = "[xcp], loc, C, sig, handler_pc, r"
have "correct_frame G hp ihp ([Init (Class D)], LT) maxl ins ?f"
proof -
  from D adr hp
  have "G, hp \vdash xcp :: \preceq Class D"
  by (auto simp add: conf_def obj_ty_def split: heap_entry.split)
  with init
  have "approx_val G hp ihp xcp (OK (Init (Class D)))"
  by (simp add: approx_val_def iconf_def)
  with len pc csi cor loc
  show ?thesis
  by (simp add: correct_frame_def consistent_init_xcp)
  (blast intro: corresponds_xcp)
qed

with cr' match meth pc frms

```

```

  show ?thesis by (unfold correct_state_def) auto
qed

```

The requirement of lemma `uncaught_xcpt_correct` (that the exception is a valid, initialized reference on the heap) is always met for welltyped instructions and conformant states:

```
lemma exec_instr_xcpt_hp:
```

```

  assumes xcpt: "fst (exec_instr (ins!pc) G hp ihp stk vars Cl sig pc r frs) = Some xcp"
  assumes wt: "wt_instr (ins!pc) G Cl rT (phi C sig) maxs (fst sig=init) (length ins)
et pc"
  assumes correct: "G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"
  shows "∃adr T. xcp = Addr adr ∧ hp adr = Some T ∧ is_init hp ihp xcp"

```

```
proof -
```

```

  note [simp] = split_beta raise_system_xcpt_def
  note [split] = split_if_asm option.split_asm

```

```
from correct have pre: "preallocated hp ihp" ..
```

```
with xcpt show ?thesis
```

```
proof (cases "ins!pc")
```

```
  case New with xcpt pre
```

```
  show ?thesis by (auto dest: new_Addr_OutOfMemory dest!: preallocatedD)
```

```
next
```

```
  case Throw
```

```
from correct obtain ST LT z where
```

```
  "((ST, LT), z) ∈ phi C sig ! pc"
```

```
  "approx_stk G hp ihp stk ST"
```

```
  by (blast elim: correct_stateE correct_frameE)
```

```
with Throw wt xcpt pre show ?thesis
```

```
  apply (unfold wt_instr_def)
```

```
  apply clarsimp
```

```
  apply (drule bspec, assumption)
```

```
  apply (clarsimp simp add: approx_val_def iconf_def)
```

```
  apply (blast dest: non_npD)+
```

```
  done
```

```
next
```

```
  case Invoke_special with xcpt pre
```

```
  show ?thesis by (auto dest: new_Addr_OutOfMemory dest!: preallocatedD)
```

```
next
```

```
  case ArrNew with xcpt pre
```

```
  show ?thesis by (auto dest: new_Addr_OutOfMemory dest!: preallocatedD)
```

```
qed (auto dest!: preallocatedD)
```

```
qed
```

```
lemma cname_of_xcp [intro]:
```

```
  "[preallocated hp ihp; xcp = Addr (XcptRef x)] ⇒ cname_of hp xcp = Xcpt x"
```

```
  by (auto elim: preallocatedE [of hp ihp x])
```

```
lemma prealloc_is_init [simp]:
  "preallocated hp ihp  $\implies$  is_init hp ihp (Addr (XcptRef x))"
  by (drule preallocatedD) blast
```

```
thm wt_jvm_progD [intro?]
```

```
lemma wt_jvm_progE [intro?]:
  "wt_jvm_prog G phi  $\implies$  ( $\bigwedge$ wt. wf_prog wt G  $\implies$  P)  $\implies$  P"
  by (simp add: wt_jvm_prog_def)
```

Finally we can state that the next state always conforms when an exception occurred:

```
lemma xcpt_correct:
  assumes wtp: "wt_jvm_prog G phi"
  assumes meth: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
  assumes wt: "wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig=init) (length ins) et
  pc"
  assumes xp: "fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = Some xcp"
  assumes s': "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
  assumes correct: "G,phi  $\vdash$ JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)  $\surd$ "

  shows "G,phi  $\vdash$ JVM state'  $\surd$ "
proof -
  from wtp obtain wfmb where wf: "wf_prog wfmb G" ..

  note xp' = meth s' xp

  note wtp
  moreover
  from xp wt correct
  obtain adr T where
    adr: "xcp = Addr adr" "hp adr = Some T" "is_init hp ihp xcp"
    by (blast dest: exec_instr_xcpt_hp)
  moreover
  note correct
  ultimately
  have "G,phi  $\vdash$ JVM find_handler G (Some xcp) hp ihp frs  $\surd$ " by (rule uncaught_xcpt_correct)
  with xp'
  have "match_exception_table G (cname_of hp xcp) pc et = None  $\implies$  ?thesis"
    (is "?m (cname_of hp xcp) = _  $\implies$  _" is "?match = _  $\implies$  _")
    by (clarsimp simp add: exec_instr_xcpt_split_beta)
  moreover
  { fix handler assume some_handler: "?match = Some handler"

    from correct meth obtain ST LT z where
      hp_ok: "G  $\vdash$ h hp  $\surd$ " and
      h_ini: "h_init G hp ihp" and
      prehp: "preallocated hp ihp" and
      class: "is_class G C" and
      phi_pc: " $((ST,LT),z) \in$  phi C sig ! pc" and
      frame: "correct_frame G hp ihp (ST, LT) maxl ins (stk, loc, C, sig, pc, r)" and
      frames: "correct_frames G hp ihp phi rT sig z r frs"
    ..
```

```

from frame obtain
  stk: "approx_stk G hp ihp stk ST" and
  loc: "approx_loc G hp ihp loc LT" and
  pc: "pc < length ins" and
  len: "length loc = length (snd sig)+maxl+1" and
  cin: "consistent_init stk loc (ST, LT) ihp" and
  cor: "fst sig = init  $\longrightarrow$ 
  corresponds stk loc (ST, LT) ihp (fst r) (PartInit C)  $\wedge$ 
  ( $\exists l$ . fst r = Addr l  $\wedge$  hp l  $\neq$  None  $\wedge$  (ihp l = PartInit C  $\vee$  ( $\exists C'$ . ihp l = Init
  (Class C'))))"
  ..

from wt obtain
  eff: " $\forall (pc', s') \in \text{set } (\text{xcpt\_eff } (\text{ins!pc}) G pc (\text{phi } C \text{ sig!pc}) \text{ et}).$ 
  pc' < length ins  $\wedge$  s'  $\subseteq$  phi C sig!pc'"
  by (simp add: wt_instr_def eff_def)

from some_handler xp'
have state':
  "state' = (None, hp, ihp, ([xcp], loc, C, sig, handler, r)#frs)"
  by (cases "ins!pc", auto simp add: raise_system_xcpt_def split_beta
  split: split_if_asm) — takes long!

let ?f' = "([xcp], loc, C, sig, handler, r)"

from eff xp'
obtain ST' where
  phi_pc': " $((ST', LT), z) \in \text{phi } C \text{ sig ! handler}$ " and
  frame': "correct_frame G hp ihp (ST',LT) maxl ins ?f'"
proof (cases "ins!pc")
  case Return — can't generate exceptions:
  with xp' have False by (simp add: split_beta split: split_if_asm)
  thus ?thesis ..
next
let ?C = "Init (Class (Xcpt OutOfMemory))"
case New
with some_handler xp'
have xcp: "xcp = Addr (XcptRef OutOfMemory)"
  by (simp add: raise_system_xcpt_def split_beta new_Addr_OutOfMemory)
with prehp have "cname_of hp xcp = Xcpt OutOfMemory" ..
with New some_handler phi_pc eff
have
  phi': " $(([?C], LT), z) \in \text{phi } C \text{ sig ! handler}$ " and
  pc': "handler < length ins"
  by (auto simp add: xcpt_eff_def match_et_imp_match)
note phi'
moreover
{ from xcp prehp
  have "G, hp  $\vdash$  xcp  $:: \preceq$  Class (Xcpt OutOfMemory)"
  by (auto simp add: conf_def obj_ty_def dest!: preallocatedD)
  with xcp prehp
  have "approx_val G hp ihp xcp (OK ?C)"
  by (simp add: iconf_def approx_val_def)
}

```

```

    with wf pc' len loc cor cin
    have "correct_frame G hp ihp ([?C],LT) maxl ins ?f'"
      by (simp add: consistent_init_xcp correct_frame_def)
        (blast intro: corresponds_xcp)
  }
  ultimately
  show ?thesis by (rule that)
next
let ?C = "Init (Class (Xcpt NullPointer))"
case Getfield
with some_handler xp'
have xcp: "xcp = Addr (XcptRef NullPointer)"
  by (simp add: raise_system_xcpt_def split_beta split: split_if_asm)
with prehp have "cname_of hp xcp = Xcpt NullPointer" ..
with Getfield some_handler phi_pc eff
have
  phi': "(([/?C],LT),z) ∈ phi C sig ! handler" and
  pc': "handler < length ins"
  by (auto simp add: xcpt_eff_def match_et_imp_match)
note phi'
moreover
{ from xcp prehp
  have "approx_val G hp ihp xcp (OK ?C)"
    by (auto simp add: iconf_def conf_def obj_ty_def approx_val_def dest!: preallocatedD)
  with wf pc' len loc cor cin
  have "correct_frame G hp ihp ([?C],LT) maxl ins ?f'"
    by (simp add: consistent_init_xcp correct_frame_def)
      (blast intro: corresponds_xcp)
}
ultimately
show ?thesis by (rule that)
next
let ?C = "Init (Class (Xcpt NullPointer))"
case Putfield
with some_handler xp'
have xcp: "xcp = Addr (XcptRef NullPointer)"
  by (simp add: raise_system_xcpt_def split_beta split: split_if_asm)
with prehp have "cname_of hp xcp = Xcpt NullPointer" ..
with Putfield some_handler phi_pc eff
have
  phi': "(([/?C],LT),z) ∈ phi C sig ! handler" and
  pc': "handler < length ins"
  by (auto simp add: xcpt_eff_def match_et_imp_match)
note phi'
moreover
{ from xcp prehp
  have "approx_val G hp ihp xcp (OK ?C)"
    by (auto simp add: iconf_def conf_def obj_ty_def approx_val_def dest!: preallocatedD)
  with wf pc' len loc cor cin
  have "correct_frame G hp ihp ([?C],LT) maxl ins ?f'"
    by (simp add: consistent_init_xcp correct_frame_def)
      (blast intro: corresponds_xcp)
}
ultimately

```



```

show ?thesis by (rule that)
next
let ?X = ClassCast
let ?C = "Init (Class (Xcpt ?X))"
case Checkcast
with some_handler xp'
have xcp: "xcp = Addr (XcptRef ?X)"
  by (simp add: raise_system_xcpt_def split_beta split: split_if_asm)
with prehp have "cname_of hp xcp = Xcpt ?X" ..
with Checkcast some_handler phi_pc eff
have
  phi': "(([?C],LT),z) ∈ phi C sig ! handler" and
  pc': "handler < length ins"
  by (auto simp add: xcpt_eff_def match_et_imp_match)
note phi'
moreover
{ from xcp prehp
  have "approx_val G hp ihp xcp (OK ?C)"
    by (auto simp add: iconf_def conf_def obj_ty_def approx_val_def dest!: preallocatedD)
  with wf pc' len loc cor cin
  have "correct_frame G hp ihp ([?C],LT) maxl ins ?f'"
    by (simp add: consistent_init_xcp correct_frame_def)
    (blast intro: corresponds_xcp)
}
ultimately
show ?thesis by (rule that)
next
case Invoke
with phi_pc eff
have
  "∀D∈set (match_any G pc et).
  the (?m D) < length ins ∧ (([Init (Class D)], LT),z) ∈ phi C sig!the (?m D)"
  by (clarsimp simp add: xcpt_eff_def) (fastsimp elim: imageE2)
moreover
from some_handler
obtain D where
  "D ∈ set (match_any G pc et)" and
  D: "G ⊢ cname_of hp xcp ≤C D" and
  "?m D = Some handler"
  by (blast dest: in_match_any)
ultimately
have
  pc': "handler < length ins" and
  phi': "(([Init (Class D)], LT), z) ∈ phi C sig ! handler"
  by auto

from xp wt correct
obtain addr T where
  xcp: "xcp = Addr addr" "hp addr = Some T" "is_init hp ihp xcp"
  by (blast dest: exec_instr_xcpt_hp)
note phi'
moreover
{ from xcp D
  have "approx_val G hp ihp xcp (OK (Init (Class D)))"

```

```

      by (auto simp add: iconf_def conf_def obj_ty_def approx_val_def dest!: preallocatedD
split: heap_entry.split)
      with wf pc' len loc cor cin
      have "correct_frame G hp ihp ([Init (Class D)],LT) maxl ins ?f'"
        by (simp add: consistent_init_xcp correct_frame_def)
          (blast intro: corresponds_xcp)
    }
    ultimately
    show ?thesis by (rule that)
  next
  case Invoke_special
  with phi_pc eff
  have
    "∀D∈set (match_any G pc et).
    the (?m D) < length ins ∧ (([Init (Class D)], LT),z) ∈ phi C sig!the (?m D)"
    by (clarsimp simp add: xcpt_eff_def) (fastsimp elim: imageE2)
  moreover
  from some_handler
  obtain D where
    "D ∈ set (match_any G pc et)" and
    D: "G ⊢ cname_of hp xcp ≼C D" and
    "?m D = Some handler"
    by (blast dest: in_match_any)
  ultimately
  have
    pc': "handler < length ins" and
    phi': "(([Init (Class D)], LT), z) ∈ phi C sig ! handler"
    by auto

  from xp wt correct
  obtain addr T where
    xcp: "xcp = Addr addr" "hp addr = Some T" "is_init hp ihp xcp"
    by (blast dest: exec_instr_xcpt_hp)
  note phi'
  moreover
  { from xcp D
    have "approx_val G hp ihp xcp (OK (Init (Class D)))"
      by (auto simp add: iconf_def conf_def obj_ty_def approx_val_def dest!: preallocatedD
split: heap_entry.split)
      with wf pc' len loc cor cin
      have "correct_frame G hp ihp ([Init (Class D)],LT) maxl ins ?f'"
        by (simp add: consistent_init_xcp correct_frame_def)
          (blast intro: corresponds_xcp)
    }
  ultimately
  show ?thesis by (rule that)
next
  case Throw
  with phi_pc eff
  have
    "∀D∈set (match_any G pc et).
    the (?m D) < length ins ∧ (([Init (Class D)], LT),z) ∈ phi C sig!the (?m D)"
    by (clarsimp simp add: xcpt_eff_def) (fastsimp elim: imageE2)
  moreover

```

```

from some_handler
obtain D where
  "D ∈ set (match_any G pc et)" and
  D: "G ⊢ cname_of hp xcp ≤C D" and
  "?m D = Some handler"
  by (blast dest: in_match_any)
ultimately
have
  pc': "handler < length ins" and
  phi': "([Init (Class D)], LT), z) ∈ phi C sig ! handler"
  by auto

from xp wt correct
obtain addr T where
  xcp: "xcp = Addr addr" "hp addr = Some T" "is_init hp ihp xcp"
  by (blast dest: exec_instr_xcpt_hp)
note phi'
moreover
{ from xcp D
  have "approx_val G hp ihp xcp (OK (Init (Class D)))"
  by (auto simp add: iconf_def conf_def obj_ty_def approx_val_def dest!: preallocatedD
split: heap_entry.split)
  with wf pc' len loc cor cin
  have "correct_frame G hp ihp ([Init (Class D)],LT) maxl ins ?f'"
  by (simp add: consistent_init_xcp correct_frame_def)
  (blast intro: corresponds_xcp)
}
ultimately
show ?thesis by (rule that)
next
let ?X1 = OutOfMemory and ?X2 = NegativeArraySize
let ?C1 = "Init (Class (Xcpt ?X1))" and ?C2 = "Init (Class (Xcpt ?X2))"
case ArrNew
with some_handler xp'
have "xcp = Addr (XcptRef ?X1) ∨ xcp = Addr (XcptRef ?X2)"
  by (simp add: raise_system_xcpt_def split_beta new_Addr_OutOfMemory split: split_if_asm)
moreover
{ assume xcp: "xcp = Addr (XcptRef ?X1)"
  with prehp have "cname_of hp xcp = Xcpt ?X1" ..
  with ArrNew some_handler phi_pc eff
  have
    phi': "([?C1], LT), z) ∈ phi C sig ! handler" and
    pc': "handler < length ins"
    by (auto simp add: xcpt_eff_def match_et_imp_match)
  note phi'
  moreover {
    from xcp prehp
    have "G, hp ⊢ xcp :: ≤ Class (Xcpt ?X1)"
    by (auto simp add: conf_def obj_ty_def dest!: preallocatedD)
    with xcp prehp
    have "approx_val G hp ihp xcp (OK ?C1)"
    by (simp add: iconf_def approx_val_def)
    with wf pc' len loc cor cin
    have "correct_frame G hp ihp ([?C1],LT) maxl ins ?f'"
  }
}

```

```

      by (simp add: consistent_init_xcp correct_frame_def)
        (blast intro: corresponds_xcp)
    }
  ultimately
  have ?thesis by (rule that)
}
moreover {
  assume xcp: "xcp = Addr (XcptRef ?X2)"
  with prehp have "cname_of hp xcp = Xcpt ?X2" ..
  with ArrNew some_handler phi_pc eff
  have
    phi': "(([[?C2], LT),z) ∈ phi C sig ! handler" and
    pc': "handler < length ins"
    by (auto simp add: xcpt_eff_def match_et_imp_match)
  note phi'
  moreover {
    from xcp prehp
    have "G, hp ⊢ xcp :: ≤ Class (Xcpt ?X2)"
      by (auto simp add: conf_def obj_ty_def dest!: preallocatedD)
    with xcp prehp
    have "approx_val G hp ihp xcp (OK ?C2)"
      by (simp add: iconf_def approx_val_def)
    with wf pc' len loc cor cin
    have "correct_frame G hp ihp ([?C2],LT) maxl ins ?f'"
      by (simp add: consistent_init_xcp correct_frame_def)
        (blast intro: corresponds_xcp)
  }
  ultimately
  have ?thesis by (rule that)
}
ultimately show ?thesis by blast
next
let ?X1 = ArrayIndexOutOfBounds and ?X2 = NullPointer
let ?C1 = "Init (Class (Xcpt ?X1))" and ?C2 = "Init (Class (Xcpt ?X2))"
case ArrLoad
with some_handler xp'
have "xcp = Addr (XcptRef ?X1) ∨ xcp = Addr (XcptRef ?X2)"
  by (simp add: raise_system_xcpt_def split_beta split: split_if_asm)
moreover
{ assume xcp: "xcp = Addr (XcptRef ?X1)"
  with prehp have "cname_of hp xcp = Xcpt ?X1" ..
  with ArrLoad some_handler phi_pc eff
  have
    phi': "(([[?C1], LT),z) ∈ phi C sig ! handler" and
    pc': "handler < length ins"
    by (auto simp add: xcpt_eff_def match_et_imp_match)
  note phi'
  moreover {
    from xcp prehp
    have "G, hp ⊢ xcp :: ≤ Class (Xcpt ?X1)"
      by (auto simp add: conf_def obj_ty_def dest!: preallocatedD)
    with xcp prehp
    have "approx_val G hp ihp xcp (OK ?C1)"
      by (simp add: iconf_def approx_val_def)
  }
}

```

```

    with wf pc' len loc cor cin
    have "correct_frame G hp ihp ([?C1],LT) maxl ins ?f'"
      by (simp add: consistent_init_xcp correct_frame_def)
        (blast intro: corresponds_xcp)
  }
  ultimately
  have ?thesis by (rule that)
}
moreover {
  assume xcp: "xcp = Addr (XcptRef ?X2)"
  with prehp have "cname_of hp xcp = Xcpt ?X2" ..
  with ArrLoad some_handler phi_pc eff
  have
    phi': "(([[?C2], LT),z) ∈ phi C sig ! handler" and
    pc': "handler < length ins"
    by (auto simp add: xcpt_eff_def match_et_imp_match)
  note phi'
  moreover {
    from xcp prehp
    have "G, hp ⊢ xcp :: ≤ Class (Xcpt ?X2)"
      by (auto simp add: conf_def obj_ty_def dest!: preallocatedD)
    with xcp prehp
    have "approx_val G hp ihp xcp (OK ?C2)"
      by (simp add: iconf_def approx_val_def)
    with wf pc' len loc cor cin
    have "correct_frame G hp ihp ([?C2],LT) maxl ins ?f'"
      by (simp add: consistent_init_xcp correct_frame_def)
        (blast intro: corresponds_xcp)
  }
  ultimately
  have ?thesis ..
}
ultimately show ?thesis by blast
next
let ?X1 = ArrayIndexOutOfBounds and ?X2 = NullPointer and ?X3 = ArrayStore
let ?C1 = "Init (Class (Xcpt ?X1))" and ?C2 = "Init (Class (Xcpt ?X2))"
let ?C3 = "Init (Class (Xcpt ?X3))"
case ArrStore
with some_handler xp'
have "xcp = Addr (XcptRef ?X1) ∨ xcp = Addr (XcptRef ?X2) ∨ xcp = Addr (XcptRef
?X3)"
  by (simp add: raise_system_xcpt_def split_beta split: split_if_asm)
moreover
{ assume xcp: "xcp = Addr (XcptRef ?X1)"
  with prehp have "cname_of hp xcp = Xcpt ?X1" ..
  with ArrStore some_handler phi_pc eff
  have
    phi': "(([[?C1], LT),z) ∈ phi C sig ! handler" and
    pc': "handler < length ins"
    by (auto simp add: xcpt_eff_def match_et_imp_match)
  note phi'
  moreover {
    from xcp prehp
    have "G, hp ⊢ xcp :: ≤ Class (Xcpt ?X1)"

```

```

    by (auto simp add: conf_def obj_ty_def dest!: preallocatedD)
  with xcp prehp
  have "approx_val G hp ihp xcp (OK ?C1)"
    by (simp add: iconf_def approx_val_def)
  with wf pc' len loc cor cin
  have "correct_frame G hp ihp ([?C1],LT) maxl ins ?f'"
    by (simp add: consistent_init_xcp correct_frame_def)
    (blast intro: corresponds_xcp)
}
ultimately
have ?thesis ..
}
moreover {
  assume xcp: "xcp = Addr (XcptRef ?X2)"
  with prehp have "cname_of hp xcp = Xcpt ?X2" ..
  with ArrStore some_handler phi_pc eff
  have
    phi': "([?C2], LT),z) ∈ phi C sig ! handler" and
    pc': "handler < length ins"
    by (auto simp add: xcpt_eff_def match_et_imp_match)
  note phi'
  moreover {
    from xcp prehp
    have "G, hp ⊢ xcp :: ≤ Class (Xcpt ?X2)"
      by (auto simp add: conf_def obj_ty_def dest!: preallocatedD)
    with xcp prehp
    have "approx_val G hp ihp xcp (OK ?C2)"
      by (simp add: iconf_def approx_val_def)
    with wf pc' len loc cor cin
    have "correct_frame G hp ihp ([?C2],LT) maxl ins ?f'"
      by (simp add: consistent_init_xcp correct_frame_def)
      (blast intro: corresponds_xcp)
  }
  ultimately
  have ?thesis ..
}
moreover {
  assume xcp: "xcp = Addr (XcptRef ?X3)"
  with prehp have "cname_of hp xcp = Xcpt ?X3" ..
  with ArrStore some_handler phi_pc eff
  have
    phi': "([?C3], LT),z) ∈ phi C sig ! handler" and
    pc': "handler < length ins"
    by (auto simp add: xcpt_eff_def match_et_imp_match)
  note phi'
  moreover {
    from xcp prehp
    have "G, hp ⊢ xcp :: ≤ Class (Xcpt ?X3)"
      by (auto simp add: conf_def obj_ty_def dest!: preallocatedD)
    with xcp prehp
    have "approx_val G hp ihp xcp (OK ?C3)"
      by (simp add: iconf_def approx_val_def)
    with wf pc' len loc cor cin
    have "correct_frame G hp ihp ([?C3],LT) maxl ins ?f'"

```

```

      by (simp add: consistent_init_xcp correct_frame_def)
        (blast intro: corresponds_xcp)
    }
  ultimately
  have ?thesis ..
}
ultimately show ?thesis by blast
next
let ?C = "Init (Class (Xcpt NullPointer))"
case ArrLength
with some_handler xp'
have xcp: "xcp = Addr (XcptRef NullPointer)"
  by (simp add: raise_system_xcpt_def split_beta split: split_if_asm)
with prehp have "cname_of hp xcp = Xcpt NullPointer" ..
with ArrLength some_handler phi_pc eff
have
  phi': "(([[?C],LT),z) ∈ phi C sig ! handler" and
  pc': "handler < length ins"
  by (auto simp add: xcpt_eff_def match_et_imp_match)
note phi'
moreover
{ from xcp prehp
  have "approx_val G hp ihp xcp (OK ?C)"
    by (auto simp add: iconf_def conf_def obj_ty_def approx_val_def dest!: preallocatedD)
  with wf pc' len loc cor cin
  have "correct_frame G hp ihp ([[?C],LT) maxl ins ?f'"
    by (simp add: consistent_init_xcp correct_frame_def)
      (blast intro: corresponds_xcp)
}
ultimately
show ?thesis ..
qed auto — the other instructions do not generate exceptions

from state' meth hp_ok class frames phi_pc' frame' h_ini prehp
have ?thesis by simp (rule correct_stateI)
}
ultimately
show ?thesis by (cases "?match") blast+
qed

```

### 4.22.3 Single Instructions

In this section we look at each single (welltyped) instruction, and prove that the state after execution of the instruction still conforms. Since we have already handled exceptions above, we can now assume, that on exception occurs for this (single step) execution.

lemmas [iff] = not\_Err\_eq

lemma Load\_correct:

```

"[[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins!pc = Load idx;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;

```

```

    Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs);
    G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply clarsimp
apply (clarsimp simp add: approx_loc_def list_all2_conv_all_nth)
apply (erule allE, erule impE, assumption)
apply simp
apply (rule conjI)
  apply (erule consistent_init_loc_nth)
  apply simp
  apply simp
apply clarsimp
apply (frule corresponds_loc_nth)
apply simp
apply assumption
apply (clarsimp simp add: corresponds_def corr_stk_def corr_loc_cons)
done

```

lemma Store\_correct:

```

"[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins!pc = Store idx;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs);
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+

```



```

  apply (rule refl)
apply clarsimp
apply (blast intro: approx_loc_imp_approx_loc_subst
        consistent_init_store
        corresponds_var_upd)
done

```

lemma LitPush\_correct:

```

"[[ wf_prog wt G;
   method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
   ins!pc = LitPush v;
   wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
   Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs);
   G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
  apply (elim correctE, assumption)
  apply (clarsimp simp add: defs1 approx_val_def iconf_def init_le_Init
        map_eq_Cons)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply clarsimp
apply (simp add: approx_val_def iconf_def)
  apply (rule conjI)
  apply (fastsimp intro: conf_litval)
  apply (rule conjI)
  apply (clarsimp simp add: is_init_def split: val.split)
  apply (rule conjI)
  apply (erule consistent_init_Init_stk)
  apply (rule impI, erule impE, assumption, elim conjE exE)
  apply (simp add: corresponds_stk_cons)
done

```

lemma widen\_Array\_Class2 [iff]:

```

"G ⊢ T.[] ≲ Class C = (C = Object)"
by (auto simp add: widen_Array_Class)

```

lemma Cast\_conf2:

```

"[[ wf_prog ok G; G,h⊢v::≲RefT rt; cast_ok G C h v;
   G⊢Class C≲T; is_class G C]]
⇒ G,h⊢v::≲T"

```

```

apply (unfold cast_ok_def)
apply (frule widen_Class)
apply (elim exE disjE)
apply (simp add: null)
apply (clarsimp simp add: conf_def obj_ty_def split: heap_entry.splits)
apply (cases v)
apply (auto intro: rtrancl_trans)
apply (cases v)
apply auto
done

```

lemmas defs1 = defs1 raise\_system\_xcpt\_def

lemma Checkcast\_correct:

```

"[[ wt_jvm_prog G phi;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins!pc = Checkcast D;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs);
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√;
  fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None ]]
⇒ G,phi ⊢JVM state'√"

```

```

apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons approx_val_def wt_jvm_prog_def
  split: split_if_asm)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply clarsimp

```

```

apply (rule conjI)
  apply (clarsimp simp add: approx_val_def iconf_def init_le_Init)
  apply (blast intro: Cast_conf2)
apply (rule conjI)
  apply (drule consistent_init_pop)
  apply (erule consistent_init_Init_stk)
apply (clarsimp simp add: init_le_Init corresponds_stk_cons)
done

```

lemma non\_np\_objD:

```

"[[a' ≠ Null; G,h ⊢ a' :: ≤ Class D; field (G, D) F = Some (D, vT); wf_prog b G]] ⇒

```

```

  ∃ a C fs. a' = Addr a ∧ h a = Some (Obj C fs) ∧ G ⊢ C ≲C D"
apply (drule non_npD, assumption)
apply clarsimp
apply (case_tac z)
  apply simp
apply clarsimp
apply (drule no_field_Object [of _ _ F])
apply simp
done

```

lemma Getfield\_correct:

```

"[[ wt_jvm_prog G phi;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins!pc = Getfield F D;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs);
  G,phi ⊢ JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) √;
  fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None ]]
⇒ G,phi ⊢ JVM state' √"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons wt_jvm_prog_def approx_val_def
      iconf_def init_le_Init2)
apply (drule bspec, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons wt_jvm_prog_def approx_val_def
      iconf_def init_le_Init2
      split: option.split split_if_asm)
apply (frule non_np_objD, rule conf_widen, assumption+)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply (clarsimp simp add: defs1 approx_val_def)
apply (rule conjI)
  apply (clarsimp simp add: iconf_def init_le_Init)
  apply (rule conjI)
  apply (drule widen_cfs_fields, assumption+)
  apply (simp add: hconf_def oconf_def lconf_def)
  apply (erule allE, erule allE, erule impE, assumption)
  apply (simp (no_asm_use))
  apply (simp add: lconf_def)
  apply (erule allE, erule allE, erule impE, assumption)
  apply clarsimp

```

```

apply (clarsimp simp: is_init_def split: val.split)
apply (simp add: h_init_def o_init_def l_init_def)
apply (erule allE, erule allE, erule impE, assumption)
apply (drule hconfD, assumption)
apply (drule widen_cfs_fields, assumption+)
apply (drule oconf_objD, assumption)
apply (clarsimp simp add: l_init_def)
apply (erule allE, erule impE) back apply blast
apply (clarsimp simp add: is_init_def)
apply (clarsimp simp add: init_le_Init corresponds_stk_cons)
apply (blast intro: consistent_init_Init_stk
          consistent_init_pop)

done

```

lemma Putfield\_correct:

```

"[[ wf_prog wt G;
   method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
   ins!pc = Putfield F D;
   wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
   Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
   G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√;
   fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None ]
⇒ G,phi ⊢JVM state'√"

```

proof -

```

assume wf: "wf_prog wt G"
assume meth: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
assume ins: "ins!pc = Putfield F D"
assume wt: "wt_instr (ins!pc) G C rT (phi C sig) maxs
           (fst sig = init) (length ins) et pc"
assume exec: "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
assume conf: "G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"
assume no_x: "fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None"

```

from ins conf meth

obtain ST LT z where

```

heap_ok: "G⊢h hp√" and
pre_alloc: "preallocated hp ihp" and
init_ok: "h_init G hp ihp" and
phi_pc: "((ST,LT),z) ∈ phi C sig!pc" and
is_class_C: "is_class G C" and
frame: "correct_frame G hp ihp (ST,LT) maxl ins (stk,loc,C,sig,pc,r)" and
frames: "correct_frames G hp ihp phi rT sig z r frs"
by (fastsimp elim: correct_stateE)

```

from phi\_pc ins wt

obtain vT oT vT' tST ST' LT' where

```

ST: "ST = (Init vT) # (Init oT) # tST" and

```

```

suc_pc: "Suc pc < length ins" and
phi_suc: "((tST, LT), z) ∈ phi C sig ! Suc pc" and
class_D: "is_class G D" and
field: "field (G, D) F = Some (D, vT')" and
v_less: "G ⊢ Init vT ≤i Init vT'" and
o_less: "G ⊢ Init oT ≤i Init (Class D)"
apply (unfold wt_instr_def eff_defs)
apply clarsimp
apply (drule bspec, assumption)
apply clarsimp
apply (drule subsetD)
apply (rule imageI, assumption)
apply (clarsimp simp add: init_le_Init2)
apply blast
done

```

from ST frame

obtain vt ot stk' where

```

stk : "stk = vt#ot#stk'" and
app_v: "approx_val G hp ihp vt (OK (Init vT))" and
app_o: "approx_val G hp ihp ot (OK (Init oT))" and
app_s: "approx_stk G hp ihp stk' tST" and
app_l: "approx_loc G hp ihp loc LT" and
consi: "consistent_init (vt#ot#stk') loc ((Init vT)#(Init oT)#tST,LT) ihp" and
corr: "fst sig = init →
corresponds (vt#ot#stk') loc (Init vT#Init oT#tST,LT) ihp (fst r) (PartInit C) ∧
(∃ l. fst r = Addr l ∧ hp l ≠ None ∧
(ihp l = PartInit C ∨ (∃ C'. ihp l = Init (Class C'))))" and
len_l: "length loc = Suc (length (snd sig) + maxl)"
by - (erule correct_frameE, simp, blast)

```

from app\_v app\_o obtain

```

"G, hp ⊢ vt :: ≤ vT" and
conf_o: "G, hp ⊢ ot :: ≤ oT" and
is_init_vo: "is_init hp ihp vt" "is_init hp ihp ot"
by (simp add: approx_val_def iconf_def)

```

with wf v\_less have conf\_v: "G, hp ⊢ vt :: ≤ vT'"

by (auto intro: conf\_widen)

from exec ins meth stk no\_x

have "ot ≠ Null" by (auto simp add: split\_beta raise\_system\_xcpt\_def)

with o\_less conf\_o field wf

obtain x C' fs oT' where

```

ot_addr: "ot = Addr x" and
hp_Some: "hp x = Some (Obj C' fs)" and
"G ⊢ (Class C') ≤ oT'"

```

```

    by (fastsimp simp add: conf_def dest!: non_np_objD intro: conf_widen)
  with o_less
  have C': "G ⊢ C' ≤C D"
    by (auto dest: widen_trans)
  with wf field
  have fields: "map_of (fields (G, C')) (F, D) = Some vT'"
    by - (rule widen_cfs_fields)

  let ?hp' = "hp(x↦Obj C' (fs((F, D)↦vt)))"
  and ?f' = "(stk', loc, C, sig, Suc pc, r)"

  from exec ins meth stk ot_addr hp_Some
  have state': "state' = Norm (?hp', ihp, ?f'#frs)"
    by (simp add: raise_system_xcpt_def)

  from hp_Some have hext: "hp ≤| ?hp'" by (rule hext_upd_obj)

  with fields hp_Some conf_v heap_ok
  have hp'_ok: "G ⊢h ?hp' √" by (blast intro: hconf_imp_hconf_field_update )

  from app_v have "is_init hp ihp vt" by (simp add: approx_val_def iconf_def)

  with init_ok hp_Some have hp'_init: "h_init G ?hp' ihp"
    by (rule h_init_field_update)

  from hp_Some heap_ok pre_alloc have pre_alloc': "preallocated ?hp' ihp"
    by (rule preallocated_field_update)

  from corr have corr':
    "fst sig = init ⟶
     corresponds stk' loc (tST, LT) ihp (fst r) (PartInit C) ∧
     (∃ l. fst r = Addr l ∧ ?hp' l ≠ None ∧
      (ihp l = PartInit C ∨ (∃ C'. ihp l = Init (Class C'))))"
    by (clarsimp simp add: corresponds_stk_cons simp del: fun_upd_apply) simp

  from consi stk have "consistent_init stk' loc (tST, LT) ihp"
    by (blast intro: consistent_init_pop)

  with wf app_l app_s len_l suc_pc hext corr'
  have f'_correct:
    "correct_frame G ?hp' ihp (tST, LT) maxl ins (stk',loc,C,sig,Suc pc,r)"
    by (simp add: correct_frame_def)
      (blast intro: approx_stk_imp_approx_stk_sup_heap
              approx_loc_imp_approx_loc_sup_heap)

  from wf frames hp_Some fields conf_v
  have "correct_frames G ?hp' ihp phi rT sig z r frs"
    by - (rule correct_frames_imp_correct_frames_field_update)

```

```

with state' ins meth is_class_C phi_suc hp'_ok hp'_init f'_correct pre_alloc'
show ?thesis by simp (rule correct_stateI)
qed

```

lemma New\_correct:

```

"[[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins!pc = New X;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√;
  fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None ] ]
⇒ G,phi ⊢JVM state'√"

```

proof -

```

assume wf: "wf_prog wt G"
assume meth: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
assume ins: "ins!pc = New X"
assume wt: "wt_instr (ins!pc) G C rT (phi C sig) maxs
            (fst sig = init) (length ins) et pc"
assume exec: "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
assume conf: "G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"
assume no_x: "fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None"

```

from conf meth

obtain ST LT z where

```

heap_ok: "G⊢h hp√" and
init_ok: "h_init G hp ihp" and
phi_pc: "( (ST,LT),z) ∈ phi C sig!pc" and
prealloc: "preallocated hp ihp" and
is_class_C: "is_class G C" and
frame: "correct_frame G hp ihp (ST,LT) maxl ins (stk,loc,C,sig,pc,r)" and
frames: "correct_frames G hp ihp phi rT sig z r frs"
..

```

from phi\_pc ins wt

obtain

```

is_class_X: "is_class G X" and
maxs: "length ST < maxs" and
suc_pc: "Suc pc < length ins" and
phi_suc: "( (UnInit X pc#ST, replace (OK (UnInit X pc)) Err LT), z) ∈ phi C sig
! Suc pc" (is "( (?ST',?LT'),_) ∈ _") and
new_type: "UnInit X pc ∉ set ST"
apply (unfold wt_instr_def eff_def eff_bool_def norm_eff_def)
apply clarsimp
apply (drule bspec, assumption)
apply clarsimp

```

```

    apply (drule subsetD, rule imageI, assumption)
  apply clarsimp
done

obtain oref xp' where
  new_Addr: "new_Addr hp = (oref, xp')"
  by (cases "new_Addr hp")
with ins no_x
obtain hp: "hp oref = None" and "xp' = None"
  by (auto dest: new_AddrD simp add: raise_system_xcpt_def)

with exec ins meth new_Addr
have state':
  "state' = Norm (hp(oref $\mapsto$ blank G X), ihp(oref := UnInit X pc),
    (Addr oref # stk, loc, C, sig, Suc pc, r) # frs)"
  (is "state' = Norm (?hp', ?ihp', ?f # frs)")
  by simp
moreover
from wf hp heap_ok is_class_X
have hp': "G  $\vdash$ h ?hp'  $\checkmark$ "
  by - (rule hconf_imp_hconf_newref,
    auto simp add: oconf_def blank_def dest: fields_is_type)
moreover
from hp heap_ok init_ok
have "h_init G ?hp' ?ihp'" by (rule h_init_newref)
moreover
from hp have sup: "hp  $\leq$ l ?hp'" by (rule hext_new)
from frame obtain
  cons: "consistent_init stk loc (ST, LT) ihp" and
  a_loc: "approx_loc G hp ihp loc LT" and
  a_stk: "approx_stk G hp ihp stk ST" and
  corr: "fst sig = init  $\longrightarrow$ 
    corresponds stk loc (ST, LT) ihp (fst r) (PartInit C)  $\wedge$ 
    ( $\exists$ l. fst r = Addr l  $\wedge$  hp l  $\neq$  None  $\wedge$ 
    (ihp l = PartInit C  $\vee$  ( $\exists$ C'. ihp l = Init (Class C'))))"
  ..
from a_loc
have a_loc': "approx_loc G hp ihp loc ?LT'" by (rule approx_loc_replace_Err)
from corr a_loc a_stk new_type hp
have corr':
  "fst sig = init  $\longrightarrow$ 
    corresponds (Addr oref#stk) loc (?ST', ?LT') ?ihp' (fst r) (PartInit C)  $\wedge$ 
    ( $\exists$ l. fst r = Addr l  $\wedge$  ?hp' l  $\neq$  None  $\wedge$ 
    (?ihp' l = PartInit C  $\vee$  ( $\exists$ C'. ?ihp' l = Init (Class C'))))"
  apply (clarsimp simp add: corresponds_stk_cons simp del: fun_upd_apply)
  apply (drule corresponds_new_val2, assumption+)
  apply (auto intro: corresponds_replace_Err)
done

```



```

from new_type have "OK (UnInit X pc)  $\notin$  set (map OK ST)  $\cup$  set ?LT'"
  by (auto simp add: replace_removes_elem)
with cons a_stk a_loc' hp
have "consistent_init (Addr oref # stk) loc (?ST',?LT') ?ihp'"
  by (blast intro: consistent_init_newref consistent_init_replace_Err)
with hp frame suc_pc wf corr' a_loc'
have "correct_frame G ?hp' ?ihp' (?ST', ?LT') maxl ins ?f"
  apply (unfold correct_frame_def)
  apply (clarsimp simp add: map_eq_Cons conf_def blank_def
    corresponds_stk_cons)
  apply (insert sup, unfold blank_def)
  apply (blast intro: approx_stk_newref
    approx_stk_imp_approx_stk_sup_heap
    approx_loc_imp_approx_loc_sup_heap
    approx_loc_newref
    approx_val_newref
    sup)
done
moreover
from hp frames wf heap_ok is_class_X
have "correct_frames G ?hp' ?ihp' phi rT sig z r frs"
  by (unfold blank_def)
  (rule correct_frames_imp_correct_frames_newref,
  auto simp add: oconf_def dest: fields_is_type)
moreover
from hp prealloc have "preallocated ?hp' ?ihp'" by (rule preallocated_newref)
ultimately
show ?thesis by simp (rule correct_stateI)
qed

```

## Method Invocation

```
lemmas [simp del] = split_paired_Ex
```

```
lemma zip_map [rule_format]:
```

```

"∀a. length a = length b  $\longrightarrow$ 
zip (map f a) (map g b) = map ( $\lambda(x,y). (f x, g y)$ ) (zip a b)"
apply (induct b)
  apply simp
  applyclarsimp
  apply (case_tac aa)
  apply simp+
done

```

```
lemma conf_ClassD:
```

```

"G,h $\vdash$  Addr r ::  $\preceq$  Class C  $\implies$ 
( $\exists$ obj. h r = Some obj  $\wedge$  G  $\vdash$  the_Class (obj_ty obj)  $\preceq$  C C)"
apply (drule conf_RefTD)

```

```

apply clarsimp
apply (case_tac obj)
apply auto
done

```

lemma the\_Class\_widen:

```

"G ⊢ the_Class (obj_ty obj) ≤C D ⇒ G ⊢ obj_ty obj ≤ Class D"
apply (case_tac obj)
apply auto
done

```

lemma Invoke\_correct:

```

"[[ wt_jvm_prog G phi;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Invoke C' mn pTs;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√;
  fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None ]]
⇒ G,phi ⊢JVM state'√"

```

proof -

```

assume wtprog: "wt_jvm_prog G phi"
assume method: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
assume ins: "ins ! pc = Invoke C' mn pTs"
assume wti: "wt_instr (ins!pc) G C rT (phi C sig) maxs
            (fst sig = init) (length ins) et pc"
assume state': "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
assume approx: "G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"
assume no_xcp: "fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None"

```

from wtprog obtain wfmb where

```
wfprog: "wf_prog wfmb G" by (simp add: wt_jvm_prog_def)
```

from approx method obtain s z where

```

heap_ok: "G ⊢ h hp√" and
init_ok: "h_init G hp ihp" and
phi_pc: "(s,z) ∈ phi C sig!pc" and
prealloc: "preallocated hp ihp" and
is_class_C: "is_class G C" and
frame: "correct_frame G hp ihp s maxl ins (stk, loc, C, sig, pc, r)" and
frames: "correct_frames G hp ihp phi rT sig z r frs"
..

```

from ins wti phi\_pc obtain apTs X ST LT D' rT body where

```
is_class: "is_class G C'" and
```

```

s: "s = (rev apTs @ X # ST, LT)" and
l: "length apTs = length pTs" and
w: "∀(x,y)∈set (zip apTs pTs). G ⊢ x ≲i (Init y)" and
mC': "method (G, C') (mn, pTs) = Some (D', rT, body)" and
pc: "Suc pc < length ins" and
eff: "norm_eff (Invoke C' mn pTs) G pc (Suc pc) (phi C sig!pc) ⊆ phi C sig!Suc pc"
and
X: "G ⊢ X ≲i Init (Class C')" and
ni: "mn ≠ init"
apply (clarsimp simp add: wt_instr_def eff_def)
apply (drule bspec, assumption)
apply clarsimp
apply blast
done

from s ins frame obtain
a_stk: "approx_stk G hp ihp stk (rev apTs @ X # ST)" and
a_loc: "approx_loc G hp ihp loc LT" and
init: "consistent_init stk loc s ihp" and
suc_l: "length loc = Suc (length (snd sig) + maxl)"
by (simp add: correct_frame_def)

from a_stk obtain opTs stk' oX where
opTs: "approx_stk G hp ihp opTs (rev apTs)" and
oX: "approx_val G hp ihp oX (OK X)" and
a_stk': "approx_stk G hp ihp stk' ST" and
stk': "stk = opTs @ oX # stk'" and
l_o: "length opTs = length apTs"
      "length stk' = length ST"
by (auto dest!: approx_stk_append_lemma)

from oX have X_oX: "G, hp, ihp ⊢ oX :: ≲i X" by (simp add: approx_val_def)
with wfprog X have oX_conf: "G, hp ⊢ oX :: ≲ (Class C')"
  by (auto simp add: approx_val_def iconf_def init_le_Init2 dest: conf_widen)
from stk' l_o l have oX_pos: "stk ! length pTs = oX" by (simp add: nth_append)
with state' method ins no_xcp oX_conf obtain ref where oX_Addr: "oX = Addr ref"
  by (auto simp add: raise_system_xcpt_def dest: conf_RefTD)
with oX_conf
obtain D obj where
hp_Some: "hp ref = Some obj" and
D_def: "D = the_Class (obj_ty obj)" and
D_le_C': "G ⊢ D ≲C C'"
by (clarsimp dest!: conf_ClassD)

from D_le_C' wfprog mC'
obtain D'' rT' mxl' mxs' ins' et' where
mD: "method (G, D) (mn, pTs) = Some (D'', rT', mxs', mxl', ins', et')"
  "G ⊢ rT' ≲ rT"

```

```

by (auto dest!: subtype_widen_methd) blast

let ?loc' = "oX # rev opTs @ replicate mxl' arbitrary"
let ?f     = "([], ?loc', D'', (mn, pTs), 0, arbitrary)"
let ?f'    = "(stk, loc, C, sig, pc, r)"

from oX_Addr oX_pos hp_Some state' method ins stk' l_o l mD D_def
have state'_val: "state' = Norm (hp, ihp, ?f# ?f' # frs)"
  by (simp add: raise_system_xcpt_def)

from is_class D_le_C' have is_class_D: "is_class G D"
  by (auto dest: subcls_is_class2)
with mD wfprog obtain mD'':
  "method (G, D'') (mn, pTs) = Some (D'', rT', mxs', mxl', ins', et')"
  "is_class G D''"
  by (auto dest: method_in_md)

from wtprog mD''
have start: "wt_start G D'' mn pTs mxl' (phi D'' (mn, pTs)) ∧ ins' ≠ []"
  by (auto dest: wt_jvm_prog_impl_wt_start)

let ?T = "OK (Init (Class D''))"
let ?LT0 = "?T # map OK (map Init pTs) @ replicate mxl' Err"

from ni start have sup_loc:
  "(([],?LT0),D''=Object) ∈ phi D'' (mn,pTs) ! 0"
  by (simp add: wt_start_def)

have c_f: "correct_frame G hp ihp ([], ?LT0) mxl' ins' ?f"
proof -
  have r:
    "approx_loc G hp ihp (replicate mxl' arbitrary) (replicate mxl' Err)"
  by (simp add: approx_loc_def approx_val_Err list_all2_def
    set_replicate_conv_if)

from wfprog mD is_class_D have "G ⊢ Class D ≼ Class D''"
  by (auto dest: method_wf_mdecl)

with hp_Some D_def oX_Addr oX X have a: "approx_val G hp ihp oX ?T"
  by (auto simp add: is_init_def init_le_Init2 the_Class_widen
    approx_val_def iconf_def conf_def)

from w l
have "∀ (x,y)∈set (zip (map (λx. x) apTs) (map Init pTs)). G ⊢ x ≼i y"
  by (simp only: zip_map) auto
hence "∀ (x,y)∈set (zip apTs (map Init pTs)). G ⊢ x ≼i y" by simp
with l
have "∀ (x,y)∈set (zip (rev apTs) (rev (map Init pTs))). G ⊢ x ≼i y"

```

```

    by (auto simp add: zip_rev)
  with wfprog l l_o opTs
  have "approx_loc G hp ihp opTs (map OK (rev (map Init pTs)))"
    by (auto intro: assConv_approx_stk_imp_approx_loc)
  hence "approx_stk G hp ihp opTs (rev (map Init pTs))"
    by (simp add: approx_stk_def)
  hence "approx_stk G hp ihp (rev opTs) (map Init pTs)"
    by (simp add: approx_stk_rev)
  hence "approx_loc G hp ihp (rev opTs) (map OK (map Init pTs))"
    by (simp add: approx_stk_def)
  with r a l_o l
  have loc: "approx_loc G hp ihp ?loc' ?LT0"
    by (auto simp add: approx_loc_append approx_stk_def)

```

```

  from l l_o have "length pTs = length opTs" by auto
  hence "consistent_init [] ?loc' ([],?LT0) ihp"
    by (blast intro: consistent_init_start)

```

```

  with start loc l_o l ni show ?thesis by (simp add: correct_frame_def)

```

qed

```

from state_val heap_ok mD'' ins method phi_pc s l
  frames c_f frame is_class_C ni init_ok prealloc sup_loc
show "G,phi ⊢JVM state'√"
  apply simp
  apply (rule correct_stateI, assumption+)
  apply clarsimp
  apply (intro exI conjI impI)
  apply assumption+
  apply (rule refl)
  apply assumption+
  done

```

qed

lemma Invoke\_special\_correct:

```

"[[ wt_jvm_prog G phi;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Invoke_special C' mn pTs;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√;
  fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None ]]
⇒ G,phi ⊢JVM state'√"

```

proof -

```

  assume wtprog: "wt_jvm_prog G phi"
  assume method: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
  assume ins: "ins ! pc = Invoke_special C' mn pTs"
  assume wti: "wt_instr (ins!pc) G C rT (phi C sig) maxs

```

```

      (fst sig = init) (length ins) et pc"
assume state': "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
assume approx: "G,phi ⊢ JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"
assume no_x:   "fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None"

from wtprog obtain wfmb where wfprog: "wf_prog wfmb G"
  by (simp add: wt_jvm_prog_def)

from approx method obtain s z where
  heap_ok: "G⊢h hp√" and
  init_ok: "h_init G hp ihp" and
  prealloc: "preallocated hp ihp" and
  phi_pc: "(s,z) ∈ phi C sig!pc" and
  is_class_C: "is_class G C" and
  frame:   "correct_frame G hp ihp s maxl ins (stk, loc, C, sig, pc, r)" and
  frames:  "correct_frames G hp ihp phi rT sig z r frs"
  ..

from ins wti phi_pc obtain apTs X ST LT rT' maxs' mxl' ins' et' where
  s: "s = (rev apTs @ X # ST, LT)" and
  l: "length apTs = length pTs" and
  is_class: "is_class G C'" and
  w: "∀(x,y)∈set (zip apTs pTs). G ⊢ x ≤i (Init y)" and
  mC': "method (G, C') (mn, pTs) = Some (C', rT', maxs', mxl', ins', et')" and
  pc: "Suc pc < length ins" and
  eff: "norm_eff (Invoke_special C' mn pTs) G pc (Suc pc) (phi C sig!pc) ⊆
        phi C sig!Suc pc" and
  X: "(∃pc. X = UnInit C' pc) ∨ (X = PartInit C ∧ G ⊢ C <C1 C' ∧ ¬z)" and
  is_init: "mn = init"
  apply (clarsimp simp add: wt_instr_def split_paired_Ex eff_def)
  apply (drule bspec, assumption)
  apply clarsimp
  apply blast
  done

from s ins frame obtain
  a_stk: "approx_stk G hp ihp stk (rev apTs @ X # ST)" and
  a_loc: "approx_loc G hp ihp loc LT" and
  init:  "consistent_init stk loc (rev apTs @ X # ST, LT) ihp" and
  corr:  "fst sig = init → corresponds stk loc s ihp (fst r) (PartInit C) ∧
        (∃l. fst r = Addr l ∧ hp l ≠ None ∧
         (ihp l = PartInit C ∨ (∃C'. ihp l = Init (Class C'))))" and
  suc_l: "length loc = Suc (length (snd sig) + maxl)"
  by (simp add: correct_frame_def)

from a_stk obtain opTs stk' oX where
  opTs: "approx_stk G hp ihp opTs (rev apTs)" and
  oX:   "approx_val G hp ihp oX (OK X)" and

```

```

a_stk': "approx_stk G hp ihp stk' ST" and
stk':   "stk = opTs @ oX # stk'" and
l_o:    "length opTs = length apTs"
        "length stk' = length ST"
by (fastsimp dest!: approx_stk_append_lemma)

from stk' l_o 1 have oX_pos: "stk ! length pTs = oX" by (simp add: nth_append)

from state' method ins no_x oX_pos have "oX ≠ Null"
  by (simp add: raise_system_xcpt_def split: split_if_asm)

from wfprog oX have oX_conf: "G, hp, ihp ⊢ oX :: ≤i X"
  by (simp add: approx_val_def)
with X
obtain ref D fs where
  oX_Addr: "oX = Addr ref" and
  hp_Some: "hp ref = Some (Obj D fs)" and
  D_le_C': "G ⊢ D ≤C C'"
  apply -
  apply (erule disjE)
  apply (fastsimp simp add: iconf_def dest!: conf_ClassD)
  apply (fastsimp simp add: iconf_def dest!: conf_ClassD dest: rtrancl_into_rtrancl)
done

let ?new = "new_Addr hp"
let ?ref' = "fst ?new"
let ?xp' = "snd ?new"
let ?hp' = "hp(?ref' ↦ blank G D)"
let ?loc' = "Addr ?ref' # rev opTs @ replicate mxl' arbitrary"
let ?ihp' = "if C' = Object then
  ihp(?ref' := Init (Class D))
else
  ihp(?ref' := PartInit C')"
let ?r' = "if C' = Object then
  (Addr ?ref', Addr ?ref')
else
  (Addr ?ref', Null)"
let ?f = "([], ?loc', C', (mn, pTs), 0, ?r')"
let ?f' = "(stk, loc, C, sig, pc, r)"

from state' method ins no_x have norm: "?xp' = None"
  by (simp add: split_beta split: split_if_asm)

with oX_Addr oX_pos hp_Some state' method ins stk' l_o 1 mC'
have state'_val: "state' = Norm (?hp', ?ihp', ?f# ?f' # frs)"
  by (simp add: raise_system_xcpt_def split_beta)

obtain ref' xp' where

```

```

    new_Addr: "new_Addr hp = (ref',xp')"
    by (cases "new_Addr hp") auto
with norm have new_ref': "hp ref' = None" by (auto dest: new_AddrD)

from is_class D_le_C' have is_class_D: "is_class G D"
  by (auto dest: subcls_is_class2)

from wtprog is_class mC'
have start: "wt_start G C' mn pTs mxl' (phi C' (mn, pTs)) ∧ ins' ≠ []"
  by (auto dest: wt_jvm_prog_impl_wt_start)

let ?T = "OK (if C' ≠ Object then PartInit C' else Init (Class C'))"
let ?LT0 = "?T # map OK (map Init pTs) @ replicate mxl' Err"

from start is_init
have LT0: "(([], ?LT0), C'=Object) ∈ phi C' (mn, pTs) ! 0"
  by (clarsimp simp add: wt_start_def)

have c_f: "correct_frame G ?hp' ?ihp' ([], ?LT0) mxl' ins' ?f"
proof -
  have r:
    "approx_loc G ?hp' ?ihp' (replicate mxl' arbitrary) (replicate mxl' Err)"
    by (simp add: approx_loc_def approx_val_Err list_all2_def
        set_replicate_conv_if)

  from new_Addr obtain fs where "?hp' ref' = Some (Obj D fs)" by (simp add: blank_def)
  hence "G, ?hp' ⊢ Addr ref' :: ≤ Class C'"
    by (auto simp add: new_Addr D_le_C' intro: conf_obj_AddrI)
  hence a: "approx_val G ?hp' ?ihp' (Addr ?ref') ?T"
    by (auto simp add: approx_val_def iconf_def new_Addr
        blank_def is_init_def)

  from w 1
  have "∀(x,y)∈set (zip (map (λx. x) apTs) (map Init pTs)). G ⊢ x ≤i y"
    by (simp only: zip_map) auto
  hence "∀(x,y)∈set (zip apTs (map Init pTs)). G ⊢ x ≤i y" by simp
  with 1
  have "∀(x,y)∈set (zip (rev apTs) (rev (map Init pTs))). G ⊢ x ≤i y"
    by (auto simp add: zip_rev)
  with wfprog 1 l_o opTs
  have "approx_loc G hp ihp opTs (map OK (rev (map Init pTs)))"
    by (auto intro: assConv_approx_stk_imp_approx_loc)
  hence "approx_stk G hp ihp opTs (rev (map Init pTs))"
    by (simp add: approx_stk_def)
  hence "approx_stk G hp ihp (rev opTs) (map Init pTs)"
    by (simp add: approx_stk_rev)
  hence "approx_loc G hp ihp (rev opTs) (map OK (map Init pTs))"
    by (simp add: approx_stk_def)
  with new_Addr new_ref'

```



```

have "approx_loc G hp ?ihp' (rev opTs) (map OK (map Init pTs))"
  by (auto dest: approx_loc_newref)
moreover
from new_Addr new_ref' have "hp ≤| ?hp'" by simp
ultimately
have "approx_loc G ?hp' ?ihp' (rev opTs) (map OK (map Init pTs))"
  by (rule approx_loc_imp_approx_loc_sup_heap)
with r a l_o l
have loc: "approx_loc G ?hp' ?ihp' ?loc' ?LT0"
  by (auto simp add: approx_loc_append)

from new_Addr new_ref' l_o l have corr':
  "corresponds [] ?loc' ([], ?LT) ?ihp' (Addr ref') (PartInit C'"
  apply (simp add: corresponds_def corr_stk_def corr_loc_cons
    split del: split_if)
  apply (rule conjI)
  apply (simp add: corr_val_def)
  apply (rule corr_loc_start)
  apply clarsimp
  apply (erule disjE)
  apply (erule imageE, erule imageE)
  apply simp
  apply (drule in_set_replicateD)
  apply assumption
  apply simp
  apply blast
done

from l_o l
have "length (rev opTs @ replicate mxl' arbitrary) =
  length (map OK (map Init pTs) @ replicate mxl' Err)" by simp
moreover
have "∀x∈set (map OK (map Init pTs) @ replicate mxl' Err).
  x = Err ∨ (∃t. x = OK (Init t))"
  by (auto dest: in_set_replicateD)
ultimately
have "consistent_init [] ?loc' ([], ?LT0) ?ihp'"
  apply (unfold consistent_init_def)
  apply (unfold corresponds_def)
  apply (simp (no_asm) add: corr_stk_def corr_loc_empty corr_loc_cons
    split del: split_if)
  apply safe
  apply (rule exI)
  apply (rule conjI)
  apply (simp (no_asm))
  apply (rule corr_loc_start)
  apply assumption+
  apply blast

```

```

    apply (rule exI)
    apply (rule conjI)
      defer
        apply (blast intro: corr_loc_start)
    apply (rule impI)
    apply (simp add: new_Addr split del: split_if)
    apply (rule conjI)
      apply (rule refl)
    apply (simp add: corr_val_def new_Addr split: split_if_asm)
  done

  with start loc l_o l corr' new_Addr new_ref'
  show ?thesis by (simp add: correct_frame_def split: split_if_asm)
qed

from a_stk a_loc init suc_l pc new_Addr new_ref' s corr
have c_f': "correct_frame G ?hp' ?ihp' (rev apTs @ X # ST, LT) maxl ins ?f'"
  apply (unfold correct_frame_def)
  apply simp
  apply (rule conjI)
    apply (rule impI)
    apply (rule conjI)
      apply (drule approx_stk_newref, assumption)
      apply (rule approx_stk_imp_approx_stk_sup_heap, assumption)
      apply simp
    apply (rule conjI)
      apply (drule approx_loc_newref, assumption)
      apply (rule approx_loc_imp_approx_loc_sup_heap, assumption)
      apply simp
    apply (rule conjI)
      apply (rule consistent_init_new_val, assumption+)
      apply (rule impI, erule impE, assumption, elim exE conjE)
      apply (rule conjI)
        apply (rule corresponds_new_val2, assumption+)
        apply simp
      apply (rule exI, rule conjI)
        apply (rule impI)
        apply assumption
        apply simp
      apply (rule impI)
      apply (rule conjI)
        apply (drule approx_stk_newref, assumption)
        apply (rule approx_stk_imp_approx_stk_sup_heap, assumption)
        apply simp
      apply (rule conjI)
        apply (drule approx_loc_newref, assumption)
        apply (rule approx_loc_imp_approx_loc_sup_heap, assumption)
        apply simp

```

```

    apply (rule conjI)
      apply (rule consistent_init_new_val, assumption+)
    apply (rule impI, erule impE, assumption, elim exE conjE)
    apply (rule conjI)
      apply (rule corresponds_new_val2, assumption+)
    apply simp
    apply (rule exI, rule conjI)
      apply (rule impI)
      apply (rule conjI)
      apply assumption
      apply simp
    apply simp
  done

from new_Addr new_ref' heap_ok is_class_D wfprog
have hp'_ok: "G ⊢ h ?hp' √"
  by (auto intro: hconf_imp_hconf_newref oconf_blank)

with new_Addr new_ref'
have ihp'_ok: "h_init G ?hp' ?ihp'"
  by (auto intro!: h_init_newref)

from hp_Some new_ref' have neq_ref: "ref ≠ ref'" by auto

with new_Addr oX_Addr X oX hp_Some
have ctor_ok:
  "constructor_ok G ?hp' ?ihp' (Addr ref) C' (C'=Object) ?r'"
  apply (unfold constructor_ok_def)
  apply (simp add: approx_val_def iconf_def)
  apply (erule disjE)
  apply (clarsimp simp add: blank_def)
  apply (elim exE conjE)
  apply (simp add: blank_def)
done

from new_Addr new_ref' frames
have c_frs: "correct_frames G ?hp' ?ihp' phi rT sig z r frs"
  by (auto simp add: blank_def
      intro: correct_frames_imp_correct_frames_newref)

from new_Addr new_ref' prealloc have prealloc': "preallocated ?hp' ?ihp'"
  by (auto intro: preallocated_newref)

from state'_val heap_ok mC' ins method phi_pc s l ctor_ok c_f' c_frs
  frames LT0 c_f is_class_C is_class new_Addr new_ref' hp'_ok ihp'_ok
  prealloc'
show "G, phi ⊢ JVM state' √"
  apply (unfold correct_state_def)

```

```

apply (simp split del: split_if)

apply (intro exI conjI impI)
  apply assumption+
  apply (rule refl)
  apply assumption+
  apply (simp add: oX_pos oX_Addr hp_Some neq_ref)
  apply assumption+
done
qed

lemmas [simp del] = map_append

lemma Return_correct_not_init:
"[[ fst sig ≠ init;
  wt_jvm_prog G phi;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Return;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
proof -
  assume wtjvm: "wt_jvm_prog G phi"
  assume mthd: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
  assume ins: "ins!pc = Return"
  assume wt: "wt_instr (ins!pc) G C rT (phi C sig) maxs
              (fst sig = init) (length ins) et pc"
  assume state: "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
  assume approx:"G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"

  from approx mthd
  obtain s z where
    heap_ok: "G ⊢h hp√" and
    init_ok: "h_init G hp ihp" and
    prealloc:"preallocated hp ihp" and
    classC: "is_class G C" and
    some: "(s,z) ∈ phi C sig ! pc" and
    frame: "correct_frame G hp ihp s maxl ins (stk, loc, C, sig, pc, r)" and
    frames: "correct_frames G hp ihp phi rT sig z r frs"
  ..

  obtain mn pTs where sig: "sig = (mn,pTs)" by (cases sig)
  moreover assume "fst sig ≠ init"
  ultimately have not_init: "mn ≠ init" by simp

  from ins some sig wt
  obtain T ST LT where

```

```

s: "s = (T # ST, LT)" and
T: "G ⊢ T ≲i Init rT"
apply (clarsimp simp add: wt_instr_def eff_def eff_bool_def norm_eff_def)
apply (drule bspec, assumption)
apply clarsimp
apply blast
done

from frame s
obtain rval stk' loc where
  stk: "stk = rval # stk'" and
  val: "approx_val G hp ihp rval (OK T)" and
  approx_stk: "approx_stk G hp ihp stk' ST" and
  approx_val: "approx_loc G hp ihp loc LT" and
  consistent: "consistent_init stk loc (T # ST, LT) ihp" and
  len: "length loc = Suc (length (snd sig) + maxl)" and
  pc: "pc < length ins"
  by (simp add: correct_frame_def) blast

from stk mthd ins state
have "frs = [] ⇒ ?thesis" by (simp add: correct_state_def)
moreover
{ fix f frs' assume frs: "frs = f#frs'"
  obtain stk0 loc0 C0 sig0 pc0 r0 where
    f: "f = (stk0, loc0, C0, sig0, pc0, r0)" by (cases f)

  let ?r' = "(stk0 ! length pTs, snd r0)"
  let ?stk' = "drop (Suc (length pTs)) stk0"
  let ?f' = "(rval#?stk',loc0,C0,sig0,Suc pc0,r0)"

  from stk mthd ins sig f frs not_init state
  have state': "state' = Norm (hp, ihp, ?f' # frs')" by (simp add: split_beta)

  from f frs frames sig
  obtain ST0 LT0 T0 z0 rT0 maxs0 maxl0 ins0 et0 C' apTs D D' rT' body' where
    class_C0: "is_class G C0" and
    methd_C0: "method (G, C0) sig0 = Some (C0, rT0, maxs0, maxl0, ins0, et0)" and
    ins0: "ins0 ! pc0 = Invoke C' mn pTs ∨
           ins0 ! pc0 = Invoke_special C' mn pTs" and
    phi_pc0: "((rev apTs @ T0 # ST0, LT0), z0) ∈ phi C0 sig0 ! pc0" and
    apTs: "length apTs = length pTs" and
    methd_C': "method (G, C') sig = Some (D', rT', body'" "G ⊢ rT ≲ rT'" and
    c_fr: "correct_frame G hp ihp (rev apTs @ T0 # ST0, LT0) maxl0 ins0
           (stk0, loc0, C0, sig0, pc0, r0)" and
    c_frs: "correct_frames G hp ihp phi rT0 sig0 z0 r0 frs'"
  apply simp
  apply (elim conjE exE)
  apply (rule that)

```

```

apply assumption+
apply simp
apply assumption+
apply simp
apply (rule conjI, rule refl)+
apply (rule refl)
apply assumption+
apply simp
apply assumption
done

```

from `c_fr` obtain

```

a_stk0: "approx_stk G hp ihp stk0 (rev apTs @ T0 # ST0)" and
a_loc0: "approx_loc G hp ihp loc0 LT0" and
cons0: "consistent_init stk0 loc0 (rev apTs @ T0 # ST0, LT0) ihp" and
corr0: "fst sig0 = init  $\longrightarrow$ 
corresponds stk0 loc0 (rev apTs @ T0 # ST0, LT0) ihp (fst r0) (PartInit C0)  $\wedge$ 
( $\exists$  l. fst r0 = Addr l  $\wedge$  hp l  $\neq$  None  $\wedge$ 
(ihp l = PartInit C0  $\vee$  ( $\exists$  C'. ihp l = Init (Class C'))))" and
pc0: "pc0 < length ins0" and
lenloc0: "length loc0 = Suc (length (snd sig0) + maxl0)"
by (unfold correct_frame_def) simp

```

from `pc0 wtjvm methd_C0` have `wt0`:

```

"wt_instr (ins0!pc0) G C0 rT0 (phi C0 sig0) maxs0
(fst sig0 = init) (length ins0) et0 pc0"
by - (rule wt_jvm_prog_impl_wt_instr)

```

from `ins0 apTs phi_pc0 not_init sig methd_C' wt0`

obtain

```

Suc_pc0: "Suc pc0 < length ins0" and
phi_Suc_pc0: "((Init rT' # ST0, LT0), z0)  $\in$  phi C0 sig0 ! Suc pc0" and
T0: "G  $\vdash$  T0  $\preceq_i$  Init (Class C')"
apply (simp add: wt_instr_def)
apply (erule disjE)
apply (clarsimp simp add: eff_def eff_bool_def norm_eff_def)
apply (drule bspec, assumption)
apply clarsimp
apply (rotate_tac -7)
apply clarsimp
apply (drule subsetD, rule imageI, assumption)
apply clarsimp
apply simp
done

```

from `wtjvm` obtain `mb` where `wf`: "`wf_prog mb G`" by (simp add: `wt_jvm_prog_def`)

from `a_stk0` obtain `apts v stk0'` where

```

stk0': "stk0 = apts @ v # stk0'" and
len:   "length apts = length apTs" and
v:     "approx_val G hp ihp v (OK T0)" and
a_stk0': "approx_stk G hp ihp stk0' ST0"
by - (drule approx_stk_append_lemma, clarsimp)

from stk0' len v a_stk0' wf apTs val T wf methd_C'
have a_stk0':
  "approx_stk G hp ihp (rval # drop (Suc (length pTs)) stk0) (Init rT'#ST0)"
  apply simp
  apply (rule approx_val_widen, assumption+)
  apply (clarsimp simp add: init_le_Init2)
  apply (erule widen_trans)
  apply assumption
done

from stk0' len apTs cons0
have cons0':
  "consistent_init (rval # drop (Suc (length pTs)) stk0) loc0
    (Init rT'#ST0, LT0) ihp"
  apply simp
  apply (drule consistent_init_append)
  apply simp
  apply (drule consistent_init_pop)
  apply (rule consistent_init_Init_stk)
  apply assumption
done

from stk0' len apTs corr0
have corr0':
  "fst sig0 = init  $\longrightarrow$ 
  corresponds (rval # drop (Suc (length pTs)) stk0) loc0
    (Init rT'#ST0, LT0) ihp (fst r0) (PartInit C0)  $\wedge$ 
  ( $\exists l. \text{fst } r0 = \text{Addr } l \wedge \text{hp } l \neq \text{None} \wedge$ 
    (ihp l = PartInit C0  $\vee$  ( $\exists C'. \text{ihp } l = \text{Init } (\text{Class } C')$ )))"
  apply clarsimp
  apply (drule corresponds_append)
  apply simp
  apply (simp add: corresponds_stk_cons)
done

with cons0' lenloc0 a_loc0 Suc_pc0 a_stk0' wf
have frame0':
  "correct_frame G hp ihp (Init rT' # ST0,LT0) maxl0 ins0 ?f'"
  by (simp add: correct_frame_def)

from state' heap_ok init_ok frame0' c_fr0 class_C0 methd_C0 phi_Suc_pc0 prealloc
have ?thesis by (simp add: correct_state_def) blast

```

```

}
ultimately
show "G,phi ⊢ JVM state'√" by (cases frs, blast+)
qed

```

lemma Return\_correct\_init:

```

" [ fst sig = init;
  wt_jvm_prog G phi;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Return;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs);
  G,phi ⊢ JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢ JVM state'√"

```

proof -

```

assume wtjvm: "wt_jvm_prog G phi"
assume mthd: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
assume ins: "ins!pc = Return"
assume wt: "wt_instr (ins!pc) G C rT (phi C sig) maxs
            (fst sig = init) (length ins) et pc"
assume state: "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
assume approx:"G,phi ⊢ JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"

```

from approx mthd

obtain s z where

```

heap_ok: "G ⊢ h hp√" and
init_ok: "h_init G hp ihp" and
prealloc:"preallocated hp ihp" and
classC: "is_class G C" and
some: "(s,z) ∈ phi C sig ! pc" and
frame: "correct_frame G hp ihp s maxl ins (stk, loc, C, sig, pc, r)" and
frames: "correct_frames G hp ihp phi rT sig z r frs"
..

```

obtain mn pTs where sig: "sig = (mn,pTs)" by (cases sig)

moreover assume sig\_in: "fst sig = init"

ultimately have is\_init: "mn = init" by simp

from ins some sig is\_init wt

obtain T ST LT where

```

s: "s = (T # ST, LT)" and
T: "G ⊢ T ≤i Init rT" and
z: "z"
apply (clarsimp simp add: wt_instr_def eff_def eff_bool_def norm_eff_def)
apply (drule bspec, assumption)
apply clarsimp
apply blast

```



```

done

from frame s sig_in
obtain rval stk' loc where
  stk: "stk = rval # stk'" and
  val: "approx_val G hp ihp rval (OK T)" and
  approx_stk: "approx_stk G hp ihp stk' ST" and
  approx_val: "approx_loc G hp ihp loc LT" and
  consistent: "consistent_init stk loc (T # ST, LT) ihp" and
  corr: "corresponds stk loc (T # ST, LT) ihp (fst r) (PartInit C)" and
  len: "length loc = Suc (length (snd sig) + maxl)"
  by (simp add: correct_frame_def) blast

from stk mthd ins state
have "frs = []  $\implies$  ?thesis" by (simp add: correct_state_def)
moreover
{ fix f frs' assume frs: "frs = f#frs'"
  obtain stk0 loc0 C0 sig0 pc0 r0 where
    f: "f = (stk0, loc0, C0, sig0, pc0, r0)" by (cases f)

  from f frs frames sig is_init
  obtain ST0 LT0 T0 z0 rT0 maxs0 maxl0 ins0 et0 C' apTs D D' rT' body' where
    class_C0: "is_class G C0" and
    methd_C0: "method (G, C0) sig0 = Some (C0, rT0, maxs0, maxl0, ins0, et0)" and
    ins0:      "ins0 ! pc0 = Invoke C' mn pTs  $\vee$ 
               ins0 ! pc0 = Invoke_special C' mn pTs" and
    phi_pc0:  "((rev apTs @ T0 # ST0, LT0), z0)  $\in$  phi C0 sig0 ! pc0" and
    apTs:     "length apTs = length pTs" and
    methd_C': "method (G, C') sig = Some (D', rT', body')" "G  $\vdash$  rT  $\preceq$  rT'" and
    ctor_ok:  "constructor_ok G hp ihp (stk0 ! length apTs) C' z r" and
    c_fr:     "correct_frame G hp ihp (rev apTs @ T0 # ST0, LT0) maxl0 ins0
               (stk0, loc0, C0, sig0, pc0, r0)" and
    c_frs:    "correct_frames G hp ihp phi rT0 sig0 z0 r0 frs'"

  apply simp
  apply (elim conjE exE)
  apply (rule that)
  apply assumption+
  apply simp
  apply simp
  apply assumption+
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
  apply assumption+
  apply simp
  apply assumption
done

```

```

from c_fr
obtain
  a_stk0: "approx_stk G hp ihp stk0 (rev apTs @ T0 # ST0)" and
  a_loc0: "approx_loc G hp ihp loc0 LT0" and
  cons0: "consistent_init stk0 loc0 (rev apTs @ T0 # ST0, LT0) ihp" and
  corr0: "fst sig0 = init  $\longrightarrow$ 
  corresponds stk0 loc0 (rev apTs @ T0 # ST0, LT0) ihp (fst r0) (PartInit C0)  $\wedge$ 
  ( $\exists$  l. fst r0 = Addr l  $\wedge$  hp l  $\neq$  None  $\wedge$ 
  (ihp l = PartInit C0  $\vee$  ( $\exists$  C'. ihp l = Init (Class C'))))" and
  pc0: "pc0 < length ins0" and
  lenloc0: "length loc0 = Suc (length (snd sig0) + maxl0)"
  by (unfold correct_frame_def) simp

from pc0 wtjvm methd_C0 have wt0:
  "wt_instr (ins0!pc0) G C0 rT0 (phi C0 sig0) maxs0
  (fst sig0 = init) (length ins0) et0 pc0"
  by - (rule wt_jvm_prog_impl_wt_instr)

let ?z' = "if  $\exists$  D. T0 = PartInit D then True else z0"
let ?ST' = "Init rT' # replace T0 (Init (theClass T0)) ST0"
let ?LT' = "replace (OK T0) (OK (Init (theClass T0))) LT0"

from ins0 apTs phi_pc0 is_init sig methd_C' wt0
obtain
  Suc_pc0: "Suc pc0 < length ins0" and
  phi_Suc_pc0: "((?ST', ?LT'), ?z')  $\in$  phi C0 sig0 ! Suc pc0" and
  T0: "( $\exists$  pc. T0 = UnInit C' pc)  $\vee$ 
  (T0 = PartInit C0  $\wedge$  G  $\vdash$  C0  $\prec$  C1 C'  $\wedge$   $\neg$ z0)"
  apply (simp add: wt_instr_def)
  apply (erule disjE)
  apply simp
  apply (clarsimp simp add: eff_def eff_bool_def norm_eff_def)
  apply (drule bspec, assumption)
  apply clarsimp
  apply (drule subsetD, rule imageI, assumption)
  apply (rotate_tac -8)
  apply (clarsimp simp add: nth_append)
  done

from a_stk0 obtain apts oX stk0' where
  stk0': "stk0 = apts @ oX # stk0'" "length apts = length apTs" and
  a_val: "approx_val G hp ihp oX (OK T0)" and
  a_stk: "approx_stk G hp ihp stk0' ST0"
  by (auto dest: approx_stk_append_lemma)

with apTs have oX_pos: "stk0!length pTs = oX" by (simp add: nth_append)

let ?c = "snd r"

```

```

let ?r'   = "if r0 = (oX,Null) then (oX, ?c) else r0"
let ?stk' = "rval#(replace oX ?c stk0'"
let ?loc' = "replace oX ?c loc0"
let ?f'   = "(?stk',?loc',CO,sig0,Suc pc0,?r'"

from stk apTs stk0' mthd ins sig f frs is_init state oX_pos
have state': "state' = Norm (hp, ihp, ?f' # frs'" by (simp add: split_beta)

from wtjvm obtain mb where "wf_prog mb G" by (simp add: wt_jvm_prog_def)

from ctor_ok z apTs oX_pos a_val T0
obtain C'' D pc' a c fs1 fs3 where
  a:      "oX = Addr a" and
  ihp_a:  "ihp a = UnInit C'' pc'  $\vee$  ihp a = PartInit D" and
  hp_a:   "hp a = Some (Obj C'' fs1)" and
  c:      "snd r = Addr c" and
  ihp_c:  "ihp c = Init (Class C'')" and
  hp_c:   "hp c = Some (Obj C'' fs3)"
by (simp add: constructor_ok_def, clarify) auto

from a_val a hp_a T0
have "G  $\vdash$  C''  $\preceq$  C C'  $\wedge$  (T0 = PartInit C0  $\longrightarrow$  G  $\vdash$  C''  $\preceq$  C C0)"
  apply -
  apply (erule disjE)
  apply (clarsimp simp add: approx_val_def iconf_def)
  apply (clarsimp simp add: approx_val_def iconf_def)
  apply (simp add: conf_def)
  apply (rule rtrancl_trans, assumption)
  apply blast
done

with c hp_c ihp_c heap_ok T0
have a_val':
  "approx_val G hp ihp (snd r) (OK (Init (theClass T0)))"
  apply (simp add: approx_val_def iconf_def is_init_def)
  apply (erule disjE)
  apply clarsimp
  apply (rule conf_obj_AddrI, assumption+)
  apply clarsimp
  apply (rule conf_obj_AddrI, assumption+)
done

from stk0' cons0 T0
have corr:
  "corresponds stk0' loc0 (ST0,LT0) ihp oX T0"
  apply simp
  apply (erule disjE)
  apply (elim exE)

```

```

    apply (drule consistent_init_append)
      apply simp
    apply (drule consistent_init_corresponds_stk_cons)
      apply blast
    apply (simp add: corresponds_stk_cons)
  apply (drule consistent_init_append)
    apply simp
  apply (drule consistent_init_corresponds_stk_cons)
    apply blast
  apply (simp add: corresponds_stk_cons)
done

from a_val a_val' a_loc0 T0 corr
have a_loc':
  "approx_loc G hp ihp ?loc' ?LT'"
  by (auto elim: approx_loc_replace simp add: corresponds_def)

from wf T methd_C' a_val a_val' a_stk corr T0 val
have a_stk':
  "approx_stk G hp ihp ?stk' ?ST'"
  apply -
  apply clarsimp
  apply (rule conjI)
    apply (clarsimp simp add: approx_val_def iconf_def init_le_Init2)
    apply (rule conf_widen, assumption+)
    apply (erule widen_trans, assumption)
  apply (unfold approx_stk_def)
  apply (erule disjE)
    apply (elim exE conjE)
    apply (drule approx_loc_replace)
      apply assumption back
    apply assumption
    apply blast
    apply (simp add: corresponds_def corr_stk_def)
    apply (simp add: replace_map_OK)
  apply (elim exE conjE)
  apply (drule approx_loc_replace)
    apply assumption back
    apply assumption
    apply blast
    apply (simp add: corresponds_def corr_stk_def)
  apply (simp add: replace_map_OK)
done

from stk0' apTs cons0
have "consistent_init stk0' loc0 (ST0,LT0) ihp"
  apply simp
  apply (drule consistent_init_append, simp)

```

```

    apply (erule consistent_init_pop)
  done

with a_stk a_loc0 a_val
have cons': "consistent_init ?stk' ?loc' (?ST', ?LT') ihp"
  apply -
  apply (rule consistent_init_Init_stk)
  apply (erule consisten_init_replace)
  apply assumption+
  apply (rule refl)
  done

from stk0' len apTs corr0
have
  "fst sig0 = init  $\longrightarrow$ 
  corresponds (rval # stk0') loc0 (Init rT'#ST0, LT0) ihp (fst r0) (PartInit C0)"
  apply clarsimp
  apply (drule corresponds_append)
  apply simp
  apply (simp add: corresponds_stk_cons)
  done

with a_stk a_loc0 a_val
have corr':
  "fst sig0 = init  $\longrightarrow$ 
  corresponds ?stk' ?loc' (?ST', ?LT') ihp (fst r0) (PartInit C0)"
  apply -
  apply clarify
  apply (simp add: corresponds_stk_cons)
  apply (erule corresponds_replace)
  apply assumption+
  apply simp
  apply blast
  done

have fst_r': "fst ?r' = fst r0" by simp

from lenloc0 a_loc' Suc_pc0 a_stk' cons' fst_r' corr' corr0
have c_fr': "correct_frame G hp ihp (?ST', ?LT') maxl0 ins0 ?f'"
  apply (unfold correct_frame_def)
  apply clarify
  apply (clarsimp simp add: map_eq_Cons split del: split_if)
  done

from c_frs
have frs'1:
  "fst sig0  $\neq$  init  $\implies$  correct_frames G hp ihp phi rT0 sig0 ?z' ?r' frs'"
  apply -
  apply (cases frs')

```

```

    apply simp
  apply (clarsimp split del: split_if)
  apply (intro exI conjI)
    apply assumption+
    apply (rule refl)
    apply assumption+
  done
moreover
have frs'2: "frs' = [] ==> correct_frames G hp ihp phi rT0 sig0 ?z' ?r' frs'"
  by simp
moreover
{ fix f'' frs''
  assume cons: "frs' = f'' # frs''"
  assume sig0_in: "fst sig0 = init"

  { assume eq: "oX = fst r0"
    with corr0 sig0_in
    have "corresponds stk0 loc0 (rev apTs @ T0 # ST0, LTO) ihp oX (PartInit C0)"
      by simp
    with stk0'
    have "corresponds (oX#stk0') loc0 (T0#ST0, LTO) ihp oX (PartInit C0)"
      apply simp
      apply (rule corresponds_append)
      apply assumption
      apply simp
    done
    with a ihp_a eq corr0 sig0_in T0
    have "ihp a = PartInit C0"
      by (clarsimp simp add: corresponds_stk_cons corr_val_def)
    with a T0 a_val
    have z0: "T0 = PartInit C0 ∧ ¬z0"
      by (auto simp add: approx_val_def iconf_def)
    from c_frs sig0_in z0 cons
    have "snd r0 = Null"
      apply -
      apply (drule correct_frames_ctor_ok)
      apply clarsimp
      apply (simp add: constructor_ok_def split_beta)
    done
  moreover
  have "r0 = (fst r0, snd r0)" by simp
  ultimately
  have eq_r0: "r0 = (oX, Null)" by (simp add: eq)
  with apTs oX_pos ctor_ok c_frs z cons
  have "correct_frames G hp ihp phi rT0 sig0 True (oX, ?c) frs'"
    apply (clarsimp split del: split_if)
    apply (intro exI conjI)
    apply assumption+
  }
}

```

```

      apply (rule refl)
      apply assumption+
    apply (rule impI, erule impE, assumption)
    apply (rule constructor_ok_pass_val)
      apply assumption
      apply simp
    apply simp
    apply assumption+
  done
with oX_pos apTs z0 eq_r0
have "correct_frames G hp ihp phi rT0 sig0 ?z' ?r' frs'" by simp
}
moreover
{ assume neq: "oX ≠ fst r0"
  with T0 stk0' corr0 sig0_in
  have "∃pc'. T0 = UnInit C' pc'"
    apply simp
    apply (erule disjE)
    apply simp
    apply clarify
    apply (drule corresponds_append)
    apply simp
    apply (simp add: corresponds_stk_cons)
  done
  with c_frs apTs oX_pos neq
  have "correct_frames G hp ihp phi rT0 sig0 ?z' ?r' frs'" by clarsimp
}
ultimately
have "correct_frames G hp ihp phi rT0 sig0 ?z' ?r' frs'" by blast
}
ultimately
have "correct_frames G hp ihp phi rT0 sig0 ?z' ?r' frs'"
  by (cases frs', blast+)

  with state' heap_ok init_ok c_frs class_C0 methd_C0 phi_Suc_pc0 c_fr' prealloc
  have ?thesis by (unfold correct_state_def) fastsimp
}
ultimately
show "G,phi ⊢JVM state'√" by (cases frs, blast+)
qed

```

lemma Return\_correct:

```

"[[ wt_jvm_prog G phi;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Return;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]]"

```

```

=> G,phi ⊢JVM state'√"
  apply (cases "fst sig = init")
  apply (rule Return_correct_init)
  apply assumption+
  apply (rule Return_correct_not_init)
  apply simp
  apply assumption+
  done

```

```
lemmas [simp] = map_append
```

```
lemma Goto_correct:
```

```

"[[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Goto branch;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ] ]
=> G,phi ⊢JVM state'√"
  apply (elim correctE, assumption)
  apply (clarsimp simp add: defs1)
  apply (rule exI)+
  apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
  apply clarsimp
  done

```

```
lemma Ifcmpeq_correct:
```

```

"[[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Ifcmpeq branch;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ] ]
=> G,phi ⊢JVM state'√"
  apply (elim correctE, assumption)
  apply (clarsimp simp add: defs1)
  apply (drule bspec, assumption)
  apply clarsimp
  apply (rule conjI, rule impI)
  apply (rule exI)+
  apply (rule conjI)
  apply (erule subsetD)

```



```

apply (erule imageE2)
apply simp
apply (rule conjI, rule refl)+
apply (rule refl)
apply clarsimp
apply (rule conjI)
  apply (drule consistent_init_pop)+
  apply assumption
apply (rule impI, erule impE, assumption, erule conjE)
  apply (drule corresponds_pop)+
  apply assumption
apply (rule impI)
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply clarsimp
apply (rule conjI)
  apply (drule consistent_init_pop)+
  apply assumption
apply (rule impI, erule impE, assumption, erule conjE)
apply (drule corresponds_pop)+
apply assumption
done

```

lemma Pop\_correct:

```

"[[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Pop;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply clarsimp

```

```

apply (fast dest: consistent_init_pop corresponds_pop)
done

```

lemma Dup\_correct:

```

"[[ wf_prog wt G;
   method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
   ins ! pc = Dup;
   wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
   Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
   G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
   apply (erule subsetD)
   apply (erule imageE2)
apply simp
apply (rule conjI, rule refl)+
apply (rule refl)
apply clarsimp
apply (rule conjI)
   apply (erule consistent_init_Dup)
apply (rule impI, erule impE, assumption, erule conjE)
apply (erule corresponds_Dup)
done

```

lemma Dup\_x1\_correct:

```

"[[ wf_prog wt G;
   method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
   ins ! pc = Dup_x1;
   wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
   Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
   G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
   apply (erule subsetD)
   apply (erule imageE2)
apply simp

```

```

apply (rule conjI, rule refl)+
apply (rule refl)
apply clarsimp
apply (rule conjI)
apply (erule consistent_init_Dup_x1)
apply (rule impI, erule impE, assumption, erule conjE)
apply (erule corresponds_Dup_x1)
done

```

**lemma Dup\_x2\_correct:**

```

"[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Dup_x2;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
apply (erule subsetD)
apply (erule imageE2)
apply simp
apply (rule conjI, rule refl)+
apply (rule refl)
apply clarsimp
apply (rule conjI)
apply (erule consistent_init_Dup_x2)
apply (rule impI, erule impE, assumption, erule conjE)
apply (erule corresponds_Dup_x2)
done

```

**lemma Swap\_correct:**

```

"[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Swap;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
⇒ G,phi ⊢JVM state'√"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons)
apply (drule bspec, assumption)

```

```

apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply clarsimp
apply (rule conjI)
  apply (erule consistent_init_Swap)
apply (rule impI, erule impE, assumption, erule conjE)
apply (erule corresponds_Swap)
done

```

lemma IAdd\_correct:

```

"[[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = IAdd;
  wt_instr (ins!pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ] ]
⇒ G,phi ⊢JVM state'√"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply (clarsimp simp add: approx_val_def iconf_def init_le_Init conf_def)
apply (simp add: is_init_def)
apply (drule consistent_init_pop)+
apply (simp add: corresponds_stk_cons)
apply (blast intro: consistent_init_Init_stk)
done

```

lemma Throw\_correct:

```

"[[ wf_prog wt G;
  method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
  ins ! pc = Throw;
  Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) ;
  G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√;

```

```

fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None ]
⇒ G,phi ⊢JVM state' √"
  by simp

```

lemma Jsr\_correct:

```

"[[wf_prog wt G;
method (G, C) sig = Some (C, rT, maxs, maxl, ins, et);
Some state' = exec (G, Norm (hp, ihp, (stk, loc, C, sig, pc, r) # frs));
G,phi ⊢JVM Norm (hp, ihp, (stk, loc, C, sig, pc, r) # frs) √;
wt_instr (ins ! pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
fst (exec_instr (ins ! pc) G hp ihp stk loc C sig pc r frs) = None; ins ! pc = Jsr b]]
⇒ G,phi ⊢JVM state' √"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply clarsimp
apply (rule conjI)
  apply (simp add: approx_val_def iconf_def is_init_def)
apply (rule conjI)
  apply (erule consistent_init_Init_stk)
apply (rule impI, erule impE, assumption, erule conjE)
apply (simp add: corresponds_def corr_stk_def corr_loc_cons)
done

```

lemma phi\_finite:

```

assumes wt: "wt_jvm_prog G phi"
assumes meth: "method (G, C) sig = Some (C, rT, maxs, maxl, ins, et)"
assumes corr: "G,phi ⊢JVM Norm (hp, ihp, (stk, loc, C, sig, pc, r) # frs) √"
shows "finite (phi C sig!pc)"
proof -
  from corr meth
  have len: "pc < length ins" and "is_class G C"
    by (unfold correct_state_def correct_frame_def) auto
  with wt meth
  have "wt_method G C (fst sig) (snd sig) rT maxs maxl ins et (phi C sig)"
    by (auto dest: method_wf_mdecl simp add: wt_jvm_prog_def wf_mdecl_def)
  with len obtain maxr where
    "set (phi C sig) ⊆ Pow (address_types G maxs maxr (length ins))"

```

```

    "pc < length (phi C sig)"
  by (clarsimp simp add: wt_method_def check_types_def states_def)
  hence "phi C sig ! pc ∈ Pow (address_types G maxs maxr (length ins))"
  by (auto intro!: nth_in)
  thus ?thesis by (auto elim: finite_subset intro: finite_address_types)
qed

```

```

lemma obj_ty_is_RefT [simp, dest!]:

```

```

  "obj_ty x = PrimT T ⇒ False"
  apply (cases x)
  apply auto
  done

```

```

lemma Ret_correct:

```

```

  "[[wt_jvm_prog G phi;
  method (G, C) sig = Some (C, rT, maxs, maxl, ins, et); Some state' = exec (G, Norm (hp,
  ihp, (stk, loc, C, sig, pc, r) # frs));
  G, phi ⊢ JVM Norm (hp, ihp, (stk, loc, C, sig, pc, r) # frs) √;
  wt_instr (ins ! pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  fst (exec_instr (ins ! pc) G hp ihp stk loc C sig pc r frs) = None; ins ! pc = Ret idx]]
  ⇒ G, phi ⊢ JVM state' √"
  apply (frule phi_finite, assumption+)
  apply (elim correctE, assumption)
  apply (clarsimp simp add: defs1 map_eq_Cons)
  apply (simp add: set_SOME_lists finite_imageI)
  apply (drule bspec, assumption)
  apply clarsimp
  apply (drule bspec)
  apply (rule UnI1)
  apply (rule imageI)
  apply (rule imageI)
  apply assumption
  apply (clarsimp simp add: theRA_def)
  apply (subgoal_tac "loc!idx = RetAddr ra")
  apply (simp add: split_def)
  apply (rule exI)+
  apply (rule conjI)
  apply (erule subsetD)
  apply simp
  apply (rule conjI, assumption)
  apply simp
  apply simp
  apply (clarsimp simp add: approx_loc_def list_all2_conv_all_nth)
  apply (erule allE, erule impE, assumption)
  apply (clarsimp simp add: approx_val_def iconf_def conf_def)
  apply (cases "loc!idx")
  apply auto

```

done

lemma conf\_ArrayD:

```
"G, hp ⊢ s :: ≤ T. [] ⇒
s = Null ∨ (∃! T' len en. s = Addr l ∧ hp l = Some (Arr T' len en) ∧ G ⊢ T' ≤ T)"
```

```
apply (cases s)
apply (auto simp add: conf_def)
apply (case_tac z)
apply (auto dest: widen_Array_Array widen_Array2)
done
```

lemma ArrLength\_correct:

```
"[[wf_prog wf_mb G;
method (G, C) sig = Some (C, rT, maxs, maxl, ins, et); Some state' = exec (G, Norm (hp,
ihp, (stk, loc, C, sig, pc, r) # frs));
G, phi ⊢ JVM Norm (hp, ihp, (stk, loc, C, sig, pc, r) # frs) √;
wt_instr (ins ! pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
fst (exec_instr (ins ! pc) G hp ihp stk loc C sig pc r frs) = None; ins ! pc = ArrLength]]
⇒ G, phi ⊢ JVM state' √"
```

```
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons split_beta)
apply (drule bspec, assumption)
applyclarsimp
apply (rule exI)+
apply (rule conjI)
apply (erule subsetD)
apply (erule imageE2)
apply simp
apply (rule conjI, rule refl)+
apply (rule refl)
apply (clarsimp simp add: approx_val_def iconf_def dest!: conf_ArrayD)
apply (erule disjE, simp)
apply (clarsimp simp add: is_init_def)
apply (rule conjI)
apply (drule consistent_init_pop)
apply (erule consistent_init_Init_stk)
apply (rule impI, erule impE, assumption, erule conjE)
apply (drule corresponds_pop)
apply (simp add: corresponds_stk_cons)
done
```

lemma conf\_IntegerD:

```
"G, hp ⊢ v :: ≤ PrimT Integer ⇒ ∃ i. v = Intg i"
apply (unfold conf_def)
apply (cases v)
apply auto
```

done

lemma not\_int:

```
"[¬ int len ≤ i; ¬ i < 0] ⇒ nat i < len"
by (simp add: linorder_not_le nat_less_iff)
```

lemma ArrLoad\_correct:

```
"[[wf_prog wf_mb G;
  method (G, C) sig = Some (C, rT, maxs, maxl, ins, et); Some state' = exec (G, Norm (hp,
ihp, (stk, loc, C, sig, pc, r) # frs));
  G, phi ⊢ JVM Norm (hp, ihp, (stk, loc, C, sig, pc, r) # frs) √;
  wt_instr (ins ! pc) G C rT (phi C sig) maxs (fst sig = init) (length ins) et pc;
  fst (exec_instr (ins ! pc) G hp ihp stk loc C sig pc r frs) = None; ins ! pc = ArrLoad]]
⇒ G, phi ⊢ JVM state' √"
apply (elim correctE, assumption)
apply (clarsimp simp add: defs1 map_eq_Cons split_beta)
apply (drule bspec, assumption)
apply clarsimp
apply (rule exI)+
apply (rule conjI)
  apply (erule subsetD)
  apply (erule imageE2)
  apply simp
  apply (rule conjI, rule refl)+
  apply (rule refl)
apply (clarsimp simp add: approx_val_def iconf_def raise_system_xcpt_def
  dest!: conf_ArrayD conf_IntegerD split: split_if_asm)
apply (clarsimp simp add: is_init_def)
apply (drule hconfD, assumption)
apply (clarsimp simp add: oconf_def)
apply (drule not_int, assumption)
apply (erule allE, erule impE, assumption)
apply clarsimp
apply (rule conjI)
  apply (erule conf_widen, assumption+)
apply (rule conjI)
  apply (clarsimp simp add: h_init_def)
  apply (erule allE, erule allE, erule impE, assumption)
  apply (clarsimp simp add: o_init_def l_init_def)
  apply (erule_tac x="nat i" in allE)
  apply (erule impE) back
  apply simp
  apply (clarsimp simp add: is_init_def)
apply (rule conjI)
  apply (drule consistent_init_pop)+
  apply (erule consistent_init_Init_stk)
apply (rule impI, erule impE, assumption, erule conjE)
```



```

apply (simp add: corresponds_stk_cons)
done

```

```

lemma iconf_Init [iff]:
  "(G, hp, ihp ⊢ v :: ≤ i Init T) = (G, hp ⊢ v :: ≤ T ∧ is_init hp ihp v)"
  by (unfold iconf_def) simp

```

```

lemma oconf_array_update:
  "[[G, hp ⊢ v :: ≤ T; G, hp ⊢ (Arr T l en) √ ] ]
  ⇒ G, hp ⊢ Arr T l (en(i↦v)) √"
  by (simp add: oconf_def lconf_def)

```

```

lemma hconf_array_update:
  "[[G ⊢ h hp √;
  hp oloc = Some (Arr T l en);
  G, hp ⊢ v :: ≤ T]] ⇒
  G ⊢ h hp(oloc ↦ Arr T l (en(i↦v))) √"
  apply (simp add: hconf_def)
  apply (force intro: oconf_imp_oconf_heap_update oconf_array_update
    simp add: obj_ty_def)
done

```

```

lemma h_init_array_update:
  "[[ h_init G hp ihp; hp x = Some (Arr T l en); is_init hp ihp v ] ]
  ⇒ h_init G (hp(x↦Arr T l (en(i↦v)))) ihp"
  apply (unfold h_init_def o_init_def l_init_def)
  apply clarsimp
  apply (rule conjI)
  apply clarify
  apply (rule conjI)
  apply (clarsimp simp add: is_init_def)
  apply clarsimp
  apply (elim allE, erule impE, assumption)
  apply (clarsimp split: split_if_asm)
  apply (elim allE, erule impE, assumption)
  apply (clarsimp simp add: is_init_def)
  apply (clarsimp split: heap_entry.split)
  apply (rule conjI)
  apply clarify
  apply (elim allE, erule impE, assumption, simp,
    elim allE,
    erule impE, rule exI, assumption)
  apply (clarsimp simp add: is_init_def)
  apply clarify

```

```

apply (elim allE, erule impE, assumption) back
apply simp
apply (erule allE, erule impE, simp)
apply (clarsimp simp add: is_init_def)
done

```

lemma ArrStore\_correct:

```

assumes wf: "wf_prog wt G"
assumes meth: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
assumes ins: "ins!pc = ArrStore"
assumes wt: "wt_instr (ins!pc) G C rT (phi C sig) maxs
              (fst sig = init) (length ins) et pc"
assumes exec: "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
assumes conf: "G,phi ⊢ JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs) √"
assumes no_x: "fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None"

```

shows "G,phi ⊢ JVM state' √"

proof -

```

from ins conf meth
obtain ST LT z where
  heap_ok: "G ⊢ h hp √" and
  pre_alloc: "preallocated hp ihp" and
  init_ok: "h_init G hp ihp" and
  phi_pc: "((ST,LT),z) ∈ phi C sig!pc" and
  is_class_C: "is_class G C" and
  frame: "correct_frame G hp ihp (ST,LT) maxl ins (stk,loc,C,sig,pc,r)" and
  frames: "correct_frames G hp ihp phi rT sig z r frs"
by (fastsimp elim: correct_stateE)

```

from phi\_pc ins wt

obtain T T' tST where

```

ST: "ST = T # Init (PrimT Integer) # Init (T'.[]) # tST" and
T: "G ⊢ T ≲i Init T'" and
suc_pc: "Suc pc < length ins" and
phi_suc: "((tST, LT),z) ∈ phi C sig ! Suc pc"
apply (unfold wt_instr_def eff_defs)
apply clarsimp
apply (drule bspec, assumption)
apply clarsimp
apply (drule subsetD)
apply (rule imageI, assumption)
apply clarsimp
apply blast
done

```

from ST frame

obtain val idx aref stk' where

```

stk : "stk = val # idx # aref # stk'" and

```

```

val : "G, hp, ihp ⊢ val :: ⋖i T" and
idx : "G, hp, ihp ⊢ idx :: ⋖i Init (PrimT Integer)" and
aref : "G, hp, ihp ⊢ aref :: ⋖i Init (T'. [])" and
app_s: "approx_stk G hp ihp stk' tST" and
app_l: "approx_loc G hp ihp loc LT" and
consi: "consistent_init stk loc (ST, LT) ihp" and
corr: "fst sig = init →
      corresponds stk loc (ST, LT) ihp (fst r) (PartInit C) ∧
      (∃ l. fst r = Addr l ∧ hp l ≠ None ∧
       (ihp l = PartInit C ∨ (∃ C'. ihp l = Init (Class C'))))" and
len_l: "length loc = Suc (length (snd sig) + maxl)"
by (clarsimp elim!: correct_frameE simp add: approx_val_def) blast

from exec ins meth stk no_x
have "aref ≠ Null" by (auto simp add: split_beta raise_system_xcpt_def)
with aref
obtain x T'' len en where
  addr: "aref = Addr x" and
  hp_Some: "hp x = Some (Arr T'' len en)"
  by (fastsimp simp add: iconf_def dest!: conf_ArrayD)
moreover
from idx obtain i where
  i: "idx = Intg i"
  by (fastsimp simp add: conf_def dest: conf_IntegerD)
moreover note exec ins meth stk no_x
ultimately
obtain val_T'': "G, hp ⊢ val :: ⋖ T''"
  and bounds: "¬i < 0 ∧ ¬int len ≤ i"
  by (auto simp add: raise_system_xcpt_def split: split_if_asm)

let ?hp' = "hp(x ↦ Arr T'' len (en(nat i ↦ val)))"
and ?f' = "(stk', loc, C, sig, Suc pc, r)"

from exec ins meth stk addr i bounds hp_Some val_T''
have state': "state' = Norm (?hp', ihp, ?f'#frs)"
  by (simp add: raise_system_xcpt_def)

from hp_Some have hext: "hp ≤ | ?hp'" by (rule hext_upd_arr)

from heap_ok hp_Some val_T''
have hp'_ok: "G ⊢ h ?hp' √" by (rule hconf_array_update)
moreover
from val T have "is_init hp ihp val" by (clarsimp simp add: init_le_Init2)
with init_ok hp_Some have hp'_init: "h_init G ?hp' ihp"
  by (rule h_init_array_update)
moreover
from hp_Some heap_ok pre_alloc have pre_alloc': "preallocated ?hp' ihp"
  by (rule preallocated_array_update)

```

```

moreover
from corr ST stk have corr':
  "fst sig = init  $\longrightarrow$ 
  corresponds stk' loc (tST, LT) ihp (fst r) (PartInit C)  $\wedge$ 
  ( $\exists l. \text{fst } r = \text{Addr } l \wedge ?hp' \ l \neq \text{None} \wedge$ 
  (ihp l = PartInit C  $\vee$  ( $\exists C'. \text{ihp } l = \text{Init (Class } C')$ )))"
  by (clarsimp simp add: corresponds_stk_cons simp del: fun_upd_apply) simp
moreover
from consi ST stk have "consistent_init stk' loc (tST, LT) ihp"
  by (blast intro: consistent_init_pop)
with wf app_l app_s len_l suc_pc hext corr' ST stk
have f'_correct:
  "correct_frame G ?hp' ihp (tST, LT) maxl ins (stk',loc,C,sig,Suc pc,r)"
  by (simp add: correct_frame_def)
  (blast intro: approx_stk_imp_approx_stk_sup_heap
  approx_loc_imp_approx_loc_sup_heap)
moreover
from frames hp_Some
have "correct_frames G ?hp' ihp phi rT sig z r frs"
  by (rule correct_frames_array_update)
moreover
note state' ins meth is_class_C phi_suc
ultimately
show ?thesis by simp (rule correct_stateI)
qed

```

**lemma** *oconf\_blank\_arr*:

```

"G, hp  $\vdash$  blank_arr T (nat l)  $\surd$ "
apply (unfold oconf_def blank_arr_def conf_def)
apply (cases T)
apply auto
apply (case_tac prim_ty)
apply auto
done

```

**lemma** *h\_init\_new\_array*:

```

" $\llbracket hp \ r = \text{None}; G \vdash h \ hp \ \surd; h\_init \ G \ hp \ ihp \rrbracket \implies h\_init \ G \ (hp(r \mapsto \text{blank\_arr } T \ l)) \ (ihp(r := T'))$ "
apply (unfold h_init_def o_init_def l_init_def blank_arr_def)
apply clarsimp
apply (rule conjI)
  apply clarsimp
  apply (rule is_init_default_val)
apply (clarsimp split: heap_entry.split)
apply (rule conjI)
  apply clarify
  apply (elim allE, erule impE, assumption)

```

```

apply simp
apply (elim allE, erule impE)
  apply fastsimp
apply clarsimp
apply (simp add: is_init_def)
apply (drule hconfD, assumption)
apply (drule oconf_objD, assumption)
apply clarsimp
apply (drule new_locD, assumption)
apply simp
apply clarify
apply (elim allE, erule impE, assumption)
apply simp
apply (elim allE, erule impE)
  apply fastsimp
apply clarsimp
apply (simp add: is_init_def)
apply (drule hconfD, assumption)
apply (drule oconf_arrD, assumption)
apply clarsimp
apply (drule new_locD, assumption)
apply simp
done

```

lemma ArrNew\_correct:

```

assumes wf: "wf_prog wt G"
assumes meth: "method (G,C) sig = Some (C,rT,maxs,maxl,ins,et)"
assumes ins: "ins!pc = ArrNew T"
assumes wt: "wt_instr (ins!pc) G C rT (phi C sig) maxs
              (fst sig = init) (length ins) et pc"
assumes exec: "Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)"
assumes conf: "G,phi ⊢ JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√"
assumes no_x: "fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs) = None"

```

shows "G,phi ⊢ JVM state' √"

proof -

```

from ins conf meth
obtain ST LT z where
  heap_ok: "G⊢h hp√" and
  pre_alloc: "preallocated hp ihp" and
  init_ok: "h_init G hp ihp" and
  phi_pc: "((ST,LT),z) ∈ phi C sig!pc" and
  is_class_C: "is_class G C" and
  frame: "correct_frame G hp ihp (ST,LT) maxl ins (stk,loc,C,sig,pc,r)" and
  frames: "correct_frames G hp ihp phi rT sig z r frs"
by (fastsimp elim: correct_stateE)

```

```

from phi_pc ins wt
obtain ST' where
  ST:      "ST = Init (PrimT Integer) # ST'" and
  type:    "is_type G T" "noRA T" "Suc (dim T) < max_dim" and
  suc_pc:  "Suc pc < length ins" and
  phi_suc: "((Init (T.[])) # ST', LT), z) ∈ phi C sig ! Suc pc" (is "((?ST',_),_) ∈ _")
  apply (unfold wt_instr_def eff_defs)
  apply clarsimp
  apply (drule bspec, assumption)
  apply clarsimp
  apply (drule subsetD)
  apply (rule imageI, assumption)
  apply clarsimp
  done

from ST frame
obtain len stk' where
  stk : "stk = len # stk'" and
  len : "G, hp, ihp ⊢ len :: ≤i Init (PrimT Integer)" and
  app_s: "approx_stk G hp ihp stk' ST'" and
  app_l: "approx_loc G hp ihp loc LT" and
  consi: "consistent_init stk loc (ST,LT) ihp" and
  corr: "fst sig = init →
        corresponds stk loc (ST,LT) ihp (fst r) (PartInit C) ∧
        (∃ l. fst r = Addr l ∧ hp l ≠ None ∧
         (ihp l = PartInit C ∨ (∃ C'. ihp l = Init (Class C'))))" and
  len_l: "length loc = Suc (length (snd sig) + maxl)"
  by (clarsimp elim!: correct_frameE simp add: approx_val_def) blast

from len obtain l where l: "len = Intg l" by (clarsimp dest!: conf_IntegerD)

obtain ref xp' where
  new_Addr: "new_Addr hp = (ref, xp'"
  by (cases "new_Addr hp")
with ins l stk no_x
obtain hp: "hp ref = None" and "xp' = None" and pos: "¬ l < 0"
  by (auto dest: new_AddrD simp add: raise_system_xcpt_def split: split_if_asm)

with exec ins meth new_Addr l stk
have state':
  "state' = Norm (hp(ref↦blank_arr T (nat l)), ihp(ref := Init (T.[])),
                 (Addr ref # stk', loc, C, sig, Suc pc, r) # frs)"
  (is "state' = Norm (?hp', ?ihp', ?f # frs)")
  by (simp add: raise_system_xcpt_def)
moreover
from hp heap_ok have hp': "G ⊢ h ?hp' √" by (intro hconf_imp_hconf_newref oconf_blank_arr)

```

```

moreover
from hp heap_ok init_ok
have "h_init G ?hp' ?ihp'" by (rule h_init_new_array)
moreover
from hp have sup: "hp ≤| ?hp'" by (rule hext_new)
from frame obtain
  cons: "consistent_init stk loc (ST, LT) ihp" and
  a_loc: "approx_loc G hp ihp loc LT" and
  a_stk: "approx_stk G hp ihp stk ST" and
  corr: "fst sig = init →
    corresponds stk loc (ST, LT) ihp (fst r) (PartInit C) ∧
    (∃l. fst r = Addr l ∧ hp l ≠ None ∧
      (ihp l = PartInit C ∨ (∃C'. ihp l = Init (Class C'))))"
..
have ref_conf: "G, hp(ref ↦ blank_arr T (nat l)) ⊢ Addr ref :: ≤ T. []"
  by (unfold conf_def blank_arr_def) auto
from corr a_loc a_stk hp ST stk
have corr':
  "fst sig = init →
    corresponds (Addr ref # stk') loc (?ST', LT) ?ihp' (fst r) (PartInit C) ∧
    (∃l. fst r = Addr l ∧ ?hp' l ≠ None ∧
      (?ihp' l = PartInit C ∨ (∃C'. ?ihp' l = Init (Class C'))))"
  apply (clarsimp simp add: corresponds_stk_cons simp del: fun_upd_apply)
  apply (auto intro: corresponds_new_val2)
done
from cons ST stk a_loc a_stk hp
have "consistent_init (Addr ref # stk') loc (?ST', LT) ?ihp'"
  by (auto intro: consistent_init_Init_stk consistent_init_new_val dest: consistent_init_pop)
with hp sup frame suc_pc wf corr' a_stk a_loc stk ST ref_conf
have "correct_frame G ?hp' ?ihp' (?ST', LT) maxl ins ?f"
  apply (unfold correct_frame_def)
  apply (clarsimp simp add: corresponds_stk_cons)
  apply (rule conjI)
  apply (clarsimp simp add: approx_val_def is_init_def)
  apply (blast intro: approx_stk_newref
    approx_stk_imp_approx_stk_sup_heap
    approx_loc_imp_approx_loc_sup_heap
    approx_loc_newref
    approx_val_newref
    sup)
done
moreover
from hp frames
have "correct_frames G ?hp' ?ihp' phi rT sig z r frs"
  by (rule correct_frames_imp_correct_frames_newref)
moreover
from hp pre_alloc have "preallocated ?hp' ?ihp'" by (rule preallocated_newref)
ultimately

```

```

  show ?thesis by simp (rule correct_stateI)
qed

```

The next theorem collects the results of the sections above, i.e. exception handling and the execution step for each instruction. It states type safety for single step execution: in well-typed programs, a conforming state is transformed into another conforming state when one instruction is executed.

```

theorem instr_correct:
  "[ wt_jvm_prog G phi;
    method (G,C) sig = Some (C,rT,maxs,maxl,ins,et);
    Some state' = exec (G, None, hp, ihp, (stk,loc,C,sig,pc,r)#frs);
    G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√ ]
  ⇒ G,phi ⊢JVM state'√"
  apply (frule wt_jvm_prog_impl_wt_instr_cor)
  apply assumption+
  apply (cases "fst (exec_instr (ins!pc) G hp ihp stk loc C sig pc r frs)")
  defer
  apply (erule xcpt_correct, assumption+)
  apply (cases "ins!pc")
  prefer 21
  apply (rule Ret_correct, assumption+)
  prefer 8
  apply (rule Invoke_correct, assumption+)
  prefer 8
  apply (rule Invoke_special_correct, assumption+)
  prefer 8
  apply (rule Return_correct, assumption+)
  prefer 5
  apply (rule Getfield_correct, assumption+)
  prefer 6
  apply (rule Checkcast_correct, assumption+)

  apply (unfold wt_jvm_prog_def)
  apply (rule Load_correct, assumption+)
  apply (rule Store_correct, assumption+)
  apply (rule LitPush_correct, assumption+)
  apply (rule New_correct, assumption+)
  apply (rule Putfield_correct, assumption+)
  apply (rule Pop_correct, assumption+)
  apply (rule Dup_correct, assumption+)
  apply (rule Dup_x1_correct, assumption+)
  apply (rule Dup_x2_correct, assumption+)
  apply (rule Swap_correct, assumption+)
  apply (rule IAdd_correct, assumption+)
  apply (rule Goto_correct, assumption+)
  apply (rule Ifcmpeq_correct, assumption+)
  apply (rule Throw_correct, assumption+)
  apply (rule Jsr_correct, assumption+)
  apply (rule ArrLoad_correct, assumption+)
  apply (rule ArrStore_correct, assumption+)
  apply (rule ArrLength_correct, assumption+)
  apply (rule ArrNew_correct, assumption+)
done

```



## 4.22.4 Main

**lemma** *correct\_state\_impl\_Some\_method*:

```
"G,phi ⊢JVM (None, hp, ihp, (stk,loc,C,sig,pc,r)#frs)√
⇒ ∃meth. method (G,C) sig = Some(C,meth)"
by (auto simp add: correct_state_def Let_def)
```

**lemma** *BV\_correct\_1* [rule\_format]:

```
"∧state. [ wt_jvm_prog G phi; G,phi ⊢JVM state√ ]
⇒ exec (G,state) = Some state' → G,phi ⊢JVM state'√"
apply (simp only: split_tupled_all)
apply (rename_tac xp hp ihp frs)
apply (case_tac xp)
  apply (case_tac frs)
    apply simp
  apply (simp only: split_tupled_all)
  apply hypsubst
  apply (frule correct_state_impl_Some_method)
  apply (force intro: instr_correct)
apply (case_tac frs)
apply simp_all
done
```

**lemma** *L0*:

```
"[ xp=None; frs≠[] ] ⇒ (∃state'. exec (G,xp,hp,ihp,frs) = Some state'"
by (clarsimp simp add: neq_Nil_conv split_beta)
```

**lemma** *L1*:

```
"[wt_jvm_prog G phi; G,phi ⊢JVM (xp,hp,ihp,frs)√; xp=None; frs≠[]]
⇒ ∃state'. exec(G,xp,hp,ihp,frs) = Some state' ∧ G,phi ⊢JVM state'√"
apply (drule L0)
apply assumption
apply (fast intro: BV_correct_1)
done
```

**theorem** *BV\_correct* [rule\_format]:

```
"[ wt_jvm_prog G phi; G ⊢ s -jvm→ t ] ⇒ G,phi ⊢JVM s√ → G,phi ⊢JVM t√"
apply (unfold exec_all_def)
apply (erule rtrancl_induct)
  apply simp
apply (auto intro: BV_correct_1)
done
```

**theorem** *BV\_correct\_implies\_approx*:

```
"[ wt_jvm_prog G phi;
  G ⊢ s0 -jvm→ (None,hp,ihp,(stk,loc,C,sig,pc,r)#frs); G,phi ⊢JVM s0 √ ]
⇒ ∃ST LT z. ((ST,LT),z) ∈ (phi C sig) ! pc ∧
  approx_stk G hp ihp stk ST ∧
  approx_loc G hp ihp loc LT"
apply (drule BV_correct)
apply assumption+
apply (clarsimp simp add: correct_state_def correct_frame_def split_def)
```

```

        split: option.splits)
apply blast
apply blast
done

```

```

lemma hconf_start:
  fixes G :: jvm_prog ("Γ")
  assumes wf: "wf_prog wf_mb Γ"
  shows "Γ ⊢ h (start_heap Γ) √"
  apply (unfold hconf_def start_heap_def)
  apply (auto simp add: blank_def oconf_def split: split_if_asm)
  apply (simp add: fields_is_type [OF _ wf is_class_xcpt [OF wf]])+
  done

```

```

lemma hinit_start:
  fixes G :: jvm_prog ("Γ")
  assumes wf: "wf_prog wf_mb Γ"
  shows "h_init Γ (start_heap Γ) start_iheap"
  apply (unfold h_init_def start_heap_def start_iheap_def)
  apply (auto simp add: blank_def o_init_def l_init_def
                    is_init_def split: split_if_asm)
  apply (auto simp add: init_vars_def map_of_map)
  done

```

```

lemma consistent_init_start:
  "consistent_init [] (Null#replicate n arbitrary) ([], OK (Init (Class C))#replicate
n Err) hp"
  apply (induct n)
  apply (auto simp add: consistent_init_def corresponds_def corr_stk_def corr_loc_def)
  done

```

```

lemma BV_correct_initial:
  fixes G :: jvm_prog ("Γ") and Phi :: prog_type ("Φ")
  shows
    "wt_jvm_prog Γ Φ ⇒ is_class Γ C ⇒ method (Γ,C) (m,[]) = Some (C, b) ⇒ m ≠ init
    ⇒ Γ, Φ ⊢ JVM start_state G C m √"
  apply (cases b)
  apply (unfold start_state_def)
  apply (unfold correct_state_def)
  apply (auto simp add: preallocated_start)
  apply (simp add: wt_jvm_prog_def hconf_start)
  apply (simp add: wt_jvm_prog_def hinit_start)
  apply (drule wt_jvm_prog_impl_wt_start, assumption+)
  apply (clarsimp simp add: wt_start_def)
  apply (auto simp add: correct_frame_def)
  apply (simp add: approx_stk_def)
  apply (rule exI, rule conjI)
  apply (rule exI, assumption)
  apply clarsimp
  apply (simp add: consistent_init_start)
  apply (auto simp add: approx_val_def iconf_def is_init_def dest!: widen_RefT)
  done

```

```

theorem typesafe:

```

```

fixes G :: jvm_prog ("Γ") and Phi :: prog_type ("Φ")
assumes welltyped: "wt_jvm_prog Γ Φ" and
        main: "is_class Γ C" "method (Γ,C) (m,[]) = Some (C, b)" "m ≠ init"
shows
  "G ⊢ start_state Γ C m -jvm→ s ⇒ Γ, Φ ⊢ JVM s √"
proof -
  from welltyped main
  have "Γ, Φ ⊢ JVM start_state Γ C m √" by (rule BV_correct_initial)
  moreover
  assume "G ⊢ start_state Γ C m -jvm→ s"
  ultimately
  show "Γ, Φ ⊢ JVM s √" using welltyped by - (rule BV_correct)
qed
end

```

## 4.23 Welltyped Programs produce no Type Errors

theory *BVNoTypeError* = *JVMDefensive* + *BVSpecTypeSafe*:

Some simple lemmas about the type testing functions of the defensive JVM:

lemma *typeof\_NoneD* [*simp*,*dest*]:

"typeof ( $\lambda v. \text{None}$ )  $v = \text{Some } x \implies \neg \text{isAddr } v$ "  
by (cases  $v$ ) auto

lemma *isRef\_def2*:

"isRef  $v = (v = \text{Null} \vee (\exists \text{loc}. v = \text{Addr } \text{loc}))$ "  
by (cases  $v$ ) (auto simp add: *isRef\_def*)

lemma *isRef* [*simp*]:

" $G, hp, ihp \vdash v :: \preceq_i \text{Init } (\text{RefT } T) \implies \text{isRef } v$ "  
by (cases  $v$ ) (auto simp add: *iconf\_def conf\_def isRef\_def*)

lemma *isIntg* [*simp*]:

" $G, hp, ihp \vdash v :: \preceq_i \text{Init } (\text{PrimT } \text{Integer}) \implies \text{isIntg } v$ "  
by (cases  $v$ ) (auto simp add: *iconf\_def conf\_def*)

declare *approx\_loc\_len* [*simp*] *approx\_stk\_len* [*simp*]

lemma *list\_all2I*:

" $\forall (x,y) \in \text{set } (\text{zip } a \ b). P \ x \ y \implies \text{length } a = \text{length } b \implies \text{list\_all2 } P \ a \ b$ "  
by (simp add: *list\_all2\_def*)

lemma *app'Store*[*simp*]:

"app' (Store  $\text{idx}, G, C', pc, \text{maxs}, rT, (ST,LT)$ ) = ( $\exists T \ ST'. ST = T\#ST' \wedge \text{idx} < \text{length } LT$ )"  
by (cases  $ST$ , auto)

lemma *app'GetField*[*simp*]:

"app' (Getfield  $F \ C, G, C', pc, \text{maxs}, rT, (ST,LT)$ ) =  
( $\exists oT \ vT \ ST'. ST = oT\#ST' \wedge \text{is\_class } G \ C \wedge$   
 $\text{field } (G,C) \ F = \text{Some } (C, vT) \wedge G \vdash oT \preceq_i (\text{Init } (\text{Class } C))$ )"  
by (cases  $ST$ , auto)

lemma *app'PutField*[*simp*]:

"app' (Putfield  $F \ C, G, C', pc, \text{maxs}, rT, (ST,LT)$ ) =  
( $\exists vT \ vT' \ oT \ ST'. ST = vT\#oT\#ST' \wedge \text{is\_class } G \ C \wedge$   
 $\text{field } (G,C) \ F = \text{Some } (C, vT') \wedge$   
 $G \vdash oT \preceq_i \text{Init } (\text{Class } C) \wedge G \vdash vT \preceq_i \text{Init } vT'$ )"  
apply rule  
defer

```

apply clarsimp
apply (cases ST)
apply simp
apply (cases "tl ST")
apply auto
done

```

```

lemma app'Checkcast[simp]:
"app' (Checkcast C, G, C', pc, maxs, rT, (ST,LT)) =
  ( $\exists rT ST'. ST = \text{Init (RefT rT)\#ST}' \wedge \text{is\_class G C}$ )"
apply rule
defer
apply clarsimp
apply (cases ST)
apply simp
apply (cases "hd ST")
defer
apply simp
apply simp
apply (case_tac ty)
apply auto
done

```

```

lemma app'Pop[simp]:
"app' (Pop, G, C', pc, maxs, rT, (ST,LT)) = ( $\exists T ST'. ST = T\#ST'$ )"
by (cases ST, auto)

```

```

lemma app'Dup[simp]:
"app' (Dup, G, C', pc, maxs, rT, (ST,LT)) =
  ( $\exists T ST'. ST = T\#ST' \wedge \text{length ST} < \text{maxs}$ )"
by (cases ST, auto)

```

```

lemma app'Dup_x1[simp]:
"app' (Dup_x1, G, C', pc, maxs, rT, (ST,LT)) =
  ( $\exists T1 T2 ST'. ST = T1\#T2\#ST' \wedge \text{length ST} < \text{maxs}$ )"
by (cases ST, simp, cases "tl ST", auto)

```

```

lemma app'Dup_x2[simp]:
"app' (Dup_x2, G, C', pc, maxs, rT, (ST,LT)) =
  ( $\exists T1 T2 T3 ST'. ST = T1\#T2\#T3\#ST' \wedge \text{length ST} < \text{maxs}$ )"
by (cases ST, simp, cases "tl ST", simp, cases "tl (tl ST)", auto)

```

```

lemma app'Swap[simp]:

```

```
"app' (Swap, G, C', pc, maxs, rT, (ST,LT)) = ( $\exists$  T1 T2 ST'. ST = T1#T2#ST')"  
by (cases ST, simp, cases "tl ST", auto)
```

```
lemma app'IAdd[simp]:
```

```
"app' (IAdd, G, C', pc, maxs, rT, (ST,LT)) =  
( $\exists$  ST'. ST = Init (PrimT Integer)#Init (PrimT Integer)#ST')"  
by (cases ST, simp, cases "tl ST", auto)
```

```
lemma app'Ifcmpeq[simp]:
```

```
"app' (Ifcmpeq b, G, C', pc, maxs, rT, (ST,LT)) =  
( $\exists$  T1 T2 ST'. ST = Init T1#Init T2#ST'  $\wedge$   $0 \leq b + \text{int } pc \wedge$   
( $\exists p. T1 = \text{PrimT } p \wedge T1 = T2$ )  $\vee$   
( $\exists r r'. T1 = \text{RefT } r \wedge T2 = \text{RefT } r'$ )))"  
apply auto  
apply (cases ST)  
apply simp  
apply (cases "tl ST")  
apply (case_tac a)  
apply auto  
apply (case_tac a)  
apply auto  
apply (case_tac aa)  
apply (case_tac ty)  
apply auto  
done
```

```
lemma app'Return[simp]:
```

```
"app' (Return, G, C', pc, maxs, rT, (ST,LT)) =  
( $\exists$  T ST'. ST = T#ST'  $\wedge$  G  $\vdash$  T  $\preceq_i$  Init rT)"  
by (cases ST, auto)
```

```
lemma app'Throw[simp]:
```

```
"app' (Throw, G, C', pc, maxs, rT, (ST,LT)) =  
( $\exists$  ST' r. ST = Init (RefT r)#ST')"  
apply (cases ST, simp)  
apply (cases "hd ST")  
apply auto  
done
```

```
lemma app'Invoke[simp]:
```

```
"app' (Invoke C mn fpTs, G, C', pc, maxs, rT, ST, LT) =  
( $\exists$  apTs X ST' mD' rT' b' z.  
ST = (rev apTs) @ X # ST'  $\wedge$  mn  $\neq$  init  $\wedge$ 
```

```

length apTs = length fpTs ∧ is_class G C ∧
(∀ (aT,fT) ∈ set(zip apTs fpTs). G ⊢ aT ≤i Init fT) ∧
method (G,C) (mn,fpTs) = Some (mD', rT', b') ∧ G ⊢ X ≤i Init (Class C))"
(is "?app ST LT = ?P ST LT")
proof
  assume "?P ST LT" thus "?app ST LT" by (auto simp add: min_def list_all2_def)
next
  assume app: "?app ST LT"
  hence 1: "length fpTs < length ST" by simp
  obtain xs ys where xs: "ST = xs @ ys" "length xs = length fpTs"
  proof -
    have "ST = take (length fpTs) ST @ drop (length fpTs) ST" by simp
    moreover from 1 have "length (take (length fpTs) ST) = length fpTs"
      by (simp add: min_def)
    ultimately show ?thesis ..
  qed
  obtain apTs where
    "ST = (rev apTs) @ ys" and "length apTs = length fpTs"
  proof -
    have "ST = rev (rev xs) @ ys" by simp
    with xs show ?thesis by - (rule that, assumption, simp)
  qed
  moreover
  from 1 xs obtain X ST' where "ys = X#ST'" by (auto simp add: neq_Nil_conv)
  ultimately
  have "ST = (rev apTs) @ X # ST'" "length apTs = length fpTs" by auto
  with app
  show "?P ST LT"
    apply (clarsimp simp add: list_all2_def min_def)
    apply ((rule exI)+, (rule conjI)?)+
    apply auto
    done
qed

lemma app'Invoke_special[simp]:
"app' (Invoke_special C mn fpTs, G, C', pc, maxs, rT, (ST,LT)) =
(∃ apTs X ST' rT' b' z.
  ST = (rev apTs) @ X # ST' ∧ mn = init ∧
  length apTs = length fpTs ∧ is_class G C ∧
  (∀ (aT,fT) ∈ set(zip apTs fpTs). G ⊢ aT ≤i (Init fT)) ∧
  method (G,C) (mn,fpTs) = Some (C, rT', b') ∧
  ((∃ pc. X = UnInit C pc) ∨ (X = PartInit C' ∧ G ⊢ C' <C1 C)))"
(is "?app ST LT = ?P ST LT")
proof
  assume "?P ST LT" thus "?app ST LT" by (auto simp add: min_def list_all2_def nth_append)
next
  assume app: "?app ST LT"
  hence 1: "length fpTs < length ST" by simp

```

```

obtain xs ys where xs: "ST = xs @ ys" "length xs = length fpTs"
proof -
  have "ST = take (length fpTs) ST @ drop (length fpTs) ST" by simp
  moreover from 1 have "length (take (length fpTs) ST) = length fpTs"
    by (simp add: min_def)
  ultimately show ?thesis ..
qed
obtain apTs where
  "ST = (rev apTs) @ ys" and "length apTs = length fpTs"
proof -
  have "ST = rev (rev xs) @ ys" by simp
  with xs show ?thesis by - (rule that, assumption, simp)
qed
moreover
from 1 xs obtain X ST' where "ys = X#ST'" by (auto simp add: neq_Nil_conv)
ultimately
have "ST = (rev apTs) @ X # ST'" "length apTs = length fpTs" by auto
with app
show "?P ST LT"
  apply (clarsimp simp add: list_all2_def min_def nth_append)
  apply ((rule exI)+, (rule conjI)?)
  apply auto
  done
qed

lemma app'ArrLoad[simp]:
  "app' (ArrLoad, G, C', pc, maxs, rT, (ST,LT)) =
  (∃ T ST'. ST=Init (PrimT Integer)#Init (T.[])#ST'"
  apply (case_tac ST)
  apply simp
  apply (case_tac list)
  apply simp
  apply (case_tac aa)
  apply (case_tac ty)
  apply simp
  apply (case_tac ref_ty)
  apply auto
  done

lemma app'ArrStore[simp]:
  "app' (ArrStore, G, C', pc, maxs, rT, (ST,LT)) =
  (∃ T T' ST'. ST=T#Init (PrimT Integer)#Init (T'.[])#ST' ∧ G ⊢ T ≤i Init T'"
  apply (case_tac ST, simp)
  apply (case_tac list, simp)
  apply (case_tac lista, simp)
  apply simp
  apply (case_tac ab)
  apply (case_tac ty)

```



```

  apply simp
  apply (case_tac ref_ty)
  apply auto
  done

lemma app'ArrLength[simp]:
  "app' (ArrLength, G, C', pc, maxs, rT, (ST,LT)) =
  (∃ T ST'. ST = Init (T.[])#ST'"
  apply (case_tac ST, simp)
  apply (case_tac a)
  apply (case_tac ty, simp)
  apply (case_tac ref_ty)
  apply auto
  done

lemma app'ArrNew[simp]:
  "app' (ArrNew T, G, C', pc, maxs, rT, (ST,LT)) =
  (∃ ST'. ST = Init (PrimT Integer)#ST' ∧ is_type G T ∧ noRA T ∧ dim T + 1 < max_dim)"
  apply (cases ST)
  apply auto
  done

declare conf_ArrayD [dest!]

The main theorem: welltyped programs do not produce type errors if they are started in a conformant state.

theorem no_type_error:
  assumes welltyped: "wt_jvm_prog G Phi" and conforms: "G,Phi ⊢ JVM s √"
  shows "exec_d G (Normal s) ≠ TypeError"
proof -
  from welltyped obtain mb where wf: "wf_prog mb G" by (fast dest: wt_jvm_progD)

  obtain xcp hp ihp frs where s [simp]: "s = (xcp, hp, ihp, frs)" by (cases s)

  from conforms have "xcp ≠ None ∨ frs = [] ⇒ check G s"
  by (unfold correct_state_def check_def) auto
  moreover {
    assume "¬(xcp ≠ None ∨ frs = [])"
    then obtain stk loc C sig pc r frs' where
      xcp [simp]: "xcp = None" and
      frs [simp]: "frs = (stk,loc,C,sig,pc,r)#frs'"
    by (clarsimp simp add: neq_Nil_conv) fast

    from conforms obtain ST LT z rT maxs maxl ins et where
      hconf: "G ⊢ h hp √" and
      class: "is_class G C" and

```

```

meth: "method (G, C) sig = Some (C, rT, maxs, maxl, ins, et)" and
phi:  "((ST,LT), z) ∈ Phi C sig ! pc" and
frame: "correct_frame G hp ihp (ST,LT) maxl ins (stk,loc,C,sig,pc,r)" and
frames: "correct_frames G hp ihp Phi rT sig z r frs'"
by simp (rule correct_stateE)

from frame obtain
stk: "approx_stk G hp ihp stk ST" and
loc: "approx_loc G hp ihp loc LT" and
init: "fst sig = init →
corresponds stk loc (ST, LT) ihp (fst r) (PartInit C) ∧
(∃ l. fst r = Addr l ∧ hp l ≠ None ∧
(ihp l = PartInit C ∨ (∃ C'. ihp l = Init (Class C'))))" and
pc: "pc < length ins" and
len: "length loc = length (snd sig) + maxl + 1"
by (rule correct_frameE)

note approx_val_def [simp]

from welltyped meth conforms
have "wt_instr (ins!pc) G C rT (Phi C sig) maxs (fst sig=init) (length ins) et pc"
by simp (rule wt_jvm_prog_impl_wt_instr_cor)
then obtain
app': "app (ins!pc) G C pc maxs (length ins) rT (fst sig=init) et (Phi C sig!pc)
" and
eff: "∀ (pc', s') ∈ set (eff (ins ! pc) G pc et (Phi C sig ! pc)). pc' < length ins"
by (simp add: wt_instr_def phi) blast

from eff
have pc': "∀ pc' ∈ set (succs (ins!pc) pc (Phi C sig!pc)). pc' < length ins"
by (simp add: eff_def) blast

from app' phi
have app:
"xcpt_app (ins!pc) G pc et ∧
app' (ins!pc, G, C, pc, maxs, rT, (ST,LT)) ∧
(fst sig = init ∧ ins ! pc = Return → z) ∧ ((∃ C m p. ins ! pc = Invoke_special
C m p ∧ ST!length p = PartInit C) → ¬ z)"
apply (clarsimp simp add: app_def)
apply (drule bspec, assumption)
apply clarsimp
done

with eff stk loc pc'
have "check_instr (ins!pc) G hp ihp stk loc C sig pc r (length ins) frs'"
proof (cases "ins!pc")
case (Getfield F C)
with app stk loc phi obtain v vT stk' where

```

```

class: "is_class G C" and
field: "field (G, C) F = Some (C, vT)" and
stk:   "stk = v # stk'" and
conf:  "G, hp, ihp ⊢ v :: ≤i Init (Class C)"
apply clarsimp
apply (blast dest: iconf_widen [OF _ _ wf])
done
from conf have isRef: "isRef v" by simp
moreover {
  assume "v ≠ Null"
  moreover
  from conf have "G, hp ⊢ v :: ≤ Class C" and "is_init hp ihp v" by auto
  moreover note field isRef wf
  ultimately
  have "∃ D vs. hp (the_Addr v) = Some (Obj D vs) ∧
        is_init hp ihp v ∧ G ⊢ D ≤C C"
    by (auto dest!: non_np_objD)
}
ultimately show ?thesis using Getfield field class stk hconf
apply clarsimp
apply (fastsimp dest!: hconfD widen_cfs_fields [OF _ _ wf] oconf_objD)
done
next
case (Putfield F C)
with app stk loc phi obtain v ref vT stk' where
class: "is_class G C" and
field: "field (G, C) F = Some (C, vT)" and
stk:   "stk = v # ref # stk'" and
confv: "G, hp, ihp ⊢ v :: ≤i Init vT" and
confr: "G, hp, ihp ⊢ ref :: ≤i Init (Class C)"
apply clarsimp
apply (blast dest: iconf_widen [OF _ _ wf])
done
from confr have isRef: "isRef ref" by simp
moreover
from confv have "is_init hp ihp v" by (simp add: iconf_def)
moreover {
  assume "ref ≠ Null"
  moreover
  from confr
  have "G, hp ⊢ ref :: ≤ Class C" and
        "is_init hp ihp ref" by auto
  moreover note field isRef wf
  ultimately have
    "∃ D vs. hp (the_Addr ref) = Some (Obj D vs)
      ∧ is_init hp ihp ref ∧ G ⊢ D ≤C C"
    by (auto dest: non_np_objD)
}

```

```

ultimately show ?thesis using Putfield field class stk confv
  by (clarsimp simp add: iconf_def)
next
case (Invoke C mn ps)
with stk app phi
show ?thesis
  apply (clarsimp simp del: app'.simps)
  apply (drule app'Invoke [THEN iffD1])
  apply (clarsimp dest!: approx_stk_append_lemma simp add: nth_append)
  apply (drule iconf_widen [OF _ _ wf], assumption)
  apply (clarsimp simp add: iconf_def)
  apply (drule non_npD, assumption)
  apply clarsimp
  apply (drule conf_ClassD)
  apply clarsimp
  apply (drule widen_methd [OF _ wf], assumption)
  apply (clarsimp simp add: approx_stk_rev [symmetric])
  apply (drule list_all2I, assumption)
  apply (unfold approx_stk_def approx_loc_def)
  apply (simp add: list_all2_approx)
  apply (drule list_all2_iconf_widen [OF wf], assumption+)
  done
next
case (Invoke_special C mn ps)
with stk app phi
show ?thesis
  apply (clarsimp simp del: app'.simps)
  apply (drule app'Invoke_special [THEN iffD1])
  apply (clarsimp dest!: approx_stk_append_lemma simp add: nth_append)
  apply (erule disjE)
  apply (clarsimp simp add: iconf_def isRef_def)
  apply (clarsimp simp add: approx_stk_rev [symmetric])
  apply (drule list_all2I, assumption)
  apply (simp add: list_all2_approx approx_stk_def approx_loc_def)
  apply (drule list_all2_iconf_widen [OF wf], assumption+)
  apply (clarsimp simp add: iconf_def isRef_def)
  apply (clarsimp simp add: approx_stk_rev [symmetric])
  apply (drule list_all2I, assumption)
  apply (unfold approx_stk_def approx_loc_def)
  apply (simp add: list_all2_approx)
  apply (drule list_all2_iconf_widen [OF wf], assumption+)
  done
next
case Return with stk app init phi meth frames
show ?thesis
  apply clarsimp
  apply (drule iconf_widen [OF _ _ wf], assumption)
  apply (clarsimp simp add: iconf_def neq_Nil_conv

```

```

        constructor_ok_def is_init_def isRef_def2)
    done
next
case (Ret idx)
from welltyped meth conforms s frs
have finite: "finite (Phi C sig!pc)" by simp (rule phi_finite)

with Ret app obtain R where
  idx: "idx < length LT" and
  R:   "LT!idx = OK (Init (RA R))"
  by clarsimp
moreover
from R idx loc have "loc!idx = RetAddr R"
  apply (simp add: approx_loc_def)
  apply (clarsimp simp add: list_all2_conv_all_nth)
  apply (erule allE, erule impE, assumption)
  apply (clarsimp simp add: iconf_def conf_def)
  apply (cases "loc!idx")
  apply auto
  done
moreover
from finite Ret pc' R
have "R < length ins"
  apply (auto simp add: set_SOME_lists finite_imageI theRA_def)
  apply (drule bspec, assumption)
  apply simp
  done
ultimately
show ?thesis using Ret loc by simp
qed auto
hence "check G s" by (simp add: check_def meth)
} ultimately
have "check G s" by blast
thus "exec_d G (Normal s) ≠ TypeError" ..
qed

```

The theorem above tells us that, in welltyped programs, the defensive machine reaches the same result as the aggressive one (after arbitrarily many steps).

**theorem** *welltyped\_aggressive\_imp\_defensive*:

```

"wt_jvm_prog G Phi  $\implies$  G,Phi  $\vdash$  JVM s  $\surd \implies$  G  $\vdash$  s -jvm $\rightarrow$  t
 $\implies$  G  $\vdash$  (Normal s) -jvmd $\rightarrow$  (Normal t)"
apply (unfold exec_all_def)
apply (erule rtrancl_induct)
  apply (simp add: exec_all_d_def)
apply simp
apply (fold exec_all_def)
apply (frule BV_correct, assumption+)
apply (drule no_type_error, assumption, drule no_type_error_commutes, simp)

```

```

apply (simp add: exec_all_d_def)
apply (rule rtrancl_trans, assumption)
apply blast
done

```

As corollary we get that the aggressive and the defensive machine are equivalent for welltyped programs (if started in a conformant state or in the canonical start state)

```

corollary welltyped_commutates:
  fixes  $G$  (" $\Gamma$ ") and  $\Phi$  (" $\Phi$ ")
  assumes "wt_jvm_prog  $\Gamma$   $\Phi$ " and " $\Gamma, \Phi \vdash_{JVM} s \checkmark$ "
  shows " $\Gamma \vdash (\text{Normal } s) \text{-jvmd} \rightarrow (\text{Normal } t) = \Gamma \vdash s \text{-jvm} \rightarrow t$ "
  by rule (erule defensive_imp_aggressive, rule welltyped_aggressive_imp_defensive)

```

```

corollary welltyped_initial_commutates:
  fixes  $G$  (" $\Gamma$ ") and  $\Phi$  (" $\Phi$ ")
  assumes "wt_jvm_prog  $\Gamma$   $\Phi$ "
  assumes "is_class  $\Gamma$   $C$ " "method ( $\Gamma, C$ ) ( $m, []$ ) = Some ( $C, b$ )" " $m \neq \text{init}$ "
  defines start: " $s \equiv \text{start\_state } \Gamma C m$ "
  shows " $\Gamma \vdash (\text{Normal } s) \text{-jvmd} \rightarrow (\text{Normal } t) = \Gamma \vdash s \text{-jvm} \rightarrow t$ "
proof -
  have " $\Gamma, \Phi \vdash_{JVM} s \checkmark$ " by (unfold start, rule BV_correct_initial)
  thus ?thesis by - (rule welltyped_commutates)
qed
end

```

## 4.24 Example Welltypings

**theory** BVExample = JVMListExample + BVSpecTypeSafe:

This theory shows type correctness of the example program in section 3.8 (p. 73) by explicitly providing a welltyping. It also shows that the start state of the program conforms to the welltyping; hence type safe execution is guaranteed.

### 4.24.1 Setup

Since the types *cnam*, *vnam*, and *mname* are anonymous, we describe distinctness of names in the example by axioms:

**axioms**

```
distinct_classes [simp]: "list_nam ≠ test_nam"
distinct_fields [simp]: "val_nam ≠ next_nam"
distinct_meth_list [simp]: "append_name ≠ init"
distinct_meth_test [simp]: "makelist_name ≠ init"
```

**declare**

```
distinct_classes [symmetric, simp]
distinct_fields [symmetric, simp]
distinct_meth_list [symmetric, simp]
distinct_meth_test [symmetric, simp]
```

Abbreviations for theorem sets we will have to use often in the proofs below:

```
lemmas name_defs = list_name_def test_name_def val_name_def next_name_def
lemmas system_defs = JVMSystemClasses_def SystemClassC_defs
lemmas class_defs = list_class_def test_class_def
```

These auxiliary proofs are for efficiency: class lookup, subclass relation, method and field lookup are computed only once:

**lemma** class\_list [simp]:

```
"class E list_name = Some list_class"
by (simp add: class_def system_defs E_def name_defs)
```

**lemma** class\_test [simp]:

```
"class E test_name = Some test_class"
by (simp add: class_def system_defs E_def name_defs)
```

**lemma** E\_classes [simp]:

```
"{C. is_class E C} =
{list_name, test_name, Xcpt NullPointerException, Xcpt ClassCast, Xcpt OutOfMemory,
Xcpt ArrayIndexOutOfBounds, Xcpt NegativeArraySize, Xcpt ArrayStore, Object}"
by (auto simp add: is_class_def class_def system_defs E_def name_defs class_defs)
```

The subclass relation spelt out:

**lemma** subcls1:

```
"subcls1 E =
```

```

{(list_name,Object), (test_name,Object), (Xcpt NullPointer, Object),
(Xcpt ClassCast, Object), (Xcpt OutOfMemory, Object), (Xcpt ArrayStore, Object),
(Xcpt ArrayIndexOutOfBounds, Object), (Xcpt NegativeArraySize, Object)}"
(is "_ = ?set")
proof
  show "?set  $\subseteq$  subcls1 E"
    apply (simp add: subcls1_def2)
    apply (unfold E_def system_defs class_def class_defs name_defs)
    apply auto
    done
  show "subcls1 E  $\subseteq$  ?set"
    apply rule
    apply (erule subcls1.elims)
    apply (drule class_SomeD)
    apply (unfold E_def class_defs system_defs)
    apply auto
    done
qed

```

The subclass relation is acyclic; hence its converse is well founded:

```

lemma notin_rtrancl:
  "(a,b)  $\in$  r*  $\implies$  a  $\neq$  b  $\implies$  ( $\bigwedge$ y. (a,y)  $\notin$  r)  $\implies$  False"
  by (auto elim: converse_rtranclE)

```

```

lemma acyclic_subcls1_E: "acyclic (subcls1 E)"
  apply (rule acyclicI)
  apply (simp add: subcls1)
  apply (auto dest!: tranclD)
  apply (auto elim!: notin_rtrancl simp add: name_defs)
  done

```

```

lemma wf_subcls1_E: "wf ((subcls1 E)-1)"
  apply (rule finite_acyclic_wf_converse)
  apply (simp add: subcls1)
  apply (rule acyclic_subcls1_E)
  done

```

Method and field lookup:

```

lemma unique: "unique E"
  by (unfold E_def class_defs system_defs name_defs) auto

```

```

lemma sys_classes_G: "sys_classes E"
  apply (rule sys_classes.intro)
  apply (clarsimp simp add: E_def)
  apply (rule unique)
  apply (rule wf_subcls1_E)
  done

```



```
lemmas lookups [simp] = sys_lemmas [OF sys_classes_G]
```

```
lemma method_append [simp]:
  "method (E, list_name) (append_name, [Class list_name]) =
  Some (list_name, PrimT Void, 3, 0, append_ins, [(1, 2, 8, Xcpt NullPointer)])"
  apply (insert class_list)
  apply (unfold list_class_def)
  apply (drule method_rec_lemma [OF _ wf_subcls1_E])
  apply (simp add: name_defs Default_ctor_def)
  done
```

```
lemma method_makelist [simp]:
  "method (E, test_name) (makelist_name, []) =
  Some (test_name, PrimT Void, 3, 2, make_list_ins, [])"
  apply (insert class_test)
  apply (unfold test_class_def)
  apply (drule method_rec_lemma [OF _ wf_subcls1_E])
  apply (simp add: name_defs Default_ctor_def)
  done
```

```
lemma method_default_ctor [simp]:
  "method (E, list_name) (init, []) =
  Some (list_name, PrimT Void, 1, 0, [Load 0, Invoke_special Object init [], Return], [])"
  apply (insert class_list)
  apply (unfold list_class_def)
  apply (drule method_rec_lemma [OF _ wf_subcls1_E])
  apply (simp add: name_defs Default_ctor_def)
  done
```

```
lemma field_val [simp]:
  "field (E, list_name) val_name = Some (list_name, PrimT Integer)"
  apply (unfold field_def)
  apply (insert class_list)
  apply (unfold list_class_def)
  apply (drule fields_rec_lemma [OF _ wf_subcls1_E])
  apply simp
  done
```

```
lemma field_next [simp]:
  "field (E, list_name) next_name = Some (list_name, Class list_name)"
  apply (unfold field_def)
  apply (insert class_list)
  apply (unfold list_class_def)
  apply (drule fields_rec_lemma [OF _ wf_subcls1_E])
  apply (simp add: name_defs distinct_fields [symmetric])
  done
```

```
lemma [simp]: "fields (E, test_name) = []"
```

```

  apply (insert class_test)
  apply (unfold test_class_def)
  apply (drule fields_rec_lemma [OF _ wf_subcls1_E])
  apply simp
done

```

```
lemmas [simp] = is_class_def
```

The next definition and three proof rules implement an algorithm to enumerate natural numbers. The command `apply (elim pc_end pc_next pc_0)` transforms a goal of the form

$$pc < n \implies P pc$$

into a series of goals

$$P (0::'a)$$

$$P (\text{Suc } 0)$$

...

$$P n$$

```
constdefs
```

```

  intervall :: "nat  $\Rightarrow$  nat  $\Rightarrow$  nat  $\Rightarrow$  bool" ("_  $\in$  [_, _]")
  "x  $\in$  [a, b)  $\equiv$  a  $\leq$  x  $\wedge$  x < b"

```

```

lemma pc_0: "x < n  $\implies$  (x  $\in$  [0, n)  $\implies$  P x)  $\implies$  P x"
  by (simp add: intervall_def)

```

```

lemma pc_next: "x  $\in$  [n0, n)  $\implies$  P n0  $\implies$  (x  $\in$  [Suc n0, n)  $\implies$  P x)  $\implies$  P x"
  apply (cases "x=n0")
  apply (auto simp add: intervall_def)
done

```

```

lemma pc_end: "x  $\in$  [n,n)  $\implies$  P x"
  by (unfold intervall_def) arith

```

#### 4.24.2 Program structure

The program is structurally wellformed:

```

lemma wf_struct:
  "wf_prog ( $\lambda G C mb. \text{True}$ ) E" (is "wf_prog ?mb E")
proof -
  note_simps [simp] = wf_mdecl_def wf_mhead_def wf_cdecl_def system_defs max_dim_def
  have "unique E" by (rule unique)
  moreover
  have "wf_syscls E" by simp
  moreover
  have "wf_cdecl ?mb E ObjectC" by simp
  moreover
  have "wf_cdecl ?mb E NullPointerC"

```

```

    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E ClassCastC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E OutOfMemoryC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E ArrayIndexOutOfBoundsC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E NegativeArraySizeC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E ArrayStoreC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E (list_name, list_class)"
    apply (auto elim!: notin_rtrancl
            simp add: wf_fdecl_def list_class_def subcls1)
    apply (auto simp add: name_defs)
    done
  moreover
  have "wf_cdecl ?mb E (test_name, test_class)"
    apply (auto elim!: notin_rtrancl
            simp add: wf_fdecl_def test_class_def subcls1)
    apply (auto simp add: name_defs)
    done
  ultimately
  show ?thesis by (simp del:_simps add: wf_prog_def E_def JVMSystemClasses_def)
qed

```

### 4.24.3 Welltypings

We show welltypings of all methods in the program  $E$ . The more interesting ones are `append_name` in class `list_name`, and `makelist_name` in class `test_name`. The rest are default constructors.

```
lemmas eff_simps [simp] = eff_def norm_eff_def xcpt_eff_def eff_bool_def
```

```
declare
```

```
  appInvoke [simp del]
  appInvoke_special [simp del]
```

```
constdefs
```

```
phi_obj_ctor :: method_type ("φo")
"φo ≡ [{([], [OK (Init (Class Object))]), True}],
  {([Init (PrimT Void)], [OK (Init (Class Object))]), True}]"
```

```
lemma check_types_lemma [simp]:
```

```
"(∀ ((a,b),x) ∈ A. b ∈ list mxr (err (init_tys E mpc)) ∧
  a ∈ list (length a) (init_tys E mpc) ∧
  length a ≤ mxs)
```

```

 $\implies$ 
OK A  $\in$  states E mxs mxr mpc"
apply (unfold states_def address_types_def)
apply simp
apply (rule subsetI)
apply clarify
apply (drule bspec, assumption)
apply blast
done

```

```

lemma Object_init [simp]:
  "wt_method E Object init [] (PrimT Void) (Suc 0) 0 [LitPush Unit, Return] []  $\varphi_0$ "
  apply (simp add: wt_method_def phi_obj_ctor_def
    wt_start_def wt_instr_def)
  apply (simp add: check_types_def init_tys_def)
  apply clarify
  apply (elim pc_end pc_next pc_0)
  apply fastsimp
  apply fastsimp
done

```

constdefs

```

phi_default_ctor :: "cname  $\Rightarrow$  method_type" (" $\varphi_c$  _")
" $\varphi_c$  C  $\equiv$  [{"], [OK (PartInit C)], False}},
  {([PartInit C], [OK (PartInit C)], False)},
  {([Init (PrimT Void)], [OK (Init (Class C))], True)}]"

```

lemma [simp]: "Ex Not" by fast

lemma [simp]: " $\exists z. z$ " by fast

```

lemma default_ctor [simp]:
  "E  $\vdash$  C  $\prec$  C1 Object  $\implies$  is_class E C  $\implies$ 
  wt_method E C init [] (PrimT Void) (Suc 0) 0
    [Load 0, Invoke_special Object init [], Return] [] ( $\varphi_c$  C)"
  apply (simp add: wt_method_def phi_default_ctor_def
    wt_start_def wt_instr_def check_types_def init_tys_def)
  apply (rule conjI)
  apply clarify
  apply (elim pc_end pc_next pc_0)
  apply simp
  apply (simp add: app_def xcpt_app_def replace_def)
  apply (simp add: Object_ctor_def)
  apply simp
  apply (fast dest: subcls1_wfD [OF _ wf_struct])
done

```

constdefs

```

phi_append :: method_type (" $\varphi_a$ ")
" $\varphi_a \equiv \text{map } (\text{image } (\lambda ST. ((\text{map } \text{Init } ST, \text{map } (\text{OK} \circ \text{Init}) [\text{Class } \text{list\_name}, \text{Class } \text{list\_name}])),
\text{False}))$ "
  [{
    [],
    { [Class list_name]},
    { [Class list_name]},
    { [Class list_name, Class list_name]},
    {[NT, Class list_name, Class list_name]},
    { [Class list_name]},
    { [Class list_name, Class list_name]},
    { [PrimT Void]},
    { [Class list_name], [Class (Xcpt NullPointer)]},
    { []},
    { [Class list_name]},
    { [Class list_name, Class list_name]},
    { []},
    { [PrimT Void]}]"

```

```
lemma wt_append [simp]:
```

```

"wt_method E list_name append_name [Class list_name] (PrimT Void) 3 0 append_ins
  [(Suc 0, 2, 8, Xcpt NullPointer)]  $\varphi_a$ "
apply (simp add: wt_method_def append_ins_def phi_append_def
  wt_start_def wt_instr_def name_defs)
apply (fold name_defs)
apply (simp add: check_types_def init_tys_def)
apply clarify
apply (elim pc_end pc_next pc_0)
apply simp
apply simp
apply (clarsimp simp add: match_exception_entry_def)
apply simp
apply simp
apply simp
apply (simp add: app_def xcpt_app_def)
apply simp
apply simp
apply simp
apply simp
apply (simp add: match_exception_entry_def)
apply simp
apply simp
done

```

Some abbreviations for readability

```
syntax
```

```

list :: ty
test :: ty
intg :: ty

```

```

void :: ty
translations
"list" == "Init (Class list_name)"
"test" == "Init (Class test_name)"
"intg" == "Init (PrimT Integer)"
"void" == "Init (PrimT Void)"

constdefs
phi_makelist :: method_type ("φm")
"φm ≡ map (λx. {(x,False)}) [
(
      [], [OK test,      Err,      Err]),
(
      [UnInit list_name 0], [OK test,      Err,      Err]),
(
  [UnInit list_name 0, UnInit list_name 0], [OK test,      Err,      Err]),
([UnInit list_name 0, UnInit list_name 0, UnInit list_name 0], [OK test, Err, Err]),
(
      [void, list, list], [OK test,      Err,      Err]),
(
      [list, list], [OK test,      Err,      Err]),
(
      [list], [OK list,      Err,      Err]),
(
      [intg, list], [OK list,      Err,      Err]),
(
      [], [OK list,      Err,      Err]),
(
      [UnInit list_name 8], [OK list,      Err,      Err]),
(
  [UnInit list_name 8, UnInit list_name 8], [OK list,      Err,      Err]),
(
      [void, list], [OK list,      Err,      Err]),
(
      [list], [OK list,      Err,      Err]),
(
      [list, list], [OK list,      Err,      Err]),
(
      [list], [OK list, OK list,      Err]),
(
      [intg, list], [OK list, OK list,      Err]),
(
      [], [OK list, OK list,      Err]),
(
      [UnInit list_name 16], [OK list, OK list,      Err]),
(
  [UnInit list_name 16, UnInit list_name 16], [OK list, OK list,      Err]),
(
      [void, list], [OK list, OK list,      Err]),
(
      [list], [OK list, OK list,      Err]),
(
      [list, list], [OK list, OK list,      Err]),
(
      [list], [OK list, OK list, OK list]),
(
      [intg, list], [OK list, OK list, OK list]),
(
      [], [OK list, OK list, OK list]),
(
      [list], [OK list, OK list, OK list]),
(
      [list, list], [OK list, OK list, OK list]),
(
      [void], [OK list, OK list, OK list]),
(
      [list, void], [OK list, OK list, OK list]),
(
      [list, list, void], [OK list, OK list, OK list]),
(
      [void, void], [OK list, OK list, OK list])
]"

lemma wt_makelist [simp]:
  "wt_method E test_name makelist_name [] (PrimT Void) 3 2 make_list_ins [] φm"
  apply (simp add: wt_method_def make_list_ins_def phi_makelist_def)
  apply (simp add: wt_start_def nat_number)
  apply (simp add: wt_instr_def name_defs)

```

```

apply (fold name_defs)
apply (simp add: check_types_def init_tys_def list_class_def)
apply clarify
apply (elim pc_end pc_next pc_0)
apply (simp add: replace_def)
apply simp
apply simp
apply (simp add: app_def xcpt_app_def replace_def)
apply simp
apply simp
apply simp
apply simp
apply (simp add: replace_def)
apply simp
apply simp
apply (simp add: app_def xcpt_app_def replace_def)
apply simp
apply simp
apply simp
apply simp
apply simp
apply (simp add: replace_def)
apply simp
apply simp
apply (simp add: app_def xcpt_app_def replace_def)
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply (simp add: app_def xcpt_app_def)
apply simp
apply simp
apply (simp add: app_def xcpt_app_def)
apply simp
done

```

The whole program is welltyped:

**constdefs**

```

Phi :: prog_type ("Φ")
"Φ C sig ≡ if C = Object ∧ sig = (init, []) then φo else
  if C = test_name ∧ sig = (makelist_name, []) then φm else
  if C = list_name ∧ sig = (append_name, [Class list_name]) then φa else
  if sig = (init, []) then φc C else []"

```

**lemma [simp]:**

```

"is_class E C = (C ∈
{Cname list_nam, Cname test_nam, Xcpt NullPointer, Xcpt ClassCast,
 Xcpt OutOfMemory, Xcpt ArrayIndexOutOfBounds, Xcpt NegativeArraySize,
 Xcpt ArrayStore, Object})"
apply (insert E_classes)
apply (auto simp add: name_defs)
done

declare is_class_def [simp del]

lemma wf_prog:
  "wt_jvm_prog E Φ"
  apply (unfold wt_jvm_prog_def)
  apply (rule wf_mb'E [OF wf_struct])
  apply (simp add: E_def)
  apply clarify
  apply (fold E_def)
  apply (simp add: system_defs class_defs)
  apply auto
  apply (auto simp add: Phi_def)
  apply (insert subcls1)
  apply (auto simp add: name_defs)
done

```

#### 4.24.4 Conformance

Execution of the program will be typesafe, because its start state conforms to the welltyping:

```

lemma "E, Φ ⊢ JVM start_state E test_name makelist_name √"
  apply (rule BV_correct_initial)
  apply (rule wf_prog)
  apply (auto simp add: is_class_def)
done

end

```



## 4.25 Example for a program with JSR

```
theory BVJSR = JVMSystemClasses + JVMSpecTypeSafe:
```

```
consts
```

```
  test_nam :: cnam
  m_name   :: mname
```

```
constdefs
```

```
  test_name :: cname
  "test_name == Cname test_nam"
```

```
  m_ins :: bytecode
  "m_ins ≡ [
  LitPush (Bool False),
  Store 1,
  Load 1,
  LitPush (Bool False),
  Ifcmpeq 6,
  LitPush (Intg 1),
  Store 3,
  Jsr 16,
  Load 3,
  Return,
  LitPush (Intg 2),
  Store 2,
  Load 1,
  LitPush (Bool False),
  Ifcmpeq 3,
  Jsr 8,
  Goto 14,
  Jsr 6,
  Goto 12,
  Store 4,
  Jsr 3,
  Load 4,
  Throw,
  Store 5,
  Load 1,
  LitPush (Bool False),
  Ifcmpeq 3,
  LitPush (Intg 3),
  Store 2,
  Ret 5,
  LitPush (Intg 4),
  Store 2,
  Load 2,
  Return]"
```

```
test_class :: "jvm_method class"
```

```
"test_class ≡
  (Object, [], [Default_ctor,
  ((m_name, []), PrimT Integer, (2, 5, m_ins, [(0,19,19,Object)])))]"
```

```
E :: jvm_prog
"E == JVMSystemClasses @ [(test_name, test_class)]"
```

#### types\_code

```
cnam ("string")
vnam ("string")
mname ("string")
loc_ ("int")
```

#### consts\_code

```
"arbitrary" ("(raise ERROR)")
"arbitrary" :: "val × val" ("{* (Unit,Unit) *}")
"arbitrary" :: "val" ("{* Unit *}")
"arbitrary" :: "cname" ("Object")

"test_nam" ("\"test\"")
"m_name" ("\"m\"")
"init" ("\"init\"")
```

#### generate\_code

```
test = "c_kil E test_name [] (PrimT Integer) False 5 10 [(0,19,19,Object)] m_ins"
```

```
ML "print_depth 100"
```

```
ML "test"
```

### 4.25.1 Setup

```
axioms not_init [simp]: "m_name ≠ init"
```

Abbreviations for theorem sets we will have to use often in the proofs below:

```
lemmas name_defs = test_name_def
```

```
lemmas system_defs = JVMSystemClasses_def SystemClassC_defs
```

```
lemmas class_defs = test_class_def
```

These auxiliary proofs are for efficiency: class lookup, subclass relation, method and field lookup are computed only once:

```
lemma class_test [simp]:
```

```
"class E test_name = Some test_class"
```

```
by (simp add: class_def system_defs E_def name_defs)
```

```
lemma E_classes [simp]:
```

```
"{C. is_class E C} =
```

```
{test_name, Xcpt NullPointer, Xcpt ClassCast, Xcpt OutOfMemory,
```

```
Xcpt ArrayIndexOutOfBounds, Xcpt NegativeArraySize,
```

```
Xcpt ArrayStore, Object}"
```

```
by (auto simp add: is_class_def class_def system_defs E_def name_defs class_defs)
```

The subclass relation spelt out:

```

lemma subcls1:
  "subcls1 E =
  {(test_name,Object), (Xcpt NullPointer, Object), (Xcpt ClassCast, Object),
  (Xcpt OutOfMemory, Object), (Xcpt ArrayStore, Object),
  (Xcpt ArrayIndexOutOfBounds, Object), (Xcpt NegativeArraySize, Object)}"
  (is "_ = ?set")
proof
  show "?set  $\subseteq$  subcls1 E"
    apply (simp add: subcls1_def2)
    apply (unfold E_def system_defs class_def class_defs name_defs)
    apply auto
    done
  show "subcls1 E  $\subseteq$  ?set"
    apply rule
    apply (erule subcls1.elims)
    apply (drule class_SomeD)
    apply (unfold E_def class_defs system_defs)
    apply auto
    done
qed

```

The subclass relation is acyclic; hence its converse is well founded:

```

lemma notin_rtrancl:
  "(a,b)  $\in$  r*  $\implies$  a  $\neq$  b  $\implies$  ( $\bigwedge$ y. (a,y)  $\notin$  r)  $\implies$  False"
  by (auto elim: converse_rtranclE)

```

```

lemma acyclic_subcls1_E: "acyclic (subcls1 E)"
  apply (rule acyclicI)
  apply (simp add: subcls1)
  apply (auto dest!: tranclD)
  apply (auto elim!: notin_rtrancl simp add: name_defs)
  done

```

```

lemma wf_subcls1_E: "wf ((subcls1 E)-1)"
  apply (rule finite_acyclic_wf_converse)
  apply (simp add: subcls1)
  apply (rule acyclic_subcls1_E)
  done

```

Method and field lookup:

```

lemma unique: "unique E"
  by (unfold E_def class_defs system_defs name_defs) auto

```

```

lemma sys_classes_G: "sys_classes E"
  apply (rule sys_classes.intro)
  apply (clarsimp simp add: E_def)
  apply (rule unique)
  apply (rule wf_subcls1_E)

```

done

lemmas lookups [simp] = sys\_lemmas [OF sys\_classes\_G]

```
lemma method_makelist [simp]:
  "method (E, test_name) (m_name, []) =
  Some (test_name, PrimT Integer, 2, 5, m_ins, [(0,19,19,Object)])"
  apply (insert class_test)
  apply (unfold test_class_def)
  apply (drule method_rec_lemma [OF _ wf_subcls1_E])
  apply (simp add: name_defs Default_ctor_def)
  done
```

```
lemma [simp]: "fields (E, test_name) = []"
  apply (insert class_test)
  apply (unfold test_class_def)
  apply (drule fields_rec_lemma [OF _ wf_subcls1_E])
  apply simp
  done
```

lemmas [simp] = is\_class\_def

The next definition and three proof rules implement an algorithm to enumerate natural numbers. The command `apply (elim pc_end pc_next pc_0)` transforms a goal of the form

$$pc < n \implies P \ pc$$

into a series of goals

$$P \ (0::'a)$$

$$P \ (Suc \ 0)$$

...

$$P \ n$$

constdefs

```
intervall :: "nat  $\Rightarrow$  nat  $\Rightarrow$  nat  $\Rightarrow$  bool" ("_  $\in$  [_, _']")
"x  $\in$  [a, b)  $\equiv$  a  $\leq$  x  $\wedge$  x < b"
```

```
lemma pc_0: "x < n  $\implies$  (x  $\in$  [0, n)  $\implies$  P x)  $\implies$  P x"
  by (simp add: intervall_def)
```

```
lemma pc_next: "x  $\in$  [n0, n)  $\implies$  P n0  $\implies$  (x  $\in$  [Suc n0, n)  $\implies$  P x)  $\implies$  P x"
  apply (cases "x=n0")
  apply (auto simp add: intervall_def)
  done
```

```
lemma pc_end: "x  $\in$  [n,n)  $\implies$  P x"
  by (unfold intervall_def) arith
```

### 4.25.2 Program structure

The program is structurally wellformed:

```

lemma wf_struct:
  "wf_prog (λG C mb. True) E" (is "wf_prog ?mb E")
proof -
  note_simps [simp] = wf_mdecl_def wf_mhead_def wf_cdecl_def system_defs max_dim_def
  have "unique E"
    by (simp add: E_def class_defs name_defs)
  moreover
  have "wf_syscls E" by simp
  moreover
  have "wf_cdecl ?mb E ObjectC" by (simp add: Object_ctor_def)
  moreover
  have "wf_cdecl ?mb E NullPointerC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E ClassCastC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E OutOfMemoryC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E ArrayIndexOutOfBoundsC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E NegativeArraySizeC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E ArrayStoreC"
    by (auto elim: notin_rtrancl simp add: name_defs subcls1)
  moreover
  have "wf_cdecl ?mb E (test_name, test_class)"
    apply (auto elim!: notin_rtrancl
      simp add: wf_fdecl_def test_class_def subcls1)
    apply (simp add: not_init [symmetric])
    done
  ultimately
  show ?thesis by (simp del:_simps add: wf_prog_def E_def JVMSystemClasses_def)
qed

```

### 4.25.3 Welltypings

We show welltypings of all methods in the program  $E$ . The more interesting ones are `append_name` in class `list_name`, and `makelist_name` in class `test_name`. The rest are default constructors.

```

lemmas eff_simps [simp] = eff_def norm_eff_def xcpt_eff_def eff_bool_def
declare
  appInvoke [simp del]
  appInvoke_special [simp del]

constdefs
  phi_obj_ctor :: method_type ("φo")

```

```
" $\varphi_o \equiv \{(\{([], \quad [OK (Init (Class Object))\}), True)\},$ 
   $\{(\{[Init (PrimT Void)], [OK (Init (Class Object))\}], True)\}$ "
```

```
lemma check_types_lemma [simp]:
```

```
"( $\forall ((a,b),x) \in A. b \in \text{list mxr (err (init_tys E mpc))} \wedge$ 
   $a \in \text{list (length a) (init_tys E mpc)} \wedge$ 
   $\text{length a} \leq \text{mxs}$ )
```

```
 $\implies$ 
```

```
OK A  $\in$  states E mxs mxr mpc"
```

```
apply (unfold states_def address_types_def)
```

```
apply simp
```

```
apply (rule subsetI)
```

```
apply clarify
```

```
apply (drule bspec, assumption)
```

```
apply blast
```

```
done
```

```
lemma Object_init [simp]:
```

```
"wt_method E Object init [] (PrimT Void) (Suc 0) 0 [LitPush Unit, Return] []  $\varphi_o$ "
```

```
apply (simp add: wt_method_def phi_obj_ctor_def
```

```
wt_start_def wt_instr_def)
```

```
apply (simp add: check_types_def init_tys_def)
```

```
apply clarify
```

```
apply (elim pc_end pc_next pc_0)
```

```
apply fastsimp
```

```
apply fastsimp
```

```
done
```

```
constdefs
```

```
phi_default_ctor :: "cname  $\Rightarrow$  method_type" (" $\varphi_c$  _")
```

```
" $\varphi_c C \equiv \{(\{([], \quad [OK (PartInit C)]), False)\},$ 
```

```
 $\{(\{[PartInit C], \quad [OK (PartInit C)]), False)\},$ 
```

```
 $\{(\{[Init (PrimT Void)], [OK (Init (Class C))\}], True)\}$ "
```

```
lemma [simp]: "Ex Not" by fast
```

```
lemma [simp]: " $\exists z. z$ " by fast
```

```
lemma default_ctor [simp]:
```

```
"E  $\vdash C <C1$  Object  $\implies$  is_class E C  $\implies$ 
```

```
wt_method E C init [] (PrimT Void) (Suc 0) 0
```

```
[Load 0, Invoke_special Object init [], Return] [] ( $\varphi_c C$ )"
```

```
apply (simp add: wt_method_def phi_default_ctor_def
```

```
wt_start_def wt_instr_def check_types_def init_tys_def)
```

```
apply (rule conjI)
```

```
apply clarify
```

```
apply (elim pc_end pc_next pc_0)
```

```
apply simp
```

```

apply (simp add: app_def xcpt_app_def replace_def Object_ctor_def)
apply simp
apply (fast dest: subcls1_wfD [OF _ wf_struct])
done

```

Some abbreviations for readability

**syntax**

```

test :: init_ty
intg :: ty
boolean :: ty

```

**translations**

```

"test" == "Init (Class test_name)"
"intg" == "Init (PrimT Integer)"
"boolean" == "Init (PrimT Boolean)"

```

**constdefs**

```

phi_m :: method_type (" $\varphi_m$ ")
" $\varphi_m \equiv \text{map (image } (\lambda x. (x, \text{False})))$ " [
  {(
    [], [OK test, Err, Err, Err, Err, Err, Err]}),
  {(
    [boolean], [OK test, Err, Err, Err, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, Err, Err, Err, Err, Err]}),
  {(
    [boolean], [OK test, OK boolean, Err, Err, Err, Err, Err]}),
  {( [boolean, boolean], [OK test, OK boolean, Err, Err, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, Err, Err, Err, Err, Err]}),
  {(
    [intg], [OK test, OK boolean, Err, Err, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, Err, OK intg, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, OK intg, OK intg, Err, OK (Init (RA 8))]),
  (
    [], [OK test, OK boolean, Err, OK intg, Err, OK (Init (RA 8))])},
  {(
    [intg], [OK test, OK boolean, OK intg, OK intg, Err, OK (Init (RA 8))]),
  (
    [intg], [OK test, OK boolean, Err, OK intg, Err, OK (Init (RA 8))])},
  {(
    [], [OK test, OK boolean, Err, Err, Err, Err, Err]}),
  {(
    [intg], [OK test, OK boolean, Err, Err, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, OK intg, Err, Err, Err, Err]}),
  {(
    [boolean], [OK test, OK boolean, OK intg, Err, Err, Err, Err]}),
  {( [boolean, boolean], [OK test, OK boolean, OK intg, Err, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, OK intg, Err, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))])},
  {(
    [], [OK test, OK boolean, OK intg, Err, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])},
  {},
  {},
  {},
  {},
  {(
    [Init (RA 8)], [OK test, OK boolean, Err, OK intg, Err, Err, Err]),
  (
    [Init (RA 16)], [OK test, OK boolean, OK intg, Err, Err, Err, Err]),
  (
    [Init (RA 18)], [OK test, OK boolean, OK intg, Err, Err, Err, Err]}),
  {(
    [], [OK test, OK boolean, Err, OK intg, Err, OK (Init (RA 8))]),

```

```

(
  (
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])},
{(
  [boolean], [OK test, OK boolean, Err, OK intg, Err, OK (Init (RA 8))]),
  (
    [boolean], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [boolean], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}),
{( [boolean, boolean], [OK test, OK boolean, Err, OK intg, Err, OK (Init (RA 8))]),
  ( [boolean, boolean], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  ( [boolean, boolean], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}),
{(
  [], [OK test, OK boolean, Err, OK intg, Err, OK (Init (RA 8))]),
  (
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}),
{(
  [intg], [OK test, OK boolean, Err, OK intg, Err, OK (Init (RA 8))]),
  (
    [intg], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [intg], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}),
{(
  [], [OK test, OK boolean, OK intg, OK intg, Err, OK (Init (RA 8))]),
  (
    [], [OK test, OK boolean, Err, OK intg, Err, OK (Init (RA 8))]),
  (
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}),
{(
  [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}),
{(
  [intg], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [intg], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}),
{(
  [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}),
{(
  [intg], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 16))]),
  (
    [intg], [OK test, OK boolean, OK intg, Err, Err, OK (Init (RA 18))])}
]"

```

```

lemma len: "length m_ins = 34"
  by (simp add: m_ins_def nat_number)

```

```

lemma wt_m [simp]:
  "wt_method E test_name m_name [] (PrimT Integer) 2 5 m_ins [(0,19,19,Object)]  $\varphi_m$ "
  apply (simp add: wt_method_def len)
  apply (rule conjI)
  apply (simp add: phi_m_def)
  apply (rule conjI)
  apply (simp add: check_types_def init_tys_def nat_number phi_m_def)
  apply (rule conjI)
  apply (simp add: wt_start_def phi_m_def nat_number name_defs)
  apply clarify
  apply (simp add: nat_number phi_m_def)
  apply (elim pc_end pc_next pc_0)
  apply (unfold wt_instr_def m_ins_def)
  apply (simp_all only: nth_Cons_Suc nth_Cons_0)
  apply simp
  apply simp
  apply simp

```



```

apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply simp
apply (simp add: set_SOME_lists theRA_def nat_number)
apply auto
done

```

The whole program is welltyped:

constdefs

```

Phi :: prog_type ("Φ")
"Φ C sig ≡ if C = Object ∧ sig = (init, []) then φo else
  if C = test_name ∧ sig = (m_name, []) then φm else
  if sig = (init, []) then φc C else []"

```

lemma [simp]:

```

"is_class E C =
(C ∈ {Cname test_nam, Xcpt NullPointer, Xcpt ClassCast, Xcpt OutOfMemory,
Xcpt OutOfMemory, Xcpt ArrayIndexOutOfBounds, Xcpt NegativeArraySize,
Xcpt ArrayStore, Object})"

```

```

  apply (insert E_classes)
  apply (auto simp add: name_defs)
done

declare is_class_def [simp del]

lemma wf_prog:
  "wt_jvm_prog E  $\Phi$ "
  apply (unfold wt_jvm_prog_def)
  apply (rule wf_mb'E [OF wf_struct])
  apply (simp add: E_def)
  apply clarify
  apply (fold E_def)
  apply (simp add: system_defs class_defs)
  apply auto
  apply (auto simp add: Phi_def)
  apply (insert subcls1)
  apply (simp_all add: not_init [symmetric])
  apply (simp add: name_defs)
  apply (simp add: name_defs)
done

```

#### 4.25.4 Conformance

Execution of the program will be typesafe, because its start state conforms to the welltyping:

```

lemma " $E, \Phi \vdash JVM \text{ start\_state } E \text{ test\_name } m\_name \checkmark$ "
  apply (rule BV_correct_initial)
  apply (rule wf_prog)
  apply (auto simp add: is_class_def)
done

```

end

# Bibliography

- [1] S. N. Freund. *Type Systems for Object-Oriented Intermediate Languages*. PhD thesis, Stanford University, 2000.
- [2] G. Klein and T. Nipkow. Verified lightweight bytecode verification. In S. Drossopoulou, S. Eisenbach, B. Jacobs, G. T. Leavens, P. Müller, and A. Poetsch-Heffter, editors, *Formal Techniques for Java Programs*. Technical Report 269, 5/2000, Fernuniversität Hagen, 2000. ECOOP2000 Workshop proceedings available from <http://www.informatik.fernuni-hagen.de/pi5/publications.html>.
- [3] G. Klein and T. Nipkow. Verified lightweight bytecode verification. *Concurrency and Computation: Practice and Experience*, 13(13):1133–1151, 2001. Invited contribution to special issue on Formal Techniques for Java.
- [4] T. Nipkow. Verified bytecode verifiers. In F. Honsell, editor, *Foundations of Software Science and Computation Structures (FOSSACS 2001)*, volume 2030, pages 347–363, 2001.
- [5] T. Nipkow, D. v. Oheimb, and C. Pusch.  $\mu$ Java: Embedding a programming language in a theorem prover. In F. L. Bauer and R. Steinbrüggen, editors, *Foundations of Secure Computation*, volume 175 of *NATO Science Series F: Computer and Systems Sciences*, pages 117–144. IOS Press, 2000.
- [6] D. von Oheimb. Axiomatic semantics for Java<sup>light</sup> in Isabelle/HOL. In S. Drossopoulou, S. Eisenbach, B. Jacobs, G. T. Leavens, P. Müller, and A. Poetsch-Heffter, editors, *Formal Techniques for Java Programs*. Technical Report 269, 5/2000, Fernuniversität Hagen, 2000. ECOOP2000 Workshop proceedings available from <http://www.informatik.fernuni-hagen.de/pi5/publications.html>.